

Chapter 2

From intuitionistic type theory to dependently typed programming

This preliminary chapter serves three purposes:

- The general theme of the dissertation is set up by describing the propositions-as-types-principle (Section 2.1) and its influence on program construction (Section 2.5).
- It is explained that, while we adopt AGDA as the expository language, we make sure that every AGDA program in this dissertation can at least in principle be translated down to well-studied aspects of type theory — no mysterious AGDA-specific feature is used. Specifically, we briefly discuss foundational aspects of pattern matching (Section 2.2) and equality (Section 2.3).
- Most AGDA-specific syntax, notational conventions, and basic constructions used throughout the dissertation are also introduced. In particular, a universe for “index-first” inductive families is constructed (Section 2.4), which is the basis of essentially all later constructions.

2.1 Propositions as types

Mathematics is all about mental constructions, that is, the intuitive grasp and manipulation of mental objects, the intuitionists say [Heyting, 1971; Dummett, 2000]. Take the natural numbers as an example. We have a distinct idea of how natural numbers are built: start from an origin 0, and form its successor 1, and then the successor of 1, which is 2, and so on. In other words, it is in our nature to be able to count, and counting is just the way the natural numbers are constructed. This construction then gives a specification of when we can immediately (i.e., directly intuitively) recognise a natural number, namely when it is 0 or a successor of some other natural number, and this specification of immediately recognisable forms is one of the conditions of forming the **set** of natural numbers in Martin-Löf’s intuitionistic type theory [Martin-Löf, 1975, 1984b; Nordström et al., 1990]. In symbols, we are justified by our intuition to have the **formation rule**

$$\frac{}{\text{Nat} : \text{Set}}$$

saying that we can conclude (below the line) that Nat is a set from no assumptions (above the line), and the two **introduction rules**

$$\frac{}{\text{zero} : \text{Nat}} \qquad \frac{n : \text{Nat}}{\text{suc } n : \text{Nat}}$$

specifying the **canonical inhabitants** of Nat, i.e., those inhabitants that are immediately recognisable as belonging to Nat, namely zero and suc n whenever n is an inhabitant of Nat. There are natural numbers which are not in canonical form (like 10^{10}) but instead encode an effective method for computing a canonical inhabitant. We accept them as **non-canonical inhabitants** of Nat, as long as they compute to a canonical form so we can see that they are indeed natural numbers. Thus, to form a set, we should be able to recognise its inhabitants, either directly or indirectly, as bearing a certain form and thus belonging to the set, so the inhabitants of the set are intuitively clear to us as a certain kind of mental construction.

What is more characteristic of intuitionism is that the intuitionistic interpretation of propositions — in particular the logical constants/connectives — follows

the same line of thought as the specification of the set of natural numbers. A proposition is an expression of its truth condition, and since intuitionistic truth follows from proofs, a proposition is clearly specified exactly when what constitutes a proof of it is determined [Martin-Löf, 1987]. What is a proof of a proposition, then? It is a piece of mental construction such that, upon inspection, the truth of the proposition is immediately recognised. For a simple example, in type theory we can formulate the formation rule for conjunctions

$$\frac{A : \text{Set} \quad B : \text{Set}}{A \wedge B : \text{Set}}$$

and the introduction rule

$$\frac{a : A \quad b : B}{(a, b) : A \wedge B}$$

saying that an immediately acceptable proof (canonical inhabitant) of $A \wedge B$ is a pair consisting of a proof (inhabitant) of A and a proof (inhabitant) of B . Any other (non-canonical) way of proving a conjunction must effectively yield a proof in the form of a pair. The relationship between a proposition and its proofs is thus exactly the same as the one between a set and its inhabitants — the proofs must be effectively recognisable as proving the proposition. Hence, in type theory, the notion of propositions and proofs is subsumed by the notion of sets and inhabitants. This is called the **propositions-as-types principle**, which reflects the observation that proofs are nothing but a certain kind of mental construction.

Notice that the notion of “effective methods” — or computation — was presumed when the notion of sets was introduced, and at some point we need to concretely specify an effective method. Since the description of every set includes an effective way to construct its canonical inhabitants, it is possible to express an effective method that mimics the construction of an inhabitant by saying that the computation has the same structure as how the inhabitant is constructed, and the computation is guaranteed to terminate since the construction of the inhabitant is finitary. For a typical example, let us look again at the natural numbers. Suppose that we have a **family of sets** $P : \text{Nat} \rightarrow \text{Set}$ indexed by inhabitants of Nat . (Since we only aim to present a casual sketch of type

theory, we take the liberty of using AGDA functions (including Set-computing ones like P above) in places where terms under contexts should have been used.) If we have an inhabitant z of P zero and a method s that, for any $n : \text{Nat}$, transforms an inhabitant of $P\ n$ to an inhabitant of $P\ (\text{suc } n)$, then we can compute an inhabitant of $P\ n$ for any given n by essentially the same counting process with which we construct n , but the counting now starts from z instead of zero and proceeds with s instead of suc . For instance, if a proof of $P\ 2$ is required, we can simply apply s to z twice, just like we apply suc to zero twice to form 2, so the computation was guided by the structure of 2. This explanation justifies the following **elimination rule**

$$\frac{P : \text{Nat} \rightarrow \text{Set} \quad z : P\ \text{zero} \quad s : (n : \text{Nat}) \rightarrow P\ n \rightarrow P\ (\text{suc } n) \quad n : \text{Nat}}{\text{Nat-elim } P\ z\ s\ n : P\ n}$$

(The type of s illustrates AGDA's syntax for dependent function types — the value n of the first argument is referred to in the types of the second argument and the result.) The symbol Nat-elim symbolises the method described above, which, given P , z , and s , transforms any natural number n into an inhabitant of the set $P\ n$. The actual computation performed by Nat-elim is stated as two **computation rules** in the form of equality judgements (see Section 2.3):

$$\frac{P : \text{Nat} \rightarrow \text{Set} \quad z : P\ \text{zero} \quad s : (n : \text{Nat}) \rightarrow P\ n \rightarrow P\ (\text{suc } n)}{\text{Nat-elim } P\ z\ s\ \text{zero} = z \in P\ \text{zero}}$$

$$\frac{P : \text{Nat} \rightarrow \text{Set} \quad z : P\ \text{zero} \quad s : (n : \text{Nat}) \rightarrow P\ n \rightarrow P\ (\text{suc } n) \quad n : \text{Nat}}{\text{Nat-elim } P\ z\ s\ (\text{suc } n) = s\ n\ (\text{Nat-elim } P\ z\ s\ n) \in P\ (\text{suc } n)}$$

From the logic perspective, predicates on Nat are a special case of Nat -indexed families of sets like P ; Nat-elim then delivers the induction principle for natural numbers, as it produces a proof of $P\ n$ for any $n : \text{Nat}$ if proofs z and s of the base case and the inductive case can be constructed. In general, the propositions-as-types principle treats logical entities as ordinary mathematical objects; the logic hence inherits the computational meaning of intuitionistic mathematics and becomes constructive.

By enabling the interplay of various sets governed by rules like the above ones, type theory is capable of formalising various mental constructions we

manipulate in mathematics in a fully computational way, making it a powerful programming language. As Martin-Löf [1984a] noted: “If programming is understood [...] as the design of the methods of computation [...], then it no longer seems possible to distinguish the discipline of programming from constructive mathematics”. Indeed, sets are easily comparable with inductive datatypes in functional programming — a formation rule names a datatype, the associated introduction rules list the constructors of the datatype, and the associated elimination rule and computation rules define a precisely typed version of primitive recursion on the datatype. Consequently, we identify “sets” with “types”, and regard them as interchangeable terms.

The uniform treatment of programs and proofs in type theory reveals new possibilities regarding proofs of program correctness. Traditional mathematical theories employ a standalone logic language for talking about some postulated objects. For example, Peano arithmetic is set up by postulating axioms about natural numbers in the language of first-order logic. Inside the postulated system of natural numbers, there is no knowledge of logic formulas or proofs (except via exotic encodings) — logic is at a higher level than the objects it is used for talking about. Programming systems based on such principle (e.g., Hoare logic) then need to have a meta-level logic language to reason about properties of programs. In **dependently typed** languages based on type theory, however, the two traditional levels are coherently integrated into one, so programs and their correctness proofs can be constructed together in the same language. For example, the proposition $\forall a : A. \exists b : B. R\ a\ b$ is interpreted as the type of a function taking $a : A$ to a pair consisting of $b : B$ and a proof of the proposition $R\ a\ b$, so the output of type B is guaranteed to be related by R to the input of type A . Checking of proof validity reduces to typechecking, and correctness proofs coexist with programs, as opposed to being separately presented at a meta-level.

The propositions-as-types principle, however, can lead to a more intimate form of program correctness by construction by blurring the distinction between programs and proofs even further; this form of program correctness — called **internalism** — is introduced in Section 2.5, which opens the central topic studied in this dissertation. Before that, we make a transition from type theory

to practical programming in AGDA, starting with its pattern matching notation.

2.2 Elimination and pattern matching

The formation rules and introduction rules for sets in type theory directly translate into inductive datatype declarations in functional programming. For example, the set of natural numbers is translated into AGDA as an inductive datatype with two constructors, with their full types displayed:

```
data Nat : Set where
  zero : Nat
  suc  : Nat → Nat
```

In type theory, computations on inductive datatypes are specified using eliminators like `Nat-elim`, whose style corresponds to **recursion schemes** [Meijer et al., 1991] in functional programming. (In particular, `Nat-elim` is a more informatively typed version of paramorphism [Meertens, 1992] on natural numbers.) One reason for making elimination as the only option is that programs in type theory are required to terminate — which is a consequence of the requirement that an inhabitant should be effectively recognisable as belonging to a set, and results in decidable typechecking — and using eliminators throughout is a straightforward way of enforcing termination. On the other hand, in functional programming, the **pattern matching** notation is widely used for defining programs on (inductive) datatypes (see, e.g., Hudak et al. [2007, Section 5]) in addition to recursion schemes. Pattern matching is vital to the clarity of functional programs because it not only allows a function to be intuitively defined by equations suggesting how the function computes, but also clearly conveys the programming strategy of splitting a problem into sub-problems by case analysis.

When it comes to dependently typed programming, the situation becomes more complicated due to the presence of **inductive families** [Dybjer, 1994], i.e., simultaneously inductively defined families of sets, like the following:

```
data _≤N_ (m : Nat) : Nat → Set where
  refl : m ≤N m
```

$$\text{step} : (n : \text{Nat}) \rightarrow m \leq_N n \rightarrow m \leq_N \text{succ } n$$

(An AGDA name with underscores (like $_ \leq_N _$) can be applied to arguments either by prefixing (like “ $_ \leq_N _ m n$ ”) or by substituting the arguments for the underscores with proper spacing (like “ $m \leq_N n$ ”).) Reading the declaration logically, the types of the two constructors `refl` and `step` give the two proof rules for establishing that one natural number is less than or equal to another. More generally, we read the declaration as a datatype parametrised by $m : \text{Nat}$ (as signified by its appearance right next to `data $_ \leq_N _$`) and indexed by Nat . For any $m : \text{Nat}$, the type family $_ \leq_N m : \text{Nat} \rightarrow \text{Set}$ as a whole is inductively populated: we have an inhabitant `refl` in the set $(_ \leq_N m) m$, and whenever we have an inhabitant $p : (_ \leq_N m) n$ for some $n : \text{Nat}$, we can make a larger inhabitant `step n p` in another set $(_ \leq_N m) (\text{succ } n)$ in the family. (From now on we usually refer to inductive families simply as datatypes, especially when emphasising their use in programming and de-emphasising the distinction with non-indexed datatypes like Nat .)

With inductive families, splitting a problem into sub-problems by case analysis often leads to nontrivial refinement of the goal type and the context, and such refinement can be tricky to handle with eliminators. Admittedly, in terms of expressive power, pattern matching and elimination are basically equivalent, as eliminators can be easily defined by dependent pattern matching, and conversely, it has been shown that dependent pattern matching can be reduced to elimination if **uniqueness of identity proofs** — or, equivalently, the **K axiom** [Streicher, 1993] — is assumed [McBride, 1999; Goguen et al., 2006]. (See Section 2.3 for more on uniqueness of identity proofs.) Nevertheless, there is a significant notational advantage of recovering pattern matching in dependently typed programming, especially with the support of an interactive development environment for managing detail about datatype indices. Below we look at an example of interactively constructing a program with pattern matching in AGDA, whose design was inspired by EPIGRAM [McBride, 2004; McBride and McKinna, 2004].

2.2.1 Pattern matching and interactive development

Suppose that we are asked to prove that \leq_N is transitive, i.e., to construct the program

$$\text{trans} : (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z$$

(The “telescopic” quantification “ $(x \ y \ z : \text{Nat}) \rightarrow$ ” is a shorthand for “ $(x : \text{Nat}) (y : \text{Nat}) (z : \text{Nat}) \rightarrow$ ”, which, in turn, is a shorthand for “ $(x : \text{Nat}) \rightarrow (y : \text{Nat}) \rightarrow (z : \text{Nat}) \rightarrow$ ”.) We define *trans* interactively by first putting pattern variables for the arguments on the left of the defining equation and then leaving an “interaction point” — also called a “goal” — on the right, which is numbered 0. AGDA then tells us that a term of type $x \leq_N z$ is expected (shown in the goal).

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ y \ z \ p \ q &= \{x \leq_N z\}_0 \end{aligned}$$

We instruct AGDA to perform case analysis on q , and there are two cases: *refl* and *step* $w \ r$ where r has type $y \leq_N w$. The original Goal 0 is split into two sub-goals, and unification is triggered for each sub-goal.

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ .z \ z \quad p^{x \leq_N z} \text{ refl} &= \{x \leq_N z\}_1 \\ \text{trans } x \ y \ .(\text{suc } w) \ p^{x \leq_N y} (\text{step } w \ r^{y \leq_N w}) &= \{x \leq_N \text{suc } w\}_2 \end{aligned}$$

In Goal 1, the type of *refl* demands that y be unified with z , and hence the pattern variable y is replaced with a “dot pattern” $.z$ indicating that the value of y is determined by unification to be z . Therefore, upon enquiry, AGDA tells us that the type of p in the context — which was originally $x \leq_N y$ — is now $x \leq_N z$ (shown in superscript next to p , which is not part of the AGDA program but only appears when interacting with AGDA). Similarly for Goal 2, z is unified with $\text{suc } w$ and the goal type is rewritten accordingly. We see that the case analysis has led to two sub-problems with different goal types and contexts, where Goal 1 is easily solvable as there is a term in the context with the right type, namely p .

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ .z \ z \quad p \quad \text{refl} &= p \end{aligned}$$

$$\text{trans } x \ y \ .(\text{suc } w) \ p^{x \leq_N y} (\text{step } w \ r^{y \leq_N w}) = \{x \leq_N \text{suc } w\}_2$$

The second goal type $x \leq_N \text{suc } w$ looks like the conclusion in the type of the term $\text{step } w : x \leq_N w \rightarrow x \leq_N \text{suc } w$, so we use this term to reduce Goal 2 to Goal 3, which now requires a term of type $x \leq_N w$.

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ .z \ z \quad p \quad \text{refl} &= p \\ \text{trans } x \ y \ .(\text{suc } w) \ p^{x \leq_N y} (\text{step } w \ r^{y \leq_N w}) &= \text{step } w \ \{x \leq_N w\}_3 \end{aligned}$$

Now we see that the induction hypothesis term $\text{trans } x \ y \ w \ p \ r : x \leq_N w$ has the right type. Solving Goal 3 with this term completes the program.

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ .z \ z \quad p \ \text{refl} &= p \\ \text{trans } x \ y \ .(\text{suc } w) \ p \ (\text{step } w \ r) &= \text{step } w \ (\text{trans } x \ y \ w \ p \ r) \end{aligned}$$

In contrast, if we stick to the default elimination approach in type theory, we would use the eliminator

$$\begin{aligned} \leq_N\text{-elim} &: (m : \text{Nat}) (P : (n : \text{Nat}) \rightarrow m \leq_N n \rightarrow \text{Set}) \rightarrow \\ &P \ m \ \text{refl} \rightarrow \\ &((n : \text{Nat}) (t : m \leq_N n) \rightarrow P \ n \ t \rightarrow P \ (\text{suc } n) \ (\text{step } n \ t)) \rightarrow \\ &(n : \text{Nat}) (t : m \leq_N n) \rightarrow P \ n \ t \end{aligned}$$

and write

$$\begin{aligned} \text{trans} &: (x \ y \ z : \text{Nat}) \rightarrow x \leq_N y \rightarrow y \leq_N z \rightarrow x \leq_N z \\ \text{trans } x \ y \ z \ p \ q &= \leq_N\text{-elim } y \ (\lambda y' _ \mapsto x \leq_N y \rightarrow x \leq_N y') \\ &\quad (\lambda p' _ \mapsto p') \ (\lambda w \ r \ ih \ p' _ \mapsto \text{step } w \ (ih \ p')) \ z \ q \ p \end{aligned}$$

We are forced to write the program in continuation passing style, where the two continuations correspond to the two clauses in the pattern matching version and likewise have more specific goal types, and the relevant context (p in this case) must be explicitly passed into the continuations in order to be refined to a more specific type. Even with interactive support, the eliminator version is inherently harder to write and understand, especially when complicated dependent types are involved. If a function definition requires more than one level of elimination, then the advantage of using pattern matching over using eliminators becomes even more apparent.

2.2.2 Pattern matching on intermediate computation

It is often the case that we need to perform pattern matching not only on an argument but also on some intermediate computation. In simply typed languages, this is usually achieved by “case expressions”, a special case being if-then-else expressions for booleans. But again, pattern matching on intermediate computation can make refinements to the goal type and the context in dependently typed languages, so case expressions — being more like eliminators — become less convenient. McBride and McKinna [2004] thus proposed **with-matching**, which generalises pattern guards [Peyton Jones, 1997] and shifts pattern matching on intermediate computations from the right of an equation to the left, thereby granting them equal status with pattern matching on arguments, in particular the power to refine contexts and goal types.

Example (*insertion into a list*). To demonstrate the syntax of **with-matching**, we give a simple example of writing the function inserting an element into a list as used in, e.g., insertion sort. (More precisely, the element is inserted at the rightmost position to the left of which all elements are strictly smaller.) First we define the usual list datatype:

```
data List (A : Set) : Set where
  []      : List A
  _::_    : A → List A → List A
```

and (throughout this dissertation) let $Val : Set$ be equipped with a decidable total preorder, i.e., there is a relation

$$\leq : Val \rightarrow Val \rightarrow Set$$

with the following operations:

```
 $\leq\text{-refl}   : \{x : Val\} \rightarrow x \leq x$ 
 $\leq\text{-trans}  : \{x\ y\ z : Val\} \rightarrow x \leq y \rightarrow y \leq z \rightarrow x \leq z$ 
 $\leq?\text{-}       : (x\ y : Val) \rightarrow Dec\ (x \leq y)$  -- see below
 $\not\leq\text{-invert} : \{x\ y : Val\} \rightarrow \neg (x \leq y) \rightarrow y \leq x$ 
```

where Dec is the following datatype witnessing whether a set is inhabited or not:

```
data Dec (A : Set) : Set where
```

```
yes : A → Dec A
no  : ¬ A → Dec A  -- ¬ A = A → ⊥, where ⊥ is the empty set
```

(Quantifications like $\{x : Val\}$ are implicit arguments to a function. They can be omitted when applying the function, and AGDA will try to infer them.) The insertion function is then written as

```
insert : Val → List Val → List Val
insert y [] = y :: []
insert y (x :: xs) with y ≤? x
insert y (x :: xs) | yes _ = y :: x :: xs
insert y (x :: xs) | no _ = x :: insert y xs
```

The result of the intermediate computation $y \leq? x : Dec (y \leq x)$ is matched against `yes` and `no` on the left-hand side of the last two equations, just like we are performing pattern matching on a new argument. (In fact, AGDA implements **with**-matching exactly by synthesising an auxiliary function with an additional argument [Norell, 2007, Section 2.3].) The witnesses carried by `yes` and `no` are ignored in this case (their names are suppressed by underscores), but in general these can be proofs that are further matched and change the context and the goal type. (Admittedly, this is a trivial example with regard to the full power of **with**-matching, but *insert* will be used in later examples in this dissertation.) □

An important application of **with**-matching is McBride and McKinna’s adaptation [2004] of Wadler’s **views** [1987] — or “customised pattern matching” — for dependently typed programming. Suppose that we wish to implement a *snoc*-list view for cons-lists, i.e., to say that a list is either empty or has the form $ys \# (y :: [])$ (where $\#$ is list append, a definition of which is shown in Section 2.4.2). We would define the following view type

```
data SnocView {A : Set} : List A → Set where
  nil  : SnocView []
  snoc : (ys : List A) (y : A) → SnocView (ys # (y :: []))
```

and write a **covering function** for the view:

```
snocView : {A : Set} (xs : List A) → SnocView xs
snocView [] = nil
snocView (x :: xs) with snocView xs
```

$$\begin{array}{lcl} \text{snocView } (x :: []) & | \text{ nil} & = \text{snoc } [] \ x \\ \text{snocView } (x :: (ys \# (y :: []))) & | \text{ snoc } \text{ys } y & = \text{snoc } (x :: \text{ys}) \ y \end{array}$$

Note that the type of *snocView* ensures that every list is covered under one constructor of *SnocView*. Also, this is a nontrivial example of **with**-matching, because performing pattern matching on the result of *snocView xs* refines *xs* in the context to either *[]* or *ys # (y :: [])*, and the refinement is propagated to the goal type. Now, for example, the function *init* which removes the last element (if any) in a list can be implemented simply as

$$\begin{array}{lcl} \text{init} : \{A : \text{Set}\} \rightarrow \text{List } A \rightarrow \text{List } A \\ \text{init } xs & \text{with } \text{snocView } xs & \\ \text{init } [] & | \text{ nil} & = [] \\ \text{init } (ys \# (y :: [])) & | \text{ snoc } \text{ys } y & = \text{ys} \end{array}$$

Views are not enough for dependently typed programming, though; they offer customised case analysis but not terminating recursion. McBride and McKinna [2004] proposed a general mechanism for invoking any programmer-defined eliminator using the pattern matching syntax, so the programmer can choose whichever recursive problem-splitting strategy they needs and express it conveniently with pattern matching. This mechanism is not implemented in AGDA, but it is implemented in EPIGRAM, and greatly improves readability of dependently typed programs. Specifically, EPIGRAM syntax makes it clear which and in what order eliminators are invoked, so programs are easily guaranteed to be based on elimination — and thus be terminating — while remaining readable. AGDA’s design does not emphasise reducibility to elimination, but almost all the recursive programs in this dissertation are written with this in mind, so termination is evident even without understanding AGDA’s termination checker in detail. Also, although AGDA provides pattern matching only for datatype constructors, all but one of the recursive programs in this dissertation use default structural induction (without need of programmer-defined elimination), so AGDA syntax suffices. (The exception is the final program in ??, where we use an explicit eliminator.)

2.3 Equality

In logic, the **intension** of a concept is its defining description, while the **extension** of the concept is the range of objects it refers to. Two concepts can differ intensionally yet agree extensionally when they use different ways to describe the same range of objects. Classical mathematics cares about extensions only (e.g., the axiom of extensionality in set theory defines that two sets are equal exactly when they have the same inhabitants, regardless of how they are described), whereas in intuitionistic mathematics, objects are given to us as mental constructions, which are inherently intensional descriptions. As a consequence, the fundamental equality for intuitionistic mathematics is intensional [Dummett, 2000, Section 1.2], since we can only compare the intensional descriptions given to us, and furthermore it might be impossible to effectively recognise whether two intensionally different constructions are extensionally the same. For example, functions describing different computational procedures are intensionally distinguished even when they always map the same input to the same output, as it is well known that pointwise equality of functions is undecidable. We can, of course, still talk about extensional equalities in intuitionistic mathematics, but they are just treated as ordinary propositions.

The fundamental equality is formulated in type theory as **judgemental equality**, a meta-level notion for determining whether two types match in type-checking, which also involves determining whether two terms match because types can contain terms. (The computation rules for Nat-elim in Section 2.1 are examples of judgemental equalities between terms.) If we take the position of intuitionistic mathematics seriously, judgemental equality would be chosen to be the intensional, syntactic equality — also called **definitional equality** — which can be implemented by reducing two types/terms to normal forms and checking whether the normal forms match. The resulting type theory is called an **intensional type theory**; its characteristic feature is decidable typechecking (enabling effective recognition of set membership) due to decidability of judgemental equality. AGDA, in particular, is intensional in this sense.

Judgemental equality — being a meta-level notion — is not an entity inside the theory. To state equality between two terms as a proposition and have proof

for that proposition inside the theory, we need **propositional equality**, which can be defined in AGDA by the following inductive family:

```
data _≡_ {A : Set} (x : A) : A → Set where
  refl : x ≡ x
```

The canonical way to prove an equality proposition $x \equiv y$ is `refl`, which is permitted when x and y are judgementally equal. Under contexts, however, it is possible to prove that two judgementally different terms are propositionally equal. For example, the following “catamorphic” identity function on natural numbers

```
id' : Nat → Nat
id' zero    = zero
id' (suc n) = suc (id' n)
```

can be shown to be pointwise equal to the polymorphic identity function

```
id : {A : Set} → A → A
id x = x
```

That is, given $n : \text{Nat}$ in the context, even though the two open terms $id\ n$ (which is definitionally just n) and $id'\ n$ are judgementally different, we can still prove $id\ n \equiv id'\ n$ by induction (elimination) on n , whose two cases instantiate n to a more specific form and make computation on $id'\ n$ happen. One might say that propositional equality — in this most basic, inductive form — is “delayed” judgemental equality as a proposition: the judgementally different terms $id\ n$ and $id'\ n$ would compute to the same canonical term — and hence become judgementally equal — after substituting a canonical natural number for n , turning them into closed terms and allowing the computation to complete. Formally, this is stated as the “reflection principle”: two propositionally equal closed terms are judgementally equal (see, e.g., Luo [1994, Section 5.1.3] and Streicher [1993, Section 1.1]).

A propositional equality satisfying the reflection principle — e.g., the AGDA one — can sometimes be too discriminating. For example, in category theory (which we use in Chapters ?? and ??), a universal function (i.e., a universal morphism in the category of sets and total functions) is unique up to extensional (i.e., pointwise) equality, but pointwise propositionally equal functions are

usually not propositionally equal themselves. (If, under the empty context, two pointwise propositionally equal functions are propositionally equal themselves, then by the reflection principle they are also judgementally equal, but the two functions can well have different intensions.) Of course, we can explicitly work with pointwise equality on functions, i.e., establishing and using properties formulated in terms of the relation

$$\begin{aligned} _ \doteq _ &: \{A\ B : \text{Set}\} \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow B) \rightarrow \text{Set} \\ _ \doteq _ \{A\} f\ g &= (x : A) \rightarrow f\ x \equiv g\ x \end{aligned}$$

(The implicit argument A is explicitly displayed in curly braces on the left-hand side of the equation since we need to refer to it on the right-hand side.) Not being able to identify extensionally equal functions propositionally, however, means that for extensionally equal functions we do not automatically get basic properties like

$$\text{cong} : \{A\ B : \text{Set}\} (f : A \rightarrow B) \{x\ y : A\} \rightarrow x \equiv y \rightarrow f\ x \equiv f\ y$$

i.e., a function maps equal arguments to equal results, and

$$\text{subst} : \{A : \text{Set}\} (P : A \rightarrow \text{Set}) \{x\ y : A\} \rightarrow x \equiv y \rightarrow P\ x \rightarrow P\ y$$

i.e., inhabitants in an indexed set can be transported to another set with an equal index. We would need to prove such properties for every entity like f and P on a case-by-case basis, which quickly becomes tedious.

Several foundational modifications to intensional type theory have been proposed to obtain a more liberal notion of equality. While this dissertation sticks to AGDA's intensional approach and does not adopt any of the alternatives, it is interesting to reflect on the relationship between these alternatives and our development.

- A simple yet radical approach is to add the **equality reflection rule** to the theory, injecting propositional equality back into judgemental equality.

$$\frac{x : A \quad y : A \quad eq : x \equiv y}{x = y \in A}$$

Extensionally equal functions become judgementally equal (and thus propositionally equal) in such a theory: Suppose that f and g are functions of type $A \rightarrow B$ and we have a proof

$$fgeq : (x : A) \rightarrow f\ x \equiv g\ x$$

Then, judgementally,

$$\begin{aligned} & f \\ = & \{ \eta\text{-expansion} \} \\ & \lambda x \mapsto f\ x \\ = & \{ \text{equality reflection} \text{ — } f\ x = g\ x \in B \text{ since } fgeq\ x : f\ x \equiv g\ x \} \\ & \lambda x \mapsto g\ x \\ = & \{ \eta\text{-contraction} \} \\ & g \quad \in A \rightarrow B \end{aligned}$$

The judgemental equality is thus able to identify pointwise equal functions and becomes extensional, and such a theory is called an **extensional type theory** (see, e.g., Nordström et al. [1990, Section 8.2]). In extensional type theory, combinators like *subst* become unnecessary, since having a proof of $x \equiv y$ means that x and y are identified judgementally and hence are regarded as the same during typechecking, so $P\ x$ and $P\ y$ are simply the same type — no explicit transportation is needed. Consequently, programs become very lightweight, with all the equality justifications moved to typing derivations at the meta-level. The downside is that typechecking in extensional type theory is undecidable, because whenever there is possibility that the equality reflection rule is needed, the typechecker would have to somehow determine whether there is a suitable equality proof, for which there is no effective procedure. This is not a big problem for proof assistants like NUPRL [Constable et al., 1985], in which the programmer instructs the proof assistant to construct typing derivations and can supply the right proof when using the equality reflection rule. (NUPRL, in fact, simply identifies judgemental equality and propositional equality and does not have the equality reflection rule explicitly.) But for programming languages like Ω MEGA [Sheard and Linger, 2007], equality reflection does present a problem, since the programmer constructs a term only, and the typing derivation has to be constructed by the typechecker, which then has to search for proofs. Ω MEGA can take hints from the programmer so the proof search is more likely to succeed, but the fundamental problem is that justification of program correctness now relies

on the proof searching algorithm and is tied to the implementation detail of a specific programming system. Since the focus of this dissertation is on dependently typed programming, extensional type theory is not a satisfactory foundation.

- Altenkirch et al. [2007] proposed a variant of intensional type theory called **observational type theory**, which defines propositional equality to be an extensional one — in particular, propositional equality on functions is pointwise equality — but retains computational behaviour (strong normalisation and canonicity) and decidable typechecking of intensional type theory. We step away from observational type theory merely for a practical reason: the theory is not implemented natively in any programming system yet. While it might be possible to model observational equality in AGDA to some extent and then construct the universes of descriptions (Section 2.4) and ornaments (??) inside the observational model, developing and programming within the nested model would be too complex to be worthwhile.
- A new direction is being pursued by **homotopy type theory** [The Univalent Foundations Program, 2013], which gives propositional equality a higher-dimensional homotopic interpretation and broadens its scope with the univalence axiom and higher inductive types, but the computational meaning of the theory remains an open problem. Its investigations into the higher dimensional structure of propositional equality might eventually lead to a systematic treatment of equality in dependently typed programming. For this dissertation, however, we confine ourselves to the zero-dimensional setting, in which we freely invoke **uniqueness of identity proofs**, which is definable in AGDA by pattern matching:

$$\begin{aligned} \text{UIP} &: \{A : \text{Set}\} \{x\ y : A\} (p\ q : x \equiv y) \rightarrow p \equiv q \\ \text{UIP refl refl} &= \text{refl} \end{aligned}$$

In particular, in definitions of specific categories in ??, we employ McBride’s “John Major” heterogeneous equality [McBride, 1999], which was shown by McBride to be equivalent to propositional equality equipped with uniqueness of identity proofs. Our identification of types and sets is thus consistent with the terminology of homotopy type theory: types on which identity proofs are unique (and hence lack higher dimensional structure) are indeed termed

“sets” by homotopy type theorists [The Univalent Foundations Program, 2013, Section 3.1].

With only AGDA’s intensional equality, we have to explicitly work with equality-like propositions (like pointwise equality on functions) and manage them with the help of **setoids** [Barthe et al., 2003] in this dissertation — Chapters ?? and ??, specifically. We will discuss to what extent the intensional approach works at the end of ??.

2.4 Universes and datatype-generic programming

Martin-Löf [1984b] introduced the notion of **universes** to support “large elimination”, i.e., arbitrary computation of sets, which is necessary for establishing the fourth Peano axiom that zero is not the successor of any natural number [Smith, 1988]. A universe (à la Tarski) is a set of codes for sets, which is equipped with a decoding function translating codes to sets. Large elimination is then computing an inhabitant in the universe (via ordinary elimination like `Nat-elim`) and decoding the result to a set. Another important purpose of universes is to support quantification over sets while precluding paradoxical formation of self-referential sets — AGDA, for example, has a universe hierarchy (`Set`, `Set1`, `Set2`, ...) for this purpose. From the programming perspective, however, the most interesting case is when the universe is supplied with an elimination rule [Nordström et al., 1990, Section 14.2]: allowing computation on universes turns out to roughly correspond to **datatype-generic programming** [Gibbons, 2007], whose idea is that the “shapes” of datatypes can be encoded so programs can be defined in terms of these shapes. Universes can encode such shapes, and since universes are just ordinary sets, datatype-generic programming becomes just ordinary programming in dependently typed languages [Altenkirch and McBride, 2003].

In this dissertation we not only use but also need to compute a class of inductive families which we call **index-first datatypes** [Chapman et al., 2010; Dagand and McBride, 2014], and hence need to construct a universe for them. Before that, we give a high-level introduction to these datatypes first.

2.4.1 Index-first datatypes

In AGDA, an inductive family is declared by listing all possible constructors and their types, all ending with one of the types in that inductive family. This conveys the idea that the index in the type of an inhabitant is synthesised in a bottom-up fashion following the construction of the inhabitant. For example, consider the following datatype of **vectors**, i.e., length-indexed lists:

```
data Vec (A : Set) : Nat → Set where
  []      : Vec A zero
  _::_   : A → {n : Nat} → Vec A n → Vec A (suc n)
```

The cons constructor `_::_` takes a vector at some index n and constructs a vector at $\text{suc } n$ — the final index is computed bottom-up from the index of the sub-vector. This synthetic view is not always helpful in dependently typed programming, though (see, e.g., McBride [2014]); also, it can yield redundant representation — the cons constructor for vectors has to store the index of the sub-vector, so the representation of a vector would be cluttered with all the intermediate lengths. If we switch to the opposite perspective, determining top-down from the targeted index what constructors should be supplied, then the representation can usually be significantly cleaned up — for a vector, if the index of its type is known to be $\text{suc } n$ for some n , then we know that its top-level constructor can only be cons and the index of the sub-vector must be n . To reflect this important reversal of logical order, Dagand and McBride [2014] proposed a new notation for index-first datatype declarations, in which we first list all possible patterns of (the indices of) the types in the inductive family, and then specify for each pattern which constructors it offers. Below we use a slightly more AGDA-like adaptation of the notation.

For a first example, natural numbers are declared by

```
indexfirst data Nat : Set where
  Nat ∋ zero
      | suc (n : Nat)
```

We use the keyword **indexfirst** to explicitly mark the declaration as an index-first one, distinguishing it from an AGDA datatype declaration. The only possible pattern of the datatype is `Nat`, which offers two constructors `zero` and `suc`, the

latter taking a recursive argument named n . We declare lists similarly, this time with a uniform parameter $A : \text{Set}$:

indexfirst data List ($A : \text{Set}$) : Set **where**

List $A \ni []$
 | $_{-}::_{-} (a : A) (as : \text{List } A)$

The declaration of vectors is more interesting, fully exploiting the power of index-first datatypes:

indexfirst data Vec ($A : \text{Set}$) : Nat \rightarrow Set **where**

Vec $A \text{ zero} \ni []$
 Vec $A (\text{suc } n) \ni _{::}_{-} (a : A) (as : \text{Vec } A n)$

Vec A is a family of types indexed by Nat, and we do pattern matching on the index, splitting the datatype into two cases Vec $A \text{ zero}$ and Vec $A (\text{suc } n)$ for some $n : \text{Nat}$. The first case only offers the nil constructor $[]$, and the second case only offers the cons constructor $_{::}_{-}$. Because the form of the index restricts constructor choice, the recursive structure of a vector $as : \text{Vec } A n$ must follow that of n , i.e., the number of cons nodes in as must match the number of successor nodes in n . We can also declare the bottom-up vector datatype in index-first style:

indexfirst data Vec' ($A : \text{Set}$) : Nat \rightarrow Set **where**

Vec' $A n \ni \text{nil } (neq : \text{zero} \equiv n)$
 | cons ($a : A$) ($m : \text{Nat}$) ($meq : \text{suc } m \equiv n$) ($as : \text{Vec}' A m$)

Besides the field m storing the length of the tail, two more fields neq and meq are inserted, demanding explicit equality proofs about the indices. When a vector of type Vec' $A n$ is demanded, we are “free” to choose between nil or cons regardless of the index n ; however, because of the equality constraints, we are indirectly forced into a particular choice.

Remark (*recursive families*). Vectors can also be defined as a “recursive family” [Capretta, 2000], i.e., an indexed family of types defined by induction on the index:

Vec : Set \rightarrow Nat \rightarrow Set
 Vec $A \text{ zero} = \top$
 Vec $A (\text{suc } n) = A \times \text{Vec } A n$

In general, however, inductive families are strictly more expressive than recursive families: with recursive families, the index in the type of a sub-object has to be structurally smaller, whereas inductive families do not have this restriction. An example is the BHeap datatype in ??.

Remark (*detagging*). The transformation from bottom-up vectors to top-down vectors is exactly what Brady et al.’s **detagging** optimisation [2004] does. With index-first datatypes, however, detagged representations are available directly, rather than arising from a compiler optimisation.

2.4.2 Universe construction

Now we proceed to construct a universe for index-first datatypes. An inductive family of type $I \rightarrow \text{Set}$ is constructed by taking the least fixed point of a base endofunctor on $I \rightarrow \text{Set}$. For example, to get index-first vectors, we would define a base functor (parametrised by $A : \text{Set}$)

$$\begin{aligned} \text{VecF } A &: (\text{Nat} \rightarrow \text{Set}) \rightarrow (\text{Nat} \rightarrow \text{Set}) \\ \text{VecF } A \text{ X zero} &= \top \\ \text{VecF } A \text{ X (suc } n) &= A \times \text{X } n \end{aligned}$$

and take its least fixed point. (\top is a singleton set whose only inhabitant is “ \blacksquare ”, and $_ \times _$ is cartesian product.) If we flip the order of the arguments of $\text{VecF } A$:

$$\begin{aligned} \text{VecF}' A &: \text{Nat} \rightarrow (\text{Nat} \rightarrow \text{Set}) \rightarrow \text{Set} \\ \text{VecF}' A \text{ zero} &= \lambda X \rightarrow \top \\ \text{VecF}' A \text{ (suc } n) &= \lambda X \rightarrow A \times \text{X } n \end{aligned}$$

we see that $\text{VecF}' A$ consists of two different “responses” to the index request, each of type $(\text{Nat} \rightarrow \text{Set}) \rightarrow \text{Set}$. It suffices to construct for such responses a universe

data RDesc ($I : \text{Set}$) : Set_1

with a decoding function specifying its semantics:

$$\llbracket _ \rrbracket : \{I : \text{Set}\} \rightarrow \text{RDesc } I \rightarrow (I \rightarrow \text{Set}) \rightarrow \text{Set}$$

Inhabitants of $\text{RDesc } I$ will be called **response descriptions**. A function of type $I \rightarrow \text{RDesc } I$, then, can be decoded to an endofunctor on $I \rightarrow \text{Set}$, so the type

$I \rightarrow \text{RDesc } I$ acts as a universe for index-first datatypes. We hence define

```
Desc : Set → Set1
Desc I = I → RDesc I
```

with decoding function

```
ℙ : {I : Set} → Desc I → (I → Set) → (I → Set)
ℙ D X i = ⟦ D i ⟧ X
```

Inhabitants of type $\text{Desc } I$ will be called **datatype descriptions**, or **descriptions** for short. Actual datatypes are manufactured from descriptions by the least fixed point operator:

```
data μ {I : Set} (D : Desc I) : I → Set where
  con : ℙ D (μ D) ⇒ μ D
```

where $_ \Rightarrow _$ is defined by

```
_⇒_ : {I : Set} → (I → Set) → (I → Set) → Set
_⇒_ {I} X Y = {i : I} → X i → Y i
```

Remark (*presentation of universes and their decoding*). We always present universes (e.g., RDesc) along with their decoding (e.g., $\llbracket _ \rrbracket$ for RDesc) to emphasise the meaning of the codes, even when the decoding is not logically tied to the codes (cf. Martin-Löf’s universe [1984b], which is inductive-recursive [Dybjer, 1998] and must present the universe and its decoding simultaneously). \square

Notation (*dependent pairs and AGDA records*). Cartesian product is a special case of Σ -types, also known as **dependent pairs**, which are defined in AGDA as a record:

```
record Σ (A : Set) (X : A → Set) : Set where
  constructor _,_
  field
    outl : A
    outr : X outl
  infixr 4 _,_
  open Σ
  syntax Σ A (λ a ↦ T) = Σ[a : A] T
```

An inhabitant of $\Sigma A X$ is a pair where the type of the second component depends on the first component; it is written by listing values for the fields like

record $\{ \text{outl} = a$
 $\quad ; \text{outr} = x \}$

(where $a : A$ and $x : X a$) — a cartesian product $A \times B$ is thus a special case, which is defined as $\Sigma A (\lambda _ \mapsto B)$. The **constructor** declaration gives rise to a constructor function

$_,_ : \{A : \text{Set}\} \{X : A \rightarrow \text{Set}\} \rightarrow (a : A) \rightarrow X a \rightarrow \Sigma A X$

which associates to the right when used as an infix operator because of the **infixr** statement below, and can be used in pattern matching. The two field declarations give rise to two projection functions, qualified by “ $\Sigma.$ ”:

$\Sigma.\text{outl} : \{A : \text{Set}\} \{X : A \rightarrow \text{Set}\} \rightarrow \Sigma A X \rightarrow A$
 $\Sigma.\text{outr} : \{A : \text{Set}\} \{X : A \rightarrow \text{Set}\} \rightarrow (p : \Sigma A X) \rightarrow X (\Sigma.\text{outl } p)$

We can drop the qualifications and refer to them simply as *outl* and *outr* due to the **open** statement. Finally, we can treat Σ as a binder and write, e.g., $\Sigma A X$ as $\Sigma[a : A] X a$, due to the **syntax** statement. \square

We now define the datatype of response descriptions — which determines the syntax available for defining base functors — and its decoding function:

data $\text{RDesc } (I : \text{Set}) : \text{Set}_1$ **where**
 $v : (is : \text{List } I) \rightarrow \text{RDesc } I$
 $\sigma : (S : \text{Set}) (D : S \rightarrow \text{RDesc } I) \rightarrow \text{RDesc } I$
 $\llbracket _ \rrbracket : \{I : \text{Set}\} \rightarrow \text{RDesc } I \rightarrow (I \rightarrow \text{Set}) \rightarrow \text{Set}$
 $\llbracket v \text{ is } \rrbracket X = \mathbb{P} \text{ is } X \quad \text{-- see below}$
 $\llbracket \sigma S D \rrbracket X = \Sigma[s : S] \llbracket D s \rrbracket X$

The operator \mathbb{P} computes the product of a finite number of types in a type family, whose indices are given in a list:

$\mathbb{P} : \{I : \text{Set}\} \rightarrow \text{List } I \rightarrow (I \rightarrow \text{Set}) \rightarrow \text{Set}$
 $\mathbb{P} [] X = \top$
 $\mathbb{P} (i :: is) X = X i \times \mathbb{P} is X$

Thus, in a response, given $X : I \rightarrow \text{Set}$, we are allowed to form dependent

sums (by σ) and the product of a finite number of types in X (via v , suggesting variable positions in the base functor).

Convention. We informally refer to the index part of a σ as a **field** of the datatype. Like Σ , we sometimes regard σ as a binder and write $\sigma[s : S] D s$ for $\sigma S (\lambda s \mapsto D s)$. \square

Example (natural numbers). The datatype of natural numbers is considered to be an inductive family trivially indexed by \top , so the declaration of Nat corresponds to an inhabitant of $\text{Desc } \top$.

```
data ListTag : Set where
  'nil  : ListTag
  'cons : ListTag

NatD : Desc  $\top$ 
NatD  $\blacksquare$  =  $\sigma$  ListTag  $\lambda$  { 'nil   $\mapsto v []$ 
                        ; 'cons  $\mapsto v (\blacksquare :: [])$  }
```

The index request is necessarily \blacksquare , and we respond with a field of type ListTag representing the constructor choices. A pattern-matching lambda function (which is syntactically distinguished by enclosing its body in curly braces) follows, which computes the trailing responses to the two possible values 'nil and 'cons for the field: if the field receives 'nil , then we are constructing zero, which takes no recursive values, so we write $v []$ to end this branch; if the ListTag field receives 'cons , then we are constructing a successor, which takes a recursive value at index \blacksquare , so we write $v (\blacksquare :: [])$. \square

Example (lists). The datatype of lists is parametrised by the element type. We represent parametrised descriptions simply as functions producing descriptions, so the declaration of lists corresponds to a function taking element types to descriptions.

```
ListD : Set  $\rightarrow$  Desc  $\top$ 
ListD A  $\blacksquare$  =  $\sigma$  ListTag  $\lambda$  { 'nil   $\mapsto v []$ 
                        ; 'cons  $\mapsto \sigma[- : A] v (\blacksquare :: [])$  }
```

$\text{ListD } A$ is the same as NatD except that, in the 'cons case, we use σ to insert a field of type A for storing an element. \square

Example (*vectors*). The datatype of vectors is parametrised by the element type and (nontrivially) indexed by Nat , so the declaration of vectors corresponds to

$$\begin{aligned} \text{VecD} &: \text{Set} \rightarrow \text{Desc Nat} \\ \text{VecD } A \text{ zero} &= v [] \\ \text{VecD } A (\text{suc } n) &= \sigma[- : A] v (n :: []) \end{aligned}$$

which is directly comparable to the index-first base functor VecF' at the beginning of this section. \square

There is no problem defining functions on the encoded datatypes, except that it has to be done with the raw representation. For example, list append is defined by

$$\begin{aligned} _ \# _ &: \mu (\text{ListD } A) \blacksquare \rightarrow \mu (\text{ListD } A) \blacksquare \rightarrow \mu (\text{ListD } A) \blacksquare \\ \text{con } ('nil, \blacksquare) \# bs &= bs \\ \text{con } ('cons, a, as, \blacksquare) \# bs &= \text{con } ('cons, a, as \# bs, \blacksquare) \end{aligned}$$

To improve readability, we define the following higher-level terms:

$$\begin{aligned} \text{List} &: \text{Set} \rightarrow \text{Set} \\ \text{List } A &= \mu (\text{ListD } A) \blacksquare \\ [] &: \{A : \text{Set}\} \rightarrow \text{List } A \\ [] &= \text{con } ('nil, \blacksquare) \\ _ :: _ &: \{A : \text{Set}\} \rightarrow A \rightarrow \text{List } A \rightarrow \text{List } A \\ a :: as &= \text{con } ('cons, a, as, \blacksquare) \end{aligned}$$

List append can then be rewritten in the usual form:

$$\begin{aligned} _ \# _ &: \text{List } A \rightarrow \text{List } A \rightarrow \text{List } A \\ [] \# ys &= ys \\ (x :: xs) \# ys &= x :: (xs \# ys) \end{aligned}$$

(AGDA supports such definitions of higher-level terms by “pattern synonyms”, which we do not explicitly use in this dissertation.) Later on, encoded datatypes are almost always accompanied by corresponding index-first datatype declarations, which are thought of as giving definitions of higher-level terms for type and data constructors — the terms List , $[]$, and $_ :: _$ above, for example, can be considered to be defined by the index-first declaration of lists given in Section 2.4.1. Index-first declarations will only be regarded in this dissertation

mutual

$$\begin{aligned}
& \text{fold} : \{I : \text{Set}\} \{D : \text{Desc } I\} \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \Rightarrow X) \rightarrow (\mu D \Rightarrow X) \\
& \text{fold } \{I\} \{D\} f \{i\} (\text{con } ds) = f (\text{mapFold } D (D i) f ds) \\
& \text{mapFold} : \{I : \text{Set}\} (D : \text{Desc } I) (D' : \text{RDesc } I) \rightarrow \\
& \quad \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \Rightarrow X) \rightarrow \llbracket D' \rrbracket (\mu D) \rightarrow \llbracket D' \rrbracket X \\
& \text{mapFold } D (\vee []) \quad f \blacksquare \quad = \blacksquare \\
& \text{mapFold } D (\vee (i :: is)) f (d , ds) = \text{fold } f d , \text{mapFold } D (\vee is) f ds \\
& \text{mapFold } D (\sigma S D') \quad f (s , ds) = s , \text{mapFold } D (D' s) f ds
\end{aligned}$$
Figure 2.1 Definition of the datatype-generic *fold* operator.

as informal hints at how encoded datatypes are presented at a higher level; we do not give a formal treatment of the elaboration process from index-first declarations to corresponding descriptions and definitions of higher-level terms. (One such treatment was given by Dagand and McBride [2012].)

Direct function definitions by pattern matching work fine for individual datatypes, but when we need to define operations and to state properties for all the datatypes encoded by the universe, it is necessary to have a generic *fold* operator parametrised by descriptions:

$$\text{fold} : \{I : \text{Set}\} \{D : \text{Desc } I\} \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \Rightarrow X) \rightarrow (\mu D \Rightarrow X)$$

The implementation of *fold*, shown in Figure 2.1, is adapted for our two-level universe from those in McBride’s original work [2011]. As McBride, we would have wished to define *fold* by

$$\begin{aligned}
& \text{fold} : \{I : \text{Set}\} \{D : \text{Desc } I\} \{X : I \rightarrow \text{Set}\} \rightarrow (\mathbb{F} D X \Rightarrow X) \rightarrow (\mu D \Rightarrow X) \\
& \text{fold } \{I\} \{D\} f \{i\} (\text{con } ds) = f (\text{mapRD } (D i) (\text{fold } f) ds)
\end{aligned}$$

where the functorial mapping *mapRD* on response structures is defined by

$$\begin{aligned}
& \text{mapRD} : \{I : \text{Set}\} (D : \text{RDesc } I) \rightarrow \\
& \quad \{X Y : I \rightarrow \text{Set}\} (g : X \Rightarrow Y) \rightarrow \llbracket D \rrbracket X \rightarrow \llbracket D \rrbracket Y \\
& \text{mapRD } (\vee []) \quad g \blacksquare \quad = \blacksquare \\
& \text{mapRD } (\vee (i :: is)) g (x , xs) = g x , \text{mapRD } (\vee is) g xs \\
& \text{mapRD } (\sigma S D) \quad g (s , xs) = s , \text{mapRD } (D s) g xs
\end{aligned}$$

AGDA does not see that this definition of *fold* is terminating, however, since the termination checker does not expand the definition of *mapRD* to see that *fold f* is applied to structurally smaller arguments. To make termination obvious to AGDA, we instead define *fold* mutually recursively with *mapFold*, which is *mapRD* specialised by fixing its argument *g* to *fold f*.

Example (*list length*). The function $\text{length} : \{A : \text{Set}\} \rightarrow \text{List } A \rightarrow \text{Nat}$ can be defined in terms of the datatype-generic *fold*:

$$\begin{aligned} \text{length} &: \{A : \text{Set}\} \rightarrow \text{List } A \rightarrow \text{Nat} \\ \text{length} &= \text{fold length-alg} \end{aligned}$$

where the algebra *length-alg* is defined by

$$\begin{aligned} \text{length-alg} &: \{A : \text{Set}\} \rightarrow \mathbb{F} (\text{ListD } A) (\text{const Nat}) \Rightarrow \text{const Nat} \\ \text{length-alg} ('nil \quad , \quad \blacksquare) &= \text{zero} \\ \text{length-alg} ('cons \, a \, n \, \blacksquare) &= \text{suc } n \end{aligned}$$

(The function *const* is defined as $\lambda x \mapsto \lambda y \mapsto x$.) □

It is helpful to form a two-dimensional image of our datatype manufacturing scheme: We manufacture a datatype by first defining a base functor, and then recursively duplicating the functorial structure by taking its least fixed point. The shape of the base functor can be imagined to stretch horizontally, whereas the recursive structure generated by the least fixed point grows vertically. This image works directly when the recursive structure is linear, like lists. (Otherwise one resorts to the abstraction of functor composition.) For example, we can typeset a list two-dimensionally like

$$\begin{array}{l} \text{con } ('cons \, a \, \\ \text{con } ('cons \, b \, \\ \text{con } ('nil \quad , \\ \quad \blacksquare) \, \blacksquare) \, \blacksquare) \end{array}$$

Ignoring the last line of trailing \blacksquare 's, things following *con* on each line — including constructor tags and list elements — are shaped by the base functor of lists, whereas the *con* nodes — aligned vertically — are generated by the least fixed point.

Remark (*first-order vs higher-order representation*). The functorial structures gen-

erated by descriptions are strongly reminiscent of **indexed containers** [Morris, 2007, Chapter 8]; this will be explored and exploited in ???. For now, it is enough to mention that we choose to stick to a first-order datatype manufacturing scheme, i.e., the datatypes we manufacture with descriptions use finite product types rather than dependent function types for branching, but it is easy to switch to a higher-order representation that is even closer to indexed containers (allowing infinite branching) by storing in v a collection of I -indices indexed by an arbitrary set S :

$$v : (S : \text{Set}) (f : S \rightarrow I) \rightarrow \text{RDesc } I$$

whose semantics is defined in terms of dependent functions:

$$\llbracket v \ S \ f \rrbracket X = (s : S) \rightarrow X (f \ s)$$

The reason for choosing to stick to first-order representations is merely to obtain a simpler equality for the manufactured datatypes (AGDA’s default equality would suffice); the examples of manufactured datatypes in this dissertation are all finitely branching and do not require the power of higher-order representation anyway. This choice, however, does complicate some subsequent datatype-generic definitions (e.g., ornaments in ???). It would probably be helpful to think of the parts involving v and \mathbb{P} in these definitions as specialisations of higher-order representations to first-order ones. \square

2.5 Externalism and internalism

The use of “such that” to describe objects that have certain properties is universal in mathematics. If the objects in question have type A , then objects with certain properties form a subset of A , and using “such that” to describe such objects means that the subset is formed by specifying a suitable predicate on A . In type theory, this can be modelled by Σ -types. In a type $\Sigma A P$, when A is interpreted as a ground set and P as a predicate on A , an inhabitant of $\Sigma A P$ is an inhabitant a of A paired with a proof that $P \ a$ holds. When programs are the objects we reason about, this style naturally suggests a distinction between programs and proofs: programs are written in the first place, and proofs are conducted afterwards with reference to existing programs and do not interfere

with their execution. This conception underlies many developments in type theory and theorem proving. For example: Luo [1994] consistently argued that proofs should not be identified with programs, one of the reasons being that logic should be regarded as independent from the objects being reasoned about. A type theory of subsets was given by Nordström et al. [1990] to suppress the second component — i.e., the proof part — of Σ -types. The proof assistant Coq [Bertot and Castéran, 2004] uses a type-theoretic foundation [Coquand and Huet, 1988; Coquand and Paulin-Mohring, 1990] which distinguishes programs and proofs, and proofs are written in a tactic-based language, differently from programs; it is also famous for the ability to extract executable portions of proof scripts into programs [Paulin-Mohring, 1989; Letouzey, 2003], as used in the COMPCERT project developing a verified C compiler [Leroy, 2009], for example.

On the other hand, having unified programs and proofs in type theory (Section 2.1), it seems a pity if the unification is not exploited to a deeper level. In Dijkstra’s proposal for program correctness by construction, proofs and programs should be conceived “hand in hand” [Dijkstra, 1976], and the unification of programs and proofs brings us unprecedentedly closer to this ideal, since we can start thinking about programs that also serve as correctness proofs themselves. The dependently typed programming community has been exploring the use of inductive families not only for defining predicates on data (like $_ \leq_{\mathbb{N}} _$) but also for representing data with embedded constraints (like Vec). Programs manipulating such datatypes would also deal with the embedded constraints and are thus correct by construction. Vector append is a classic (albeit somewhat trivial) example: Defining addition on natural numbers as

$$\begin{aligned} _ + _ &: \text{Nat} \rightarrow \text{Nat} \rightarrow \text{Nat} \\ \text{zero} \quad + \, n &= n \\ (\text{suc } m) + n &= \text{suc } (m + n) \end{aligned}$$

vector append is then defined by

$$\begin{aligned} _ \mathbin{++} _ &: \{A : \text{Set}\} \{m \, n : \text{Nat}\} \rightarrow \text{Vec } A \, m \rightarrow \text{Vec } A \, n \rightarrow \text{Vec } A \, (m + n) \\ [] \quad \mathbin{++} \, ys &= ys \\ (x :: xs) \mathbin{++} \, ys &= x :: (xs \mathbin{++} \, ys) \end{aligned}$$

The program for vector append looks exactly like the one for list append except

for the more informative type, which makes it possible for the program to meet its specification — the length of the result of `append` should be the sum of the lengths of the two input lists — by construction. For list `append`, whose type uses the plain list datatype, we need to separately produce the following proof:

```

append-length :
  {A : Set} (xs ys : List A) → length (xs ++ ys) ≡ length xs + length ys
append-length []      ys = refl
append-length (x :: xs) ys = cong suc (append-length xs ys)

```

But by switching to the vector datatype, the proof dissolves into the typing of the program and needs no separate handling. This is possible because list `append` and the proof *append-length* share the same structure; consequently, by careful type design, the vector `append` program alone is able to carry both of them simultaneously. We propose to call this programming style **internalism**, suggesting that proofs are internalised in programs, while the traditional proving-after-programming style is called **externalism**. Externalism is necessary when we wish to show that an existing program satisfies additional properties, especially when the proofs are complicated and do not follow the structure of the program. On the other hand, writing a (simply typed) program and an externalist proof following the structure of the program is like stating the same thing twice: even though the programmer knows the meaning of the program, they have to first state the meaningless symbol manipulation aspect and then explain its meaning via a separate proof, doubling the effort. In contrast, internalism is programming with informative datatypes so as to give precise description of meaningful computations, so explanations via separate proofs become unnecessary. As McBride [2004] aptly put it, internalism makes programs and their explanations via proofs “not merely coexist but coincide”. In addition, by encoding meanings in datatypes, semantic considerations are (at least partially) reduced to syntax and can be aided mechanically — AGDA’s interactive development environment (Section 2.2.1) is one form of such aid. With interactive development, internalist types not only passively rule out nonsensical programs, but can actively provide helpful information to guide program construction. We will see several examples of such **type-directed programming** in this dissertation.

Internalism comes with its own problems, however. For one: internalist type design is difficult, yet there is almost no effective guideline or discipline for such design. Carelessly designed internalist types can lead to less natural programs. A simple example is when we switch the order of the arguments of the addition in the type of vector append,

$$\begin{aligned} _ \mathbin{++} _ &: \{A : \text{Set}\} \{m\ n : \text{Nat}\} \rightarrow \text{Vec } A\ m \rightarrow \text{Vec } A\ n \rightarrow \text{Vec } A\ (n + m) \\ [] &\mathbin{++} ys = \text{subst } (\text{Vec } A) \{n \equiv n + \text{zero}\}_0\ ys \\ (x :: xs) \mathbin{++} ys &= \text{subst } (\text{Vec } A) \{\text{suc } (n + m) \equiv n + \text{suc } m\}_1\ (x :: (xs \mathbin{++} ys)) \end{aligned}$$

we would be forced to perform two type-casts that could have been avoided. Not only does the program look ugly, but this can also cause a cascade effect when we need to write other programs whose types depend on this program (e.g., externalist proofs about it), which would have to deal with the type-casts. McBride [2012] gave a more interesting example: to prove the polynomial testing principle (two polynomial functions of degree n are pointwise equal if they agree at $n + 1$ different points), McBride started with a datatype of encodings of polynomial functions indexed by degree but, after trying to program with the datatype, quickly found out that the datatype should instead be indexed by an arbitrary upper bound of the degree so a relaxed form of the polynomial testing principle can be naturally programmed (two polynomial functions of degree at most n are pointwise equal if they agree at $n + 1$ different points). Manageability is another issue, especially when the only tool we have is the primitive language of datatype declarations: writing a datatype declaration with a sophisticated property internalised is comparable to programming a sophisticated algorithm in assembly language, and understanding the meaning of a complicated datatype declaration takes thorough reading and some inductive guessing and reasoning. In short, the complexity of internalist types has made type design a nontrivial programming problem, and we are in serious lack of type-level programming support.

Internalist library design also poses a problem: Since internalism requires differently indexed versions of the same data structure, an internalist library should provide more or less the same set of operations for all possible variants of the data structure. Without a way to manage these formally unrelated datatypes and operations modularly, an ad hoc library would need to duplicate

the same structure and logic for all the variants and becomes hard to expand. For example, suppose that we have constructed a library for lists that include vectors, ordered lists, and ordered vectors, and now wish to add a new flavour of lists, say, association lists indexed with the list of keys. Operations need to be reimplemented not only for such key-indexed lists, but also for key-indexed vectors, ordered key-indexed lists, and ordered key-indexed vectors, even though key-indexing is the only new feature. An ideal structure for such a library would be having a separate module for each of the properties about length, ordering, and key-indexing. These modules can be developed independently, and there would be a way to assemble components in these modules at will — for example, ordered vectors and related operations would be synthesised from the components in the modules about length and ordering. This ideal library structure calls for some form of composability of internalist datatypes and operations.

Composability has never been a problem for externalism, however. In an externalist list library, we would have only one basic list datatype and several predicates on lists about length, ordering, key-indexing, etc. Lists are “promoted” to vectors, ordered lists, or ordered vectors by simply pairing the list datatype with the length predicate, the ordering predicate, or the pointwise conjunction of the two predicates, respectively. Common operations are implemented for basic lists only, and their properties regarding length or ordering are proved independently and invoked when needed. Can we somehow introduce this beneficial composability to internalism as well? The answer is yes, because there are isomorphisms between externalist and internalist datatypes to be exploited.

To illustrate, let us go through a small case study about upgrading the *insert* function on lists (Section 2.2.2) for vectors, ordered lists, and ordered vectors. The externalist would define vectors as a Σ -type,

$$\begin{aligned} \text{ExtVec} &: \text{Set} \rightarrow \text{Nat} \rightarrow \text{Set} \\ \text{ExtVec } A \ n &= \Sigma[xs : \text{List } A] \ \text{length } xs \equiv n \end{aligned}$$

prove that *insert* increases the length of a list by one,

$$\begin{aligned} \text{insert-length} &: (y : \text{Val}) \{n : \text{Nat}\} (xs : \text{List } \text{Val}) \rightarrow \\ &\quad \text{length } xs \equiv n \rightarrow \text{length } (\text{insert } y \ xs) \equiv \text{suc } n \end{aligned}$$

and define insertion on vectors as

$$\begin{aligned} \text{insert}_{EV} &: \text{Val} \rightarrow \{n : \text{Nat}\} \rightarrow \text{ExtVec Val } n \rightarrow \text{ExtVec Val } (\text{succ } n) \\ \text{insert}_{EV} \ y \ (xs, \text{len}) &= \text{insert } y \ xs, \text{insert-length } y \ xs \ \text{len} \end{aligned}$$

which processes the list and the length proof by *insert* and *insert-length* respectively. Similarly for ordered lists (indexed with a lower bound), the externalist would use the Σ -type

$$\begin{aligned} \text{ExtOrdList} &: \text{Val} \rightarrow \text{Set} \\ \text{ExtOrdList } b &= \Sigma[xs : \text{List Val}] \ \text{Ordered } b \ xs \end{aligned}$$

where the Ordered predicate is defined by

$$\begin{aligned} \text{indexfirst data Ordered} &: \text{Val} \rightarrow \text{List Val} \rightarrow \text{Set} \text{ where} \\ \text{Ordered } b \ [] &\ni \text{nil} \\ \text{Ordered } b \ (x :: xs) &\ni \text{cons } (\text{leq} : b \leq x) \ (\text{ord} : \text{Ordered } x \ xs) \end{aligned}$$

Insertion on ordered lists is then

$$\begin{aligned} \text{insert}_{EO} &: (y : \text{Val}) \ \{b : \text{Val}\} \rightarrow \text{ExtOrdList } b \rightarrow \\ &\quad \{b' : \text{Val}\} \rightarrow b' \leq y \rightarrow b' \leq b \rightarrow \text{ExtOrdList } b' \\ \text{insert}_{EO} \ y \ (xs, \text{ord}) \ b' \leq y \ b' \leq b &= \text{insert } y \ xs, \text{insert-ordered } y \ xs \ \text{ord } b' \leq y \ b' \leq b \end{aligned}$$

where *insert-ordered* proves that *insert* preserves ordering:

$$\begin{aligned} \text{insert-ordered} &: (y : \text{Val}) \ \{b : \text{Val}\} \ (xs : \text{List Val}) \rightarrow \text{Ordered } b \ xs \rightarrow \\ &\quad \{b' : \text{Val}\} \rightarrow b' \leq y \rightarrow b' \leq b \rightarrow \text{Ordered } b' \ (\text{insert } y \ xs) \end{aligned}$$

Now the externalist has arrived at a modular list library (albeit a tiny one), which contains

- a basic module consisting of the basic list datatype and insertion on basic lists, and
- two independent upgrading modules about length and ordering, each consisting of a predicate on lists and a related proof about insertion.

It is easy to mix all three modules and get ordered vectors and insertion on them. The Σ -type uses the pointwise conjunction of the two predicates,

$$\begin{aligned} \text{ExtOrdVec} &: \text{Val} \rightarrow \text{Nat} \rightarrow \text{Set} \\ \text{ExtOrdVec } b \ n &= \Sigma[xs : \text{List Val}] \ \text{Ordered } b \ xs \times \text{length } xs \equiv n \end{aligned}$$

and insertion simply uses *insert-ordered* and *insert-length* to process the two proofs bundled with a list:

$$\begin{aligned}
 \text{insert}_{\text{EOV}} &: (y : \text{Val}) \{b : \text{Val}\} \{n : \text{Nat}\} \rightarrow \text{ExtOrdVec } b \ n \rightarrow \\
 &\quad \{b' : \text{Val}\} \rightarrow b' \leq y \rightarrow b' \leq b \rightarrow \text{ExtOrdVec } b' \ (\text{suc } n) \\
 \text{insert}_{\text{EOV}} \ y \ (xs, \text{ord}, \text{len}) \ b' \leq y \ b' \leq b &= \text{insert} \quad y \ xs, \\
 &\quad \text{insert-ordered } y \ xs \ \text{ord} \ b' \leq y \ b' \leq b, \\
 &\quad \text{insert-length } y \ xs \ \text{len}
 \end{aligned}$$

(Note that the variable of type $b' \leq y$ (with spaces around ' \leq ') is named $b' \leq y$ (without spaces) — a subtle but useful convention.) This is the kind of library we are looking for, except that the types are all externalist. The externalist and internalist types are not unrelated, however. For example, internalist and externalist vectors are related by the indexed family of **conversion isomorphisms**:

$$\text{Vec-iso } A : (n : \text{Nat}) \rightarrow \text{Vec } A \ n \cong \text{ExtVec } A \ n$$

Fixing $n : \text{Nat}$, the left-to-right direction of the isomorphism

$$\text{Iso.to } (\text{Vec-iso } A \ n) : \text{Vec } A \ n \rightarrow \Sigma [xs : \text{List } A] \ \text{length } xs \equiv n$$

computes the underlying list of a vector and a proof that the list has length n , and the right-to-left direction

$$\text{Iso.from } (\text{Vec-iso } A \ n) : (\Sigma [xs : \text{List } A] \ \text{length } xs \equiv n) \rightarrow \text{Vec } A \ n$$

promotes a list to a vector when there is a proof that the list has length n . (The definition of \cong , which is actually an instance of a record datatype `Iso`, appears in ??.) To get insertion on internalist vectors, we convert the input vector to its externalist representation, make $\text{insert}_{\text{EV}}$ do the work, and convert the result back to the internalist representation; more formally, the operation

$$\text{insert}_V : \text{Val} \rightarrow \{n : \text{Nat}\} \rightarrow \text{Vec } \text{Val} \ n \rightarrow \text{Vec } \text{Val} \ (\text{suc } n)$$

is defined by the commutative diagram:

$$\begin{array}{ccc}
 \text{Vec Val } n & \xrightarrow{\text{insert}_V y} & \text{Vec Val (suc } n) \\
 \downarrow \text{Iso.to (Vec-iso Val } n) & & \uparrow \text{Iso.from (Vec-iso Val (suc } n)) \\
 \Sigma[xs : \text{List Val}] & \xrightarrow[\text{insert-length } y _]{\text{insert } y *} & \Sigma[xs : \text{List Val}] \\
 \text{length } xs \equiv n & & \text{length } xs \equiv \text{suc } n
 \end{array}$$

(The $_*_$ operator is defined by $(f * g) (x, y) = (f x, g y)$, and the underscore leaves out the term for AGDA to infer.) Similarly, we can get insertion for internalist ordered lists and ordered vectors (definitions to appear in ??) from the externalist library by suitable conversion isomorphisms of the same form as *Vec-iso*. It is due to these conversion isomorphisms between internalist and externalist representations that we can **analyse** internalist datatypes into externalist components, which can then be modularly processed. This analysis of internalist datatypes and its application to modular library structuring is explored in ?? (in particular, the insertion example is concluded in ??).

The interconnection between internalism and externalism (in the form of conversion isomorphisms) also sheds some light on supporting internalist type design. The **synthetic** direction of the interconnection goes from basic types and predicates to internalist types. It is conceivable that, for externalist predicates of some particular form, we can manufacture corresponding internalist types on the other side of the interconnection. The externalist side of the interconnection is usually kept non-dependently typed, so it is possible to use existing non-dependently typed calculi to derive suitable externalist predicates from specifications, which are then used to manufacture datatypes on the internalist side for type-directed programming. ?? presents one such approach, using relational calculus [Bird and de Moor, 1997] as a design language for internalist datatypes. Rather than improvising internalist types and hoping that they will work, we write specifications in the form of relational programs, which are amenable to algebraic transformation and can be much more concise and readable than the language of datatype declarations, making it easier to arrive at helpful and comprehensible internalist types.

To sum up: While internalism offers type-directed program construction and reduces the burden of producing correctness proofs, additional or more complicated properties of existing programs can only be established by externalism, which naturally gives rise to a modular organisation. Both internalism and externalism are here to stay, since the two styles serve different purposes and have their own advantages and disadvantages. Exploiting their interconnection can lead to interesting programming patterns, which the rest of this dissertation explores.

Bibliography

- Thorsten ALTENKIRCH and Conor McBRIDE [2003]. Generic programming within dependently typed programming. In *IFIP TC2/WG2.1 Working Conference on Generic Programming*, pages 1–20. Kluwer, B.V. doi: 10.1007/978-0-387-35672-3_1. ↗ page 18
- Thorsten ALTENKIRCH, Conor McBRIDE, and Wouter SWIERSTRA [2007]. Observational equality, now! In *Programming Languages meets Program Verification, PLPV’07*, pages 57–68. ACM. doi: 10.1145/1292597.1292608. ↗ page 17
- Gilles BARTHE, Venanzio CAPRETTA, and Olivier PONS [2003]. Setoids in type theory. *Journal of Functional Programming*, 13(2):261–293. doi: 10.1017/S0956796802004501. ↗ page 18
- Yves BERTOT and Pierre CASTÉLAN [2004]. *Interactive Theorem Proving and Program Development — Coq’Art: The Calculus of Inductive Constructions*. Springer-Verlag. ISBN: 978-3540208549. ↗ page 29
- Richard BIRD and Oege DE MOOR [1997]. *Algebra of Programming*. Prentice-Hall. ISBN: 978-0135072455. ↗ page 35
- Edwin BRADY, Conor McBRIDE, and James MCKINNA [2004]. Inductive families need not store their indices. In *Types for Proofs and Programs*, volume 3085 of *Lecture Notes in Computer Science*, pages 115–129. Springer-Verlag. doi: 10.1007/978-3-540-24849-1_8. ↗ page 21
- Venanzio CAPRETTA [2000]. Recursive families of inductive types. In *Theorem Proving in Higher Order Logics*, volume 1869 of *Lecture Notes in Computer Science*, pages 73–89. Springer-Verlag. doi: 10.1007/3-540-44659-1_5. ↗ page 20

- James CHAPMAN, Pierre-Évariste DAGAND, Conor McBRIDE, and Peter MORRIS [2010]. The gentle art of levitation. In *International Conference on Functional Programming*, ICFP'10, pages 3–14. ACM. doi: 10.1145/1863543.1863547. ↱ page 18
- R. L. CONSTABLE, S. F. ALLEN, H. M. BROMLEY, W. R. CLEAVELAND, J. F. CREMER, R. W. HARPER, D. J. HOWE, T. B. KNOBLOCK, N. P. MENDLER, P. PANANGADEN, J. T. SASAKI, and S. F. SMITH [1985]. *Implementing Mathematics with the Nuprl Proof Development System*. Prentice-Hall. ISBN: 978-1468059106. ↱ page 16
- Thierry COQUAND and Gérard HUET [1988]. The Calculus of Constructions. *Information and Computation*, 76(2–3):95–120. doi: 10.1016/0890-5401(88)90005-3. ↱ page 29
- Thierry COQUAND and Christine PAULIN-MOHRING [1990]. Inductively defined types. In *International Conference on Computer Logic*, volume 417 of *Lecture Notes in Computer Science*, pages 50–66. Springer-Verlag. doi: 10.1007/3-540-52335-9_47. ↱ page 29
- Pierre-Évariste DAGAND and Conor McBRIDE [2012]. Elaborating inductive definitions. arXiv:1210.6390. ↱ page 26
- Pierre-Évariste DAGAND and Conor McBRIDE [2014]. Transporting functions across ornaments. *Journal of Functional Programming*, 24(2–3):316–383. doi: 10.1017/S0956796814000069. ↱ pages 18 and 19
- Edsger W. DIJKSTRA [1976]. *A Discipline of Programming*. Prentice-Hall. ISBN: 978-0132158718. ↱ page 29
- Michael DUMMETT [2000]. *Elements of Intuitionism*. Oxford University Press, second edition. ISBN: 978-0198505242. ↱ pages 2 and 13
- Peter DYBJER [1994]. Inductive families. *Formal Aspects of Computing*, 6(4):440–465. doi: 10.1007/BF01211308. ↱ page 6
- Peter DYBJER [1998]. A general formulation of simultaneous inductive-recursive definitions in type theory. *Journal of Symbolic Logic*, 65(2):525–549. doi: 10.2307/2586554. ↱ page 22

- Jeremy GIBBONS [2007]. Datatype-generic programming. In *Spring School on Datatype-Generic Programming*, volume 4719 of *Lecture Notes in Computer Science*, pages 1–71. Springer-Verlag. doi: 10.1007/978-3-540-76786-2_1. ↱ page 18
- Healfdene GOGUEN, Conor McBRIDE, and James McKINNA [2006]. Eliminating dependent pattern matching. In *Algebra, Meaning, and Computation*, volume 4060 of *Lecture Notes in Computer Science*, pages 521–540. Springer-Verlag. doi: 10.1007/11780274_27. ↱ page 7
- Arend HEYTING [1971]. *Intuitionism: An Introduction*. Amsterdam: North-Holland Publishing, third revised edition. ISBN: 978-0720422399. ↱ page 2
- Paul HUDAK, John HUGHES, Simon PEYTON JONES, and Philip WADLER [2007]. A history of Haskell: Being lazy with class. In *History of Programming Languages*, HOPL-III, pages 1–55. ACM. doi: 10.1145/1238844.1238856. ↱ page 6
- Xavier LEROY [2009]. A formally verified compiler back-end. *Journal of Automated Reasoning*, 43(4):363–446. doi: 10.1007/s10817-009-9155-4. ↱ page 29
- Pierre LETOUZEY [2003]. A new extraction for Coq. In *Types for Proofs and Programs*, volume 2646 of *Lecture Notes in Computer Science*, pages 200–219. Springer-Verlag. doi: 10.1007/3-540-39185-1_12. ↱ page 29
- Zhaohui LUO [1994]. *Computation and Reasoning: A Type Theory for Computer Science*. Clarendon Press. ISBN: 978-0198538356. ↱ pages 14 and 29
- Per MARTIN-LÖF [1975]. An intuitionistic theory of types: Predicative part. In *Logic Colloquium '73*, volume 80 of *Studies in Logic and the Foundations of Mathematics*, pages 73–118. Elsevier B.V. doi: 10.1016/S0049-237X(08)71945-1. ↱ page 2
- Per MARTIN-LÖF [1984a]. Constructive mathematics and computer programming. *Philosophical Transactions of the Royal Society of London*, 312(1522):501–518. doi: 10.1098/rsta.1984.0073. ↱ page 5
- Per MARTIN-LÖF [1984b]. *Intuitionistic Type Theory*. Bibliopolis, Napoli. ↱ pages 2, 18, and 22

- Per MARTIN-LÖF [1987]. Truth of a proposition, evidence of a judgement, validity of a proof. *Synthese*, 73(3):407–420. doi: 10.1007/BF00484985. ↗ page 3
- Conor McBRIDE [1999]. *Dependently Typed Functional Programs and their Proofs*. Ph.D. thesis, University of Edinburgh. ↗ pages 7 and 17
- Conor McBRIDE [2004]. Epigram: Practical programming with dependent types. In *Advanced Functional Programming*, volume 3622 of *Lecture Notes in Computer Science*, pages 130–170. Springer-Verlag. doi: 10.1007/11546382_3. ↗ pages 7 and 30
- Conor McBRIDE [2011]. Ornamental algebras, algebraic ornaments. URL: <https://personal.cis.strath.ac.uk/conor.mcbride/pub/OAA0/LitOrn.pdf>. ↗ page 26
- Conor McBRIDE [2012]. A polynomial testing principle. URL: <https://personal.cis.strath.ac.uk/conor.mcbride/PolyTest.pdf>. ↗ page 31
- Conor McBRIDE [2014]. How to keep your neighbours in order. URL: <https://personal.cis.strath.ac.uk/conor.mcbride/Pivotal.pdf>. ↗ page 19
- Conor McBRIDE and James McKINNA [2004]. The view from the left. *Journal of Functional Programming*, 14(1):69–111. doi: 10.1017/S0956796803004829. ↗ pages 7, 10, 11, and 12
- Lambert MEERTENS [1992]. Paramorphisms. *Formal Aspects of Computing*, 4(5):413–424. doi: 10.1007/BF01211391. ↗ page 6
- Erik MEIJER, Maarten FOKKINGA, and Ross PATERSON [1991]. Functional programming with bananas, lenses, envelopes and barbed wire. In *Functional Programming Languages and Computer Architecture*, number 523 in *Lecture Notes in Computer Science*, pages 124–144. Springer-Verlag. doi: 10.1007/3540543961_7. ↗ page 6
- Peter MORRIS [2007]. *Constructing Universes for Generic Programming*. Ph.D. thesis, University of Nottingham. ↗ page 28

- Bengt NORDSTRÖM, Kent PETERSON, and Jan M. SMITH [1990]. *Programming in Martin-Löf's Type Theory: An Introduction*. Oxford University Press. ISBN: 978-0198538141. ↗ pages 2, 16, 18, and 29
- Ulf NORELL [2007]. *Towards a Practical Programming Language based on Dependent Type Theory*. Ph.D. thesis, Chalmers University of Technology. ↗ page 11
- Christine PAULIN-MOHRING [1989]. Extracting F_ω 's programs from proofs in the Calculus of Constructions. In *Principles of Programming Languages*, pages 89–104. ACM. doi: 10.1145/75277.75285. ↗ page 29
- Simon PEYTON JONES [1997]. A new view of guards. URL: <http://research.microsoft.com/en-us/um/people/simonpj/Haskell/guards.html>. ↗ page 10
- Tim SHEARD and Nathan LINGER [2007]. Programming in Ω mega. In *Central-European Functional Programming School*, volume 5161 of *Lecture Notes in Computer Science*, pages 158–227. Springer-Verlag. doi: 10.1007/978-3-540-88059-2_5. ↗ page 16
- Jan M. SMITH [1988]. The independence of Peano's fourth axiom from Martin-Löf's type theory without universes. *Journal of Symbolic Logic*, 53(3):840–845. doi: 10.2307/2274575. ↗ page 18
- Thomas STREICHER [1993]. Investigations into intensional type theory. Habilitation thesis, Ludwig Maximilian Universität. ↗ pages 7 and 14
- THE UNIVALENT FOUNDATIONS PROGRAM [2013]. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Institute for Advanced Study, Princeton. URL: <http://homotopytypetheory.org/book/>. ↗ pages 17 and 18
- Philip WADLER [1987]. Views: A way for pattern matching to cohabit with data abstraction. In *Principles of Programming Languages*, pages 307–313. ACM. doi: 10.1145/41625.41653. ↗ page 11

Todo list