

Chapter 4

Categorical organisation of the ornament–refinement framework

?? left some obvious holes in the theory of ornaments. For instance:

- When it comes to composition of ornaments, the following **sequential composition** is probably the first that comes to mind (rather than parallel composition), which is evidence that the ornamental relation is transitive:

$$\begin{aligned} _ \odot _ &: \{I J K : \text{Set}\} \{e : J \rightarrow I\} \{f : K \rightarrow J\} \rightarrow \\ &\{D : \text{Desc } I\} \{E : \text{Desc } J\} \{F : \text{Desc } K\} \rightarrow \\ &\text{Orn } e D E \rightarrow \text{Orn } f E F \rightarrow \text{Orn } (e \circ f) D F \end{aligned}$$

-- definition in Figure 4.4

Correspondingly, we expect that

$$\text{forget } (O \odot P) \quad \text{and} \quad \text{forget } O \circ \text{forget } P$$

are extensionally equal. That is, the sequential compositional structure of ornaments corresponds to the compositional structure of forgetful functions. We wish to state such correspondences in concise terms.

- While parallel composition of ornaments has a sensible definition (??), it is defined by case analysis at the microscopic level of individual fields. Such a microscopic definition is difficult to comprehend, and so are any subsequent definitions and proofs. It is desirable to have a macroscopic characterisation

of parallel composition, so the nature of parallel composition is immediately clear, and subsequent definitions and proofs can be done in a more abstract manner.

- The ornamental conversion isomorphisms (??) and the modularity isomorphisms (??) were left unimplemented. Both sets of isomorphisms are about the optimised predicates (??), which are defined in terms of parallel composition with singleton ornaments (??). We thus expect that the existence of these isomorphisms can be explained in terms of properties of parallel composition and singleton ornaments.

A lightweight organisation of the ornament–refinement framework in basic category theory [Mac Lane, 1998] can help to fill up all these holes. In more detail:

- Categories and functors are abstractions for compositional structures and structure-preserving maps between them. Facts about translations between ornaments, refinements, and functions can thus be neatly organised under the categorical language (Section 4.1).
- Parallel composition merges two compatible ornaments and does nothing more; in other words, it computes the least informative ornament that contains the information of both ornaments. Characterisation of such **universal constructions** is a speciality of category theory; in our case, parallel composition can be shown to satisfy some **pullback properties** (Section 4.2).
- Universal constructions are unique up to isomorphism, so it is convenient to establish isomorphisms about universal constructions. The pullback properties of parallel composition can thus help to construct the ornamental conversion isomorphisms (Section 4.3.1) and the modularity isomorphisms (Section 4.3.2).




Figure 4.1 Categories and functors for the ornament-refinement framework.

4.1 Categories and functors

A first approximation of a category is a (directed multi-) **graph**, which consists of a set of objects (nodes) and a collection of sets of morphisms (edges) indexed with their source and target objects:

record Graph $\{l\ m : \text{Level}\} : \text{Set}(\text{succ}(l \sqcup m))$ **where**
field

Object : Set *l*

$_ \Rightarrow _$: *Object* \rightarrow *Object* \rightarrow Set *m*

For example, the underlying graph of the category **IFUN** of (small) sets and (total) functions is just

IFUN-graph : Graph

IFUN-graph = **record** $\{ \text{Object} = \text{Set}$
 $\ ; _ \Rightarrow _ = \lambda A\ B \mapsto A \rightarrow B \}$

A category is a graph whose morphisms are equipped with a monoid-like compositional structure — there is a morphism composition operator of type

$_ \cdot _ : \{X\ Y\ Z : \text{Object}\} \rightarrow (Y \Rightarrow Z) \rightarrow (X \Rightarrow Y) \rightarrow (X \Rightarrow Z)$

which has identities and is associative.

Syntactic remark (*universe polymorphism*). Many definitions in this chapter

(like `Graph` above) employ Agda’s universe polymorphism [Harper and Pollack, 1991], so the definitions can be instantiated at suitable levels of the `Set` hierarchy as needed. (For example, the type of `Fun-graph` is implicitly instantiated as `Graph {1} {1}`, since both `Set` and any $A \rightarrow B$ (where $A, B : \text{Set}$) are of type `Set1`.) We will give the first few universe-polymorphic definitions with full detail about the levels, but will later suppress the syntactic noise wherever possible. \square

Before we move on to the definition of categories, though, special attention must be paid to equality on morphisms, which is usually coarser than definitional equality — in `Fun`, for example, it is necessary to identify functions up to extensional equality (so uniqueness of morphisms in universal properties would make sense). One ad hoc way to achieve this in Agda’s intensional setting is to use **setoids** [Barthe et al., 2003] — sets with an explicitly specified equivalence relation — to represent sets of morphisms. The type of setoids can be defined as a record which contains a carrier set, an equivalence relation on the set, and the three laws for the equivalence relation:

```
record Setoid {c d : Level} : Set (suc (c  $\sqcup$  d)) where
  field
```

```
  Carrier : Set c
```

```
   $\approx$  : Carrier  $\rightarrow$  Carrier  $\rightarrow$  Set d
```

```
  refl : {x : Carrier}  $\rightarrow$  x  $\approx$  x
```

```
  sym : {x y : Carrier}  $\rightarrow$  x  $\approx$  y  $\rightarrow$  y  $\approx$  x
```

```
  trans : {x y z : Carrier}  $\rightarrow$  x  $\approx$  y  $\rightarrow$  y  $\approx$  z  $\rightarrow$  x  $\approx$  z
```

For example, we can define a setoid of functions that uses extensional equality:

```
FunSetoid : Set  $\rightarrow$  Set  $\rightarrow$  Setoid
```

```
FunSetoid A B = record { Carrier = A  $\rightarrow$  B
                        ;  $\approx$  =  $\dot{=}$ 
                        ; proofs of laws }
```

Proofs of the three laws are omitted from the presentation.

Similarly, we can define the type of categories as a record containing a set of objects, a collection of **setoids** of morphisms indexed by source and target

```

record Category {l m n : Level} : Set (suc (l ⊔ m ⊔ n)) where
  field
    Object      : Set l
    Morphism    : Object → Object → Setoid {m} {n}
    _⇒_         : Object → Object → Set m
    X ⇒ Y      = Setoid.Carrier (Morphism X Y)
    _≈_         : {X Y : Object} → (X ⇒ Y) → (X ⇒ Y) → Set n
    _≈_ {X} {Y} = Setoid._≈_ (Morphism X Y)
  field
    _·_         : {X Y Z : Object} → (Y ⇒ Z) → (X ⇒ Y) → (X ⇒ Z)
    id          : {X : Object} → (X ⇒ X)
    id-l       : {X Y : Object} (f : X ⇒ Y) →
                  id · f ≈ f
    id-r       : {X Y : Object} (f : X ⇒ Y) →
                  f · id ≈ f
    assoc      : {X Y Z W : Object} (f : Z ⇒ W) (g : Y ⇒ Z) (h : X ⇒ Y) →
                  (f · g) · h ≈ f · (g · h)
    cong-l     : {X Y Z : Object} {f g : Y ⇒ Z} (h : X ⇒ Y) →
                  f ≈ g → f · h ≈ g · h
    cong-r     : {X Y Z : Object} (h : Y ⇒ Z) {f g : X ⇒ Y} →
                  f ≈ g → h · f ≈ h · g

```

Figure 4.2 Definition of categories.

objects, the composition operator on morphisms, the identity morphisms, and the identity and associativity laws for composition. The definition is shown in Figure 4.2. Two notations are introduced to improve readability: $X \Rightarrow Y$ is defined to be the carrier set of the setoid of morphisms from X to Y , and $f \approx g$ is defined to be the equivalence between the morphisms f and g as specified by the setoid to which f and g belong. The last two laws *cong-l* and *cong-r* require morphism composition to preserve the equivalence on morphisms; they are given in this form to work better with the equational reasoning combinators commonly used in Agda (see, for example, the AoPA library [Mu et al., 2009]).

Now we can define the category $\mathbb{F}\text{UN}$ of sets and functions as

```

FUN : Category
FUN = record { Object      = Set
                ; Morphism = FunSetoid
                ;  $\_ \cdot \_$     =  $\_ \circ \_$ 
                ; id      =  $\lambda x \mapsto x$ 
                ; proofs of laws }

```

Another important category that we will make use of is $\mathbb{F}\text{AM}$, the category of indexed families of sets and indexed families of functions, which is useful for talking about componentwise structures. An object in $\mathbb{F}\text{AM}$ has type $\Sigma[I : \text{Set}] I \rightarrow \text{Set}$, i.e., it is a set I and a family of sets indexed by I ; a morphism from (J, Y) to (I, X) is a function $e : J \rightarrow I$ and a family of functions from $Y j$ to $X (e j)$ for each $j : J$.

```

FAM : Category
FAM = record
  { Object      =  $\Sigma[I : \text{Set}] I \rightarrow \text{Set}$ 
    ; Morphism =  $\lambda (J, Y) (I, X) \mapsto$  record
      { Carrier =  $\Sigma[e : J \rightarrow I] Y \Rightarrow (X \circ e)$ 
        ;  $\_ \approx \_$    =  $\lambda (e, u) (e', u') \mapsto$ 
           $(e \doteq e') \times ((j : J) \rightarrow u \{j\} \cong u' \{j\})$ 
        ; proofs of laws }
    ;  $\_ \cdot \_$       =  $\lambda (e, u) (f, v) \mapsto (e \circ f), (\lambda \{k\} \mapsto u \{f k\} \circ v \{k\})$ 

```

```

record Functor {l m n l' m' n' : Level}
  (C : Category {l} {m} {n}) (D : Category {l'} {m'} {n'}) :
  Set (l ⊔ m ⊔ n ⊔ l' ⊔ m' ⊔ n') where
  field
    object      : Object C → Object D
    morphism    : {X Y : Object C} → X ⇒C Y → object X ⇒D object Y
    equiv-preserving : {X Y : Object C} {f g : X ⇒C Y} →
                        f ≈C g → morphism f ≈D morphism g
    id-preserving  : {X : Object C} →
                        morphism (id C {X}) ≈D id D {object X}
    comp-preserving : {X Y Z : Object C} (f : Y ⇒C Z) (g : X ⇒C Y) →
                        morphism (f ·C g) ≈D (morphism f ·D morphism g)

```

Figure 4.3 Definition of functors. Subscripts are used to indicate to which category an operator belongs.

```

; id    = (λ x ↦ x) , (λ {i} x ↦ x)
; proofs of laws }

```

Note that the equivalence on morphisms is defined to be componentwise extensional equality, which is formulated with the help of McBride’s “John Major” heterogeneous equality $_ \approx _$ [McBride, 1999] — the equivalence $_ \cong _$ is defined by $g \cong h = \forall x \rightarrow g\ x \cong h\ x$. (Given $y : Y\ j$ for some $j : J$, the types of $u\ \{j\}\ y$ and $u'\ \{j\}\ y$ are not definitionally equal but only provably equal, so it is necessary to employ heterogeneous equality.)

Categories are graphs with a compositional structure, and **functors** are transformations between categories that preserve the compositional structure. The definition of functors is shown in Figure 4.3: a functor consists of two mappings, one on objects and the other on morphisms, where the morphism part preserves all structures on morphisms, including equivalence, identity, and composition. For example, we have two functors from \mathbb{FAM} to \mathbb{FUN} , one summing components together

```

COM : Functor IFAM IFUN  -- the comprehension functor
COM = record { object      = λ (I , X) ↦ Σ I X
              ; morphism   = λ (e , u) ↦ e * u
              ; proofs of laws }

```

and the other extracting the index part.

```

FAMF : Functor IFAM IFUN  -- the family fibration functor
FAMF = record { object      = λ (I , X) ↦ I
              ; morphism   = λ (e , u) ↦ e
              ; proofs of laws }

```

The functor laws should be proved for both functors alongside their object and morphism maps. In particular, we need to prove that the morphism part preserves equivalence: for COM, this means we need to prove, for all $e : J \rightarrow I$, $u : Y \rightrightarrows (X \circ e)$ and $f : J \rightarrow I$, $v : Y \rightrightarrows (X \circ f)$, that

$$(e \doteq f) \times ((j : J) \rightarrow u \{j\} \cong v \{j\}) \rightarrow (e * u \doteq f * v)$$

and for FAMF we need to prove

$$(e \doteq f) \times ((j : J) \rightarrow u \{j\} \cong v \{j\}) \rightarrow (e \doteq f)$$

both of which can be easily discharged.

Categories for refinements and ornaments

Some constructions in ?? can now be organised under several categories and functors. For a start, we already saw that refinements are interesting only because of their intensional contents; this is reflected in an isomorphism of categories between the category IFAM and the category IFREF of type families and refinement families (i.e., there are two functors back and forth inverse to each other). An object in IFREF is an indexed family of sets as in IFAM, and a morphism from (J, Y) to (I, X) consists of a function $e : J \rightarrow I$ on the indices and a refinement family of type FRefinement $e X Y$. As for the equivalence on morphisms, it suffices to use extensional equality on the index functions and componentwise extensional equality on refinement families, where extensional

equality on refinements means extensional equality on their forgetful functions (extracted by `Refinement.forget`), which we have shown in ?? to be the core of refinements. Formally:

```

IFREF : Category
IFREF = record
  { Object      =  $\Sigma[I : \text{Set}] I \rightarrow \text{Set}$ 
    ; Morphism =  $\lambda (J, Y) (I, X) \mapsto$  record
      { Carrier =  $\Sigma[e : J \rightarrow I] \text{FRefinement } e \ X \ Y$ 
        ;  $\_ \approx \_$  =  $\lambda (e, rs) (e', rs') \mapsto$ 
           $(e \doteq e') \times$ 
           $((j : J) \rightarrow \text{Refinement.forget } (rs \ (\text{ok } j)) \cong$ 
             $\text{Refinement.forget } (rs' \ (\text{ok } j)))$ 
        ; proofs of laws }
    ; proofs of laws }
```

Note that a refinement family from $X : I \rightarrow \text{Set}$ to $Y : J \rightarrow \text{Set}$ is deliberately cast as a morphism in the opposite direction from (J, Y) to (I, X) ; think of this as suggesting the direction of the forgetful functions of refinements. **IFREF** is no more powerful than **IFAM** since **IFREF** ignores the intensional contents of refinements by using an extensional equality, and consequently there are two functors between **IFREF** and **IFAM** that are inverse to each other, forming an isomorphism of categories:

- We have a forgetful functor **FREFF** : **Functor IFREF IFAM** which is identity on objects and componentwise `Refinement.forget` on morphisms (which preserves equivalence automatically):

```

FREFF : Functor IFREF IFAM
FREFF = record
  { object      = id
    ; morphism =  $\lambda (e, rs) \mapsto e, (\lambda j \mapsto \text{Refinement.forget } (rs \ (\text{ok } j)))$ 
    ; proofs of laws }
```

Note that **FREFF** remains a familiar covariant functor rather than a contravariant one because of our choice of morphism direction.

- Conversely, there is a functor $\mathbf{FREFC} : \mathbf{Functor} \mathbb{FAM} \mathbb{FREF}$ whose object part is identity and whose morphism part is componentwise *canonRef*:

$\mathbf{FREFC} : \mathbf{Functor} \mathbb{FAM} \mathbb{FREF}$

$\mathbf{FREFC} = \mathbf{record}$

$\{ \text{object} = id$
 $; \text{morphism} = \lambda (e, u) \mapsto e, (\lambda (ok\ j) \mapsto canonRef\ (u\ \{j\}))$
 $; \text{proofs of laws} \}$

The two functors \mathbf{FREF} and \mathbf{FREFC} are inverse to each other by definition.

There is another category \mathbf{ORN} , which has objects of type $\Sigma[I : \mathbf{Set}] \text{ Desc } I$, i.e., descriptions paired with index sets, and morphisms from (J, E) to (I, D) of type $\Sigma[e : J \rightarrow I] \text{ Orn } e\ D\ E$, i.e., ornaments paired with index erasure functions. To complete the definition of \mathbf{ORN} :

- We need to devise an equivalence on ornaments

$OrnEq : \{I\ J : \mathbf{Set}\} \{ef : J \rightarrow I\} \{D : \text{Desc } I\} \{E : \text{Desc } J\} \rightarrow$
 $\text{Orn } e\ D\ E \rightarrow \text{Orn } f\ D\ E \rightarrow \mathbf{Set}$

such that it implies extensional equality of e and f and that of ornamental forgetful functions:

$OrnEq\text{-forget} : \{I\ J : \mathbf{Set}\} \{ef : J \rightarrow I\} \{D : \text{Desc } I\} \{E : \text{Desc } J\} \rightarrow$
 $(O : \text{Orn } e\ D\ E) (P : \text{Orn } f\ D\ E) \rightarrow OrnEq\ O\ P \rightarrow$
 $(e \doteq f) \times ((j : J) \rightarrow forget\ O\ \{j\} \cong forget\ P\ \{j\})$

The actual definition of $OrnEq$ is deferred until ??.

- Morphism composition is sequential composition $-\odot-$, which merges two successive batches of modifications in a straightforward way. The definition is shown in Figure 4.4. There is also a family of **identity ornaments**:

$idOrn : \{I : \mathbf{Set}\} (D : \text{Desc } I) \rightarrow \text{Orn } id\ D\ D$
 $idOrn\ \{I\}\ D\ (ok\ i) = idROrn\ (D\ i)$

where

$\mathbb{E}\text{-refl} : (is : \mathbf{List}\ I) \rightarrow \mathbb{E}\ id\ is\ is$
 $\mathbb{E}\text{-refl}\ [] = []$
 $\mathbb{E}\text{-refl}\ (i :: is) = refl :: \mathbb{E}\text{-refl}\ is$

$$\begin{aligned}
\mathbb{E}\text{-trans} &: \{I\ J\ K : \text{Set}\} \{e : J \rightarrow I\} \{f : K \rightarrow J\} \rightarrow \\
&\quad \{is : \text{List } I\} \{js : \text{List } J\} \{ks : \text{List } K\} \rightarrow \\
&\quad \mathbb{E}\ e\ js\ is \rightarrow \mathbb{E}\ f\ ks\ js \rightarrow \mathbb{E}\ (e \circ f)\ ks\ is \\
\mathbb{E}\text{-trans} \quad [] \quad [] \quad &= [] \\
\mathbb{E}\text{-trans } \{e := e\} (eeq :: eeqs) (feq :: feqs) &= \text{trans } (\text{cong } e\ feq)\ eeq :: \\
&\quad \mathbb{E}\text{-trans } eeqs\ feqs \\
scROrn &: \{I\ J\ K : \text{Set}\} \{e : J \rightarrow I\} \{f : K \rightarrow J\} \rightarrow \\
&\quad \{D : \text{RDesc } I\} \{E : \text{RDesc } J\} \{F : \text{RDesc } K\} \rightarrow \\
&\quad \text{ROrn } e\ D\ E \rightarrow \text{ROrn } f\ E\ F \rightarrow \text{ROrn } (e \circ f)\ D\ F \\
scROrn (\vee eeqs) (\vee feqs) &= \vee (\mathbb{E}\text{-trans } eeqs\ feqs) \\
scROrn (\vee eeqs) (\Delta\ T\ P) &= \Delta[t : T]\ scROrn (\vee eeqs) (P\ t) \\
scROrn (\sigma\ S\ O) (\sigma\ .S\ P) &= \sigma[s : S]\ scROrn (O\ s) (P\ s) \\
scROrn (\sigma\ S\ O) (\Delta\ T\ P) &= \Delta[t : T]\ scROrn (\sigma\ S\ O) (P\ t) \\
scROrn (\sigma\ S\ O) (\nabla\ s\ P) &= \nabla[s]\ scROrn (O\ s) P \\
scROrn (\Delta\ T\ O) (\sigma\ .T\ P) &= \Delta[t : T]\ scROrn (O\ t) (P\ t) \\
scROrn (\Delta\ T\ O) (\Delta\ U\ P) &= \Delta[u : U]\ scROrn (\Delta\ T\ O) (P\ u) \\
scROrn (\Delta\ T\ O) (\nabla\ t\ P) &= scROrn (O\ t) P \\
scROrn (\nabla\ s\ O) P &= \nabla[s]\ scROrn O P \\
-\odot- &: \{I\ J\ K : \text{Set}\} \{e : J \rightarrow I\} \{f : K \rightarrow J\} \rightarrow \\
&\quad \{D : \text{Desc } I\} \{E : \text{Desc } J\} \{F : \text{Desc } K\} \rightarrow \\
&\quad \text{Orn } e\ D\ E \rightarrow \text{Orn } f\ E\ F \rightarrow \text{Orn } (e \circ f)\ D\ F \\
-\odot- \{f := f\} O\ P\ (\text{ok } k) &= scROrn (O\ (\text{ok } (f\ k))) (P\ (\text{ok } k))
\end{aligned}$$

Figure 4.4 Definitions for sequential composition of ornaments.

$$\begin{aligned}
idROrn &: (E : RDesc\ I) \rightarrow ROrn\ id\ E\ E \\
idROrn\ (\vee\ is) &= \vee\ (\mathbb{E}\text{-}refl\ is) \\
idROrn\ (\sigma\ S\ E) &= \sigma[s : S]\ idROrn\ (E\ s)
\end{aligned}$$

which simply use σ and \vee everywhere to express that a description is identical to itself. Unsurprisingly, the identity ornaments serve as identity of sequential composition.

To summarise:

$ORN : \text{Category}$

$ORN = \mathbf{record}$

$$\begin{aligned}
\{ \text{Object} &= \Sigma[I : \text{Set}]\ \text{Desc}\ I \\
; \text{Morphism} &= \lambda(J, E)\ (I, D) \mapsto \mathbf{record} \\
&\quad \{ \text{Carrier} = \Sigma[e : J \rightarrow I]\ \text{Orn}\ e\ D\ E \\
&\quad ; _ \approx _ = \lambda(e, O)\ (f, P) \mapsto \text{OrnEq}\ O\ P \\
&\quad ; \text{proofs of laws} \} \\
; _ \cdot _ &= \lambda(e, O)\ (f, P) \mapsto (e \circ f), (O \odot P) \\
; id &= \lambda\{I, D\} \mapsto id, idOrn\ D \\
; \text{proofs of laws} &\}
\end{aligned}$$

A functor $IND : \text{Functor}\ ORN\ \mathbb{FAM}$ can then be constructed, which gives the ordinary semantics of descriptions and ornaments: the object part of IND decodes a description (I, D) to its least fixed point $(I, \mu\ D)$, and the morphism part translates an ornament (e, O) to the forgetful function $(e, \text{forget}\ O)$, the latter preserving equivalence by virtue of $\text{OrnEq}\text{-forget}$.

$IND : \text{Functor}\ ORN\ \mathbb{FAM}$

$$\begin{aligned}
IND = \mathbf{record} \{ \text{object} &= \lambda(I, D) \mapsto I, \mu\ D \\
; \text{morphism} &= \lambda(e, O) \mapsto e, \text{forget}\ O \\
; \text{proofs of laws} &\}
\end{aligned}$$

To translate ORN to \mathbb{FREF} , i.e., datatype declarations to refinements, a naive way is to use the composite functor

$$ORN \xrightarrow{IND} \mathbb{FAM} \xrightarrow{\mathbb{FREFC}} \mathbb{FREF}$$

The resulting refinements would then use the canonical promotion predicates. However, the whole point of incorporating ORN in the framework is that we can construct an alternative functor RSEM directly from ORN to FREF. The functor RSEM is extensionally equal to the above composite functor, but intentionally very different. While its object part still takes the least fixed point of a description, its morphism part is the refinement semantics of ornaments given in ??, whose promotion predicates are the optimised predicates and have a more efficient representation.

RSEM : Functor ORN FAM

RSEM = **record** { *object* = $\lambda (I, D) \mapsto I, \mu D$
 ; *morphism* = $\lambda (e, O) \mapsto e, RSem\ O$
 ; **proofs of laws** }

Categorical isomorphisms

So far switching to the categorical language offers no obvious benefits.

Define the type of isomorphisms between two objects X and Y in C as

record Iso C X Y : Set **where**

field

to : $X \Rightarrow Y$

from : $Y \Rightarrow X$

from-to-inverse : $from \cdot to \approx id$

to-from-inverse : $to \cdot from \approx id$

(The relation $_ \cong _$ is formally defined as Iso FUN.)

functors preserve isomorphisms (a quick demonstration of preorder reasoning); TBC

4.2 Pullback properties of parallel composition

One of the great advantages of category theory is the ability to formulate the idea of **universal constructions** generically and concisely, which we will use to give parallel composition a useful macroscopic characterisation. An intuitive way to understand the idea of a universal construction is to think of it as a “best” solution to some specification. More precisely, the specification is represented as a category whose objects are all possible solutions and whose morphisms are evidence of how the solutions “compare” with each other, and a “best” solution is a **terminal object** in this category, meaning that it is “evidently better” than all objects in the category. For the actual definition: an object in a category C is **terminal** when it satisfies the **universal property** that for every object X' there is a unique morphism from X' to X , i.e., the setoid *Morphism* $X' X$ has a unique inhabitant:

name scoping

Terminal $C : \text{Object} \rightarrow \text{Set } _$

Terminal $C X = (X' : \text{Object}) \rightarrow \text{Singleton } (\text{Morphism } X' X)$

where *Singleton* is defined by

Singleton $: (S : \text{Setoid}) \rightarrow \text{Set } _$

Singleton $S = \text{Setoid.Carrier } S \times ((x\ y : \text{Setoid.Carrier } S) \rightarrow x \approx_S y)$

The uniqueness condition ensures that terminal objects are unique up to (a unique) isomorphism — that is, if two objects are both terminal in C , then there is an isomorphism between them:

terminal-iso $C : (X\ Y : \text{Object}) \rightarrow \text{Terminal } C X \rightarrow \text{Terminal } C Y \rightarrow \text{Iso } C X Y$

terminal-iso $C X Y\ tX\ tY =$

let $f : X \Rightarrow Y$

$f = \text{outl } (tY\ X)$

$g : Y \Rightarrow X$

$g = \text{outl } (tX\ Y)$

in record $\{ to = f$

$; from = g$

$; from\text{-}to\text{-}inverse = \text{outr } (tX\ X) (g \cdot f) id$

$$; \text{to-from-inverse} = \text{outr } (tY \ Y) \ (f \cdot g) \ \text{id} \}$$

Thus, to prove that two constructions are isomorphic, one way would be to prove that they are universal in the same sense, i.e., they are both terminal objects in the same category. This is the main method we use to construct the ornamental conversion isomorphisms in Section 4.3.1 and the modularity isomorphisms in Section 4.3.2, both involving parallel composition. The goal of the rest of this section, then, is to find suitable universal properties that characterise parallel composition, preparing for Sections 4.3.1 and 4.3.2.

As said earlier, parallel composition computes the least informative ornament that contains the information of two compatible ornaments, and this is exactly a categorical **product**. Below we construct the definition of categorical products step by step. Let C be a category and L, R two objects in C . A **span** over L and R is defined by

record Span $C \ L \ R$: Set _ **where**

constructor span

field

M : Object

l : $M \Rightarrow L$

r : $M \Rightarrow R$

or diagrammatically:

$$L \xleftarrow{l} M \xrightarrow{r} R$$

If we interpret a morphism $X \Rightarrow Y$ as evidence that X is more informative than Y , then a span over L and R is essentially an object which is more informative than both L and R . Spans over the same objects can be “compared”: define a morphism between two spans by

record SpanMorphism $C \ L \ R$ ($s \ s'$: Span $C \ L \ R$) : Set _ **where**

constructor spanMorphism

field

m : Span. $M \ s \Rightarrow \text{Span}.M \ s'$

triangle-l : Span. $l \ s' \cdot m \approx \text{Span}.l \ s$

triangle-r : Span. $r \ s' \cdot m \approx \text{Span}.r \ s$

or diagrammatically (abbreviating $\text{Span}.l\ s'$ to l' and so forth):

$$\begin{array}{ccccc} & & M & & \\ & l & \downarrow m & r & \\ L & \swarrow & & \searrow & R \\ & l' & M' & r' & \end{array}$$

where the two triangles are required to commute. Thus a span s is more informative than another span s' when $\text{Span}.M\ s$ is more informative than $\text{Span}.M\ s'$ and the morphisms factorise appropriately. We can then form a category of spans over L and R :

$\text{SpanCategory}\ C\ L\ R : \text{Category}$

$\text{SpanCategory}\ C\ L\ R = \mathbf{record}$

$\{ \text{Object} = \text{Span}\ C\ L\ R$

$; \text{Morphism} =$

$\lambda s\ s' \mapsto \mathbf{record}$

$\{ \text{Carrier} = \text{SpanMorphism}\ C\ L\ R\ s\ s'$

$; _ \approx _ = \lambda f\ g \mapsto \text{SpanMorphism}.m\ f \approx \text{SpanMorphism}.m\ g$

$; \text{proofs of laws} \}$

$; \text{proofs of laws} \}$

diagram commutativity not yet defined

morphism equivalence and proof irrelevance

and a product of L and R is a terminal object in this category:

$\text{Product}\ C\ L\ R : \text{Span}\ C\ L\ R \rightarrow \text{Set}\ _$

$\text{Product}\ C\ L\ R = \text{Terminal}\ (\text{SpanCategory}\ C\ L\ R)$

In particular, a product of L and R contains the least informative object that is more informative than both L and R .

product diagram; “morphism relevance”?

We thus aim to characterise parallel composition as a product of two compatible ornaments. This means that ornaments should be the objects of some category, but so far we only know that ornaments are morphisms of the category ORN . We are thus directed to construct a category whose objects are morphisms in an ambient category C , so when we use ORN as the ambient category, parallel composition can be characterised as a product in the derived category. Such a category is in general a **comma category** [Mac Lane, 1998, § II.6], whose

objects are morphisms with arbitrary source and target objects, but here we should restrict ourselves to a special case called a **slice category**, since we seek to form products of only compatible ornaments (whose less informative end coincide) rather than arbitrary ones. A slice category is parametrised with an ambient category C and an object B in C , and has

- objects: all the morphisms in C with target B ,

record $\text{Slice } C \ B : \text{Set} _ \text{where}$

constructor slice

field

$T : \text{Object}$

$s : T \Rightarrow B$

and

- morphisms: mediating morphisms giving rise to commutative triangles,

record $\text{SliceMorphism } C \ B \ (s \ s' : \text{Slice } C \ B) : \text{Set} _ \text{where}$

constructor sliceMorphism

field

$m : \text{Slice}.T \ s \Rightarrow \text{Slice}.T \ s'$

$\text{triangle} : \text{Slice}.s \ s' \cdot m \approx \text{Slice}.s \ s$

or diagrammatically:

$$\begin{array}{ccc} \text{objects} & \begin{array}{c} T \\ s \downarrow \\ B \end{array} & \text{and} \quad \text{morphisms} \quad \begin{array}{ccc} & T & \xrightarrow{m} T' \\ s \swarrow & & \searrow s' \\ & B & \end{array} \end{array}$$

The definitions above are assembled into the definition of slice categories in much the same way as span categories:

$\text{SliceCategory } C \ B : \text{Category}$

$\text{SliceCategory } C \ B = \text{record}$

$\{ \text{Object} = \text{Slice } C \ B$

$; \text{Morphism} =$

$\lambda s \ s' \mapsto \text{record}$

$$\begin{aligned}
& \{ \text{Carrier} = \text{SliceMorphism } C \ B \ s \ s' \\
& ; _ \approx _ = \lambda f \ g \mapsto \text{SliceMorphism}.m \ f \approx \text{SliceMorphism}.m \ g \\
& ; \text{proofs of laws} \} \\
& ; \text{proofs of laws} \}
\end{aligned}$$


Objects in a slice category are thus morphisms with a common target, and when the ambient category is ORN , they are exactly the compatible ornaments that can be composed in parallel.

We have arrived at the characterisation of parallel composition as a product in a slice category on top of ORN . The composite term “product in a slice category” has become a multi-layered concept and can be confusing; to facilitate comprehension, we give several new definitions that can sometimes deliver better intuition. Let C be an ambient category and X an object in C . We refer to spans over two slices $f, g : \text{Slice } C \ X$ alternatively as **squares** over f and g :

$\text{Square } C \ f \ g : \text{Set } _$

$\text{Square } C \ f \ g = \text{Span } (\text{SliceCategory } C \ X) \ f \ g$

since diagrammatically a square looks like



In a square q , we will refer to the object $\text{Slice}.T \ (\text{Span}.M \ q)$, i.e., the node W in the diagrams above, as the **vertex** of q :

$\text{vertex} : \text{Square } C \ f \ g \rightarrow \text{Object}$

$\text{vertex} = \text{Slice}.T \circ \text{Span}.M$

A product of f and g is alternatively referred to as a **pullback** of f and g ; that is, it is a square over f and g satisfying

$\text{Pullback } C \ f \ g : \text{Square } C \ f \ g \rightarrow \text{Set } _$

$\text{Pullback } C \ f \ g = \text{Product } (\text{SliceCategory } C \ X) \ f \ g$

Equivalently, if we define the **square category** over f and g as

$\text{SquareCategory } C f g : \text{Category}$

$\text{SquareCategory } C f g = \text{SpanCategory } (\text{SliceCategory } C X) f g$

then a pullback of f and g is a terminal object in the square category over f and g — indeed, $\text{Product } (\text{SliceCategory } C X) f g$ is definitionally equal to $\text{Terminal } (\text{SquareCategory } C f g)$. This means that, by *terminal-iso*, there is an isomorphism between any two pullbacks p and q of the same slices f and g :

diagram of pullback

$\text{Iso } (\text{SquareCategory } C f g) p q$

Subsequently, since there is a forgetful functor from $\text{SquareCategory } C f g$ to C whose object part is *vertex*, and functors preserve isomorphisms, we also have an isomorphism

$$\text{Iso } C (\text{vertex } p) (\text{vertex } q) \quad (4.1)$$

which is what we actually use in Sections 4.3.1 and 4.3.2.

pullback diagram; pullback preservation

We are now ready to state precisely the pullback properties for parallel composition that we make use of later. We could attempt to establish that, for any two ornaments $O : \text{Orn } e D E$ and $P : \text{Orn } f D F$ where $D : \text{Desc } I$, $E : \text{Desc } J$, and $F : \text{Desc } K$, the following square in ORN is a pullback:

$$\begin{array}{ccc}
 e \bowtie f, [O \otimes P] & \xrightarrow{\text{outr}, \text{diffOrn-r } O P} & K, F \\
 \downarrow \text{outl}, \text{diffOrn-l } O P & \searrow \text{pull}, [O \otimes P] & \downarrow f, P \\
 J, E & \xrightarrow{e, O} & I, D
 \end{array} \quad (4.2)$$

The Agda term for this square is

```

pc-square : Square ORN (slice (J, E) (e, O)) (slice (K, F) (f, P))
pc-square = span (slice (e ⋈ f, [O ⊗ P]) (pull, [O ⊗ P]))
               (sliceMorphism (outl, diffOrn-l O P) { }0)
               (sliceMorphism (outr, diffOrn-r O P) { }1)

```

where Goal 0 has type $\text{OrnEq } (O \odot \text{diffOrn-l } O P) [O \otimes P]$ and Goal 1 has type

$OrnEq (P \odot diffOrn-r O P) \lceil O \otimes P \rceil$, both of which can be discharged. Comparing the commutative diagram (4.2) and the Agda term *pc-square*, it should be obvious how concise the categorical language can be — the commutative diagram expresses the structure of the Agda term in a clean and visually intuitive way. Since terms like *pc-square* can be reconstructed from commutative diagrams and the categorical definitions, from now on we will present commutative diagrams as representations of the corresponding Agda terms and omit the latter. The pullback property of (4.2) is not too useful by itself, though: ORN is a quite restricted category, so a universal property established in ORN has limited applicability. Instead, we are more interested in the pullback property of the image of (4.2) under IND in \mathbb{FAM} :

$$\begin{array}{ccc}
 & \text{outr}, \text{forget } (diffOrn-r O P) & \\
 e \bowtie f, \mu \lceil O \otimes P \rceil & \xrightarrow{\quad} & K, \mu F \\
 \downarrow \text{outl}, \text{forget } (diffOrn-l O P) & \lrcorner & \downarrow f, \text{forget } P \\
 & \text{pull}, \text{forget } \lceil O \otimes P \rceil & \\
 J, \mu E & \xrightarrow{e, \text{forget } O} & I, \mu D
 \end{array} \tag{4.3}$$

We assert that the above square is a pullback by marking its vertex with “ \lrcorner ”. The proof of its universal property boils down to, very roughly speaking, datatype-generic construction of an inverse to

$$\text{forget } (diffOrn-l O P) \triangle \text{forget } (diffOrn-r O P)$$

which involves tricky manipulation of equality proofs but is achievable. After the pullback square (4.3) is established in \mathbb{FAM} , since the functor COM is pullback-preserving, we also get a pullback square in \mathbb{FUN} :

$$\begin{array}{ccc}
 & \text{outr} * \text{forget } (diffOrn-r O P) & \\
 \Sigma (e \bowtie f) (\mu \lceil O \otimes P \rceil) & \xrightarrow{\quad} & \Sigma K (\mu F) \\
 \downarrow \text{outl} * \text{forget } (diffOrn-l O P) & \lrcorner & \downarrow f * \text{forget } P \\
 & \text{pull} * \text{forget } \lceil O \otimes P \rceil & \\
 \Sigma J (\mu E) & \xrightarrow{e * \text{forget } O} & \Sigma I (\mu D)
 \end{array} \tag{4.4}$$

mention commutativity, associativity; should probably refactor the pullback square in \mathbb{FAM} into the pullback square in \mathbf{ORN} and pullback preservation of \mathbf{IND}

4.3 Consequences

4.3.1 The ornamental conversion isomorphisms

We restate the ornamental conversion isomorphisms as follows: for any ornament $O : \mathbf{Orn} \, e \, D \, E$ where $D : \mathbf{Desc} \, I$ and $E : \mathbf{Desc} \, J$, we have

$$\mu \, E \, j \cong \Sigma[x : \mu \, D \, (e \, j)] \, \mathbf{OptP} \, O \, (\mathbf{ok} \, j) \, x$$

for all $j : J$. Since the optimised predicates $\mathbf{OptP} \, O$ are defined by parallel composition of O and the singleton ornament $S = \mathbf{singleton} \, O \, D$, the isomorphism expands to

$$\mu \, E \, j \cong \Sigma[x : \mu \, D \, (e \, j)] \, \mu \, [O \otimes [S]] \, (\mathbf{ok} \, j, \mathbf{ok} \, (e \, j), x) \quad (4.5)$$

How do we derive this from the pullback properties for parallel composition? It turns out that the pullback property in \mathbf{FUN} (4.4) can help.

- First, observe that we have the following pullback square:

$$\begin{array}{ccc}
 (\mu \, E \, j) \triangleq (\mathbf{singleton} \, O \, D) & \xrightarrow{\quad} & \Sigma(\Sigma I(\mu D))(\mu [S]) \\
 \downarrow \text{id} \quad \lrcorner & \searrow e * \mathbf{forget} \, O & \downarrow \mathbf{outl} * \mathbf{forget} \, [S] \\
 \Sigma J(\mu E) & \xrightarrow[e * \mathbf{forget} \, O]{} & \Sigma I(\mu D)
 \end{array} \quad (4.6)$$

Viewing pullbacks as products of slices, since a singleton ornament does not add information to a datatype, the vertical slice on the right-hand side

$$s = \mathbf{slice}(\Sigma(\Sigma I(\mu D))(\mu [S]))(\mathbf{outl} * \mathbf{forget} \, [S])$$

behaves like a “multiplicative unit”: any (compatible) slice s' alone gives rise to a product of s and s' . As a consequence, we have the bottom-left type

$\Sigma J (\mu E)$ as the vertex of the pullback. This pullback square is over the same slices as the pullback square (4.4) with P substituted by $\lceil S \rceil$, so by (4.1) we obtain an isomorphism

$$\Sigma J (\mu E) \cong \Sigma (e \bowtie \text{outl}) (\mu \lfloor O \otimes \lceil S \rceil \rfloor) \quad (4.7)$$

- To get from (4.7) to (4.5), we need to look more closely into the construction of (4.7). The right-to-left direction of (4.7) is obtained by applying the universal property of (4.6) to the square (4.4) (with P substituted by $\lceil S \rceil$), so it is the unique mediating morphism m that makes the following diagram commute:

$$\begin{array}{ccccc}
 & \Sigma (e \bowtie \text{outl}) (\mu \lfloor O \otimes \lceil S \rceil \rfloor) & & & \\
 \text{outl} * \text{forget} (\text{diffOrn-l } O P) \swarrow & & \downarrow m & & \searrow \text{outr} * \text{forget} (\text{diffOrn-r } O P) \\
 \Sigma J (\mu E) & & & & \Sigma (\Sigma I (\mu D)) (\mu \lfloor S \rfloor) \\
 \nwarrow id & & \downarrow & & \nearrow (e * \text{forget } O) \Delta \\
 & \Sigma J (\mu E) & & & (\text{singleton} \circ \text{forget } O \circ \text{outr})
 \end{array}$$

From the left commuting triangle, we see that, extensionally, the morphism m is just $\text{outl} * \text{forget} (\text{diffOrn-l } O P)$.

- This leads us to the following general lemma: if there is an isomorphism

$$\Sigma K X \cong \Sigma L Y$$

whose right-to-left direction is extensionally equal to some $f * g$, then we have

$$X k \cong \Sigma [l : f^{-1} k] Y (\text{und } l)$$

for all $k : K$. For a justification: fixing $k : K$, an element of the form $(k, x) : \Sigma K X$ must correspond, under the given isomorphism, to some element $(l, y) : \Sigma L Y$ such that $f l \equiv k$, so the set $X k$ corresponds to exactly the sum of the sets $Y l$ such that $f l \equiv k$.

- Specialising the lemma above for (4.7), we get

$$\mu E j \cong \Sigma [jix : \text{outl}^{-1} j] \mu \lfloor O \otimes \lceil S \rceil \rfloor (\text{und } jix) \quad (4.8)$$

for all $j : J$. Finally, observe that a canonical element of type $\text{outl}^{-1} j$ must be of the form $\text{ok } (j, \text{ok } (e j, x))$ for some $x : \mu D (e j)$, so we perform a change of variables for the summation, turning the right-hand side of (4.8) into

$$\Sigma[x : \mu D (e j)] \mu [O \otimes [S]] (\text{ok } j, \text{ok } (e j, x))$$

and arriving at (4.5).

Formalisation detail. There is a twist when it comes to formalisation of the proof in Agda, however, due to Agda's intensionality: It is possible to formalise the lemma and the change of variables individually and chain them together, but the resulting isomorphisms would have a very complicated definition due to suspended type casts. If we use them to construct the refinement family in the morphism part of RSEM , it would be rather difficult to prove that the morphism part of RSEM preserves equivalence. We are thus forced to fuse all the above reasoning into one step to get a clean Agda definition such that RSEM preserves equivalence automatically, but the idea is still essentially the same. \square

4.3.2 The modularity isomorphisms

The other important family of isomorphisms we should construct from the pullback properties of parallel composition is the modularity isomorphisms, which is restated as follows: Suppose that there are descriptions $D : \text{Desc } I$, $E : \text{Desc } J$ and $F : \text{Desc } K$, and ornaments $O : \text{Orn } e D E$, and $P : \text{Orn } f D F$. Then we have

$$\text{OptP } [O \otimes P] (\text{ok } (j, k)) x \cong \text{OptP } O j x \times \text{OptP } P k x$$

for all $i : I, j : e^{-1} i, k : f^{-1} i$, and $x : \mu D i$. The isomorphism expands to

$$\begin{aligned} & \mu [[O \otimes P] \otimes [S]] (\text{ok } (j, k), \text{ok } (i, x)) \\ & \cong \mu [O \otimes [S]] (j, \text{ok } (i, x)) \times \mu [P \otimes [S]] (k, \text{ok } (i, x)) \end{aligned} \quad (4.9)$$

where again $S = \text{singleton} \circ D$. A quick observation is that they are componentwise isomorphisms between the two families of sets

$$M = \mu \llbracket O \otimes P \rrbracket \otimes \llbracket S \rrbracket$$

and

$$N = \lambda (\text{ok } (j, k), \text{ok } (i, x)) \mapsto \mu \llbracket O \otimes \llbracket S \rrbracket \rrbracket (j, \text{ok } (i, x)) \times \mu \llbracket P \otimes \llbracket S \rrbracket \rrbracket (k, \text{ok } (i, x))$$

both indexed by $\text{pull} \bowtie \text{outl}$ where pull has type $e \bowtie f \rightarrow I$ and outl has type $\Sigma I X \rightarrow I$. This is just an isomorphism in \mathbb{FAM} between $(\text{pull} \bowtie \text{outl}, M)$ and $(\text{pull} \bowtie \text{outl}, N)$ whose index part (i.e., the isomorphism obtained under the functor FAMF) is identity. Thus we seek to prove that both $(\text{pull} \bowtie \text{outl}, M)$ and $(\text{pull} \bowtie \text{outl}, N)$ are vertices of pullbacks of the same slices.

- We look at $(\text{pull} \bowtie \text{outl}, N)$ first. For fixed i, j, k , and x , the set

$$N (\text{ok } (j, k), \text{ok } (i, x))$$


along with the cartesian projections is a product, which trivially extends to a pullback since there is a forgetful function from each of the two component sets to the **singleton** set $\mu \llbracket S \rrbracket (i, x)$, as shown in the following diagram:

$$\begin{array}{ccc} N (\text{ok } (j, k), \text{ok } (i, x)) & \xrightarrow{\text{outr}} & \mu \llbracket P \otimes \llbracket S \rrbracket \rrbracket (k, \text{ok } (i, x)) \\ \text{outl} \downarrow \lrcorner & & \downarrow \text{forget } (\text{diffOrn-r } P \llbracket S \rrbracket) \\ \mu \llbracket O \otimes \llbracket S \rrbracket \rrbracket (j, \text{ok } (i, x)) & \xrightarrow{\text{forget } (\text{diffOrn-r } O \llbracket S \rrbracket)} & \mu \llbracket S \rrbracket (i, x) \end{array}$$

Note that this pullback square is possible because of the common x in the indices of the two component sets — otherwise they cannot project to the same singleton set. Collecting all such pullback squares together, we get the following pullback square in \mathbb{FAM} :

$$\begin{array}{ccc} \text{pull} \bowtie \text{outl}, N & \xrightarrow{-, \text{outr}} & f \bowtie \text{outl}, \mu \llbracket P \otimes \llbracket S \rrbracket \rrbracket \\ \downarrow \lrcorner \text{, outl} & & \downarrow \text{outr, forget } (\text{diffOrn-r } P \llbracket S \rrbracket) \\ e \bowtie \text{outl}, \mu \llbracket O \otimes \llbracket S \rrbracket \rrbracket & \xrightarrow{\text{outr, forget } (\text{diffOrn-r } O \llbracket S \rrbracket)} & \Sigma I (\mu D), \mu \llbracket S \rrbracket \end{array} \quad (4.10)$$

- Next we prove that $(pull \bowtie outl, M)$ is also the vertex of a pullback of the same slices as (4.10). This second pullback arises as a consequence of the following lemma (illustrated in the diagram below): In any category, consider the objects X, Y , their product $X \Leftarrow X \boxtimes Y \Rightarrow Y$, and products of each of the three objects X, Y , and $X \boxtimes Y$ with an object Z . (All the projections are shown as solid arrows in the diagram.) Then $(X \boxtimes Y) \boxtimes Z$ is the vertex of a pullback of the two projections $X \boxtimes Z \Rightarrow Z$ and $Y \boxtimes Z \Rightarrow Z$.



We again intend to view a pullback as a product of slices, and instantiate the lemma in $SliceCategory \mathbb{FAM} (I, \mu D)$, substituting all the objects by slices consisting of relevant ornamental forgetful functions in (4.9). The substitutions are as follows:

$$\begin{aligned}
 X &\mapsto \text{slice } _ \ (_, \text{forget } O) \\
 Y &\mapsto \text{slice } _ \ (_, \text{forget } P) \\
 X \boxtimes Y &\mapsto \text{slice } _ \ (_, \text{forget } [O \otimes P]) \\
 Z &\mapsto \text{slice } _ \ (_, \text{forget } [S]) \\
 X \boxtimes Z &\mapsto \text{slice } _ \ (_, \text{forget } [O \otimes [S]]) \\
 Y \boxtimes Z &\mapsto \text{slice } _ \ (_, \text{forget } [P \otimes [S]]) \\
 (X \boxtimes Y) \boxtimes Z &\mapsto \text{slice } _ \ (_, \text{forget } [[O \otimes P] \otimes [S]])
 \end{aligned}$$

where $X \boxtimes Y$, $X \boxtimes Z$, $Y \boxtimes Z$, and $(X \boxtimes Y) \boxtimes Z$ indeed give rise to products in $SliceCategory \mathbb{FAM} (I, \mu D)$, i.e., pullbacks in \mathbb{FAM} , by instantiating (4.3). What we get out of this instantiation of the lemma is a pullback in $SliceCategory \mathbb{FAM} (I, \mu D)$ rather than \mathbb{FAM} . This is easy to fix, since there is a forgetful functor from any $SliceCategory C B$ to C whose object part is $\text{Slice}.T$, and it is pullback-preserving. We thus get a pullback in \mathbb{FAM} of the same slices as (4.10) whose vertex is $(pull \bowtie outl, M)$.

Having the two pullbacks, by (4.1) we get an isomorphism in \mathbb{FAM} between $(pull \bowtie outl, M)$ and $(pull \bowtie outl, N)$, whose index part can be shown to be identity, so there are componentwise isomorphisms between M and N in \mathbb{FUN} , arriving at (4.9).

4.4 Discussion

elimination of arbitrariness of type-theoretic constructions; functor-level abstraction; compare with purely categorical approach

Bibliography

Gilles BARTHE, Venanzio CAPRETTA, and Olivier PONS [2003]. Setoids in type theory. *Journal of Functional Programming*, 13(2):261–293. doi: 10.1017/S0956796802004501. ↗ page 4

Robert HARPER and Robert POLLACK [1991]. Type checking with universes. *Theoretical Computer Science*, 89(1):107–136. doi: 10.1016/0304-3975(90)90108-T. ↗ page 4

Saunders MAC LANE [1998]. *Categories for the Working Mathematician*. Springer-Verlag, second edition. ↗ pages 2 and 16

Conor McBRIDE [1999]. *Dependently Typed Functional Programs and their Proofs*. Ph.D. thesis, University of Edinburgh. ↗ page 7

Shin-Cheng MU, Hsiang-Shang KO, and Patrik JANSSON [2009]. Algebra of Programming in Agda: Dependent types for relational program derivation. *Journal of Functional Programming*, 19(5):545–579. doi: 10.1017/S0956796809007345. ↗ page 6

Todo list

functors preserve isomorphisms (a quick demonstration of preorder reasoning); TBC	13
name scoping	14
diagram commutativity not yet defined	16
morphism equivalence and proof irrelevance	16
product diagram; “morphism relevance”?	16
diagram of pullback	19
pullback diagram; pullback preservation	19
mention commutativity, associativity; should probably refactor the pullback square in \mathbb{FAM} into the pullback square in \mathbb{ORN} and pullback preservation of \mathbb{IND}	20
elimination of arbitrariness of type-theoretic constructions; compare with purely categorical approach	26