Analysis and synthesis of inductive families

Hsiang-Shang Ko

19 November 2013

Contents

Chapter 1

Introduction

"datatypes" for inductive families

Chapter 2

From intuitionistic type theory to dependently typed programming

We start with an introduction to intuitionistic type theory [Martin-Löf, 1984] and dependently typed programming [Altenkirch et al., 2005; McBride, 2004] using the Agda language [Norell, 2007, 2009; Bove and Dybjer, 2009]. Intuitionistic type theory was developed by Martin-Löf to serve as a foundation of intuitionistic mathematics like Bishop's renowned work on constructive analysis [Bishop and Bridges, 1985]. While originated from intuitionistic type theory, dependently typed programming is more concerned with mechanisation and practicalities, and is influenced by the program construction movement. It has thus departed from the mathematical traditions considerably, and deviations can be found from syntactic presentations to the underlying philosophy.

2.1 Datatypes and universe construction

Central to *datatype-generic programming* is the idea that the definitional structure of datatypes can be coded as first-class entities and thus become ordinary parameters to programs. The same idea is also found in Martin-Löf's Type Theory [Martin-Löf, 1984], in which a set of codes for datatypes is called a *uni-*

verse (à la Tarski), and there is a decoding function translating codes to actual types. Type theory being the foundation of dependently typed languages, universe construction can be done directly in such languages, so datatype-generic programming becomes just ordinary programming in the dependently typed world [Altenkirch and McBride, 2003]. In this section we construct a universe of *index-first datatypes* [Chapman et al., 2010; Dagand and McBride, 2012b], on which a second universe of *ornaments*, to be constructed in Section 3.2, will depend.

present codes along with their interpretation; not induction-recursion [Dybjer, 1998] though

2.1.1 High-level introduction to index-first datatypes

In Agda, an inductive family is declared by listing all possible constructors and their types, all ending with one of the types in that inductive family. This conveys the idea that the index in the type of an inhabitant is synthesised in a *bottom-up* fashion following the construction of the inhabitant. Consider vectors, for example: the cons constructor takes a vector at some index n and constructs a vector at suc n — the final index is computed bottom-up from the index of the sub-vector. This approach can yield redundant representation, though — the cons constructor for vectors has to store the index of the sub-vector, so the representation of a vector would be cluttered with all the intermediate lengths. If we switch to the opposite perspective, determining *top-down* from the targeted index what constructors should be supplied, then the representation can usually be significantly cleaned up — for a vector, if the index of its type is known to be suc n for some n, then we know that its top-level constructor can only be cons and the index of the sub-vector must be n. To reflect this important reversal of logical order, Dagand and McBride [2012b] proposed a new notation for index-first datatype declarations, in which we first list all possible patterns of (the indices of) the types in the inductive family, and then specify for each pattern which constructors it offers. Below we follow Ko

and Gibbons's slightly more Agda-like adaptation of the notation [2013].

Index-first declarations of simple datatypes look almost like Haskell data declarations. For example, natural numbers are declared by

```
indexfirst data Nat : Set where
  Nat ∋ zero
      | suc (n : Nat)
```

We use the keyword **indexfirst** to explicitly mark the declaration as an index-first one. The only possible pattern of the datatype is Nat, which offers two constructors zero and suc, the latter taking a recursive argument named n. We declare lists similarly, this time with a uniform parameter A : Set:

```
indexfirst data List (A : Set) : Set where
  List A ∋ []
         | _::_ (a : A) (as : List A)
```

The declaration of vectors is more interesting, fully exploiting the power of index-first datatypes:

```
indexfirst data Vec (A : Set) : Nat → Set where
  Vec A zero    ∋ []
  Vec A (suc n) ∋ _::_ (a : A) (as : Vec A n)
```

Vec A is a family of types indexed by Nat, and we do pattern matching on the index, splitting the datatype into two cases Vec A zero and Vec A (suc n) for some n : Nat. The first case only offers the nil constructor [], and the second case only offers the cons constructor _::_. Because the form of the index restricts constructor choice, the recursive structure of a vector as : Vec A n must follow that of n, i.e., the number of cons nodes in as must match the number of successor nodes in n. We can also declare the bottom-up vector datatype in index-first style:

```
indexfirst data Vec (A : Set) : Nat → Set where
  Vec A n ∋ nil (neq : n ≡ zero)
          | cons (a : A) {m : Nat}
                 (as : Vec A m) (meq : n ≡ suc m)
```

Besides the field m storing the length of the tail, two more fields *neq* and *meq* are inserted, demanding explicit equality proofs about the indices. When a vector of type Vec A n is demanded, we are "free" to choose between nil or cons regardless of the index n; however, because of the equality constraints, we are indirectly forced into a particular choice.

Remark (*detagging*)**.** The transformation from bottom-up vectors to top-down vectors is exactly what Brady et al.'s *detagging* optimisation [2004] does. With index-first datatypes, however, detagged representations are available directly, rather than arising from a compiler optimisation. (*End of remark.*)

Remark (*bidirectional typechecking*)**.**

TBC

(*End of remark.*)

2.1.2 Universe construction

Now we proceed to construct a universe for index-first datatypes. An inductive family of type $I \to \mathsf{Set}$ is constructed by taking the least fixed point of a base endofunctor on $I \to \mathsf{Set}$. For example, to get index-first vectors, we would define a base functor (parametrised by $A : \mathsf{Set}$)

$$\begin{aligned} &\mathit{VecF}\ A\ :\ (\mathsf{Nat} \to \mathsf{Set}) \to (\mathsf{Nat} \to \mathsf{Set}) \\ &\mathit{VecF}\ A\ X\ \mathsf{zero} \quad\ = \top \\ &\mathit{VecF}\ A\ X\ (\mathsf{suc}\ n) \ = A \times X\ n \end{aligned}$$

and take its least fixed point. If we flip the order of arguments of $\mathit{VecF}\ A$:

$$\begin{aligned} &\mathit{VecF}'\ A\ :\ \mathsf{Nat} \to (\mathsf{Nat} \to \mathsf{Set}) \to \mathsf{Set} \\ &\mathit{VecF}'\ A\ \mathsf{zero} \quad\ = \lambda\ X \to \top \\ &\mathit{VecF}'\ A\ (\mathsf{suc}\ n) \ = \lambda\ X \to A \times X\ n \end{aligned}$$

we see that $\mathit{VecF}'\ A$ consists of two different "responses" to the index request, each of type $(\mathsf{Nat} \to \mathsf{Set}) \to \mathsf{Set}$. It suffices to construct for such responses a universe

$$\textbf{data}\ \mathsf{RDesc}\ (I\ :\ \mathsf{Set})\ :\ \mathsf{Set}_1$$

with a decoding function specifying its semantics:

$$[\![\_]\!]\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{RDesc}\ I \to (I \to \mathsf{Set}) \to \mathsf{Set}$$

Inhabitants of $\mathsf{RDesc}\ I$ will be called *response descriptions*. A function of type $I \to \mathsf{RDesc}\ I$, then, can be decoded to an endofunctor on $I \to \mathsf{Set}$, so the type $I \to \mathsf{RDesc}\ I$ acts as a universe for index-first datatypes. We hence define

$$\begin{aligned}&\mathsf{Desc}\ :\ \mathsf{Set} \to \mathsf{Set}_1\\ &\mathsf{Desc}\ I\ =\ I \to \mathsf{RDesc}\ I\end{aligned}$$

with decoding function

$$\begin{aligned}&\mathbb{F}\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{Desc}\ I \to (I \to \mathsf{Set}) \to (I \to \mathsf{Set})\\ &\mathbb{F}\ D\ X\ i\ =\ [\![\ D\ i\ ]\!]\ X\end{aligned}$$

Inhabitants of type $\mathsf{Desc}\ I$ will be called *datatype descriptions*, or *descriptions* for short. Actual datatypes are manufactured from descriptions by the least fixed point operator:

$$\begin{aligned}&\textbf{data}\ \mu\ \{I\ :\ \mathsf{Set}\}\ (D\ :\ \mathsf{Desc}\ I)\ :\ I \to \mathsf{Set}\ \textbf{where}\\ &\quad \mathsf{con}\ :\ \mathbb{F}\ D\ (\mu\ D) \rightrightarrows \mu\ D\end{aligned}$$

We now define the datatype of response descriptions — which determines the syntax available for defining base functors — and its decoding function:

$$\begin{aligned}&\textbf{data}\ \mathsf{RDesc}\ (I\ :\ \mathsf{Set})\ :\ \mathsf{Set}_1\ \textbf{where}\\ &\quad \mathsf{v}\ :\ (is\ :\ \mathsf{List}\ I) \to \mathsf{RDesc}\ I\\ &\quad \sigma\ :\ (S\ :\ \mathsf{Set})\ (D\ :\ S \to \mathsf{RDesc}\ I) \to \mathsf{RDesc}\ I\\ &[\![\_]\!]\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{RDesc}\ I \to (I \to \mathsf{Set}) \to \mathsf{Set}\\ &[\![\ \mathsf{v}\ is\quad]\!]\ X\ =\ \mathbb{P}\ is\ X\quad \text{-- see below}\\ &[\![\ \sigma\ S\ D\ ]\!]\ X\ =\ \Sigma\langle\, s : S\,\rangle\ [\![\ D\ s\ ]\!]\ X\end{aligned}$$

The operator $\mathbb{P}$ computes the product of a finite number of types in a type family, whose indices are given in a list:

$$\begin{aligned}&\mathbb{P}\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{List}\ I \to (I \to \mathsf{Set}) \to \mathsf{Set}\\ &\mathbb{P}\ [\,]\qquad X\ =\ \top\\ &\mathbb{P}\ (i :: is)\ X\ =\ X\ i \times \mathbb{P}\ is\ X\end{aligned}$$

Thus, in a response, given $X : I \to \mathsf{Set}$, we are allowed to form dependent sums (by σ) and the product of a finite number of types in X (via v, suggesting variable positions in the base functor).

Convention. We will informally refer to the index part of a σ as a *field*. Like Σ, we regard σ as a binder and write $\sigma\langle s : S \rangle\ D\ s$ for $\sigma\ S\ (\lambda s \mapsto D\ s)$. (*End of convention*.)

Example (*natural numbers*)**.** The datatype of natural numbers is considered to be an inductive family trivially indexed by $\top$, so the declaration of Nat corresponds to an inhabitant of Desc $\top$.

```
data ListTag : Set where 'nil 'cons : ListTag

NatD : Desc ⊤
NatD ▪ = σ ListTag λ { 'nil  ↦ v []
                     ; 'cons ↦ v (▪ :: []) }
```

The index request is necessarily ▪, and we respond with a field of type ListTag representing the constructor choices. If the field receives 'nil, then we are constructing zero, which takes no recursive values, so we write v [] to end this branch; if the ListTag field receives 'cons, then we are constructing a successor, which takes a recursive value at index ▪, so we write v (▪ :: []). (*End of example*.)

Example (*lists*)**.** The datatype of lists is parametrised by the element type. We represent parametrised descriptions simply as functions producing descriptions, so the declaration of lists corresponds to a function taking element types to descriptions.

```
ListD : Set → Desc ⊤
ListD A ▪ = σ ListTag λ { 'nil  ↦ v []
                        ; 'cons ↦ σ⟨ _ : A ⟩ v (▪ :: []) }
```

ListD A is the same as *NatD* except that, in the 'cons case, we use σ to insert a field of type A for storing an element. (*End of example*.)

Example (*vectors*)**.** The datatype of vectors is parametrised by the element type and (non-trivially) indexed by Nat, so the declaration of vectors corre-

sponds to

$$
\begin{aligned}
&\mathit{VecD} \;:\; \mathsf{Set} \to \mathsf{Desc}\ \mathsf{Nat} \\
&\mathit{VecD}\ A\ \mathsf{zero} \;=\; \mathsf{v}\ [\,] \\
&\mathit{VecD}\ A\ (\mathsf{suc}\ n) \;=\; \sigma\langle\ \_ : A\ \rangle\ \mathsf{v}\ (n :: [\,])
\end{aligned}
$$

which is directly comparable to the index-first base functor VecF' at the beginning of Section 2.1.2. (*End of example.*)

There is no problem defining functions on the encoded datatypes except that it has to be done with the raw representation. For example, list append is defined by

$$
\begin{aligned}
&\_{+\!\!+}\_ \;:\; \mu\ (\mathit{ListD}\ A)\ \blacksquare \to \mu\ (\mathit{ListD}\ A)\ \blacksquare \to \mu\ (\mathit{ListD}\ A)\ \blacksquare \\
&\mathsf{con}\ (\text{`}\mathsf{nil}\quad ,\qquad\quad \blacksquare)\ {+\!\!+}\ \mathit{bs} \;=\; \mathit{bs} \\
&\mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ \mathit{as}\ ,\ \blacksquare)\ {+\!\!+}\ \mathit{bs} \;=\; \mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ \mathit{as} {+\!\!+} \mathit{bs}\ ,\ \blacksquare)
\end{aligned}
$$

To improve readability, we define the following higher-level terms:

$$
\begin{aligned}
&\mathsf{List} \;:\; \mathsf{Set} \to \mathsf{Set} \\
&\mathsf{List}\ A \;=\; \mu\ (\mathit{ListD}\ A)\ \blacksquare \\
&[\,] \;:\; \{A \;:\; \mathsf{Set}\} \to \mathsf{List}\ A \\
&[\,] \;=\; \mathsf{con}\ (\text{`}\mathsf{nil}\ ,\ \blacksquare) \\
&\_::\_ \;:\; \{A \;:\; \mathsf{Set}\} \to A \to \mathsf{List}\ A \to \mathsf{List}\ A \\
&a :: \mathit{as} \;=\; \mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ \mathit{as}\ ,\ \blacksquare)
\end{aligned}
$$

List append can then be rewritten in the usual form (assuming that the terms $[\,]$ and $\_::\_$ can be used in pattern matching):

$$
\begin{aligned}
&\_{+\!\!+}\_ \;:\; \mathsf{List}\ A \to \mathsf{List}\ A \to \mathsf{List}\ A \\
&[\,] \qquad {+\!\!+}\ \mathit{bs} \;=\; \mathit{bs} \\
&(a :: \mathit{as})\ {+\!\!+}\ \mathit{bs} \;=\; a :: (\mathit{as} {+\!\!+} \mathit{bs})
\end{aligned}
$$

Later on, when an encoded datatype is defined, we almost always supply a corresponding index-first datatype declaration immediately afterwards, which is thought of as giving definitions of higher-level terms for type and data constructors — the terms List, $[\,]$, and $\_::\_$ above, for example, can be considered to be defined by the index-first declaration of lists given in Section 2.1.1. Index-first declarations will only be regarded in this thesis as informal hints

```
mutual
  fold : {I : Set} {D : Desc I} →
         {X : I → Set} → (𝕀 D X ⇉ X) → (μ D ⇉ X)
  fold {I} {D} f {i} (con ds)  =  f (mapFold D (D i) f ds)

  mapFold : {I : Set} (D : Desc I) (D′ : RDesc I) →
            {X : I → Set} → (𝕀 D X ⇉ X) → ⟦ D′ ⟧ (μ D) → ⟦ D′ ⟧ X
  mapFold D (v [])        f ▪        =  ▪
  mapFold D (v (i :: is)) f (d , ds) =  fold f d , mapFold D (v is) f ds
  mapFold D (σ S D′)      f (s , ds) =  s , mapFold D (D′ s) f ds
```

Figure 2.1 Definition of the datatype-generic *fold* operator.

at how encoded datatypes are presented at a higher level; we do not give a formal treatment of the elaboration process from index-first declarations to corresponding descriptions and definitions of higher-level terms. (One such treatment was given by Dagand and McBride [2012a].)

Direct function definitions by pattern matching work fine for individual datatypes, but when we need to define operations and to state properties for all the datatypes encoded by the universe, it is necessary to have a generic *fold* operator parametrised by descriptions:

```
fold : {I : Set} {D : Desc I} →
       {X : I → Set} → (𝕀 D X ⇉ X) → (μ D ⇉ X)
```

There is also a generic *induction* operator, which can be used to prove generic propositions about all encoded datatypes and subsumes *fold*, but *fold* is much easier to use when the full power of *induction* is not required. The implementations of both operators are adapted for our two-level universe from those in McBride's original work [2011]. We look at the implementation of the *fold* operator only, which is shown in Figure 2.1. As McBride, we would have wished to define *fold* by

```
fold {I} {D} f {i} (con ds)  =  f (mapRD (D i) (fold f) ds)
```

where the functorial mapping *mapRD* on response structures is defined by

$$
\begin{array}{l}
\mathit{mapRD} \;:\; \{I \;:\; \mathsf{Set}\}\;(D \;:\; \mathsf{RDesc}\;I) \to \\
\qquad\qquad \{X\;Y \;:\; I \to \mathsf{Set}\}\;(g \;:\; X \rightrightarrows Y) \to [\![\; D \;]\!]\; X \to [\![\; D \;]\!]\; Y \\
\mathit{mapRD}\;(\mathsf{v}\;[\,])\qquad g\;\blacksquare \qquad\quad = \blacksquare \\
\mathit{mapRD}\;(\mathsf{v}\;(i :: \mathit{is}))\;g\;(x\;,\;\mathit{xs}) \;=\; g\;x\;,\;\mathit{mapRD}\;(\mathsf{v}\;\mathit{is})\;g\;\mathit{xs} \\
\mathit{mapRD}\;(\sigma\;S\;D)\qquad g\;(s\;,\;\mathit{xs}) \;=\; s\;,\;\mathit{mapRD}\;(D\;s)\;g\;\mathit{xs}
\end{array}
$$

Agda does not see that this definition of *fold* is terminating, however, since the termination checker does not expand the definition of *mapRD* to see that *fold f* is applied to structurally smaller arguments. To make termination obvious, we instead define *fold* mutually recursively with *mapFold*, which is *mapRD* specialised by fixing its argument g to *fold f*.

It is helpful to form a two-dimensional image of our datatype manufacturing scheme: we manufacture a datatype by first defining a base functor, and then recursively duplicating the functorial structure by taking its least fixed point. The shape of the base functor can be imagined to stretch horizontally, whereas the recursive structure generated by the least fixed point grows vertically. This image works directly when the recursive structure is linear, like lists. (Otherwise one resorts to the abstraction of functor composition.) For example, we can typeset a list two-dimensionally like

```
con (‘cons , a ,
con (‘cons , b ,
con (‘nil  ,
  ▪) , ▪) , ▪)
```

Ignoring the last line of trailing ▪'s, things following con on each line — including constructor tags and list elements — are shaped by the base functor of lists, whereas the con nodes, aligned vertically, are generated by the least fixed point. This two-dimensional metaphor will be referred to in later explanations.

Remark (*first-order vs higher-order representation*)**.** The functorial structures generated by descriptions are strongly reminiscent of *indexed containers* [Altenkirch and Morris, 2009]; this will be explored and exploited in Chapter 6. For now, it is enough to mention that we choose to stick to a first-order datatype man-

ufacturing scheme, i.e., the datatypes we manufacture with descriptions use finite product types rather than dependent function types for branching, but it is easy to switch to a higher-order representation that is even closer to indexed containers (allowing infinite branching) by storing in v a collection of I-indices indexed by an arbitrary set S:

$$\mathsf{v} : (S : \mathsf{Set})\ (f : S \to I) \to \mathsf{RDesc}\ I$$

whose semantics is defined in terms of dependent functions:

$$[\![\ \mathsf{v}\ S\ f\ ]\!]\ X\ =\ (s : S) \to X\ (f\ s)$$

The reason for choosing to stick to first-order representation is simply to obtain a simpler equality for the manufactured datatypes (Agda's default equality would suffice); the examples of manufactured datatypes in this thesis are all finitely branching and do not require the power of higher-order representation anyway. This choice, however, does complicate some subsequent datatype-generic definitions (e.g., ornaments). It would probably be helpful to think of the parts involving v and $\mathbb{P}$ in these definitions as specialisations of higher-order representations to first-order ones. (*End of remark.*)

2.2 Internalism vs externalism

Chapter 3

Refinements and ornaments

This chapter begins our exploration of the interconnection between internalism and externalism by looking at the analytic direction, i.e., the decomposition of sophisticated types into basic types and predicates on them. (The synthetic direction will have to wait until Chapter 5.) The purpose of such decomposition is for internalist datatypes and operations to take a round trip to the externalist world so as to harvest composability there. For example, consider the insertion operation on ordered vectors:

$$\begin{aligned} \mathit{ovinsert} \;:\; & (x \;:\; \mathit{Val}) \to \{b \;:\; \mathit{Val}\}\ \{n \;:\; \mathsf{Nat}\} \to \mathsf{OrdVec}\ b\ n \to \\ & \{b' \;:\; \mathit{Val}\} \to b' \leqslant x \to b' \leqslant b \to \mathsf{OrdVec}\ b'\ (\mathsf{suc}\ n) \end{aligned}$$

As long as OrdVec is formally unrelated to OrdList and Vec, we cannot reuse insertion on OrdList and Vec but can only reimplement *ovinsert* completely. Here one way to relate the three datatypes is to switch to externalism so it becomes apparent that the datatypes have common ingredients. If we change the appearances of OrdVec in the type of *ovinsert* to its externalist counterpart,

$$\begin{aligned} \mathit{ovinsert}' \;:\; & (x \;:\; \mathit{Val}) \to \{b \;:\; \mathit{Val}\}\ \{n \;:\; \mathsf{Nat}\} \to \\ & \quad \Sigma\langle\, xs : \mathsf{List}\ \mathit{Val}\,\rangle\ \ \mathsf{Ordered}\ b\ xs \times \mathit{length}\ xs \;\equiv\; n \to \\ & \{b' \;:\; \mathit{Val}\} \to b' \leqslant x \to b' \leqslant b \to \\ & \quad \Sigma\langle\, xs : \mathsf{List}\ \mathit{Val}\,\rangle\ \ \mathsf{Ordered}\ b'\ xs \times \mathit{length}\ xs \;\equiv\; \mathsf{suc}\ n \end{aligned}$$

then, given the three functions

$$
\begin{aligned}
&\mathit{insert} \;\; : \mathit{Val} \to \mathsf{List}\ \mathit{Val} \to \mathsf{List}\ \mathit{Val} \\
&\mathit{oinsert}' \; : (x \; : \mathit{Val}) \to \{b \; : \mathit{Val}\} \to \\
&\qquad\qquad (\mathit{xs} \; : \mathsf{List}\ \mathit{Val}) \to \mathsf{Ordered}\ b\ \mathit{xs} \to \\
&\qquad\qquad \{b' \; : \mathit{Val}\} \to b' \leqslant x \to b' \leqslant b \to \mathsf{Ordered}\ b'\ (\mathit{insert}\ x\ \mathit{xs}) \\
&\mathit{vinsert}' \; : (x \; : \mathit{Val}) \to \{n \; : \mathsf{Nat}\} \to \\
&\qquad\qquad (\mathit{xs} \; : \mathsf{List}\ \mathit{Val}) \to \mathit{length}\ \mathit{xs} \equiv n \to \\
&\qquad\qquad \mathit{length}\ (\mathit{insert}\ x\ \mathit{xs}) \equiv \mathsf{suc}\ n
\end{aligned}
$$

we can easily combine them to form *ovinsert′*:

$$
\begin{aligned}
\mathit{ovinsert}'\ x\ (\mathit{xs}\ ,\ \mathit{ord\text{-}xs}\ ,\ \mathit{len\text{-}xs})\ b'{\leqslant}x\ b'{\leqslant}b \;=\; & \mathit{insert}\ x\ \mathit{xs}\ , \\
& \mathit{oinsert}'\ x\ \mathit{xs}\ \mathit{ord\text{-}xs}\ b'{\leqslant}x\ b'{\leqslant}b\ , \\
& \mathit{vinsert}'\ x\ \mathit{xs}\ \mathit{len\text{-}xs}
\end{aligned}
$$

All that is left is converting *ovinsert′* to *ovinsert*, which involves switching from the externalist representation back to OrdVec with the help of the family of *conversion isomorphisms*

$$\mathsf{OrdVec}\ b\ n \;\cong\; \Sigma\langle\, \mathit{xs} : \mathsf{List}\ \mathit{Val} \,\rangle\ \mathsf{Ordered}\ b\ \mathit{xs} \times \mathit{length}\ \mathit{xs} \equiv n$$

for all $b \; : \; \mathit{Val}$ and $n \; : \; \mathsf{Nat}$. Note that the three functions *insert*, *oinsert′*, and *vinsert′* are reusable components that can go into a library of list datatypes — insertion for OrdList and Vec can also be composed from the three functions in the same way as insertion for OrdVec with the help of appropriate conversion isomorphisms.

This chapter develops the abstractions and constructions that facilitate the above externalist composition of internalist operations as follows:

- Conversion isomorphisms are axiomatised as *refinements* (Section 3.1).
- Refinements are coordinated by *upgrades* (Section 3.1.2) to enable switching between internalist and externalist representations in function types.
- A class of refinements are conveniently synthesised by marking differences between datatypes with *ornaments* (Section 3.2), which relate datatype descriptions that are vertically the same but horizontally different.

TBC (should probably sneak in the term “function upgrading” somewhere)

3.1 Refinements

3.1.1 Refinements between individual types

residual view, partitioning view (fine vs coarse)

A *refinement* from a type X to a type Y is a *promotion predicate* $P : X \to \mathsf{Set}$ and a *conversion isomorphism* $i : Y \cong \Sigma\, X\, P$.

universe polymorphism

```
record Refinement (X Y : Set) : Set₁ where
  field
    P : X → Set
    i : Y ≅ Σ X P
  forget : Y → X
  forget = outl ∘ Iso.to i
```

Refinements are not guaranteed to be interesting in general. For example, Y can be chosen to be $\Sigma\, X\, P$ and the conversion isomorphism simply the identity. We will, however, only be interested in refinements from basic types to their more informative — often internalist — variants. The conversion isomorphism tells us that the inhabitants of Y exactly correspond to the inhabitants of X bundled with more information, i.e., proofs that the promotion predicate P is satisfied. Computationally, any inhabitant of Y can be decomposed (by $\mathsf{Iso}.to\ i$) into an underlying value $x : X$ and a proof that x satisfies the promotion predicate P (which we will call a *promotion proof* for x), and conversely, if $x : X$ satisfies P, then it can be promoted (by $\mathsf{Iso}.from\ i$) to an inhabitant of Y. We denote the forgetful computation of the underlying value, i.e., $\mathsf{outl} \circ \mathsf{Iso}.to\ i$, as $\mathsf{Refinement}.forget$, which, as we will see in Chapter 4, is actually the core of a refinement.

Example (*refinement from lists to ordered lists*)**.** Suppose $A : \mathsf{Set}$ is equipped with an ordering $\_\leqslant_A\_$. Fixing $a : A$, there is a refinement from $\mathsf{List}\ A$ to $\mathsf{OrdList}\ A\ \_\leqslant_A\_\ a$ whose promotion predicate is $\mathsf{Ordered}\ A\ \_\leqslant_A\_\ a$, since we have an isomorphism of type

$$\mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ a\ \cong\ \Sigma\ (\mathsf{List}\ A)\ (\mathsf{Ordered}\ A\ \_{\leqslant_A}\_\ a)$$

as shown in Section 2.2. An ordered list of type $\mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ a$ can be decomposed into a list $xs\ :\ \mathsf{List}\ A$ and a proof of type $\mathsf{Ordered}\ A\ \_{\leqslant_A}\_\ a\ xs$ that the list xs is ordered and bounded below by a; conversely, a list satisfying $\mathsf{Ordered}\ A\ \_{\leqslant_A}\_\ a$ can be promoted to an ordered list of type $\mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ a$. (*End of example*.)

Example (*refinement from natural numbers to lists*)**.** Let $A\ :\ \mathsf{Set}$. We have a refinement from Nat to List A

$$\mathit{Nat\text{-}List}\ A\ :\ \mathsf{Refinement}\ \mathsf{Nat}\ (\mathsf{List}\ A)$$

for which Vec A serves as the promotion predicate — there is a conversion isomorphism of type

$$\mathsf{List}\ A\ \cong\ \Sigma\ \mathsf{Nat}\ (\mathsf{Vec}\ A)$$

whose decomposing direction computes from a list its length and a vector containing the same elements. We might say that a natural number $n\ :\ \mathsf{Nat}$ is an incomplete list — the list elements are missing from the successor nodes of n. To promote n to a List A, we need to supply a vector of type Vec $A\ n$, i.e., n elements of type A. This example helps to emphasise that the notion of refinements is *proof-relevant*: a basic element can have more than one promotion proofs, and consequently the more informative type in a refinement can have more elements than the basic type does. (*End of example*.)

a type refines another in the sense of being more informative rather than merely being a subset

Coherence-based definition of refinements

There is an alternative definition of refinements which, instead of the conversion isomorphism, postulates the forgetful computation and characterises the promotion predicate in term of it:

record $\mathsf{Refinement}'\ (X\ Y\ :\ \mathsf{Set})\ :\ \mathsf{Set}_1$ **where**
  field

$$
\begin{aligned}
P \quad &: X \to \mathsf{Set} \\
\mathit{forget} \; &: Y \to X \\
p \quad &: (x : X) \to P\, x \cong \Sigma \langle\, y : Y \,\rangle\ \mathit{forget}\, y \equiv x
\end{aligned}
$$

We say that $x : X$ and $y : Y$ are *in coherence* when $\mathit{forget}\, y \equiv x$, i.e., the underlying basic element in y is x. This definition then requires that the promotion proofs for any $x : X$ exactly correspond to those more informative elements in coherence with x. The two definitions of refinements are equivalent (and in fact isomorphic, which can be easily shown once we define a sensible equivalence on refinements in Chapter 4). Of particular importance is the direction from Refinement to Refinement′:

$$
\begin{aligned}
&\mathit{toRefinement}' : \{X\ Y : \mathsf{Set}\} \to \mathsf{Refinement}\ X\ Y \to \mathsf{Refinement}'\ X\ Y \\
&\mathit{toRefinement}'\ r = \mathbf{record}\ \{\ P \quad = \mathsf{Refinement}.P\ r \\
&\qquad\qquad\qquad\qquad\quad ;\ \mathit{forget} = \mathsf{Refinement}.\mathit{forget}\ r \\
&\qquad\qquad\qquad\qquad\quad ;\ p \quad = \mathit{coherence}\ r\ \}
\end{aligned}
$$

where *coherence* is a proof (which is deferred until Chapter 4) that the promotion predicate of any refinement of type Refinement $X\ Y$ is pointwise isomorphic to the *canonical promotion predicate* $\lambda\, x \mapsto \Sigma \langle\, y : Y \,\rangle\ \mathit{forget}\, y \equiv x$:

$$
\begin{aligned}
&\mathit{coherence} : \\
&\quad \{X\ Y : \mathsf{Set}\}\ (r : \mathsf{Refinement}\ X\ Y) \to \\
&\quad (x : X) \to \mathsf{Refinement}.P\ r\ x \cong \Sigma \langle\, y : Y \,\rangle\ \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x
\end{aligned}
$$

We prefer the definition of refinements in terms of conversion isomorphisms because it is more concise and directly applicable to function upgrading. The coherence-based definition, however, is easier to generalise for function types, as we will see below.

TBC

3.1.2 Upgrades

Refinements are less useful when we move on to function types: the requirement that a conversion isomorphism exists between related function types is

too strong, even when we have extensional equality for functions so isomorphisms between function types make more sense. For example, it is not — and should not be — possible to have a refinement from the function type $\mathsf{Nat} \to \mathsf{Nat}$ to the function type $\mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$, despite that the component types Nat and List Nat are related by a refinement: If such a refinement existed, we would be able to extract from any function $f : \mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$ an "underlying" function of type $\mathsf{Nat} \to \mathsf{Nat}$ which has roughly the same behaviour as f. However, the behaviour of a function taking a list may depend essentially on the list elements, which is not available to a function taking only a natural number. For example, a function of type $\mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$ might compute the sum s of the input list and emit a list of length s whose elements are all zero. We cannot hope to write a function of type $\mathsf{Nat} \to \mathsf{Nat}$ that reproduces the corresponding behaviour on natural numbers.

Comparison (*type theory in colour*)**.** Bernardy and Guilhem [2013]

TBC

(*End of comparison.*)

It is only the decomposing direction of refinements that causes problem in the case of function types, however; the promoting direction is perfectly valid for function types. For example, to promote the function

$$
\begin{aligned}
&\mathit{double} : \mathsf{Nat} \to \mathsf{Nat} \\
&\mathit{double}\ \mathsf{zero} \quad\ = \mathsf{zero} \\
&\mathit{double}\ (\mathsf{suc}\ n) \ = \mathsf{suc}\ (\mathsf{suc}\ (\mathit{double}\ n))
\end{aligned}
$$

to a function of type $\mathsf{List}\ A \to \mathsf{List}\ A$ for some fixed $A : \mathsf{Set}$, we can use

$$Q = \lambda f \mapsto (n : \mathsf{Nat}) \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (\mathit{double}\ n)$$

as the promotion predicate: Consider the refinement from Nat to List A. Given a promotion proof of type $Q\ \mathit{double}$, say

Explain the meaning of this (scoping).

$$
\begin{aligned}
&\mathit{duplicate}' : (n : \mathsf{Nat}) \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (\mathit{double}\ n) \\
&\mathit{duplicate}'\ \mathsf{zero}\ \ [] \quad\ = [] \\
&\mathit{duplicate}'\ (\mathsf{suc}\ n)\ (x :: xs) \ = x :: x :: \mathit{duplicate}'\ n\ xs
\end{aligned}
$$

we can synthesise a function $\mathit{duplicate} : \mathsf{List}\ A \to \mathsf{List}\ A$ by

definition of $*$

$$\mathit{duplicate} = \mathsf{Iso}.\mathit{from}\ i \circ (\mathit{double} * \mathit{duplicate}'\ \_) \circ \mathsf{Iso}.\mathit{to}\ i$$

i.e., we decompose the input list into the underlying natural number and a vector of elements, process the two parts separately with *double* and $\mathit{duplicate}'$, and finally combine the results back to a list. The relationship between the promoted function *duplicate* and the underlying function *double* is characterised by the coherence property [Dagand and McBride, 2012b]

definition of pointwise equality

$$\mathit{double} \circ \mathit{length} \doteq \mathit{length} \circ \mathit{duplicate}$$

or as a commutative diagram:



which states that *duplicate* preserves length as computed by *double*, or in more generic terms, processes the recursive structure (i.e., nil and cons nodes) of its input in the same way as *double* does.

We thus define *upgrades* to capture the promoting direction and the coherence property abstractly. An upgrade from $X : \mathsf{Set}$ to $Y : \mathsf{Set}$ is a promotion predicate $P : X \to \mathsf{Set}$, a coherence property $C : X \to Y \to \mathsf{Set}$ relating basic elements of type X and promoted elements of type Y, an upgrading (promoting) operation $u : (x : X) \to P\ x \to Y$, and a coherence proof $c : (x : X)\ (p : P\ x) \to C\ x\ (u\ x\ p)$ saying that the result of promoting a basic element $x : X$ must be in coherence with x.

record Upgrade $(X\ Y : \mathsf{Set}) : \mathsf{Set}_1$ **where**
  field
    $P : X \to \mathsf{Set}$
    $C : X \to Y \to \mathsf{Set}$
    $u : (x : X) \to P\ x \to Y$
    $c : (x : X)\ (p : P\ x) \to C\ x\ (u\ x\ p)$

Like refinements, arbitrary upgrades are not guaranteed to be interesting, but we will only use the upgrades synthesised by the combinators we define below specifically for deriving coherence properties and upgrading operations for function types from refinements between component types.

Upgrades from refinements

As we said, upgrades amount to only the promoting direction of refinements. This is most obvious when we look at the coherence-based refinements, of which upgrades are a direct generalisation: we get from $\mathsf{Refinement}'$ to $\mathsf{Upgrade}$ by abstracting the notion of coherence and weakening the isomorphism to only the left-to-right computation. Any coherence-based refinement can thus be weakened to an upgrade,

$$
\begin{aligned}
&\mathit{toUpgrade}' \;:\; \{X\ Y \;:\; \mathsf{Set}\} \to \mathsf{Refinement}'\ X\ Y \to \mathsf{Upgrade}\ X\ Y \\
&\mathit{toUpgrade}'\ r \;=\; \mathbf{record}\ \{\ P \;=\; \mathsf{Refinement}'.P\ r \\
&\qquad\qquad\qquad\qquad ;\ C \;=\; \lambda\, x\ y \mapsto \mathsf{Refinement}'.\mathit{forget}\ r\ y \equiv x \\
&\qquad\qquad\qquad\qquad ;\ u \;=\; \lambda\, x \mapsto \mathsf{outl} \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement}'.p\ r\ x) \\
&\qquad\qquad\qquad\qquad ;\ c \;=\; \lambda\, x \mapsto \mathsf{outr} \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement}'.p\ r\ x)\ \}
\end{aligned}
$$

and consequently any refinement gives rise to an upgrade.

$$
\begin{aligned}
&\mathit{toUpgrade} \;:\; \{X\ Y \;:\; \mathsf{Set}\} \to \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ X\ Y \\
&\mathit{toUpgrade} \;=\; \mathit{toUpgrade}' \circ \mathit{toRefinement}'
\end{aligned}
$$

Composition of upgrades

The most representative combinator for upgrades is the following one for synthesising upgrades between function types:

$$
\begin{aligned}
&\_\rightharpoonup\_ \;:\; \{X\ Y\ Z\ W \;:\; \mathsf{Set}\} \to \\
&\qquad \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ Z\ W \to \mathsf{Upgrade}\ (X \to Z)\ (Y \to W)
\end{aligned}
$$

Note that there should be a *refinement* between the source types X and Y, rather than just an upgrade. (As a consequence, we can produce upgrades between

curried multi-argument function types but not between higher-order function types.) This is because, as we see in the *double–duplicate* example, we need the ability to decompose the source type Y.

Let r : Refinement X Y and s : Upgrade Z W. The upgrading operation takes a function $f : X \to Z$ and combines it with a promotion proof to get a function $g : Y \to W$, which should transform underlying values in coherence with f. That is, as g takes $y : Y$ to $g\ y : W$ at the more informative level, correspondingly at the underlying level the value Refinement.*forget* $r\ y : X$ underlying y should be taken by f to a value in coherence with $g\ y$. We thus define the statement "g is in coherence with f" as

$$(x : X)\ (y : Y) \to \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x \to \mathsf{Upgrade}.C\ s\ (f\ x)\ (g\ y)$$

As for the type of promotion proofs, since we already know that the underlying values are transformed by f, the missing information is only how the residual parts are transformed — that is, we need to know for any $x : X$ how a promotion proof for x is transformed to a promotion proof for $f\ x$. The type of promotion proofs for f is thus

$$(x : X) \to \mathsf{Refinement}.P\ r\ x \to \mathsf{Upgrade}.P\ s\ (f\ x)$$

Having determined the coherence property and the promotion predicate, it is then easy to construct the upgrading operation and the coherence proof. In particular, following the *double–duplicate* example, the upgrading operation breaks an input $y : Y$ into its underlying value $x = \mathsf{Refinement}.\mathit{forget}\ r\ y : X$ and a promotion proof for x, computes a promotion proof q for $f\ x : Z$ using the given promotion proof for f, and upgrades $f\ x$ to an inhabitant of type W using q. To sum up, the complete definition of $\_\rightharpoonup\_$ is

$$\begin{aligned}
&_\rightharpoonup_ : \{X\ Y\ Z\ W : \mathsf{Set}\} \to \\
&\qquad \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ Z\ W \to \mathsf{Upgrade}\ (X \to Z)\ (Y \to W) \\
&r \rightharpoonup s = \mathbf{record} \\
&\quad \{\ P = \lambda f \mapsto (x : X) \to \mathsf{Refinement}.P\ r\ x \to \mathsf{Upgrade}.P\ s\ (f\ x) \\
&\quad ;\ C = \lambda f\ g \mapsto (x : X)\ (y : Y) \to \\
&\qquad\qquad \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x \to \mathsf{Upgrade}.C\ s\ (f\ x)\ (g\ y) \\
&\quad ;\ u = \lambda f\ h \mapsto \mathsf{Upgrade}.u\ s\ _ \circ \mathit{uncurry}\ h \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement}.i\ r)
\end{aligned}$$

$$; c \;=\; \lambda\,\{\, f\ h\ \_\ y\ \mathsf{refl} \;\mapsto\; \mathbf{let}\ (x\ ,\ p) \;=\; \mathsf{Iso}.to\ (\mathsf{Refinement}.i\ r)\ y \\ \mathbf{in}\ \ \mathsf{Upgrade}.c\ s\ (f\ x)\ (h\ x\ p)\ \}\ \}$$

Example (*upgrade from* $\mathsf{Nat} \to \mathsf{Nat}$ *to* $\mathsf{List}\ A \to \mathsf{List}\ A$)**.** Using the $\_\rightharpoonup\_$ combinator on the refinement

$$r \;=\; \mathit{Nat\text{-}List}\ A \;:\; \mathsf{Refinement}\ \mathsf{Nat}\ (\mathsf{List}\ A)$$

and the upgrade derived from r, we get an upgrade

$$u \;=\; r \rightharpoonup \mathit{toUpgrade}\ r \;:\; \mathsf{Upgrade}\ (\mathsf{Nat} \to \mathsf{Nat})\ (\mathsf{List}\ A \to \mathsf{List}\ A)$$

The type $\mathsf{Upgrade}.P\ u\ \mathit{double}$ is exactly the type of $\mathit{duplicate}'$, and the type $\mathsf{Upgrade}.C\ u\ \mathit{double}\ \mathit{duplicate}$ is exactly the coherence property satisfied by *double* and *duplicate*. (*End of example.*)

Comparison (*functional ornaments*)**.**

Dagand and McBride [2012b], origin of coherence property, no need to construct a universe

We can define more combinators for upgrades, like the ones in Figure 3.1. (*End of comparison.*)

3.1.3 Refinement families

When we move on to consider refinements between indexed families of types, refinement relationship exists not only between the member types but also between the index sets: a type family $X\ :\ I \to \mathsf{Set}$ is refined by another type family $Y\ :\ J \to \mathsf{Set}$ when

- at the index level, there is a refinement r from I to J, and
- at the member type level, there is a refinement from $X\ i$ to $Y\ j$ whenever $i\ :\ I$ underlies $j\ :\ J$, i.e., $\mathsf{Refinement}.\mathit{forget}\ r\ j \equiv i$.

In short, each type $X\ i$ is refined by a collection of types in Y, the underlying values of their indices all being i. We will not exploit the complete refinement

```
  -- the upgraded function type has an extra argument
new : {X : Set} (I : Set) {Y : I → Set} →
      (∀ i → Upgrade X (Y i)) → Upgrade X ((i : I) → Y i)
new I u = record { P = λ x ↦ ∀ i → Upgrade.P (u i) x
                 ; C = λ x y ↦ ∀ i → Upgrade.C (u i) x (y i)
                 ; u = λ x p i ↦ Upgrade.u (u i) x (p i)
                 ; c = λ x p i ↦ Upgrade.c (u i) x (p i) }

syntax new I (λ i ↦ u) = ∀⁺⟨ i : I ⟩ u

  -- implicit version of new
new′ : {X : Set} (I : Set) {Y : I → Set} →
       (∀ i → Upgrade X (Y i)) → Upgrade X ({i : I} → Y i)
new′ I u = record { P = λ x ↦ ∀ {i} → Upgrade.P (u i) x
                  ; C = λ x y ↦ ∀ {i} → Upgrade.C (u i) x (y {i})
                  ; u = λ x p {i} ↦ Upgrade.u (u i) x (p {i})
                  ; c = λ x p {i} ↦ Upgrade.c (u i) x (p {i}) }

syntax new′ I (λ i ↦ u) = ∀⁺⟪ i : I ⟫ u

  -- the underlying and the upgraded function types
  -- have a common argument
fixed : (I : Set) {X : I → Set} {Y : I → Set} →
        (∀ i → Upgrade (X i) (Y i)) → Upgrade ((i : I) → X i) ((i : I) → Y i)
fixed I u = record { P = λ f ↦ ∀ i → Upgrade.P (u i) (f i)
                   ; C = λ f g ↦ ∀ i → Upgrade.C (u i) (f i) (g i)
                   ; u = λ f h i ↦ Upgrade.u (u i) (f i) (h i)
                   ; c = λ f h i ↦ Upgrade.c (u i) (f i) (h i) }

syntax fixed I (λ i ↦ u) = ∀⟨ i : I ⟩ u

  -- implicit version of fixed
fixed′ : (I : Set) {X : I → Set} {Y : I → Set} →
         (∀ i → Upgrade (X i) (Y i)) → Upgrade ({i : I} → X i) ({i : I} → Y i)
fixed′ I u = record { P = λ f ↦ ∀ {i} → Upgrade.P (u i) (f {i})
                    ; C = λ f g ↦ ∀ {i} → Upgrade.C (u i) (f {i}) (g {i})
                    ; u = λ f h {i} ↦ Upgrade.u (u i) (f {i}) (h {i})
                    ; c = λ f h {i} ↦ Upgrade.c (u i) (f {i}) (h {i}) }

syntax fixed′ I (λ i ↦ u) = ∀⟪ i : I ⟫ u
```

Figure 3.1 More combinators for upgrades.

structure on indices, though, so in the actual definition of *refinement families* below, the index-level refinement degenerates into the forgetful computation.

```
FRefinement : {I J : Set} (e : J → I) (X : I → Set) (Y : J → Set) → Set₁
FRefinement {I} e X Y = {i : I} (j : e ⁻¹ i) → Refinement (X i) (Y (und j))
```

Example (*refinement family from ordered lists to ordered vectors*)**.** The datatype OrdList A $\_\leqslant_A\_$: $A \to$ Set is a family of types into which ordered lists are classified according to their lower bound. For each type of ordered lists having a particular lower bound, we can further classify them by their length, yielding OrdVec A $\_\leqslant_A\_$: $A \to$ Nat $\to$ Set. This further classification is captured as a refinement family of type

```
FRefinement outl (OrdList A _≤A_) (uncurry (OrdVec A _≤A_))
```

which consists of refinements from OrdList A $\_\leqslant_A\_$ b to OrdVec A $\_\leqslant_A\_$ b n for all b : A and n : Nat. (*End of example.*)

3.2 Ornaments

One possible way to establish relationships between datatypes is to write conversion functions. Conversions that involve only modifications of horizontal structures like copying, projecting away, or assigning default values to fields, however, may instead be stated at the level of datatype declarations, i.e., in terms of natural transformations between base functors. For example, a list is a natural number whose successor nodes are decorated with elements, and to convert a list to its length, we simply discard those elements. The essential information in this conversion is just that the elements associated with cons nodes should be discarded, which is described by the following natural transformation between the two base functors $\mathbb{F}$ (*ListD* A) and $\mathbb{F}$ *NatD*:

```
erase : {A : Set} {X : ⊤ → Set} → 𝔽 (ListD A) X ⇉ 𝔽 NatD X
erase (‘nil   , ▪) = ‘nil   , ▪   -- ‘nil copied
erase (‘cons , a , x , ▪) = ‘cons , x , ▪   -- ‘cons copied, a discarded,
                                            -- and x retained
```

The transformation can then be lifted to work on the least fixed points.

$$\mathit{length} : \{A : \mathsf{Set}\} \to \mu\ (\mathit{ListD}\ A) \rightrightarrows \mu\ \mathit{NatD}$$
$$\mathit{length}\ \{A\} = \mathit{fold}\ (\mathsf{con} \circ \mathit{erase}\ \{A\}\ \{\mu\ \mathit{NatD}\})$$

Our goal in this section is to construct a universe for such horizontal natural transformations between the base functors arising as decodings of descriptions. The inhabitants of this universe are called *ornaments*. By encoding the relationship between datatype descriptions as a universe, whose inhabitants are analysable syntactic objects, we will not only be able to derive conversion functions between datatypes, but even compute new datatypes that are related to old ones in prescribed ways, which is something we cannot achieve if we simply write the conversion functions directly.

3.2.1 Universe construction

The definition of ornaments has the same two-level structure as that of datatype descriptions. We have an upper-level datatype Orn of ornaments

$$\mathsf{Orn} : \{I\ J : \mathsf{Set}\}\ (e : J \to I)\ (D : \mathsf{Desc}\ I)\ (E : \mathsf{Desc}\ J) \to \mathsf{Set}_1$$
$$\mathsf{Orn}\ e\ D\ E = \{i : I\}\ (j : e^{-1}\ i) \to \mathsf{ROrn}\ e\ (D\ i)\ (E\ (\mathit{und}\ j))$$

which is defined in terms of a lower-level datatype ROrn of *response ornaments*, while ROrn contains the actual encoding of horizontal transformations and is decoded by the function *erase*:

$$\textbf{data}\ \mathsf{ROrn}\ \{I\ J : \mathsf{Set}\}\ (e : J \to I) : \mathsf{RDesc}\ I \to \mathsf{RDesc}\ J \to \mathsf{Set}_1$$
$$\mathit{erase} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{RDesc}\ I\}\ \{E : \mathsf{RDesc}\ J\} \to$$
$$\quad \mathsf{ROrn}\ e\ D\ E \to \{X : I \to \mathsf{Set}\} \to [\![\ E\ ]\!]\ (X \circ e) \to [\![\ D\ ]\!]\ X$$

The datatype Orn is parametrised by an erasure function $e : J \to I$ on the index sets and relates two datatype descriptions $D : \mathsf{Desc}\ I$ and $E : \mathsf{Desc}\ J$ such that from any ornament $O : \mathsf{Orn}\ e\ D\ E$ we can derive a forgetful map:

$$\mathit{forget}\ O : \mu\ E \rightrightarrows \mu\ D \circ e$$

By design, this forgetful map necessarily preserves the recursive structure of its input. In terms of the two-dimensional metaphor mentioned towards the

end of Section 2.1.2, an ornament describes only how the horizontal shapes change, and the forgetful map — which is a *fold* — simply applies the changes to each vertical level; it never alters the vertical structure. For example, the *length* function discards elements associated with cons nodes, shrinking the list horizontally to a natural number, but keeps the vertical structure (i.e., the con nodes) intact. Look more closely: Given $y : \mu\ E\ j$, we should transform it into an inhabitant of type $\mu\ D\ (e\ j)$. Deconstructing y into con ys where $ys : [\![\ E\ j\ ]\!]\ (\mu\ E)$ and assuming that the $(\mu\ E)$–inhabitants at the recursive positions of ys have been inductively transformed into $(\mu\ D \circ e)$–inhabitants, we horizontally modify the resulting structure of type $[\![\ E\ j\ ]\!]\ (\mu\ D \circ e)$ to one of type $[\![\ D\ (e\ j)\ ]\!]\ (\mu\ D)$, which can then be wrapped by con to an inhabitant of type $\mu\ D\ (e\ j)$. The above steps are performed by the *ornamental algebra* induced by O:

$$
\begin{aligned}
&\mathit{ornAlg} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{Desc}\ I\}\ \{E : \mathsf{Desc}\ J\} \\
&\qquad (O : \mathsf{Orn}\ e\ D\ E) \to \mathbb{F}\ E\ (\mu\ D \circ e) \rightrightarrows \mu\ D \circ e \\
&\mathit{ornAlg}\ O\ \{j\} \ = \ \mathsf{con} \circ \mathit{erase}\ (O\ (\mathsf{ok}\ j))
\end{aligned}
$$

where the horizontal modification — a transformation from $[\![\ E\ j\ ]\!]\ (X \circ e)$ to $[\![\ D\ (e\ j)\ ]\!]\ X$, natural in X — is decoded by *erase* from a response ornament relating $D\ (e\ j)$ and $E\ j$. The forgetful function is then defined by

$$\mathit{forget}\ O \ = \ \mathit{fold}\ (\mathit{ornAlg}\ O)$$

Hence an ornament of type Orn $e\ D\ E$ contains, for each index request j, a response ornament of type ROrn $e\ (D\ (e\ j))\ (E\ j)$ to cope with all possible horizontal structures that can occur in a $(\mu\ E)$–inhabitant. The definition of Orn given above is a restatement of this in an intensionally more flexible form.

connection to refinement families

Now we look at the definitions of ROrn and *erase*, followed by explanations of the four cases.

```
data ROrn {I J : Set} (e : J → I) : RDesc I → RDesc J → Set₁ where
  v : {js : List J} {is : List I} (eqs : 𝔼 e js is) → ROrn e (v is) (v js)
  σ : (S : Set) {D : S → RDesc I} {E : S → RDesc J}
      (O : (s : S) → ROrn e (D s) (E s)) → ROrn e (σ S D) (σ S E)
  Δ : (T : Set) {D : RDesc I} {E : T → RDesc J}
```

$$
\begin{aligned}
&\quad (O \,:\, (t \,:\, T) \to \mathsf{ROrn}\ e\ D\ (E\ t)) \to \mathsf{ROrn}\ e\ D\ (\sigma\ T\ E)\\
\nabla \,:\,\ &\{S \,:\, \mathsf{Set}\}\ (s \,:\, S)\ \{D \,:\, S \to \mathsf{RDesc}\ I\}\ \{E \,:\, \mathsf{RDesc}\ J\}\\
&\quad (O \,:\, \mathsf{ROrn}\ e\ (D\ s)\ E) \to \mathsf{ROrn}\ e\ (\sigma\ S\ D)\ E
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{erase} \,:\, \{I\ J \,:\, \mathsf{Set}\}\ \{e \,:\, J \to I\}\ \{D \,:\, \mathsf{RDesc}\ I\}\ \{E \,:\, \mathsf{RDesc}\ J\} \to\\
&\quad \mathsf{ROrn}\ e\ D\ E \to \{X \,:\, I \to \mathsf{Set}\} \to [\![\ E\ ]\!]\ (X \circ e) \to [\![\ D\ ]\!]\ X\\
&\mathit{erase}\ (\mathsf{v}\ [\,]\qquad\quad)\ \blacksquare\qquad\quad = \blacksquare\\
&\mathit{erase}\ (\mathsf{v}\ (\mathsf{refl} :: \mathit{eqs}))\ (x\ ,\ \mathit{xs}) = x\ ,\ \mathit{erase}\ (\mathsf{v}\ \mathit{eqs})\ \mathit{xs} \quad \text{-- } x \text{ retained}\\
&\mathit{erase}\ (\sigma\ S\ O)\qquad (s\ ,\ \mathit{xs}) = s\ ,\ \mathit{erase}\ (O\ s)\quad \mathit{xs} \quad \text{-- } s \text{ copied}\\
&\mathit{erase}\ (\Delta\ T\ O)\qquad (t\ ,\ \mathit{xs}) = \quad \mathit{erase}\ (O\ t)\quad \mathit{xs} \quad \text{-- } t \text{ discarded}\\
&\mathit{erase}\ (\nabla\ s\ O)\qquad\quad \mathit{xs} = s\ ,\ \mathit{erase}\ O\qquad \mathit{xs} \quad \text{-- } s \text{ inserted}
\end{aligned}
$$

The first two cases v and σ of ROrn relate response descriptions that have the same top-level constructor, and the transformations decoded from them preserve horizontal structure.

- The v case of ROrn states that a response description $\mathsf{v}\ \mathit{js}$ refines another response description $\mathsf{v}\ \mathit{is}$, i.e., when $[\![\ \mathsf{v}\ \mathit{js}\ ]\!]\ (X \circ e)$ can be transformed into $[\![\ \mathsf{v}\ \mathit{is}\ ]\!]\ X$. The source type $[\![\ \mathsf{v}\ \mathit{js}\ ]\!]\ (X \circ e)$ expands to a product of types of the form $X\ (e\ j)$ for some $j \,:\, J$ and the target type $[\![\ \mathsf{v}\ \mathit{is}\ ]\!]\ X$ to a product of types of the form $X\ i$ for some $i \,:\, I$. There are no horizontal contents and thus no horizontal modifications to make, and the input values should be preserved. We thus demand that *js* and *is* have the same number of elements and the corresponding pairs of indices $e\ j$ and i are equal; that is, we demand a proof of $\mathit{map}\ e\ \mathit{js} \equiv \mathit{is}$ (where *map* is the usual functorial mapping on lists). To make it easier to analyse a proof of $\mathit{map}\ e\ \mathit{js} \equiv \mathit{is}$ in the v case of *erase*, we instead define the proposition inductively as $\mathbb{E}\ e\ \mathit{js}\ \mathit{is}$, where the datatype $\mathbb{E}$ is defined by

  $$
  \begin{aligned}
  &\mathbf{data}\ \mathbb{E}\ \{I\ J \,:\, \mathsf{Set}\}\ (e \,:\, J \to I) \,:\, \mathsf{List}\ J \to \mathsf{List}\ I \to \mathsf{Set}\ \mathbf{where}\\
  &\quad [\,] \quad\ \,:\, \mathbb{E}\ e\ [\,]\ [\,]\\
  &\quad \_::\_ \,:\, \{j \,:\, J\}\ \{i \,:\, I\}\ (\mathit{eq} \,:\, e\ j \equiv i) \to\\
  &\qquad\quad \{\mathit{js} \,:\, \mathsf{List}\ J\}\ \{\mathit{is} \,:\, \mathsf{List}\ I\}\ (\mathit{eqs} \,:\, \mathbb{E}\ e\ \mathit{js}\ \mathit{is}) \to \mathbb{E}\ e\ (j :: \mathit{js})\ (i :: \mathit{is})
  \end{aligned}
  $$

- The σ case of ROrn states that $\sigma\ S\ E$ refines $\sigma\ S\ D$, i.e., that both response descriptions start with the same field of type S. The intended semantics —

the σ case of *erase* — is to preserve (copy) the value of this field. To be able to transform the rest of the input structure, we should demand that, for any value $s : S$ of the field, the remaining response description $E\ s$ refines the other remaining response description $D\ s$.

The other two cases Δ and ∇ of ROrn deal with mismatching fields in the two response descriptions being related and prompt *erase* to perform nontrivial horizontal transformations.

- The Δ case of ROrn states that $\sigma\ T\ E$ refines D, the former having an additional field of type T whose value is not retained — the Δ case of *erase* discards the value of this field. We still need to transform the rest of the input structure, so the Δ constructor demands that, for every possible value $t : T$ of the field, the response description D is refined by the remaining response description $E\ t$.
- Conversely, the ∇ case of ROrn states that E refines $\sigma\ S\ D$, the latter having an additional field of type S. The value of this field needs to be restored by the ∇ case of *erase*, so the ∇ constructor demands a default value $s : S$ for the field. To be able to continue with the transformation, the ∇ constructor also demands that the response description E refines the remaining response description $D\ s$.

Convention. Again we regard Δ as a binder and write $\Delta\langle t : T\rangle\ O\ t$ for $\Delta\ T\ (\lambda t \mapsto O\ t)$. Also, even though ∇ is not a binder, we write $\nabla\langle s\rangle\ O$ for $\nabla\ s\ O$ to save the parentheses around O when O is a complex expression. (*End of convention*.)

Example (*ornament from natural numbers to lists*)**.** For any $A : \mathsf{Set}$, there is an ornament from the description *NatD* of natural numbers to the description *ListD A* of lists:

```
NatD-ListD A : Orn ! NatD (ListD A)
NatD-ListD A (ok ▪) = σ ListTag λ { 'nil  ↦ v []
                                  ; 'cons ↦ Δ⟨ _ : A ⟩ v (refl :: []) }
```

There is only one response ornament in *NatD-ListD A* since the datatype of

lists is trivially indexed. The constructor tag is preserved (σ ListTag), and, in the cons case, the list element field is marked as additional by Δ. Consequently, the forgetful function

$$\mathit{forget}\ (\mathit{NatD\text{-}ListD}\ A)\ \{\blacksquare\}\ :\ \mathsf{List}\ A \to \mathsf{Nat}$$

discards all list elements from a list and returns its underlying natural number, i.e., its length. (*End of example.*)

Example (*ornament from lists to vectors*)**.** Again for any A : Set, there is an ornament from the description *ListD A* of lists to the description *VecD A* of vectors:

$$\begin{aligned}
&\mathit{ListD\text{-}VecD}\ A\ :\ \mathsf{Orn}\ !\ (\mathit{ListD}\ A)\ (\mathit{VecD}\ A)\\
&\mathit{ListD\text{-}VecD}\ A\ (\mathsf{ok\ zero}\quad)\ =\ \nabla\langle\ \text{`}\mathsf{nil}\ \rangle\quad \mathsf{v}\ []\\
&\mathit{ListD\text{-}VecD}\ A\ (\mathsf{ok}\ (\mathsf{suc}\ n))\ =\ \nabla\langle\ \text{`}\mathsf{cons}\ \rangle\ \sigma\langle\ _ : A\ \rangle\ \mathsf{v}\ (\mathsf{refl} :: [])\ \}
\end{aligned}$$

The response ornaments are indexed by Nat, since Nat is the index set of the datatype of vectors. We do pattern matching on the index request, resulting in two cases. In both cases, the constructor tag field exists for lists but not for vectors (since the constructor choice for vectors is determined from the index), so ∇ is used to insert the appropriate tag; in the suc case, the list element field is preserved by σ. Consequently, the forgetful function

$$\mathit{forget}\ (\mathit{ListD\text{-}VecD}\ A)\ :\ \{n\ :\ \mathsf{Nat}\} \to \mathsf{Vec}\ A\ n \to \mathsf{List}\ A$$

computes the underlying list of a vector. (*End of example.*)

Remark (*vertical invariance of ornamental relationship*)**.** It is worth emphasising again that ornaments encode only horizontal transformations, so datatypes related by ornaments necessarily have the same recursion patterns (as enforced by the v constructor) — ornamental relationship exists between list-like datatypes but not between lists and binary trees, for example. (*End of remark.*)

3.2.2 Ornamental descriptions

There is apparent similarity between, e.g., the description *ListD A* and the ornament *NatD-ListD A*, which is typical: frequently we define a new description

```
data ROrnDesc {I : Set} (J : Set) (e : J → I) : RDesc I → Set₁ where
  v : {is : List I} (js : ℙ is (InvImage e)) → ROrnDesc J e (v is)
  σ : (S : Set) {D : S → RDesc I}
      (OD : (s : S) → ROrnDesc J e (D s)) → ROrnDesc J e (σ S D)
  Δ : (T : Set) {D : RDesc I} (OD : T → ROrnDesc J e D) → ROrnDesc J e D
  ∇ : {S : Set} (s : S) {D : S → RDesc I}
      (OD : ROrnDesc J e (D s)) → ROrnDesc J e (σ S D)

und-ℙ : {I J : Set} {e : J → I} (is : List I) → ℙ is (InvImage e) → List J
und-ℙ []       ▪       = []
und-ℙ (i :: is) (j , js) = und j :: und-ℙ is js

toRDesc : {I J : Set} {e : J → I} {D : RDesc I} → ROrnDesc J e D → RDesc J
toRDesc (v {is} js) = v (und-ℙ is js)
toRDesc (σ S OD)    = σ⟨ s : S ⟩ toRDesc (OD s)
toRDesc (Δ T OD)    = σ⟨ t : T ⟩ toRDesc (OD t)
toRDesc (∇ s OD)    = toRDesc OD

toEq-ℙ : {I J : Set} {e : J → I}
         (is : List I) (js : ℙ is (InvImage e)) → 𝔼 e (und-ℙ is js) is
toEq-ℙ []       ▪       = []
toEq-ℙ (i :: is) (j , js) = toEq j :: toEq-ℙ is js

toROrn : {I J : Set} {e : J → I} {D : RDesc I} →
         (OD : ROrnDesc J e D) → ROrn e D (toRDesc OD)
toROrn (v js)    = v (toEq-ℙ _ js)
toROrn (σ S OD)  = σ⟨ s : S ⟩ toROrn (OD s)
toROrn (Δ T OD)  = Δ⟨ t : T ⟩ toROrn (OD t)
toROrn (∇ s OD)  = ∇⟨ s ⟩ (toROrn OD)

OrnDesc : {I : Set} (J : Set) (e : J → I) (D : Desc I) → Set₁
OrnDesc J e D = {i : I} (j : e ⁻¹ i) → ROrnDesc J e (D i)

⌊_⌋ : {I J : Set} {e : J → I} {D : Desc I} → OrnDesc J e D → Desc J
⌊ OD ⌋ j = toRDesc (OD (ok j))

⌈_⌉ : {I J : Set} {e : J → I} {D : Desc I}
      (OD : OrnDesc J e D) → Orn e D ⌊ OD ⌋
⌈ OD ⌉ (ok j) = toROrn (OD (ok j))
```

Figure 3.2 Definitions for ornamental descriptions.

(e.g. *ListD A*), intending it to be a more refined version of an existing one (e.g., *NatD*), and then immediately write an ornament from the latter to the former (e.g., *NatD-ListD A*). The syntactic structures of the new description and of the ornament are essentially the same, however, so the effort is duplicated. It would be more efficient if we could use the existing description as a template and just write a "relative description" specifying how to "patch" the template, and afterwards from this "relative description" extract a new description and an ornament from the template to the new description.

Ornamental descriptions are designed for this purpose; the definitions are shown in Figure 3.2 and closely follow the definitions for ornaments, having a upper-level type OrnDesc of ornamental descriptions which refers to a lower-level datatype ROrnDesc of response ornamental descriptions. An ornamental description looks like an annotated description, on which we can use a greater variety of constructors to mark differences from the template description. We think of an ornamental description

$$OD \;:\; \mathsf{OrnDesc}\ J\ e\ D$$

as simultaneously denoting a new description of type Desc J and an ornament from the template description D to the new description, and use floor and ceiling brackets $\lfloor \_ \rfloor$ and $\lceil \_ \rceil$ to resolve ambiguity: the new description is

$$\lfloor OD \rfloor \;:\; \mathsf{Desc}\ J$$

and the ornament is

$$\lceil OD \rceil \;:\; \mathsf{Orn}\ e\ D\ \lfloor OD \rfloor$$

Example (*ordered lists as an ornamentation of lists*)**.** Given $A \;:\; \mathsf{Set}$ with an ordering relation $\_\leqslant_A\_ \;:\; A \to A \to \mathsf{Set}$, we can define ordered lists on A by an ornamental description, using the description of lists as the template:

$$\begin{array}{l}
\mathit{OrdListOD}\ A\ _\leqslant_A_ \;:\; \mathsf{OrnDesc}\ A\ !\ (\mathit{ListD}\ A) \\
\mathit{OrdListOD}\ A\ _\leqslant_A_\ (\mathsf{ok}\ b)\ = \\
\quad \sigma\ \mathsf{ListTag}\ \lambda\ \{\ \text{`nil} \quad\quad \mapsto\ \mathsf{v}\ \blacksquare \\
\qquad\qquad\quad ;\ \text{`cons} \quad \mapsto\ \sigma\langle\, a : A \,\rangle\ \Delta\langle\, \mathit{leq} : b \leqslant_A a \,\rangle\ \mathsf{v}\ (a\ ,\ \blacksquare)\ \}
\end{array}$$

indexfirst data OrdList $A\ \_\leqslant_A\_ \;:\; A \to \mathsf{Set}$ **where**

$$\begin{aligned}&\mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ b\ \ni\ \mathsf{nil}\\ &\qquad\qquad\qquad\qquad\mid\ \mathsf{cons}\ (a\ :\ A)\ (\mathit{leq}\ :\ b \leqslant_A a)\ (\mathit{as}\ :\ \mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ a)\end{aligned}$$

If we read $\mathit{OrdListOD}\ A\ \_{\leqslant_A}\_$ as an annotated description, we can think of the *leq* field as being marked as additional (relative to the description of lists) by using Δ rather than σ. To decode $\mathit{OrdListOD}\ A\ \_{\leqslant_A}\_$ to an ordinary description of ordered lists, we write

$$\lfloor \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rfloor\ :\ \mathsf{Desc}\ A$$

and

$$\lceil \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rceil\ :\ \mathsf{Orn}\ !\ (\mathit{ListD}\ A)\ \lfloor \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rfloor$$

is an ornament from lists to ordered lists. (*End of example.*)

Example (*singleton ornamentation*)**.** Consider the following *singleton datatype* for lists:

$$\begin{aligned}&\textbf{indexfirst data}\ \mathsf{ListS}\ A\ :\ \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\ &\quad \mathsf{ListS}\ A\ []\ \ni\ \mathsf{nil}\\ &\quad \mathsf{ListS}\ A\ (x :: xs)\ \ni\ \mathsf{cons}\ (s\ :\ \mathsf{ListS}\ A\ xs)\end{aligned}$$

For each type $\mathsf{ListS}\ A\ xs$, there is exactly one (canonical) inhabitant (hence the name "singleton datatype" [Monnier and Haguenauer, 2010]), which has the same vertical structure as *xs* and is devoid of any horizontal contents. We can encode the datatype as an ornamental description relative to $\mathit{ListD}\ A$:

$$\begin{aligned}&\mathit{ListSOD}\ :\ (A\ :\ \mathsf{Set}) \to \mathsf{OrnDesc}\ (\mathsf{List}\ A)\ !\ (\mathit{ListD}\ A)\\ &\mathit{ListSOD}\ A\ (\mathsf{ok}\ [])\ =\ \nabla\langle\ \text{`}\mathsf{nil}\ \rangle\ \mathsf{v}\ \blacksquare\\ &\mathit{ListSOD}\ A\ (\mathsf{ok}\ (x :: xs))\ =\ \nabla\langle\ \text{`}\mathsf{cons}\ \rangle\ \nabla\langle\ x\ \rangle\ \mathsf{v}\ (\mathsf{ok}\ xs\ ,\ \blacksquare)\end{aligned}$$

which does pattern matching on the index request, in each case restricts the constructor choice to the one matched against, and in the cons case deletes the element field and sets the index of the recursive position to be the value of the tail in the pattern. In general, we can define a parametrised ornamental description

$$\mathit{singletonOD}\ :\ \{I\ :\ \mathsf{Set}\}\ (D\ :\ \mathsf{Desc}\ I) \to \mathsf{OrnDesc}\ (\Sigma\ I\ (\mu\ D))\ \mathsf{outl}\ D$$

called the *singleton ornamental description*, which delivers a singleton datatype as an ornamentation of any datatype. The complete definition is

$$
\begin{aligned}
&\mathit{erode} : \{I : \mathsf{Set}\}\ (D : \mathsf{RDesc}\ I)\ \{J : I \to \mathsf{Set}\} \to \\
&\qquad [\![\ D\ ]\!]\ J \to \mathsf{ROrnDesc}\ (\Sigma\ I\ J)\ \mathsf{outl}\ D \\
&\mathit{erode}\ (\mathsf{v}\ \mathit{is})\quad \mathit{js} \quad = \mathsf{v}\ (\mathbb{P}\text{-}\mathit{map}\ (\lambda\ \{i\}\ j \mapsto \mathsf{ok}\ (i\ ,\ j))\ \mathit{is}\ \mathit{js}) \\
&\mathit{erode}\ (\sigma\ S\ D)\ (s\ ,\ \mathit{js}) \ = \nabla\langle\ s\ \rangle\ \ \mathit{erode}\ (D\ s)\ \mathit{js}
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{singletonOD} : \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I) \to \mathsf{OrnDesc}\ (\Sigma\ I\ (\mu\ D))\ \mathsf{outl}\ D \\
&\mathit{singletonOD}\ D\ (\mathsf{ok}\ (i\ ,\ \mathsf{con}\ \mathit{ds})) \ = \ \mathit{erode}\ (D\ i)\ \mathit{ds}
\end{aligned}
$$

where

$$
\begin{aligned}
&\mathbb{P}\text{-}\mathit{map} : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\} \to (X \rightrightarrows Y) \to \\
&\qquad (\mathit{is} : \mathsf{List}\ I) \to \mathbb{P}\ \mathit{is}\ X \to \mathbb{P}\ \mathit{is}\ Y \\
&\mathbb{P}\text{-}\mathit{map}\ f\ [\,]\qquad \blacksquare \qquad = \ \blacksquare \\
&\mathbb{P}\text{-}\mathit{map}\ f\ (i :: \mathit{is})\ (x\ ,\ \mathit{xs}) \ = \ f\ x\ ,\ \mathbb{P}\text{-}\mathit{map}\ f\ \mathit{is}\ \mathit{xs}
\end{aligned}
$$

Note that *erode* deletes all fields (i.e., horizontal contents), drawing default values from the index request, retaining only the vertical structure. We will see in Section 3.3 that singleton ornamentation plays a key role in the ornament–refinement framework. (*End of example*.)

Remark (*ornaments as relations*)**.** We define ornaments as relations between descriptions (indexed with an erasure function), whereas the original ornaments [McBride, 2011; Dagand and McBride, 2012b] are rebranded as ornamental descriptions. One obvious advantage of relational ornaments is that they can arise between existing descriptions, whereas ornamental descriptions always produce (definitionally) new descriptions at the more informative end. A consequence is that there can be multiple ornaments between a pair of descriptions. For example, consider the following description of a datatype consisting of two fields of the same type:

$$
\begin{aligned}
&\mathit{SquareD} : (A : \mathsf{Set}) \to \mathsf{Desc}\ \top \\
&\mathit{SquareD}\ A\ \blacksquare \ = \ \sigma\langle\ \_ : A\ \rangle\ \ \sigma\langle\ \_ : A\ \rangle\ \ \mathsf{v}\ [\,]
\end{aligned}
$$

Between *SquareD A* and itself, we have the identity ornament

$$\lambda\ \{\ \blacksquare \ \mapsto\ \ \sigma\langle\ \_ : A\ \rangle\ \ \sigma\langle\ \_ \ : \ A\ \rangle\ \ \mathsf{v}\ [\,]\ \}$$

and the "swapping" ornament

$$\lambda\ \{\ \blacksquare \ \mapsto\ \ \Delta\langle\ x : A\ \rangle\ \ \Delta\langle\ y : A\ \rangle\ \ \nabla\langle\ y\ \rangle\ \ \nabla\langle\ x\ \rangle\ \ \mathsf{v}\ [\,]\ \}$$

whose forgetful function swaps the two fields.

The other advantage of relational ornaments is that they allow new datatypes to arise at the less informative end. For example, *coproduct of signatures* as used in, e.g., data types à la carte [Swierstra, 2008], can be implemented naturally with relational ornaments but not with ornamental descriptions. In more detail: Consider (a simplistic version of) *tagged descriptions* [Chapman et al., 2010], which are descriptions that, for any index request, always respond with a constructor field first. A tagged description with index set $I : \mathsf{Set}$ thus consists of a family of types $C : I \to \mathsf{Set}$, where each $C\ i$ is the set of constructor tags for the index request $i : I$, and a family of subsequent response descriptions for each constructor tag.

$$\mathsf{TDesc} : \mathsf{Set} \to \mathsf{Set}_1$$
$$\mathsf{TDesc}\ I\ =\ \Sigma\langle\, C : I \to \mathsf{Set} \,\rangle\ ((i : I) \to C\ i \to \mathsf{RDesc}\ I)$$

Tagged descriptions are decoded to ordinary descriptions by

$$\lfloor \_ \rfloor_T : \{I : \mathsf{Set}\} \to \mathsf{TDesc}\ I \to \mathsf{Desc}\ I$$
$$\lfloor C\ ,\ D \rfloor_T\ i\ =\ \sigma\ (C\ i)\ (D\ i)$$

We can then define binary coproduct of tagged descriptions, which sums the corresponding constructor fields, as follows:

coproduct-related definitions

$$\_\oplus\_ : \{I : \mathsf{Set}\} \to \mathsf{TDesc}\ I \to \mathsf{TDesc}\ I \to \mathsf{TDesc}\ I$$
$$(C\ ,\ D) \oplus (C'\ ,\ D')\ =\ (\lambda\, i \mapsto C\ i + C'\ i)\ ,\ (\lambda\, i \mapsto D\ i \triangledown D'\ i)$$

Now given two tagged descriptions $tD\ =\ (C\ ,\ D)$ and $tD'\ =\ (C'\ ,\ D')$ of type $\mathsf{TDesc}\ I$, there are two ornaments from $\lfloor tD \oplus tD' \rfloor_T$ to $\lfloor tD \rfloor_T$ and $\lfloor tD' \rfloor_T$

$$inlOrn\ :\ \mathsf{Orn}\ id\ \lfloor tD \oplus tD' \rfloor_T\ \lfloor tD \rfloor_T$$
$$inlOrn\ (\mathsf{ok}\ i)\ =\ \Delta\langle\, c\ : C\ i \,\rangle\ \ \nabla\langle\, \mathsf{inl}\ c \,\rangle\ \ idOrn\ (D\ i\ c)$$
$$inrOrn\ :\ \mathsf{Orn}\ id\ \lfloor tD \oplus tD' \rfloor_T\ \lfloor tD' \rfloor_T$$
$$inrOrn\ (\mathsf{ok}\ i)\ =\ \Delta\langle\, c' : C'\ i \,\rangle\ \ \nabla\langle\, \mathsf{inr}\ c' \,\rangle\ \ idOrn\ (D'\ i\ c')$$

whose forgetful functions perform suitable injection of constructor tags. Note that the synthesised new description $\lfloor\ tD\ \oplus\ tD'\ \rfloor_T$ is at the less informative end of *inlOrn* and *inrOrn*. (This, of course, is not a complete implementation

of data types à la carte and requires more engineering for practical use.) (*End of remark.*)

3.2.3 Parallel composition of ornaments

intro — analysis for composability

The generic scenario is illustrated below: Chapter 4



Given three descriptions D : Desc I, E : Desc J, and F : Desc K and two ornaments O : Orn e D E and P : Orn e D F independently specifying how D is refined to E and F, we can compute an ornamental description

$O \otimes P$: OrnDesc $(e \bowtie f)$ *pull* D

Intuitively, since both O and P encode modifications to the same base description D, we can commit all modifications encoded by O and P to D to get a new description $\lfloor O \otimes P \rfloor$, and encode all these modifications in one ornament $\lceil O \otimes P \rceil$. (This merging of two sets of modifications is best characterised by a category-theoretic pullback, which we defer until Chapter 4.) The forgetful function of the ornament $\lceil O \otimes P \rceil$ removes all modifications, taking $\mu \lfloor O \otimes P \rfloor$ all the way back to the base datatype μD; there are also two *difference ornaments*

diffOrn-l O P : Orn outl E $\lfloor O \otimes P \rfloor$ -- left difference ornament
diffOrn-r O P : Orn outr F $\lfloor O \otimes P \rfloor$ -- right difference ornament

which give rise to "less forgetful" functions taking $\mu\ \lfloor O \otimes P \rfloor$ to $\mu\ E$ and $\mu\ F$, such that both

$$\mathit{forget}\ O \circ \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$$

and

$$\mathit{forget}\ P \circ \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P)$$

are extensionally equal to $\mathit{forget}\ \lceil O \otimes P \rceil$.

Example (*ordered vectors*)**.** Consider the two ornaments $\lceil \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rceil$ from lists to ordered lists and $\mathit{ListD\text{-}VecD}\ A$ from lists to vectors. Composing them in parallel gives us an ornamental description from which we can decode (i) a new datatype of ordered vectors

$$\begin{aligned}
&\mathsf{OrdVec}\ A\ _{\leqslant_A}_ : A \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{OrdVec}\ A\ _{\leqslant_A}_\ b\ n\ = \\
&\quad \mu\ \lfloor \lceil \mathit{OrdListOD}\ A\ _{\leqslant_A}_ \rceil \otimes \mathit{ListD\text{-}VecD}\ A \rfloor\ (\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ n)) \\
&\textbf{indexfirst data}\ \mathsf{OrdVec}\ A\ _{\leqslant_A}_ : A \to \mathsf{Nat} \to \mathsf{Set}\ \textbf{where} \\
&\quad \mathsf{OrdVec}\ A\ _{\leqslant_A}_\ b\ \mathsf{zero} \quad \ni\ \mathsf{nil} \\
&\quad \mathsf{OrdVec}\ A\ _{\leqslant_A}_\ b\ (\mathsf{suc}\ n) \ \ni\ \mathsf{cons}\ (a : A)\ (\mathit{leq} : b \leqslant_A a) \\
&\qquad\qquad\qquad\qquad\qquad\qquad (\mathit{as} : \mathsf{OrdVec}\ A\ _{\leqslant_A}_\ a\ n)
\end{aligned}$$

and (ii) an ornament whose forgetful function converts ordered vectors to plain lists, retaining the list elements. The forgetful functions of the difference ornaments convert ordered vectors to ordered lists and vectors, removing only length and ordering information respectively. (*End of example*.)

The complete definitions for parallel composition are shown in Figure 3.3. The core definition is *pcROD*, which analyses and merges the modifications encoded by two response ornaments into a response ornamental description at the level of individual fields. Below are some representative cases of *pcROD*.

- When both response ornaments use σ, both of them preserve the same field in the base description — no modification is made. Consequently, the field is preserved in the resulting response ornamental description as well.

  $$\mathit{pcROD}\ (\sigma\ S\ O)\ (\sigma\ .S\ P)\ =\ \sigma \langle\, s : S \,\rangle\ \mathit{pcROD}\ (O\ s)\ (P\ s)$$

```
pc-𝔼 : {I J K : Set} {e : J → I} {f : K → I} →
       {is : List I} {js : List J} {ks : List K} →
       𝔼 e js is → 𝔼 f ks is → ℙ is (InvImage pull)
pc-𝔼          []          []          = ▪
pc-𝔼 {e := e} {f} (eeq :: eeqs) (feq :: feqs) = ok (fromEq e eeq , fromEq f feq) ,
                                                pc-𝔼 eeqs feqs

mutual
  pcROD : {I J K : Set} {e : J → I} {f : K → I}
          {D : RDesc I} {E : RDesc J} {F : RDesc K} →
          ROrn e D E → ROrn f D F → ROrnDesc (e ⋈ f) pull D
  pcROD (v eeqs) (v feqs)   = v (pc-𝔼 eeqs feqs)
  pcROD (v eeqs) (Δ T P)    = Δ ⟨ t : T ⟩ pcROD (v eeqs) (P t)
  pcROD (σ S O) (σ .S P)    = σ ⟨ s : S ⟩ pcROD (O s)    (P s)
  pcROD (σ f O) (Δ T P)     = Δ ⟨ t : T ⟩ pcROD (σ f O) (P t)
  pcROD (σ S O) (∇ s P)     = ∇ ⟨ s ⟩     pcROD (O s)    P
  pcROD (Δ T O) P           = Δ ⟨ t : T ⟩ pcROD (O t)    P
  pcROD (∇ s O) (σ S P)     = ∇ ⟨ s ⟩     pcROD O        (P s)
  pcROD (∇ s O) (Δ T P)     = Δ ⟨ t : T ⟩ pcROD (∇ s O) (P t)
  pcROD (∇ s O) (∇ s′ P)    = Δ (s ≡ s′) (pcROD-double∇ O P)

  pcROD-double∇ :
    {I J K S : Set} {e : J → I} {f : K → I}
    {D : S → RDesc I} {E : RDesc J} {F : RDesc K} {s s′ : S} →
    ROrn e (D s) E → ROrn f (D s′) F →
    s ≡ s′ → ROrnDesc (e ⋈ f) pull (σ S D)
  pcROD-double∇ {s := s} O P refl = ∇ ⟨ s ⟩ pcROD O P

_⊗_ : {I J K : Set} {e : J → I} {f : K → I}
      {D : Desc I} {E : Desc J} {F : Desc K} →
      Orn e D E → Orn f D F → OrnDesc (e ⋈ f) pull D
(O ⊗ P) (ok (j , k)) = pcROD (O j) (P k)
```

Figure 3.3 Definitions for parallel composition of ornaments.

- When one of the response ornaments uses Δ to mark the addition of a new field, that field would be added into the resulting response ornamental description, like in

  $$\mathit{pcROD}\ (\Delta\ T\ O)\ P\ =\ \Delta\langle\, t : T\,\rangle\ \mathit{pcROD}\ (O\ t)\ P$$

- If one of the response ornaments retains a field by σ and the other deletes it by ∇, the only modification to the field is deletion, and thus the field is deleted in the resulting response ornamental description, like in

  $$\mathit{pcROD}\ (\sigma\ S\ O)\ (\nabla\ s\ P)\ =\ \nabla\langle\, s\,\rangle\ \mathit{pcROD}\ (O\ s)\ P$$

- The most interesting case is when both response ornaments encode deletion: we would add an equality field demanding that the default values supplied in the two response ornaments be equal,

  $$\mathit{pcROD}\ (\nabla\ s\ O)\ (\nabla\ s'\ P)\ =\ \Delta\ (s\ \equiv\ s')\ (\mathit{pcROD\text{-}double}\nabla\ O\ P)$$

  and then *pcROD-double*∇ puts the deletion into the resulting response ornamental description after matching the proof of the equality field with refl.

  $$\mathit{pcROD\text{-}double}\nabla\ \{s := s\}\ O\ P\ \mathsf{refl}\ =\ \nabla\langle\, s\,\rangle\ \mathit{pcROD}\ O\ P$$

  It might seem bizarre that two deletions results in a new field (and a deletion), but consider this informally described scenario: A field $\sigma\ S$ in the base response description is refined by two independent response ornaments

  $$\Delta\langle\, t : T\,\rangle\quad\nabla\langle\, g\ t\,\rangle$$

  and

  $$\Delta\langle\, u : U\,\rangle\ \nabla\langle\, h\ u\,\rangle$$

  That is, instead of S-values, the response descriptions at the more informative end of the two response ornaments use T- and U-values at this position, which are erased to their underlying S-value by $g\ :\ T \to S$ and $h\ :\ U \to S$ respectively. Composing the two response ornaments in parallel, we get

  $$\Delta\langle\, t : T\,\rangle\ \Delta\langle\, u : U\,\rangle\ \Delta\langle\, \_ : g\ t\ \equiv\ h\ u\,\rangle\ \nabla\langle\, g\ t\,\rangle$$

where the added equality field completes the construction of a set-theoretic pullback of g and h. Here indeed we need a pullback: When we have an actual value for the field $\sigma\ S$, which gets refined to values of types T and U, the generic way to mix the two refining values is to store them both, as a product. If we wish to retrieve the underlying value of type S, we can either extract the value of type T and apply g to it or extract the value of type U and apply h to it, and through either path we should get the same underlying value. So the product should really be a pullback to ensure this. Chapter 4

Example (*ornamental description of ordered vectors*)**.** Composing the ornaments $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ and *ListD-VecD A* in parallel yields the following ornamental description relative to *ListD A*:

$$\begin{aligned}\lambda\ \{\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ \mathsf{zero}\quad))\ \mapsto\ \nabla\langle\ \text{`nil}\ \rangle\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\mathsf{suc}\ n)))\ \mapsto\ \nabla\langle\ \text{`cons}\ \rangle\ \sigma\langle\ a : A\ \rangle\\ &\qquad\qquad\qquad\qquad\Delta\langle\ \_ : b \leqslant_A a\ \rangle\ \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ a\ ,\ \mathsf{ok}\ n)\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates modifications from $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ and darker box from *ListD-VecD A*. (*End of example.*)

Finally, the definitions for left difference ornament are shown in Figure 3.4. Left difference ornament has the same structure as parallel composition, but records only modifications from the right-hand side ornament. For example, the case

$$\mathit{diffROrn\text{-}l}\ (\sigma\ S\ O)\ (\nabla\ s\ P)\ =\ \nabla\langle\ s\ \rangle\ \mathit{diffROrn\text{-}l}\ (O\ s)\ P$$

is the same as the corresponding case of *pcROD*, since the deletion comes from the right-hand side response ornament, whereas the case

$$\mathit{diffROrn\text{-}l}\ (\Delta\ T\ O)\ P\ =\ \sigma\langle\ t : T\ \rangle\ \mathit{diffROrn\text{-}l}\ (O\ t)\ P$$

produces σ (a preservation) rather than Δ (a modification) as in the corresponding case of *pcROD*, since the addition comes from the left-hand side response ornament. We can then see that the composition of the forgetful functions

$$\mathit{forget}\ O \circ \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$$

is indeed extensionally equal to $\mathit{forget}\ \lceil\ O \otimes P\ \rceil$, since $\mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$ removes modifications encoded in the right-hand side ornament and then

```
diff-𝔼-l :
  {I J K : Set} {e : J → I} {f : K → I} →
  {is : List I} {js : List J} {ks : List K} →
  (eeqs : 𝔼 e js is) (feqs : 𝔼 f ks is) → 𝔼 outl (und-ℙ is (pc-𝔼 eeqs feqs)) js
diff-𝔼-l          []          []          = []
diff-𝔼-l {e := e} (eeq :: eeqs) (feq :: feqs) = und-fromEq e eeq :: diff-𝔼-l eeqs feqs
mutual
  diffROrn-l :
    {I J K : Set} {e : J → I} {f : K → I} →
    {D : RDesc I} {E : RDesc J} {F : RDesc K} →
    (O : ROrn e D E) (P : ROrn f D F) → ROrn outl E (toRDesc (pcROD O P))
  diffROrn-l (v eeqs) (v feqs)   = v (diff-𝔼-l eeqs feqs)
  diffROrn-l (v eeqs) (Δ T P)    = Δ ⟨ t : T ⟩ diffROrn-l (v eeqs) (P t)
  diffROrn-l (σ S O) (σ .S P)    = σ ⟨ s : S ⟩ diffROrn-l (O s)    (P s)
  diffROrn-l (σ S O) (Δ T P)     = Δ ⟨ t : T ⟩ diffROrn-l (σ S O) (P t)
  diffROrn-l (σ S O) (∇ s P)     = ∇ ⟨ s ⟩     diffROrn-l (O s)    P
  diffROrn-l (Δ T O) P           = σ ⟨ t : T ⟩ diffROrn-l (O t)    P
  diffROrn-l (∇ s O) (σ S P)     =             diffROrn-l O        (P s)
  diffROrn-l (∇ s O) (Δ T P)     = Δ ⟨ t : T ⟩ diffROrn-l (∇ s O) (P t)
  diffROrn-l (∇ s O) (∇ s′ P)    = Δ (s ≡ s′) (diffROrn-l-double∇ O P)
  diffROrn-l-double∇ :
    {I J K S : Set} {e : J → I} {f : K → I} →
    {D : S → RDesc I} {E : RDesc J} {F : RDesc K} {s s′ : S} →
    (O : ROrn e (D s) E) (P : ROrn f (D s′) F) (eq : s ≡ s′) →
    ROrn outl E (toRDesc (pcROD-double∇ O P eq))
  diffROrn-l-double∇ O P refl = diffROrn-l O P
diffOrn-l :
  {I J K : Set} {e : J → I} {f : K → I} →
  {D : Desc I} {E : Desc J} {F : Desc K} →
  (O : Orn e D E) (P : Orn f D F) → Orn outl E ⌊ O ⊗ P ⌋
diffOrn-l O P (ok (j , k)) = diffROrn-l (O j) (P k)
```

Figure 3.4 Definitions for left difference ornament.

forget O removes modifications encoded in the left-hand side ornament. Right difference ornament is defined analogously and is omitted from the presentation.

3.3 Refinement semantics of ornaments

Every ornament $O : \mathsf{Orn}\ e\ D\ E$ induces a refinement family from $\mu\ D$ to $\mu\ E$. That is, we can construct a function

$$RSem : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{Desc}\ I\}\ \{E : \mathsf{Desc}\ J\} \to \mathsf{Orn}\ e\ D\ E \to \mathsf{FRefinement}\ e\ (\mu\ D)\ (\mu\ E)$$

which is called the *refinement semantics* of ornaments.

intro

3.3.1 Optimised predicates

Our most important task for now is to construct a promotion predicate

$$\mathsf{OptP} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{Desc}\ I\}\ \{E : \mathsf{Desc}\ J\} \to (O : \mathsf{Orn}\ e\ D\ E)\ \{i : I\}\ (j : e^{-1}\ i)\ (x : \mu\ D\ i) \to \mathsf{Set}$$

which is called the *optimised predicate* for the ornament O. Given $x : \mu\ D\ i$, a proof of type $\mathsf{OptP}\ O\ j\ x$ contains the necessary information for complementing x and forming an inhabitant y of type $\mu\ E\ (und\ j)$ with the same recursive structure — the proof is the "horizontal" difference between y and x, speaking in terms of the two-dimensional metaphor. Such a proof should have the same vertical structure as x, and, at each recursive node, store horizontally only those data marked as modified by the ornament. For example, if we are promoting the natural number

Optimised in what sense?

$$two = \mathsf{con}\ (\text{`cons}\ ,\ \mathsf{con}\ (\text{`cons}\ ,$$

$$\begin{array}{l}\mathsf{con}\ (\text{`}\mathsf{nil}\quad ,\\ \quad \blacksquare)\ ,\ \blacksquare)\ ,\ \blacksquare)\ :\ \mu\ \mathit{NatD}\ \blacksquare\end{array}$$

to a list, an optimised promotion proof would look like

$$\begin{array}{l}p\ =\ \mathsf{con}\ (a\ ,\\ \qquad \mathsf{con}\ (a'\ ,\\ \qquad \mathsf{con}\ (\\ \qquad\quad \blacksquare)\ ,\ \blacksquare)\ ,\ \blacksquare)\ :\ \mathsf{OptP}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathsf{ok}\ \blacksquare)\ \mathit{two}\end{array}$$

where a and a' are some elements of type A, so we get a list by zipping together *two* and r node by node:

$$\begin{array}{l}\mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a\ \ ,\\ \mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a'\ ,\\ \mathsf{con}\ (\text{`}\mathsf{nil}\quad ,\\ \quad \blacksquare)\ ,\ \blacksquare)\ ,\ \blacksquare)\ :\ \mu\ (\mathit{ListD}\ A)\ \blacksquare\end{array}$$

Note that p contains only values of the field marked as additional by Δ in the ornament *NatD-ListD A*. The constructor tags are essential for determining the recursive structure of p, but instead of being stored in p, they are derived from *two*, which is part of the index of the type of p. In general, here is how we compute an ornamental description for such proofs, using D as the template: we incorporate the modifications made by O, and delete the fields that already exist in D, whose default values are derived in the index-first fashion from the inhabitant being promoted, which appears in the index of the type of a proof. The deletion is independent of O and can be performed by the *singleton ornament* for D (Section 3.2.2), so the desired ornamental description is produced by the parallel composition of O and $\lceil \mathit{singletonOD}\ D \rceil$:

$$\begin{array}{l}\mathit{OptPOD}\ :\ \{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to\\ \qquad\qquad \mathsf{Orn}\ e\ D\ E \to \mathsf{OrnDesc}\ (e \bowtie \mathsf{outl})\ \mathit{pull}\ D\\ \mathit{OptPOD}\ \{D := D\}\ O\ =\ O \otimes \lceil \mathit{singletonOD}\ D \rceil\end{array}$$

where outl has type $\Sigma\ I\ (\mu\ D) \to I$. The optimised predicate, then, is the least fixed point of the description.

$$\begin{array}{l}\mathsf{OptP}\ :\ \{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to\\ \qquad (O\ :\ \mathsf{Orn}\ e\ D\ E)\ \{i\ :\ I\}\ (j\ :\ e^{-1}\ i)\ (x\ :\ \mu\ D\ i) \to \mathsf{Set}\end{array}$$

$$\mathsf{OptP}\ O\ \{i\}\ j\ d\ =\ \mu\ \lfloor \mathit{OptPOD}\ O \rfloor\ (j\ ,\ (\mathsf{ok}\ (i\ ,\ d)))$$

Example (*index-first vectors as an optimised predicate*)**.** The optimised predicate for the ornament *NatD-ListD A* from natural numbers to lists is the datatype of index-first vectors. Expanding the definition of the ornamental description *OptPOD* (*NatD-ListD A*) relative to *NatD*:

$$\begin{aligned}\lambda\ \{\ &(\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ \mathsf{zero}\ )))\ \mapsto\ \nabla\langle\ \text{`}\mathsf{nil}\ \rangle\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ \mathsf{suc}\ n)))\ \mapsto\ \nabla\langle\ \text{`}\mathsf{cons}\ \rangle\ \Delta\langle\ \_ : A\ \rangle\\ &\qquad\qquad \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ n))\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates contributions from the ornament *NatD-ListD A* and darker box from the singleton ornament $\lceil \mathit{singletonOD}\ \mathit{NatD} \rceil$, we see that the ornamental description indeed yields the datatype of index-first vectors (albeit indexed by a more heavily packaged datatype of natural numbers). (*End of example.*)

Example (*predicate characterising ordered lists*)**.** The optimised predicate for the ornament $\lceil\ \mathit{OrdListOD}\ A\ \_\leqslant_A\_\ \rceil$ from lists to ordered lists is given by the ornamental description *OptPOD* $\lceil\ \mathit{OrdListOD}\ A\ \_\leqslant_A\_\ \rceil$ relative to *ListD A*, which expands to

$$\begin{aligned}\lambda\ \{\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ []\quad\ )))\ \mapsto\ \nabla\langle\ \text{`}\mathsf{nil}\ \rangle\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ a :: as)))\ \mapsto\ \nabla\langle\ \text{`}\mathsf{cons}\ \rangle\ \nabla\langle\ a\ \rangle\ \Delta\langle\ \mathit{leq} : b \leqslant_A a\ \rangle\\ &\qquad\qquad \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ a\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ as))\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates contributions from $\lceil\ \mathit{OrdListOD}\ A\ \_\leqslant_A\_\ \rceil$ and darker box from $\lceil \mathit{singletonOD}\ (\mathit{ListD}\ A) \rceil$.

$$\begin{aligned}&\textbf{indexfirst data}\ \mathsf{Ordered}\ A\ \_\leqslant_A\_\ :\ A \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\ &\quad \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ []\qquad\ \ni\ \mathsf{nil}\\ &\quad \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ (a :: as)\ \ni\ \mathsf{cons}\ (\mathit{leq}\ :\ b \leqslant_A a)\ (o\ :\ \mathsf{Ordered}\ A\ \_\leqslant_A\_\ a\ as)\end{aligned}$$

A proof of $\mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ as$ consists of exactly the inequality proofs necessary for ensuring that *as* is ordered and bounded below by *b* — the representation of such a proof is optimised, justifying the name "optimised predicate". (*End of example.*)

Example (*inductive length predicate on lists*)**.** The optimised predicate for the ornament *ListD-VecD A* from lists to vectors is produced by the ornamental description *OptPOD* (*ListD-VecD A*) relative to *ListD A*:

$$
\begin{aligned}
\lambda\,\{\ & (\mathsf{ok}\ (\mathsf{ok}\ \mathsf{zero}\ \ \ , \mathsf{ok}\ (\blacksquare\ , []\ \ \ \ ))) \mapsto \Delta\langle\ \_ : \text{`nil} \equiv \text{`nil}\ \rangle\ \nabla\langle\ \text{`nil}\ \rangle\ \mathsf{v}\ \blacksquare \\
& ;\ (\mathsf{ok}\ (\mathsf{ok}\ \mathsf{zero}\ \ \ , \mathsf{ok}\ (\blacksquare\ , a :: as))) \mapsto \Delta\ (\text{`nil} \equiv \text{`cons})\ \lambda\ () \\
& ;\ (\mathsf{ok}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ , \mathsf{ok}\ (\blacksquare\ , []\ \ \ \ ))) \mapsto \Delta\ (\text{`cons} \equiv \text{`nil})\ \lambda\ () \\
& ;\ (\mathsf{ok}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ , \mathsf{ok}\ (\blacksquare\ , a :: as))) \mapsto \Delta\langle\ \_ : \text{`cons} \equiv \text{`cons}\ \rangle\ \nabla\langle\ \text{`cons}\ \rangle \\
& \qquad\qquad \nabla\langle\ a\ \rangle\ \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ n\ , \mathsf{ok}\ (\blacksquare\ , as))\ , \blacksquare)\ \}
\end{aligned}
$$

where lighter box indicates contributions from *ListD-VecD A* and darker box from $\lceil$ *singletonOD* (*ListD A*) $\rceil$. Both ornaments perform pattern matching and accordingly restrict constructor choices by ∇, so the resulting four cases all start with an equality field demanding that the constructor choices specified by the two ornaments are equal.

- In the first and last cases, where the specified constructor choices match, the equality proof obligation can be successfully discharged and the response ornamental description can continue after installing the constructor choice by ∇;
- in the middle two cases, where the specified constructor choices mismatch, the equality is obviously unprovable and the rest of the response ornamental description is (extensionally) the empty function $\lambda\ ()$.

Thus, in effect, the ornamental description produces the following inductive length predicate on lists:

$$
\begin{aligned}
&\textbf{indexfirst data}\ \mathsf{Length}\ A\ :\ \mathsf{Nat} \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where} \\
&\quad \mathsf{Length}\ A\ \mathsf{zero}\ \ \ [] \ \ \ \ \ni\ \mathsf{nil} \\
&\quad \mathsf{Length}\ A\ \mathsf{zero}\ \ \ (a :: as)\ \not\ni \\
&\quad \mathsf{Length}\ A\ (\mathsf{suc}\ n)\ [] \ \ \ \ \not\ni \\
&\quad \mathsf{Length}\ A\ (\mathsf{suc}\ n)\ (a :: as)\ \ni\ \mathsf{cons}\ (l\ :\ \mathsf{Length}\ A\ n\ as)
\end{aligned}
$$

where $\not\ni$ indicates that a case is uninhabited. (*End of example.*)

We have thus determined the promotion predicate used by the refinement semantics of ornaments to be the optimised predicate:

$$
\begin{aligned}
&\mathit{RSem} \;:\; \{I\,J \;:\; \mathsf{Set}\}\;\{e \;:\; J \to I\}\;\{D \;:\; \mathsf{Desc}\;I\}\;\{E \;:\; \mathsf{Desc}\;J\} \to \\
&\qquad \mathsf{Orn}\;e\;D\;E \to \mathsf{FRefinement}\;e\;(\mu\;D)\;(\mu\;E) \\
&\mathit{RSem}\;O\;j \;=\; \mathbf{record}\;\{\;P \;=\; \mathsf{OptP}\;O\;j \\
&\qquad\qquad\qquad\qquad ;\; i \;=\; \mathit{ornPromIso}\;O\;j\;\}
\end{aligned}
$$

We call *ornPromIso* the *ornamental conversion isomorphisms*, whose type is

$$
\begin{aligned}
&\mathit{ornPromIso}\;: \\
&\quad \{I\,J \;:\; \mathsf{Set}\}\;\{e \;:\; J \to I\}\;\{D \;:\; \mathsf{Desc}\;I\}\;\{E \;:\; \mathsf{Desc}\;J\}\;(O \;:\; \mathsf{Orn}\;e\;D\;E) \to \\
&\quad \{i \;:\; I\}\;(j \;:\; e^{-1}\;i) \to \mu\;E\;(\mathit{und}\;j) \;\cong\; \Sigma\langle\, x \;:\; \mu\;D\;i\,\rangle\;\;\mathsf{OptP}\;O\;j\;x
\end{aligned}
$$

The construction of *ornPromIso* will be deferred until Chapter 4.

3.3.2 Predicate swapping for parallel composition

An ornament describes differences between two datatypes, and the optimised predicate for the ornament is the datatype of differences between inhabitants of the two datatypes. To promote an inhabitant from the less informative end to the more informative end of the ornament using its refinement semantics, we give a proof that the object satisfies the optimised predicate for the ornament. If, however, the ornament is a parallel composition, say $\lceil\, O \otimes P \,\rceil$, then the differences recorded in the ornament are simply collected from the component ornaments O and P. Consequently, it should suffice to give separate proofs that the inhabitant satisfies the optimised predicates for O and P, instead of a proof that it satisfies the monolithic optimised predicate induced by $\lceil O \otimes P \rceil$. We are thus led to prove that the optimised predicate for $\lceil\, O \otimes P \,\rceil$ amounts to the pointwise conjunction of the optimised predicates for O and P. More precisely: if $O \;:\; \mathsf{Orn}\;e\;D\;E$ and $P \;:\; \mathsf{Orn}\;f\;D\;F$ where $D \;:\; \mathsf{Desc}\;I$, $E \;:\; \mathsf{Desc}\;J$, and $F \;:\; \mathsf{Desc}\;K$, then we expect the existence of the *modularity isomorphisms*

$$\mathsf{OptP}\;\lceil O \otimes P \rceil\;(\mathsf{ok}\;(j\,,\,k))\;x \;\cong\; \mathsf{OptP}\;O\;j\;x \times \mathsf{OptP}\;P\;k\;x$$

for all $i \;:\; I$, $j \;:\; e^{-1}\;i$, $k \;:\; f^{-1}\;i$, and $x \;:\; \mu\;D\;i$.

Example (*promotion predicate from lists to ordered vectors*)**.** The optimised predicate for the ornament $\lceil \lceil\, \mathit{OrdListOD}\;A\;\_{\leqslant_A}\_ \,\rceil \otimes \mathit{ListD\text{-}VecD}\;A \,\rceil$ from lists to

ordered vectors is

$$\begin{array}{l}\textbf{indexfirst data}\ \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ :\ A \to \mathsf{Nat} \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\ \quad \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ \mathsf{zero}\quad [\,]\quad\ \ni\ \mathsf{nil}\\ \quad \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ \mathsf{zero}\quad (a :: as)\ \not\ni\\ \quad \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ (\mathsf{suc}\ n)\ [\,]\quad\ \not\ni\\ \quad \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ (\mathsf{suc}\ n)\ (a :: as)\\ \qquad \ni\ \mathsf{cons}\ (leq\ :\ b \leqslant_A a)\ (ol\ :\ \mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ a\ n\ as)\end{array}$$

which is monolithic and inflexible. We can avoid using this predicate directly by exploiting the modularity isomorphisms

$$\mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ n\ as\ \cong\ \mathsf{Ordered}\ A\ \_{\leqslant_A}\_\ b\ as\ \times\ \mathsf{Length}\ A\ n\ as$$

for all $b : A$, $n : \mathsf{Nat}$, and $as : \mathsf{List}\ A$ — to promote a list to an ordered vector, we can prove that it satisfies Ordered and Length instead of OrderedLength. Promotion proofs from lists to ordered vectors can thus be divided into ordering and length aspects and carried out separately. (*End of example.*)

Along with the ornamental conversion isomorphisms, the construction of the modularity isomorphisms will be deferred until Chapter 4. Here we deal with a practical issue regarding composition of modularity isomorphisms: for example, to get pointwise isomorphisms between the optimised predicate for $\lceil\, O \otimes \lceil\, P \otimes Q \,\rceil \,\rceil$ and the pointwise conjunction of the optimised predicates for O, P, and Q, we need to instantiate the modularity isomorphisms twice and compose the results appropriately, a procedure which quickly becomes tedious. What we need is an auxiliary mechanism that helps with organising computation of composite predicates and isomorphisms following the parallel compositional structure of ornaments, in the same spirit as the upgrade mechanism (Section 3.1.2) helping with organising computation of coherence properties and proofs following the syntactic structure of function types.

We thus define the following auxiliary datatype Swap, parametrised with a refinement whose promotion predicate is to be swapped for a new one:

$$\begin{array}{l}\textbf{record}\ \mathsf{Swap}\ \{X\ Y\ :\ \mathsf{Set}\}\ (r\ :\ \mathsf{Refinement}\ X\ Y)\ :\ \mathsf{Set}_1\ \textbf{where}\\ \quad \textbf{field}\end{array}$$

$$
\begin{aligned}
&Q \;:\; X \to \mathsf{Set} \\
&i \;\;:\; (x \;:\; X) \to \mathsf{Refinement}.P\; r\; x \;\cong\; Q\; x
\end{aligned}
$$

An inhabitant of Swap r consists of a new promotion predicate for r and a proof that the new predicate is pointwise isomorphic to the original one in r. The actual swapping is done by the function

$$
\begin{aligned}
&\mathit{toRefinement} \;:\; \{X\; Y \;:\; \mathsf{Set}\}\; \{r \;:\; \mathsf{Refinement}\; X\; Y\} \to \mathsf{Swap}\; r \to \mathsf{Refinement}\; X\; Y \\
&\mathit{toRefinement}\; s \;=\; \mathbf{record}\; \{\; P \;=\; \mathsf{Swap}.Q\; s \\
&\qquad\qquad\qquad\qquad\quad ;\; i \;\;=\; \{\;\}_0 \;\}
\end{aligned}
$$

where Goal 0 is the new conversion isomorphism

$$
Y \;\cong\; \Sigma\; X\; (\mathsf{Refinement}.P\; r) \;\cong\; \Sigma\; X\; (\mathsf{Swap}.Q\; s)
$$

constructed by using transitivity and product of isomorphisms to compose $\mathsf{Refinement}.i\; r$ and $\mathsf{Swap}.i\; s$. We can then define the datatype FSwap of "swap families" in the usual way:

$$
\begin{aligned}
&\mathsf{FSwap} \;:\; \{I\; J \;:\; \mathsf{Set}\}\; \{e \;:\; J \to I\}\; \{X \;:\; I \to \mathsf{Set}\}\; \{Y \;:\; J \to \mathsf{Set}\} \to \\
&\qquad (rs \;:\; \mathsf{FRefinement}\; e\; X\; Y) \to \mathsf{Set}_1 \\
&\mathsf{FSwap}\; rs \;=\; \{i \;:\; I\}\; (j \;:\; e^{-1}\; i) \to \mathsf{Swap}\; (rs\; j)
\end{aligned}
$$

and provide the following combinator on swap families, which says that if there are alternative promotion predicates for the refinement semantics of O and P, then the pointwise conjunction of the two predicates is an alternative promotion predicate for the refinement semantics of $\lceil O \otimes P \rceil$:

$$
\begin{aligned}
&\otimes\text{-}\mathit{FSwap} \;:\; \{I\; J\; K \;:\; \mathsf{Set}\}\; \{e \;:\; J \to I\}\; \{f \;:\; K \to I\} \to \\
&\qquad\qquad \{D \;:\; \mathsf{Desc}\; I\}\; \{E \;:\; \mathsf{Desc}\; J\}\; \{F \;:\; \mathsf{Desc}\; K\} \to \\
&\qquad\qquad (O \;:\; \mathsf{Orn}\; e\; D\; E)\; (P \;:\; \mathsf{Orn}\; f\; D\; F) \to \\
&\qquad\qquad \mathsf{FSwap}\; (\mathit{RSem}\; O) \to \mathsf{FSwap}\; (\mathit{RSem}\; P) \to \mathsf{FSwap}\; (\mathit{RSem}\; \lceil O \otimes P \rceil) \\
&\otimes\text{-}\mathit{FSwap}\; O\; P\; ss\; ts\; (\mathsf{ok}\; (j\; ,\; k)) \;= \\
&\quad \mathbf{record}\; \{\; Q \;=\; \lambda\, x \mapsto \mathsf{Swap}.Q\; (ss\; j)\; x \times \mathsf{Swap}.Q\; (ts\; k)\; x \\
&\qquad\quad ;\; i \;\;=\; \lambda\, x \mapsto \{\;\}_1 \;\}
\end{aligned}
$$

Goal 1 is straightforwardly discharged by composing the modularity isomorphisms and the isomorphisms in ss and ts:

$$\begin{aligned}\mathsf{OptP}\ \lceil O \otimes P \rceil\ (\mathsf{ok}\ (j\ ,\ k))\ x &\cong \mathsf{OptP}\ O\ j\ x \quad \times \mathsf{OptP}\ P\ k\ x \\ &\cong \mathsf{Swap}.Q\ (ss\ j)\ x \times \mathsf{Swap}.Q\ (ts\ k)\ x\end{aligned}$$

Example (*modular promotion predicate for the parallel composition of three ornaments*)**.** To use the pointwise conjunction of the optimised predicates for ornaments O, P, and Q as an alternative promotion predicate for $\lceil O \otimes \lceil P \otimes Q \rceil \rceil$, we use the swap family

$$\otimes\text{-}FSwap\ O\ \lceil P \otimes Q \rceil\ id\text{-}FSwap\ (\otimes\text{-}FSwap\ P\ Q\ id\text{-}FSwap\ id\text{-}FSwap)$$

where

$$id\text{-}FSwap\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ :\ I \to \mathsf{Set}\}\ \{rs\ :\ \mathsf{FRefinement}\ X\ Y\} \to \mathsf{FSwap}\ rs$$

simply retains the original promotion predicate in rs. (*End of example.*)

Example (*swapping the promotion predicate from lists to ordered vectors*)**.** The swap family

$$\otimes\text{-}FSwap\ \lceil OrdListOD\ A\ \_\leqslant_A\_ \rceil\ (ListD\text{-}VecD\ A)\ id\text{-}FSwap\ (Length\text{-}FSwap\ A)$$

yields a refinement family from lists to ordered vectors using

$$\lambda\, b\ n\ as\ \mapsto\ \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ as\ \times\ length\ as\ \equiv\ n$$

as the promotion predicate, where

$$Length\text{-}FSwap\ A\ :\ \mathsf{FSwap}\ (RSem\ (ListD\text{-}VecD\ A))$$

swaps $\mathsf{Length}\ A$ for $\lambda\, n\ as\ \mapsto\ length\ as\ \equiv\ n$. (*End of example.*)

3.3.3 Resolution of the list insertion example

3.4 Two examples about heaps

To further demonstrate the use of the ornament–refinement framework, we look at two dependently typed heap data structures adapted from Okasaki's work [1999]. The first example about *binomial heaps* shows that Okasaki's idea of *numerical representations* can be elegantly captured by ornaments and the



Figure 3.5 *Left:* a binomial heap of size 3 consisting of two binomial trees storing elements x, y, and z. *Right:* the result of inserting an element w into the heap. (Note that the digits of the underlying binary numbers are ordered with the least significant digit first.)

coherence properties computed with upgrades, and the second example about *leftist heaps* demonstrates the power of parallel composition of ornaments by treating heap ordering and leftist balancing properties modularly.

Postulate operations on *Val* like $\_\leqslant_?\_$, $\leqslant$-*refl*, $\leqslant$-*trans*, and $\not\leqslant$-*invert* in Chapter 2.

3.4.1 Binomial heaps

We are all familiar with the idea of *positional number systems*, in which we represent numbers as a list of digits. Each position in a list of digits is associated with a weight, and the interpretation of the list is the weighted sum of the digits. (For example, the weights used for binary numbers are powers of 2.) Some container data structures and associated operations strongly resemble positional representations of natural numbers and associated operations. For example, a *binomial heap* (illustrated in Figure 3.5) can be thought of as a binary number in which every 1-digit stores a *binomial tree* — the actual place for storing elements — whose size is exactly the weight of the digit. The number of elements stored in a binomial heap is therefore exactly the value of the



Figure 3.6 *Left:* inductive definition of binomial trees. *Right:* decomposition of binomial trees of ranks 1 to 3.

underlying binary number. Inserting a new element into a binomial heap is analogous to incrementing a binary number, with carrying corresponding to combining smaller binomial trees into larger ones. Okasaki thus proposed to design container data structures by analogy with positional representations of natural numbers, and called such data structures *numerical representations*. Using an ornament, it is easy to express the relationship between a numerically represented container datatype (e.g., binomial heaps) and its underlying numeric datatype (e.g., binary numbers). But the ability to express the relationship alone is not too surprising. What is more interesting is that the ornament can give rise to upgrades such that

- the coherence properties of the upgrades semantically characterise the resemblance between container operations and corresponding numeric operations, and
- the promotion predicates give the precise types of the container operations that guarantee such resemblance.

We use insertion into binomial heaps as an example, which is presented in detail below.

Binomial trees

The basic building blocks of binomial heaps are *binomial trees*, in which elements are stored. Binomial trees are defined inductively on their *rank*, which is a natural number (see Figure 3.6):

- a binomial tree of rank 0 is a single node storing an element of type *Val*, and
- a binomial tree of rank $1 + r$ consists of two binomial trees of rank r, with one attached under the other's root node.

From this definition we can readily deduce that a binomial tree of rank r has 2^r elements. To actually define binomial trees as a datatype, however, an alternative view is more useful: a binomial tree of rank r is constructed by attaching binomial trees of ranks 0 to $r - 1$ under a root node. (Figure 3.6 shows how binomial trees of ranks 1 to 3 can be decomposed according to this view.) We thus define the datatype $\mathsf{BTree} : \mathsf{Nat} \to \mathsf{Set}$ — which is indexed with the rank of binomial trees — as follows: for any rank $r : \mathsf{Nat}$, the type $\mathsf{BTree}\ r$ has a field of type *Val* — which is the root node — and r recursive positions indexed from $r - 1$ down to 0. This is directly encoded as a description:

$$
\begin{aligned}
&\mathit{BTreeD} : \mathsf{Desc\ Nat} \\
&\mathit{BTreeD}\ r = \sigma\langle \_ : \mathit{Val} \rangle\ \mathsf{v}\ (\mathit{descend}\ r) \\
&\mathsf{BTree} : \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{BTree} = \mu\ \mathit{BTreeD}
\end{aligned}
$$

where *descend r* is a list from $r - 1$ down to 0:

$$
\begin{aligned}
&\mathit{descend} : \mathsf{Nat} \to \mathsf{List\ Nat} \\
&\mathit{descend}\ \mathsf{zero} = [\,] \\
&\mathit{descend}\ (\mathsf{suc}\ n) = n :: \mathit{descend}\ n
\end{aligned}
$$

Note that, in *BTreeD*, we are exploiting the full computational power of Desc, computing the list of recursive indices from the index request. Due to this, it is tricky to wrap up *BTreeD* as an index-first datatype declaration, so we will skip this step and work directly with the raw representation, which looks

reasonably intuitive anyway: a binomial tree of type BTree r is of the form con $(x\ ,\ ts)$ where $x\ :\ Val$ is the root element and $ts\ :\ \mathbb{P}\ (descend\ r)$ BTree is a series of sub-trees.

The most important operation on binomial trees is combining two smaller binomial trees of the same rank into a larger one, which corresponds to carrying in positional arithmetic. Given two binomial trees of the same rank r, one can be *attach*ed under the root of the other, forming a single binomial tree of rank $1 + r$ — this is exactly the inductive definition of binomial trees.

```
attach : {r : Nat} → BTree r → BTree r → BTree (suc r)
attach t (con (y , us)) = con (y , t , us)
```

For use in binomial heaps, though, we should ensure that elements in binomial trees are in *heap order*, i.e., the root of any binomial tree (including sub-trees) is the minimum element in the tree. This is achieved by comparing the roots of two binomial trees before deciding which one is to be attached to the other:

```
link : {r : Nat} → BTree r → BTree r → BTree (suc r)
link t u with root t ⩽? root u
link t u | yes _ = attach u t
link t u | no  _ = attach t u
```

where *root* extracts the root element of a binomial tree:

```
root : {r : Nat} → BTree r → Val
root (con (x , ts)) = x
```

If we always build binomial trees of positive rank by *link*, then the elements in any binomial tree we build would be in heap order. This is a crucial assumption in binomial heaps (which is not essential to our development, though).

From binary numbers to binomial heaps

The datatype Bin : Set of binary numbers is just a specialised datatype of lists of binary digits:

```
data BinTag : Set where ‘nil ‘zero ‘one : BinTag
```

```
BinD : Desc ⊤
BinD ▪ = σ BinTag λ { 'nil  ↦ v []
                    ; 'zero ↦ v (▪ :: [])
                    ; 'one  ↦ v (▪ :: []) }

indexfirst data Bin : Set where
  Bin ∋ nil
      | zero (b : Bin)
      | one  (b : Bin)
```

The intended interpretation of binary numbers is given by

```
toNat : Bin → Nat
toNat nil      = 0
toNat (zero b) = 0 + 2 * toNat b
toNat (one  b) = 1 + 2 * toNat b
```

That is, the list of digits of a binary number of type Bin starts from the least significant digit, and the i-th digit (counting from 0) has weight 2^i. We refer to the position of a digit as its rank, i.e., the i-th digit is said to have rank i.

As stated in the beginning, binomial heaps are binary numbers whose 1-digits are decorated with binomial trees of matching rank, which can be expressed straightforwardly as an ornamentation of binary numbers. To ensure that the binomial trees in binomial heaps have the right rank, the datatype BHeap : Nat → Set is indexed with a "starting rank": if a binomial heap of type BHeap r is nonempty (i.e., not nil), then its first digit has rank r (and stores a binomial tree of rank r when the digit is one), and the rest of the heap is indexed with $1 + r$.

```
BHeapOD : OrnDesc Nat
BHeapOD (ok r) = σ BinTag λ { 'nil  ↦ v ▪
                            ; 'zero ↦ v (ok (suc r) , ▪)
                            ; 'one  ↦ Δ⟨ t : BTree r ⟩ v (ok (suc r) , ▪) }

indexfirst data BHeap : Nat → Set where
  BHeap r ∋ nil
          | zero (h : BHeap (suc r))
```

$$\mid \ \mathsf{one}\ \ (t\ :\ \mathsf{BTree}\ r)\ (h\ :\ \mathsf{BHeap}\ (\mathsf{suc}\ r))$$

In applications, we would use binomial heaps of type BHeap 0, which encompasses binomial heaps of all sizes.

Increment and insertion, in coherence

Increment of binary numbers is defined by

$$\begin{aligned}
&\mathit{incr}\ :\ \mathsf{Bin} \to \mathsf{Bin}\\
&\mathit{incr}\ \mathsf{nil} &&= \mathsf{one}\ \mathsf{nil}\\
&\mathit{incr}\ (\mathsf{zero}\ b) &&= \mathsf{one}\ b\\
&\mathit{incr}\ (\mathsf{one}\ \ b) &&= \mathsf{zero}\ (\mathit{incr}\ b)
\end{aligned}$$

The corresponding operation on binomial heaps is insertion of a binomial tree into a binomial heap (of matching rank), whose direct implementation is

$$\begin{aligned}
&\mathit{insT}\ :\ \{r\ :\ \mathsf{Nat}\} \to \mathsf{BTree}\ r \to \mathsf{BHeap}\ r \to \mathsf{BHeap}\ r\\
&\mathit{insT}\ t\ \mathsf{nil} &&= \mathsf{one}\ t\ \mathsf{nil}\\
&\mathit{insT}\ t\ (\mathsf{zero}\ \ h) &&= \mathsf{one}\ t\ h\\
&\mathit{insT}\ t\ (\mathsf{one}\ u\ h) &&= \mathsf{zero}\ (\mathit{insT}\ (\mathit{link}\ t\ u)\ h)
\end{aligned}$$

Conceptually, *incr* puts a 1-digit into the least significant position of a binary number, triggering a series of carries, i.e., summing 1-digits of smaller ranks into 1-digits of larger ranks; *insT* follows the pattern of *incr*, but since 1-digits now have to store a binomial tree of matching rank, *insT* takes an additional binomial tree as input and *link*s binomial trees of smaller ranks into binomial trees of larger ranks whenever carrying happens. Having defined *insT*, inserting a single element into a binomial heap of type BHeap 0 is then inserting, by *insT*, a rank-0 binomial tree (i.e., a single node) storing the element into the heap.

$$\begin{aligned}
&\mathit{insert}\ :\ \mathit{Val} \to \mathsf{BHeap}\ 0 \to \mathsf{BHeap}\ 0\\
&\mathit{insert}\ x\ =\ \mathit{insT}\ (\mathsf{con}\ (x\ ,\ \blacksquare))
\end{aligned}$$

It is apparent that the program structure of *insT* strongly resembles that of *incr* — they manipulate the list-of-binary-digits structure in the same way.

But can we characterise the resemblance semantically? It turns out that the coherence property of the following upgrade from the type of *incr* to that of *insT* is an appropriate answer:

$$
\begin{aligned}
&\mathit{upg} \;:\; \mathsf{Upgrade}\;(\mathsf{Bin} \to \mathsf{Bin})\;(\{r \;:\; \mathsf{Nat}\} \to \mathsf{BTree}\;r \to \mathsf{BHeap}\;r \to \mathsf{BHeap}\;r)\\
&\mathit{upg} \;=\; \forall^{+}\langle\!\langle\, r : \mathsf{Nat} \,\rangle\!\rangle\;\; \forall^{+}\langle\, \_ : \mathsf{BTree}\;r \,\rangle\\
&\qquad\quad \mathbf{let}\;\mathit{ref} \;:\; \mathsf{Refinement}\;\mathsf{Bin}\;(\mathsf{BHeap}\;r)\\
&\qquad\qquad\quad \mathit{ref} \;=\; \mathit{RSem}\;\lceil \mathit{BHeapOD} \rceil\;(\mathsf{ok}\;r)\\
&\qquad\quad \mathbf{in}\;\; \mathit{ref} \rightharpoonup \mathit{toUpgrade}\;\mathit{ref}
\end{aligned}
$$

The upgrade *upg* says that, compared to the type of *incr*, the type of *insT* has two new arguments — the implicit argument $r \;:\; \mathsf{Nat}$ and the explicit argument of type $\mathsf{BTree}\;r$ — and that the two occurrences of $\mathsf{BHeap}\;r$ in the type of *insT* refine the corresponding occurrences of Bin in the type of *incr* using the refinement semantics of the ornament $\lceil \mathit{BHeapOD} \rceil\;(\mathsf{ok}\;r)$ from Bin to $\mathsf{BHeap}\;r$. The type $\mathsf{Upgrade}.C\;\mathit{upg}\;\mathit{incr}\;\mathit{insT}$ (which states that *incr* and *insT* are in coherence with respect to *upg*) expands to

$$
\begin{aligned}
&\{r \;:\; \mathsf{Nat}\}\;(t \;:\; \mathsf{BTree}\;r)\;(b \;:\; \mathsf{Bin})\;(h \;:\; \mathsf{BHeap}\;r) \to\\
&\mathit{toBin}\;h \;\equiv\; b \to \mathit{toBin}\;(\mathit{insT}\;t\;h) \;\equiv\; \mathit{incr}\;b
\end{aligned}
$$

where *toBin* extracts the underlying binary number of a binomial heap:

$$
\begin{aligned}
&\mathit{toBin} \;:\; \{r \;:\; \mathsf{Nat}\} \to \mathsf{BHeap}\;r \to \mathsf{Bin}\\
&\mathit{toBin} \;=\; \mathit{forget}\;\lceil \mathit{BHeapOD} \rceil
\end{aligned}
$$

That is, given a binomial heap $h \;:\; \mathsf{BHeap}\;r$ whose underlying binary number is $b \;:\; \mathsf{Bin}$, after inserting a binomial tree into h by *insT*, the underlying binary number of the result is $\mathit{incr}\;b$. This says exactly that *insT* manipulates the underlying binary number in the same way as *incr* does.

We have seen that the coherence property of *upg* is appropriate for characterising the resemblance of *incr* and *insT*; proving that it holds for *incr* and *insT* is a separate matter, though. We can, however, avoid doing the implementation of insertion and the coherence proof separately: instead of implementing *insT* directly, we can implement insertion with a more precise type in the first place such that, from this more precisely typed version, we can derive *insT* that satisfies the coherence property automatically. The above process is fully supported

by the mechanism of upgrades. Specifically, the more precise type for insertion is given by the promotion predicate of *upg* (applied to *incr*), the more precisely typed version of insertion acts as a promotion proof of *incr* (with respect to *upg*), and the promotion gives us *insT*, accompanied by a proof that *insT* is in coherence with *incr*.

Let BHeap′ be the optimised predicate for the ornament from Bin to BHeap *r*:

```
BHeap′ : Nat → Bin → Set
BHeap′ r b = OptP ⌈BHeapOD⌉ (ok r) b

indexfirst data BHeap′ : Nat → Bin → Set where
  BHeap′ r nil       ∋ nil
  BHeap′ r (zero b)  ∋ zero (h : BHeap′ (suc r) b)
  BHeap′ r (one  b)  ∋ one  (t : BTree r) (h : BHeap′ (suc r) b)
```

Here a more helpful interpretation is that BHeap′ is a datatype of binomial heaps additionally indexed with the underlying binary number. The type Upgrade.*P upg incr* of promotion proofs for *incr* then expands to

```
{r : Nat} → BTree r → (b : Bin) → BHeap′ r b → BHeap′ r (incr b)
```

A function of this type is explicitly required to transform the underlying binary number structure of its input in the same way as *incr* does. Insertion can now be implemented as

```
insT′ : {r : Nat} → BTree r → (b : Bin) → BHeap′ r b → BHeap′ r (incr b)
insT′ t nil       nil       = one t nil
insT′ t (zero b) (zero  h)  = one t h
insT′ t (one  b) (one u h)  = zero (insT′ (link t u) h)
```

which is very much the same as the original *insT*. It is interesting to note that all the constructor choices for binomial heaps in *insT*′ are actually completely determined by the types. This fact is easier to observe if we desugar *insT*′ to the raw representation:

```
insT′ : {r : Nat} → BTree r → (b : Bin) → BHeap′ r b → BHeap′ r (incr b)
insT′ t (con (‘nil   , ▪)) h                = con (t , con ▪          , ▪)
insT′ t (con (‘zero , b , ▪)) (con (  h , ▪)) = con (t , h            , ▪)
```

$$insT'\ t\ (\mathsf{con}\ (\text{`}\mathsf{one}\ ,\ b\ ,\ \blacksquare))\ (\mathsf{con}\ (u\ ,\ h\ ,\ \blacksquare))\ =\ \mathsf{con}\ (\quad insT'\ (link\ t\ u)\ b\ h\ ,\ \blacksquare)$$

in which no constructor tags for binomial heaps are present. This means that the types would instruct which constructors to use when programming *insT′*, establishing the coherence property by construction. Finally, since *insT′* is a promotion proof for *incr*, we can invoke the upgrading operation of *upg* and get *insT*:

$$\begin{aligned} &insT\ :\ \{r\ :\ \mathsf{Nat}\} \to \mathsf{BTree}\ r \to \mathsf{BHeap}\ r \to \mathsf{BHeap}\ r \\ &insT\ =\ \mathsf{Upgrade}.u\ upg\ incr\ insT' \end{aligned}$$

which is automatically in coherence with *incr*:

$$\begin{aligned} &incr\text{-}insT\text{-}coherence\ :\ \{r\ :\ \mathsf{Nat}\}\ (t\ :\ \mathsf{BTree}\ r)\ (b\ :\ \mathsf{Bin})\ (h\ :\ \mathsf{BHeap}\ r) \to \\ &\qquad\qquad toBin\ h\ \equiv\ b \to toBin\ (insT\ t\ h)\ \equiv\ incr\ b \\ &incr\text{-}insT\text{-}coherence\ =\ \mathsf{Upgrade}.c\ upg\ incr\ insT' \end{aligned}$$

Summary

We define Bin, *incr*, and then BHeap as an ornamentation of Bin, describe in *upg* how the type of *insT* is an upgraded version of the type of *incr*, and implement *insT′*, whose type is supplied by *upg*. We can then derive *insT*, the coherence property of *insT* with respect to *incr*, and its proof, all automatically by *upg*. Compared to Okasaki's implementation, besides rank-indexing, which elegantly transfers the management of rank-related invariants to the type system, the extra work is only the straightforward markings of the differences between Bin and BHeap (in *BHeapOD*) and between the type of *incr* and that of *insT* (in *upg*). The reward is huge in comparison: we get a coherence property that precisely characterises the structural behaviour of insertion with respect to increment, and an enriched function type that guides the implementation of insertion such that the coherence property is satisfied by construction. From straightforward markings to nontrivial types and programs — this clearly demonstrates the power of the ornament–refinement framework.



Figure 3.7 Datatypes about leftist heaps and their ornamental relationships.

3.4.2 Leftist heaps

Our second example is about treating the ordering and balancing properties of *leftist heaps* modularly. In Okasaki's words:

> Leftist heaps [...] are heap-ordered binary trees that satisfy the *leftist property*: the rank of any left child is at least as large as the rank of its right sibling. The rank of a node is defined to be the length of its *right spine* (i.e., the rightmost path from the node in question to an empty node).

From this passage we can immediately analyse the concept of leftist heaps into three: leftist heaps (i) are binary trees that (ii) are heap-ordered and (iii) satisfy the leftist property. This suggests that there is a basic datatype of binary trees together with two ornamentations, one expressing heap ordering and the other the leftist property. The datatype of leftist heaps is then synthesised by composing the two ornamentations in parallel. All the datatypes about leftist heaps and their ornamental relationships are shown in Figure 3.7.

Datatypes leading to leftist heaps

The basic datatype Tree : Set of "skeletal" binary trees, which consist of empty nodes and internal nodes not storing any elements, is defined by

```
data TreeTag : Set where ‘nil ‘node : TreeTag

TreeD : Desc ⊤
TreeD ▪ = σ TreeTag λ { ‘nil  ↦ v []
                      ; ‘node ↦ v (▪ :: ▪ :: []) }

indexfirst data Tree : Set where
  Tree ∋ nil
       | node (t : Tree) (u : Tree)
```

Leftist trees — skeletal binary trees satisfying the leftist property — are then an ornamented version of Tree. The datatype LTree : Nat → Set of leftist trees is indexed with the rank of the root of the trees. The constructor choices can be determined from the rank: the only node that can have rank zero is the empty node nil; otherwise, when the rank of a node is non-zero, it must be an internal node constructed by the node constructor, which enforces the leftist property.

```
LTreeOD : OrnDesc Nat ! TreeD
LTreeOD (ok zero   ) = ∇⟨ ‘nil ⟩ v ▪
LTreeOD (ok (suc r)) = ∇⟨ ‘node ⟩ Δ⟨ l : Nat ⟩ Δ⟨ r⩽l : r ⩽ l ⟩ v (ok l , ok r , ▪)

indexfirst data LTree : Nat → Set where
  Tree zero    ∋ nil
  Tree (suc r) ∋ node {l : Nat} (r⩽l : r ⩽ l) (t : Tree l) (u : Tree r)
```

Independently, *heap-ordered trees* are also an ornamented version of Tree. The datatype Heap : *Val* → Set of heap-ordered trees can be regarded as a generalisation of ordered lists: in a heap-ordered tree, every path from the root to an empty node is an ordered list.

```
HeapOD : OrnDesc Val ! TreeD
HeapOD (ok b) =
  σ TreeTag λ { ‘nil  ↦ v ▪
              ; ‘node ↦ Δ⟨ x : Val ⟩ Δ⟨ b⩽x : b ⩽ x ⟩ v (ok x , ok x , ▪) }
```

```
indexfirst data Heap : Val → Set where
  Heap b ∋ nil
         | node (x : Val) (b⩽x : b ⩽ x) (t : Heap x) (u : Heap x)
```

Composing the two ornaments in parallel gives us exactly the datatype of leftist heaps.

```
LHeapOD : OrnDesc (! ⋈ !) pull TreeD
LHeapOD = ⌈HeapOD⌉ ⊗ ⌈LTreeOD⌉

indexfirst data LHeap : Val → Nat → Set where
  LHeap b zero    ∋ nil
  LHeap b (suc r) ∋ node (x : Val) (b⩽x : b ⩽ x)
                         {l : Nat} (r⩽l : r ⩽ l) (t : Heap x l) (u : Heap x r)
```

Operations on leftist heaps

The analysis of leftist heaps as the parallel composition of the two ornamentations allows us to talk about heap ordering and the leftist property independently. For example, a useful operation on heap-ordered trees is relaxing the lower bound. It can be regarded as an upgraded version of the identity function on Tree, since it leaves the tree structure intact, changing only the ordering information. With the help of the optimised predicate for ⌈*HeapOD*⌉,

```
Heap′ : Val → Set
Heap′ b = OptP ⌈HeapOD⌉ (ok b)

indexfirst data Heap′ : Val → Tree → Set where
  Heap′ b nil        ∋ nil
  Heap′ b (node t u) ∋ node (x : Val) (b⩽x : b ⩽ x)
                            (t′ : Heap x t) (u′ : Heap x u)
```

we can give the type of bound-relaxing in predicate form, stating explicitly in the type that the underlying tree structure is unchanged:

```
relax : {b b′ : Val} → b′ ⩽ b → {t : Tree} → Heap′ b t → Heap′ b′ t
relax b′⩽b {nil      } nil                = nil
relax b′⩽b {node _ _} (node x b⩽x t u) = node x (⩽-trans b′⩽b b⩽x) t u
```

Since the identity function on LTree can also be seen as an upgraded version of the identity function on Tree, we can combine *relax* and the predicate form of the identity function on LTree to get bound-relaxing on leftist heaps, which modifies only the heap-ordering portion of a leftist heap:

```
lhrelax : {b b′ : Val} → b′ ⩽ b → {r : Nat} → LHeap b r → LHeap b′ r
lhrelax {b} {b′} b′⩽b {r} = Upgrade.u upg id (λ _ ↦ relax b′⩽b ∗ id)
  where
    ref : (b″ : Val) → Refinement Tree (LHeap b″ r)
    ref b″ = toRefinement
                 (⊗-FSwap ⌈HeapOD⌉ ⌈LTreeOD⌉ id-FSwap id-FSwap
                   (ok (ok b″ , ok r)))
    upg : Upgrade (Tree → Tree) (LHeap b r → LHeap b′ r)
    upg = ref b ⇀ toUpgrade (ref b′)
```

In general, non-modifying heap operations do not depend on the leftist property and can be implemented for heap-ordered trees and later lifted to work with leftist heaps, relieving us of the unnecessary work of dealing with the leftist property when it is simply to be ignored. For another example, converting a leftist heap to a list of its elements has nothing to do with the leftist property. In fact, it even has nothing to do with heap ordering, but only with the internal labelling. Hence we can define the *internally labelled trees* as an ornamentation of skeletal binary trees:

```
ITreeOD : Set → OrnDesc ⊤ ! TreeD
ITreeOD A ▪ = σ TreeTag λ { ‘nil  ↦ v ▪
                          ; ‘node ↦ Δ⟨ _ : A ⟩ v (ok tt, ok tt, ▪) }

indexfirst data ITree (A : Set) : Set where
  ITree A ∋ nil
          | node (x : A) (t : ITree A) (u : ITree A)
```

on which we can do preorder traversal:

```
preorder : {A : Set} → ITree A → List A
preorder nil          = []
preorder (node x t u) = x :: preorder t ++ preorder u
```

We have an ornament from internally labelled trees to heap-ordered trees:

```
ITreeD-HeapD : Orn ! ⌊ITreeOD Val⌋ ⌊HeapOD⌋
ITreeD-HeapD (ok b) =
  σ TreeTag λ { 'nil  ↦ v []
              ; 'node ↦ σ⟨ x : Val ⟩ Δ⟨ _ : b ⩽ x ⟩ v (refl :: refl :: []) }
```

So, to get a list of the elements of a leftist heap (whose first element is the minimum one), we convert the leftist heap to an internally labelled tree and then invoke *preorder*.

```
toList : {b : Val} {r : Nat} → LHeap b r → List Val
toList = preorder ∘ forget (ITreeD-HeapD ⊙ diffOrn-l ⌈HeapOD⌉ ⌈LTreeOD⌉)
```

For modifying operations, however, we need to consider both heap ordering and the leftist property at the same time, so we should program directly with the composite datatype of leftist heaps. For example, a key operation is merging two heaps:

```
merge : {b₀ : Val} {r₀ : Nat} → LHeap b₀ r₀ →
        {b₁ : Val} {r₁ : Nat} → LHeap b₁ r₁ →
        {b  : Val} → b ⩽ b₀ → b ⩽ b₁ → Σ⟨ r : Nat ⟩ LHeap b r
```

with which we can easily implement insertion of a new element (by merging with a singleton heap) and deletion of the minimum element (by deleting the root and merging the two sub-heaps). The definition of *merge* is shown in Figure 3.8. It is a more precisely typed version of Okasaki's implementation, split into two mutually recursive functions to make it clear to Agda's termination checker that we are doing two-level induction on the ranks of the two input heaps. When one of the ranks is zero, meaning that the corresponding heap is nil, we simply return the other heap (whose bound is suitably relaxed) as the result. When both ranks are nonzero, meaning that both heaps are nonempty, we compare the roots of the two heaps and recursively merge the heap with the larger root into the right branch of the heap with the smaller root. The recursion is structural because the rank of the right branch of a nonempty heap is strictly smaller. There is a catch, however: the rank of the new right subheap resulting from the recursive merging might be larger than that of the left

```
makeT : (x : Nat) → {r₀ : Nat} (h₀ : LHeap x r₀) →
                    {r₁ : Nat} (h₁ : LHeap x r₁) → Σ⟨ r : Nat ⟩ LHeap x r
makeT x {r₀} h₀ {r₁} h₁ with r₀ ⩽? r₁
makeT x {r₀} h₀ {r₁} h₁ | yes r₀⩽r₁ = suc r₀ ,
                                       node x ⩽-refl r₀⩽r₁          h₁ h₀
makeT x {r₀} h₀ {r₁} h₁ | no  r₀⩽̸r₁ = suc r₁ ,
                                       node x ⩽-refl (⩽̸-invert r₀⩽̸r₁) h₀ h₁

mutual

  merge : {b₀ : Val} {r₀ : Nat} → LHeap b₀ r₀ →
          {b₁ : Val} {r₁ : Nat} → LHeap b₁ r₁ →
          {b  : Val} → b ⩽ b₀ → b ⩽ b₁ → Σ⟨ r : Nat ⟩ LHeap b r
  merge {b₀} {zero  } nil h₁ b⩽b₀ b⩽b₁ = _ , lhrelax b⩽b₁ h₁
  merge {b₀} {suc r₀} h₀ h₁ b⩽b₀ b⩽b₁ = merge′ h₀ h₁ b⩽b₀ b⩽b₁

  merge′ : {b₀ : Val} {r₀ : Nat} → LHeap b₀ (suc r₀) →
           {b₁ : Val} {r₁ : Nat} → LHeap b₁ r₁       →
           {b  : Val} → b ⩽ b₀ → b ⩽ b₁ → Σ⟨ r : Nat ⟩ LHeap b r
  merge′ h₀                                {b₁} {zero  } nil
         b⩽b₀ b⩽b₁ = _ , lhrelax b⩽b₀ h₀
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁)
         b⩽b₀ b⩽b₁ with x₀ ⩽? x₁
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁)
         b⩽b₀ b⩽b₁ | yes x₀⩽x₁ =
           _ , lhrelax (⩽-trans b⩽b₀ b₀⩽x₀)
                 (outr (makeT x₀ t₀
                         (outr (merge u₀ (node x₁ x₀⩽x₁ r₁⩽l₁ t₁ u₁)
                                 ⩽-refl ⩽-refl))))
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁)
         b⩽b₀ b⩽b₁ | no  x₀⩽̸x₁ =
           _ , lhrelax (⩽-trans b⩽b₁ b₁⩽x₁)
                 (outr (makeT x₁ t₁
                         (outr (merge′ (node x₀ (⩽̸-invert x₀⩽̸x₁) r₀⩽l₀ t₀ u₀) u₁
                                 ⩽-refl ⩽-refl)))))
```

Figure 3.8 Merging two leftist heaps. Proof terms about ordering are coloured grey to aid comprehension (taking inspiration from — but not really employing — *type theory in colour* [Bernardy and Guilhem, 2013]).

sub-heap, violating the leftist property, so there is a helper function *makeT* that swaps the sub-heaps when necessary.

Weight-biased leftist heaps

Another advantage of separating the leftist property and heap ordering is that we can swap the leftist property for another balancing property. The non-modifying operations, previously defined for heap-ordered trees, can be upgraded to work with the new balanced heap datatype in the same way, while the modifying operations are reimplemented with respect to the new balancing property. For example, the leftist property requires that the *rank* of the left sub-tree is at least that of the right one; we can replace "rank" with "size" in its statement and get the *weight-biased leftist property*. This is again codified as an ornamentation of skeletal binary trees:

```
WLTreeOD : OrnDesc Nat ! TreeD
WLTreeOD (ok zero    ) = ∇⟨ 'nil ⟩ v ▪
WLTreeOD (ok (suc n)) = ∇⟨ 'node ⟩ Δ⟨ l : Nat ⟩ Δ⟨ r : Nat ⟩
                          Δ⟨ _ : r ⩽ l ⟩ Δ⟨ _ : n ≡ l + r ⟩ v (ok l , ok r , ▪)

indexfirst data WLTree : Nat → Set where
  WLTree zero    ∋ nil
  WLTree (suc n) ∋ node {l : Nat} {r : Nat}
                        (r⩽l : r ⩽ l) (n≡l+r : n ≡ l + r)
                        (t : WLTree l) (u : WLTree r)
```

which can be composed in parallel with the heap-ordering ornament $\lceil HeapOD \rceil$ and gives us weight-biased leftist heaps.

```
WLHeapD : Desc (! ⋈ !)
WLHeapD = ⌊ ⌈ HeapOD ⌉ ⊗ ⌈ WLTreeOD ⌉ ⌋

indexfirst data WLHeap : Val → Nat → Set where
  WLHeap b zero    ∋ nil
  WLHeap b (suc n) ∋ node (x : Val) (b⩽x : b ⩽ x)
                          {l : Nat} {r : Nat}
```

$$(r{\leqslant}l \;:\; r \leqslant l)\;(n{\equiv}l{+}r \;:\; n \equiv l + r)$$
$$(t \;:\; \mathsf{WLHeap}\; x\; l)\;(u \;:\; \mathsf{WLHeap}\; x\; r)$$

The weight-biased leftist property makes it possible to reimplement merging in a single, top-down pass rather than two passes: With the original rank-biased leftist property, recursive calls to *merge* are determined top-down by comparing root elements, and the helper function *makeT* swaps a recursively computed sub-heap with the other sub-heap if the rank of the former is larger; the rank of a recursively computed sub-heap, however, is not known before a recursive call returns (which is reflected by the existential quantification of the rank index in the result type of *merge*), so during the whole merging process *makeT* does the swapping in a second bottom-up pass. On the other hand, with the weight-biased leftist property, the merging operation has type

$$\begin{aligned} \mathit{wmerge} \;:\; &\{b_0 \;:\; \mathit{Val}\}\;\{n_0 \;:\; \mathsf{Nat}\} \to \mathsf{WLHeap}\; b_0\; n_0 \to \\ &\{b_1 \;:\; \mathit{Val}\}\;\{n_1 \;:\; \mathsf{Nat}\} \to \mathsf{WLHeap}\; b_1\; n_1 \to \\ &\{b \;\;:\; \mathit{Val}\} \to b \leqslant b_0 \to b \leqslant b_1 \to \mathsf{WLHeap}\; b\; (n_0 + n_1) \end{aligned}$$

The implementation of *wmerge* is largely similar to *merge* and is omitted here. For *wmerge*, however, the weight of a recursively computed sub-heap is known before the recursive merging is actually performed (so the weight index can be given explicitly in the result type of *wmerge*). The counterpart of *makeT* can thus determine before a recursive call whether to do the swapping or not, and the whole merging process requires only one top-down pass.

Do we need a summary here?

3.5 Discussion

summary of the three-level architecture of ornaments, refinements, and upgrades; bundle; why ornaments?

Chapter 4

Categorical organisation of the ornament–refinement framework

4.1 Formalisation of categories

In this section we formalise some basic category-theoretic terms in Agda, establishing vocabulary for Sections 4.2 and 4.3.

4.1.1 Definitions of categories and functors

We will define a category to be a set of objects and sets of morphisms indexed by source and target, together with the usual laws. Special attention must be paid to equality on morphisms, though, which is usually coarser than definitional equality — for example, in the category of sets and (total) functions, it is necessary to identify functions up to extensional equality, so uniqueness of morphisms in universal properties would make sense. One simple way to achieve this in Agda's intensional setting is to use *setoids* Barthe et al. [2003] — i.e., sets with an explicitly specified equivalence relation — to represent sets of morphisms. Subsequently, all operations on morphisms should respect the equivalence.

In Agda, the type of setoids can be defined as a record which contains a carrier set, an equivalence relation on the set, and the three laws for the equivalence relation:[1]

```
record Setoid {c d : Level} : Set (suc (c ⊔ d)) where
  field
    Carrier : Set c
    _≈_     : Carrier → Carrier → Set d
    refl'   : ∀ {x} → x ≈ x
    sym     : ∀ {x y} → x ≈ y → y ≈ x
    trans   : ∀ {x y z} → x ≈ y → y ≈ z → x ≈ z
```

For example, we can define a setoid of functions that uses extensional equality:

```
FunSetoid : Set → Set → Setoid
FunSetoid A B = record { Carrier = A → B
                       ; _≈_     = _≐_
                       ; proofs-of-laws}
```

where $\_ \doteq \_$ is defined by $f \doteq g = \forall\, x \to f\ x \equiv g\ x$. Proofs of the three laws are omitted from the paper.

Similarly, we can define the type of categories as a record containing a set of objects, a collection of setoids of morphisms indexed by source and target (so morphisms with the same source and target — and only such morphisms — can be compared for equality), the composition operator on morphisms, the identity morphisms, and the laws of categories. The definition is shown in Figure 4.1. Two notations are introduced to improve readability: $X ==> Y$ is defined to be the carrier set of the setoid of morphisms from X to Y, and $f \approx g$ is defined to be the equivalence between the morphisms f and g as specified by the setoid to which f and g belong. The last two laws $cong-l$ and $cong-r$ require composition of morphisms to respect the equivalence on

[1]The definition of setoids uses Agda's universe polymorphism, so the definition can be instantiated at suitable levels of the Set hierarchy as needed. We will give the first few universe-polymorphic definitions with full detail about the levels, but will later switch to writing just 'Set _' to suppress the noise.

```
record Category {l m n : Level} : Set (suc (l ⊔ m ⊔ n)) where
  field
    Object      : Set l
    Morphism    : Object → Object → Setoid {m} {n}
  _==>_ : Object → Object → Set m
  X ==> Y = Setoid.Carrier (Morphism X Y)
  _≈_ : ∀ {X Y} → X ==> Y → X ==> Y → Set n
  _≈_ {X} {Y} = Setoid._≈_ (Morphism X Y)
  field
    _∆_   : ∀ {X Y Z} → Y ==> Z → X ==> Y → X ==> Z
    id    : ∀ {X} → X ==> X
    id-l  : ∀ {X Y} (f : X ==> Y) → id ∆ f ≈ f
    id-r  : ∀ {X Y} (f : X ==> Y) → f ∆ id ≈ f
    assoc : ∀ {X Y Z W}
            (f : Z ==> W) (g : Y ==> Z) (h : X ==> Y) →
            (f ∆ g) ∆ h ≈ f ∆ (g ∆ h)
    cong-l : ∀ {X Y Z} {f g : Y ==> Z} (h : X ==> Y) →
             f ≈ g → f ∆ h ≈ g ∆ h
    cong-r : ∀ {X Y Z} (h : Y ==> Z) {f g : X ==> Y} →
             f ≈ g → h ∆ f ≈ h ∆ g
```

Figure 4.1 Definition of categories.

morphisms; they are given in this form to work better with the equational reasoning combinators commonly used in Agda (see, for example, the AoPA library Mu et al. [2009]). Now we can define the category *Fun* of sets and (total) functions as

$$
\begin{aligned}
&\mathit{Fun} \;:\; \mathit{Category}\\
&\mathit{Fun} \;=\; \mathbf{record}\; \{\; \mathit{Object} \;\; = \;\; \mathsf{Set}\\
&\qquad\qquad\quad ;\; \mathit{Morphism} \;=\; \mathit{FunSetoid}\\
&\qquad\qquad\quad ;\; \_\Delta\_ \;=\; \_\circ\_\\
&\qquad\qquad\quad ;\; \mathit{id} \;\; = \;\; \lambda\, x \mapsto x\\
&\qquad\qquad\quad ;\; \mathit{proofs} - \mathit{of} - \mathit{laws}\}
\end{aligned}
$$

Another important category that we will make use of is *Fam*, the category of families of sets and families of functions, which is useful for talking about componentwise structures. An object in *Fam* has type $\Sigma\langle\, I : \mathsf{Set}\,\rangle\; I \to \mathsf{Set}$, i.e., it is a set I and a family of sets indexed by I; a morphism from $(J\,,\,Y)$ to $(I\,,\,X)$ is a function $e \;:\; J \to I$ and a family of functions from $Y\; j$ to $X\; (e\; j)$ for each $j \;:\; J$.

$$
\begin{aligned}
&\mathit{Fam} \;:\; \mathit{Category}\\
&\mathit{Fam} \;=\; \mathbf{record}\\
&\quad \{\; \mathit{Object} \;\; = \;\; \Sigma\langle\, I : \mathsf{Set}\,\rangle\; I \to \mathsf{Set}\\
&\quad ;\; \mathit{Morphism} \;=\\
&\qquad \lambda\,(J\,,\,Y)\;(I\,,\,X) \mapsto \mathbf{record}\\
&\qquad\quad \{\; \mathit{Carrier} \;=\; \Sigma\langle\, e : J \to I\,\rangle\; Y \rightrightarrows (X \circ e)\\
&\qquad\quad ;\; \_\approx\_ \;\; = \;\; \lambda\,(e\,,\,u)\;(e'\,,\,u') \mapsto\\
&\qquad\qquad\qquad\qquad (e \doteq e') \times (\forall\; \{j\} \to u\; \{j\}\; \mathit{JMEq'}\; u'\; \{j\})\\
&\qquad\quad ;\; \mathit{proofs} - \mathit{of} - \mathit{laws}\}\\
&\quad ;\; \_\Delta\_ \;=\; \lambda\,(e\,,\,u)\;(f\,,\,v) \mapsto (e \circ f)\,,\,(\lambda\,\{k\} \mapsto u\; \{f\; k\} \circ v\; \{k\})\\
&\quad ;\; \mathit{id} \;\; = \;\; (\lambda\, x \mapsto x)\,,\,(\lambda\,\{i\}\; x \mapsto x)\\
&\quad ;\; \mathit{proofs} - \mathit{of} - \mathit{laws}\}
\end{aligned}
$$

Note that the equivalence on morphisms is defined to be componentwise extensional equality, which is formulated with the help of McBride's "John Major" heterogeneous equality *_JMEq_* McBride [1999] — the equivalence $\mathit{\_JMEq'\_}$ is defined by $g\; \mathit{JMEq'}\; h \;=\; \forall\; x \to g\; x\; \mathit{JMEq}\; h\; x$. (Given $y \;:\; Y\; j$ for some $j \;:\; J$,

```
record Functor
  {l m n l′ m′ n′ : Level}
  (C : Category {l} {m} {n}) (D : Category {l′} {m′} {n′}) :
  Set (l ⊔ m ⊔ n ⊔ l′ ⊔ m′ ⊔ n′) where
  field
    object      : CObject → DObject
    morphism    : ∀ {X Y} → X = C => Y → object X = D => object Y
    ≈ −respecting :
      ∀ {X Y} {f g : X = C => Y} →
      f ≈ C g → morphism f ≈ D morphism g
    id − preserving :
      ∀ {X} → morphism (idC {X}) ≈ D idD {object X}
    comp − preserving :
      ∀ {X Y Z} (f : Y = C => Z) (g : X = C => Y) →
      morphism (f ΔC g) ≈ D (morphism f ΔD morphism g)
```

Figure 4.2 Definition of functors.

the types of $u\ \{j\}\ y$ and $u'\ \{j\}\ y$ are not definitionally equal but only provably equal, so it is necessary to employ heterogeneous equality.)

We will also need functors, whose definition is shown in Figure 4.2: a functor consists of two mappings, one on objects and the other on morphisms, where the morphism part respects equivalence on morphisms and preserves identity and composition. For example, we have two forgetful functors from *Fam* to *Fun*, one summing components together

```
FamF : Functor Fam Fun
FamF = record { object      = λ (I , X) ↦ Σ I X
              ; morphism    = λ (e , u) ↦ e ** u
              ; proofs − of − laws}
```

and the other extracting the index part.

```
FamI : Functor Fam Fun
```

$$\begin{aligned} \mathit{FamI} \;=\; \textbf{record}\; \{\; & \mathit{object} \;\;\;\; = \lambda\,(I\,,\,X) \mapsto I \\ & ;\; \mathit{morphism} \;=\; \lambda\,(e\,,\,u\,) \mapsto e \\ & ;\; \mathit{proofs} - \mathit{of} - \mathit{laws}\} \end{aligned}$$

The functor laws should be proved for both functors alongside their object and morphism maps. In particular, we need to prove that the morphism part respects equivalence: for *FamF* this means we need to prove, for all $e : J \to I$, $u : Y \rightrightarrows (X \circ e)$, $f : J \to I$, and $v : Y \rightrightarrows (X \circ f)$, that

$$(e \doteq f) \times (\forall\, \{j\} \to u\, \{j\}\; \mathit{JMEq}'\; v\, \{j\}) \to (e \ast\ast\ u \doteq f \ast\ast\ v)$$

and for *FamI* we need to prove

$$(e \doteq f) \times (\forall\, \{j\} \to u\, \{j\}\; \mathit{JMEq}'\; v\, \{j\}) \to (e \doteq f)$$

both of which can be easily discharged.

4.1.2 Definition of pullbacks

We will define a pullback to be a product in the suitable slice category, where a product is defined to be a terminal object in the suitable span category. Below we give definitions of all these terms in logical order. Let C : *Category* in what follows.

- A *slice category* based on C is parametrised by an object B; its objects are those morphisms in C with target B and its morphisms are mediating morphisms giving rise to commutative triangles — diagrammatically,

  $$\text{objects}\;\; \begin{array}{c} T \\ {\scriptstyle s}\big\downarrow \\ B \end{array} \qquad \text{and} \qquad \text{morphisms}\;\; \begin{array}{ccc} T & \xrightarrow{\;m\;} & T' \\ & {\scriptstyle s}\searrow \quad \swarrow{\scriptstyle s'} & \\ & B & \end{array}$$

  The slice objects and morphisms are defined in Agda as two records; they are shown in the upper half of Figure 4.3 along with the definition of slice categories. Note that the equivalence on slice morphisms is defined as only the equivalence on the mediating morphisms, essentially achieving proof-irrelevance.

- *Span categories* are similar: parametrised by two objects L and R, a span category has

  $$\text{objects} \quad L \xleftarrow{l} M \xrightarrow{r} R \qquad \text{and} \quad \text{morphisms} \quad \begin{array}{ccccc} & & l & M & r & & \\ & \swarrow & & \downarrow m & & \searrow & \\ L & & & & & & R \\ & \nwarrow & & \downarrow & & \nearrow & \\ & & l' & M' & r' & & \end{array}$$

  The Agda definitions are shown in the lower half of Figure 4.3, and are similar to those for slice categories.

- An object X in C is *terminal* if it satisfies the universal property that for every object Y there is a unique morphism from Y to X:

  $$\begin{aligned} &\mathit{Terminal}\ C\ :\ \mathit{Object} \to \mathsf{Set}\ \_ \\ &\mathit{Terminal}\ C\ X\ = \\ &\quad (Y\ :\ \mathit{Object}) \to \Sigma\langle f : Y ==> X \rangle\ \mathit{Unique}\ (\mathit{Morphism}\ Y\ X)\ f \end{aligned}$$

  where uniqueness is defined relative to a setoid:

  $$\begin{aligned} &\mathit{Unique}\ :\ (S\ :\ \mathit{Setoid}) \to \mathit{Carrier\_S} \to \mathsf{Set}\ \_ \\ &\mathit{Unique}\ S\ x\ =\ (y\ :\ \mathit{Carrier\_S}) \to x \approx S\ y \end{aligned}$$

- A *product* of two objects X and Y in C is then a $\mathit{Span}\ C\ X\ Y$ that is terminal in $\mathit{SpanCategory}\ C\ X\ Y$:

  $$\begin{aligned} &\mathit{Product}\ C\ X\ Y\ :\ \mathit{Span}\ C\ X\ Y \to \mathsf{Set}\ \_ \\ &\mathit{Product}\ C\ X\ Y\ =\ \mathit{Terminal}\ (\mathit{SpanCategory}\ C\ X\ Y) \end{aligned}$$

- A *pullback* of two slices $f, g\ :\ \mathit{Slice}\ C\ X$ is a product of f and g in $\mathit{SliceCategory}\ C\ X$: Define the type of *squares* based on f and g as

  $$\begin{aligned} &\mathsf{Square}\ C\ f\ g\ :\ \mathsf{Set}\ \_ \\ &\mathsf{Square}\ C\ f\ g\ =\ \mathit{Span}\ (\mathit{SliceCategory}\ C\ X)\ f\ g \end{aligned}$$

  or diagrammatically,

  

  which is the same as

  

In a square q, we will refer to the object $\mathit{Slice.T}\ (\mathit{Span.M}\ q)$, i.e., the node W in the diagrams above, as the *vertex* of q. A pullback of f and g is then a square based on f and g that satisfies

$$\mathit{Pullback}\ C\ f\ g\ :\ \mathsf{Square}\ C\ f\ g \to \mathsf{Set}\ \_$$
$$\mathit{Pullback}\ C\ f\ g\ =\ \mathit{Product}\ (\mathit{SliceCategory}\ C\ X)\ f\ g$$

Equivalently, if we define the *square category* based on f and g as

$$\mathit{SquareCategory}\ C\ f\ g\ :\ \mathit{Category}$$
$$\mathit{SquareCategory}\ C\ f\ g\ =$$
$$\quad \mathit{SpanCategory}\ (\mathit{SliceCategory}\ C\ X)\ f\ g$$

then a pullback of f and g is a terminal object in the square category based on f and g — indeed, $\mathit{Product}\ (\mathit{SliceCategory}\ C\ X)\ f\ g$ is definitionally equal to $\mathit{Terminal}\ (\mathit{SquareCategory}\ C\ f\ g)$.

The most important category-theoretic fact that we will use in this paper is that the vertices of any two pullbacks of the same slices are isomorphic. Define the type of isomorphisms between two objects X and Y in C as

$$\textbf{record}\ \mathsf{Iso}\ C\ X\ Y\ :\ \mathsf{Set}\ \_\ \textbf{where}$$
$$\quad \textbf{field}$$
$$\quad\quad \mathit{to}\ \ \ :\ X \Longrightarrow Y$$
$$\quad\quad \mathit{from}\ :\ Y \Longrightarrow X$$
$$\quad\quad \mathit{from} - \mathit{to} - \mathit{inverse}\ :\ \mathit{from}\ \Delta\ \mathit{to} \approx \mathit{id}$$
$$\quad\quad \mathit{to} - \mathit{from} - \mathit{inverse}\ :\ \mathit{to}\ \Delta\ \mathit{from} \approx \mathit{id}$$

(The isomorphism relation $\_ \cong \_$ we used in Section 3.2 is formally defined as $\mathsf{Iso}\ \mathit{Fun}$.) Then we can formulate the following lemma: If $p, q\ :\ \mathsf{Square}\ C\ f\ g$ are both pullbacks, then we have an isomorphism

$$\mathsf{Iso}\ C\ (\mathit{Slice.T}\ (\mathit{Span.M}\ p))\ (\mathit{Slice.T}\ (\mathit{Span.M}\ q))$$

4.2 Categorical organisation of the ornament–refinement framework

We proceed to organise the ornament–refinement framework under several concrete categories and functors, aiming to clarify the overall structure of the framework, and then derive useful pullback properties from parallel composition of ornaments.

4.2.1 The category of type families and refinement families

We will see that the category *FRef* of type families and refinement families has a very close relationship to the category *Fam*. An object in *FRef* is an indexed family of sets as in *Fam*, and a morphism from $(J\ ,\ Y)$ to $(I\ ,\ X)$ consists of a function $e\ :\ J \to I$ on the indices and a refinement family of type $\mathsf{FRefinement}\ e\ X\ Y$. As for the equivalence on morphisms, it suffices to use extensional equality on the index functions and componentwise equivalence on refinement families, where the equivalence on refinements is defined to be extensional equality on their forgetful functions (extracted by $\mathsf{Refinement}.\mathit{forget}$). In Agda:

$$
\begin{array}{l}
\mathit{FRef}\ :\ \mathit{Category} \\
\mathit{FRef}\ =\ \mathbf{record} \\
\quad \{\ \mathit{Object}\ \ \ \ \ =\ \Sigma \langle\, I : \mathsf{Set} \,\rangle\ I \to \mathsf{Set} \\
\quad ;\ \mathit{Morphism}\ = \\
\qquad \lambda\,(J\ ,\ Y)\ (I\ ,\ X)\ \mapsto\ \mathbf{record} \\
\qquad\quad \{\ \mathit{Carrier}\ =\ \Sigma \langle\, e : J \to I \,\rangle\ \mathsf{FRefinement}\ e\ X\ Y \\
\qquad\quad ;\ \_ \approx \_\ \ \ = \\
\qquad\qquad \lambda\,(e\ ,\ rs)\ (e'\ ,\ rs')\ \mapsto \\
\qquad\qquad\quad (e\ \doteq\ e')\ \times \\
\qquad\qquad\quad (\forall\ j \to \mathsf{Refinement}.\mathit{forget}\ (rs\ \ (\mathsf{ok}\ j))\ \mathit{JMEq}' \\
\qquad\qquad\qquad\qquad\quad \mathsf{Refinement}.\mathit{forget}\ (rs'\ (\mathsf{ok}\ j))) \\
\qquad\quad ;\ \mathit{proofs} - \mathit{of} - \mathit{laws}\}
\end{array}
$$

$$; \mathit{proofs-of-laws}\}$$

Two facts support our choice of refinement equivalence: (i) under this definition, if two refinements are equivalent, then their promotion predicates are pointwise isomorphic, i.e., we have

$$\begin{aligned}
&\mathit{forget-iso} :\\
&\quad \{X\ Y : \mathsf{Set}\}\ (r\ s : \mathsf{Refinement}\ X\ Y) \to\\
&\quad (\mathsf{Refinement}.\mathit{forget}\ r \doteq \mathsf{Refinement}.\mathit{forget}\ s) \to\\
&\quad \forall\ x \to \mathsf{Refinement}.P\ r\ x \cong \mathsf{Refinement}.P\ s\ x
\end{aligned}$$

and (ii) we get a forgetful functor $\mathit{FRefF} : \mathit{Functor\ FRef\ Fam}$ which is identity on objects and componentwise $\mathsf{Refinement}.\mathit{forget}$ on morphisms, the latter respecting equivalence automatically.

$$\begin{aligned}
&\mathit{FRefF} : \mathit{Functor\ FRef\ Fam}\\
&\mathit{FRefF} = \mathbf{record}\\
&\quad \{\ \mathit{object} \quad = \lambda\,(I\ ,\ X) \mapsto I\ ,\ X\\
&\quad ;\ \mathit{morphism} =\\
&\qquad \lambda\,(e\ ,\ rs) \mapsto e\ ,\ (\lambda\,\{j\} \mapsto \mathsf{Refinement}.\mathit{forget}\ (rs\ (\mathsf{ok}\ j)))\\
&\quad ;\ \mathit{proofs-of-laws}\}
\end{aligned}$$

Note that a refinement family from $X : I \to \mathsf{Set}$ to $Y : J \to \mathsf{Set}$ is deliberately cast as a morphism in the opposite direction from $(J\ ,\ Y)$ to $(I\ ,\ X)$, so *FRefF* remains a familiar covariant functor rather than a contravariant one. Think of this as suggesting the direction of the forgetful functions of refinements.

The above discussion suggests that the essential ingredient of a refinement is just its forgetful function. Indeed, from any function we can construct a *canonical refinement*:

$$\begin{aligned}
&\mathit{canonRef} : \{X\ Y : \mathsf{Set}\} \to (Y \to X) \to \mathsf{Refinement}\ X\ Y\\
&\mathit{canonRef}\ \{X\}\ \{Y\}\ f = \mathbf{record}\\
&\quad \{\ P = \lambda\,x \mapsto \Sigma\langle\, y : Y \,\rangle\ f\ y \equiv x\\
&\quad ;\ i = \mathbf{record}\ \{\ \mathit{to} \quad = (\mathit{split}\ (f)\ (\mathit{split}\ ((\lambda\,y \mapsto y))\ (\lambda\,y \mapsto \mathsf{refl})))\\
&\qquad\qquad\qquad ;\ \mathit{from} = \mathsf{outl} \circ \mathsf{outr}\\
&\qquad\qquad\qquad ;\ \mathit{proofs-of-laws}\}\}
\end{aligned}$$

(The operator $split - op$ is defined by $(split\ (g)\ (h)) \;=\; \lambda\, x \;\mapsto\; (g\ x\ ,\ h\ x)$.) The canonical promotion predicate is very simplistic: to promote some $x : X$ to type Y, we are required to supply a complete $y : Y$ such that x can be recovered from y (rather than only the necessary information that augments x to an element of Y). Any refinement r : Refinement X Y is equivalent to $canonRef$ (Refinement.$forget\ r$), so by $forget - iso$ we have

$$\mathsf{Refinement}.P\ r\ x \;\cong\; \Sigma\langle\, y : Y \,\rangle\ \mathsf{Refinement}.forget\ r\ y \equiv x$$

for all $x : X$. That is, a promotion predicate is always pointwise isomorphic to the canonical promotion predicate. Thus all the refinement mechanism provides is a convenient way of expressing intensional (representational) optimisations of the canonical promotion predicate — extensionally, *FRef* is no more powerful than *Fam*. This is reflected in the existence of a functor $FRefC : Functor\ Fam\ FRef$, whose object part is identity and whose morphism part is componentwise *canonRef*:

$$\begin{aligned}
&FRefC : Functor\ Fam\ FRef\\
&FRefC \;=\; \mathbf{record}\\
&\quad \{\ object \;\;\;\; = \lambda\,(I\ ,\ X) \mapsto I\ ,\ X\\
&\quad ;\ morphism \;=\; \lambda\,(e\ ,\ u\ ) \mapsto e\ ,\ (\lambda\,(\mathsf{ok}\ j) \mapsto canonRef\ (u\ \{j\}))\\
&\quad ;\ proofs - of - laws\}
\end{aligned}$$

FRefC is strictly inverse to *FRefF*, forming an isomorphism (not merely an equivalence) of categories between *FRef* and *Fam*.

4.2.2 The category of descriptions and ornaments

The category Orn has objects of type $\Sigma\langle\, I : \mathsf{Set}\,\rangle\ \mathsf{Desc}\ I$, i.e., descriptions paired with index sets, and morphisms from $(J\ ,\ E)$ to $(I\ ,\ D)$ of type $\Sigma\langle\, e : J \to I\,\rangle\ \mathsf{Orn}\ e\ D\ E$, i.e., ornaments paired with index erasure functions. We also need to devise an equivalence on ornaments

$$\begin{aligned}
&OrnEq\ :\\
&\quad \{I\ J\ :\ \mathsf{Set}\}\ \{e\ e'\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to\\
&\quad \mathsf{Orn}\ e\ D\ E \to \mathsf{Orn}\ e'\ D\ E \to \mathsf{Set}
\end{aligned}$$

such that it implies extensional equality of e and e' and that of ornamental forgetful functions:

$$
\begin{aligned}
&\mathit{OrnEq} - \mathit{forget} \ : \\
&\quad \{I\ J\ :\ \mathsf{Set}\}\ \{e\ e'\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \\
&\quad (O\ :\ \mathsf{Orn}\ e\ D\ E)\ (P\ :\ \mathsf{Orn}\ e'\ D\ E) \to \mathit{OrnEq}\ O\ P \to \\
&\quad (e\ \doteq\ e') \times (\forall\ \{j\} \to \mathit{forget}\ O\ \{j\}\ \mathit{JMEq}'\ \mathit{forget}\ P\ \{j\})
\end{aligned}
$$

We omit the detail of *OrnEq* from the paper (which depends on the detail of the universe of ornaments). Morphism composition is sequential composition, and there is a family of *identity ornaments*

$$\mathit{idOrn}\ :\ \{I\ :\ \mathsf{Set}\}\ \{D\ :\ \mathsf{Desc}\ I\} \to \mathsf{Orn}\ (\lambda\, i \mapsto i)\ D\ D$$

such that $\mathit{idOrn}\ \{I\}\ \{D\}$ simply expresses that D is identical to itself. Unsurprisingly, the identity ornaments serve as identity of sequential composition. To summarise:

$$
\begin{aligned}
&\textsc{Orn}\ :\ \mathit{Category} \\
&\textsc{Orn}\ =\ \mathbf{record} \\
&\quad \{\ \mathit{Object}\ \ \ \ \ \ =\ \Sigma\langle\, I : \mathsf{Set}\,\rangle\ \mathsf{Desc}\ I \\
&\quad ;\ \mathit{Morphism}\ = \\
&\qquad \lambda\,(J\ ,\ E)\ (I\ ,\ D) \mapsto\ \mathbf{record} \\
&\qquad\quad \{\ \mathit{Carrier}\ =\ \Sigma\langle\, e : J \to I\,\rangle\ \mathsf{Orn}\ e\ D\ E \\
&\qquad\quad ;\ \_ \approx \_\ \ =\ \lambda\,(e\ ,\ O)\ (e'\ ,\ O') \mapsto\ \mathit{OrnEq}\ O\ O' \\
&\qquad\quad ;\ \mathit{proofs} - \mathit{of} - \mathit{laws}\} \\
&\quad ;\ \_\Delta\_\ =\ \lambda\,(e\ ,\ O)\ (f\ ,\ P) \mapsto\ (e \circ f)\ ,\ (O \odot P) \\
&\quad ;\ \mathit{id}\ \ \ \ =\ (\lambda\, i \mapsto i)\ ,\ \mathit{idOrn} \\
&\quad ;\ \mathit{proofs} - \mathit{of} - \mathit{laws}\}
\end{aligned}
$$

A functor $\mathit{Ind}\ :\ \mathit{Functor}\ \textsc{Orn}\ \mathit{Fam}$ can then be constructed, which gives the ordinary semantics of descriptions and ornaments: the object part of *Ind* decodes a description $(I\ ,\ D)$ to its least fixed point $(I\ ,\ \mu\ D)$, and the morphism part translates an ornament $(e\ ,\ O)$ to the forgetful function $(e\ ,\ \mathit{forget}\ O)$, the latter respecting equivalence by virtue of $\mathit{OrnEq} - \mathit{forget}$.

$$
\begin{aligned}
&\mathit{Ind}\ :\ \mathit{Functor}\ \textsc{Orn}\ \mathit{Fam} \\
&\mathit{Ind}\ =\ \mathbf{record}\ \{\ \mathit{object}\ \ \ \ \ \ =\ \lambda\,(I\ ,\ D) \mapsto\ I\ ,\ \mu\ D
\end{aligned}
$$

$$
\begin{aligned}
&; \mathit{morphism} \;=\; \lambda\,(e\;,\,O) \;\mapsto\; e\;,\mathit{forget}\;O \\
&; \mathit{proofs} - \mathit{of} - \mathit{laws}\}
\end{aligned}
$$

4.2.3 Pullback properties for parallel composition

We are now ready to state the pullback properties for parallel composition of ornaments. With suitable choices of encoding for the universes, we could attempt to establish that, for any two ornaments O : Orn e D E and P : Orn f D F where D : Desc I, E : Desc J, and F : Desc K, the following square in $\mathbb{O}$rn is a pullback:

$$
\begin{array}{ccc}
e \bowtie f\;,\,\lfloor O \otimes P \rfloor & \xrightarrow{\;\mathsf{outr}\;,\,\mathit{diffOrn\text{-}r}\;O\;P\;} & K\;,\,F \\
{\scriptstyle \mathsf{outl}\;,\,\mathit{diffOrn\text{-}l}\;O\;P}\downarrow & \searrow^{\mathit{pull},\,\lceil O \otimes P \rceil} & \downarrow{\scriptstyle f\;,\,P} \\
J\;,\,E & \xrightarrow[e\;,\,O]{} & I\;,\,D
\end{array}
$$

This square is encoded in Agda as

$$
\begin{aligned}
&\mathit{pc} - \mathit{square} \;:\; \mathsf{Square}\;\mathbb{O}\mathrm{rn}\;(\mathit{slice}\;(J\;,\,E)\;(e\;,\,O))\;(\mathit{slice}\;(K\;,\,F)\;(f\;,\,P)) \\
&\mathit{pc} - \mathit{square} \;=\; \mathit{span}\;(\mathit{slice}\;(e \bowtie f\;,\,\lfloor O \otimes P \rfloor)\;(\mathit{pull},\,\lceil O \otimes P \rceil)) \\
&\qquad\qquad (\mathit{sliceMorphism}\;(\mathsf{outl}\;,\,\mathit{diffOrn\text{-}l}\;O\;P)\;(\mathit{hole}\;()\;(1))) \\
&\qquad\qquad (\mathit{sliceMorphism}\;(\mathsf{outr}\;,\,\mathit{diffOrn\text{-}r}\;O\;P)\;(\mathit{hole}\;()\;(2)))
\end{aligned}
$$

where goal 1 has type $\mathit{OrnEq}\;(O \odot \mathit{diffOrn\text{-}l}\;O\;P)\;\lceil O \otimes P \rceil$ and goal 2 has type $\mathit{OrnEq}\;(P \odot \mathit{diffOrn\text{-}r}\;O\;P)\;\lceil O \otimes P \rceil$, both of which can be discharged.[2] The pullback property of $\mathit{pc} - \mathit{square}$, i.e., $\mathit{Pullback}\;\mathbb{O}\mathrm{rn}\;\_\;\_\;\mathit{pc} - \mathit{square}$, is not too useful by itself, though: $\mathbb{O}$rn is quite a restricted category, so a universal property established in $\mathbb{O}$rn has limited applicability. Instead, we are more interested in the pullback property of the image of the above square under *Ind* in *Fam*, which is stated in the follow theorem. If O : Orn e D E and

[2] Since the structure of Agda terms like $\mathit{pc} - \mathit{square}$ can be reconstructed from commutative diagrams and the categorical definitions, in the rest of the paper we will present only the commutative diagrams and omit the underlying Agda terms.

P : Orn f D F where D : Desc I, E : Desc J, and F : Desc K, then the following square in *Fam* is a pullback.

$$\begin{array}{ccc} & & \xrightarrow{\mathsf{outr}\ ,\ forget\ (diffOrn\text{-}r\ O\ P)} \\ e \bowtie f\ ,\ \mu\ \lfloor O \otimes P \rfloor & \longrightarrow & K\ ,\ \mu\ F \\ \downarrow{\scriptstyle \mathsf{outl}\ ,\ forget\ (diffOrn\text{-}l\ O\ P)} & \lrcorner\ \searrow{\scriptstyle pull,\ forget\ \lceil O \otimes P \rceil} & \downarrow{\scriptstyle f\ ,\ forget\ P} \\ J\ ,\ \mu\ E & \xrightarrow[e\ ,\ forget\ O]{} & I\ ,\ \mu\ D \end{array}$$

The proof of the universal property boils down to, very roughly speaking, construction of an inverse to

$$(split\ (forget\ (diffOrn\text{-}l\ O\ P))\ (forget\ (diffOrn\text{-}r\ O\ P)))$$

which involves tricky manipulation of equality proofs but is achievable. After the pullback property is established in *Fam*, since *FamF* is pullback-preserving, we also get a pullback square in *Fun*. If O : Orn e D E and P : Orn f D F where D : Desc I, E : Desc J, and F : Desc K, then the following square in *Fun* is a pullback.

$$\begin{array}{ccc} & & \xrightarrow{\mathsf{outr}\ **\ forget\ (diffOrn\text{-}r\ O\ P)} \\ \Sigma\ (e \bowtie f)\ (\mu\ \lfloor O \otimes P \rfloor) & \longrightarrow & \Sigma\ K\ (\mu\ F) \\ \downarrow{\scriptstyle \mathsf{outl}\ **\ forget\ (diffOrn\text{-}l\ O\ P)} & \lrcorner\ \searrow{\scriptstyle pull\ **\ forget\ \lceil O \otimes P \rceil} & \downarrow{\scriptstyle f\ **\ forget\ P} \\ \Sigma\ J\ (\mu\ E) & \xrightarrow[e\ **\ forget\ O]{} & \Sigma\ I\ (\mu\ D) \end{array}$$

To translate Orn to *FRef*, i.e., datatype declarations to refinements, a naive way is to use the composite functor

$$\textsc{Orn} \xrightarrow{\ Ind\ } Fam \xrightarrow{\ FRefC\ } FRef$$

The resulting refinements would then use canonical promotion predicates. However, the whole point of incorporating Orn in the framework is that we can construct an alternative functor *RSem* directly from Orn to *FRef*. The functor *RSem* is extensionally equal to the above composite functor, but intensionally very different. Its object part still takes the least fixed point of a

description, but its morphism part is the refinement semantics of ornaments given in Section 3.1, whose promotion predicates have a more efficient representation.

$$
\begin{array}{l}
\mathit{RSem} \;:\; \mathit{Functor}\ \mathbb{O}\textsc{rn}\ \mathit{FRef} \\
\mathit{RSem} \;=\; \mathbf{record} \\
\quad \{\ \mathit{object} \quad\;\; = \;\lambda\,(I\,,\,D) \mapsto I\,,\,\mu\,D \\
\quad ;\ \mathit{morphism} \;= \\
\qquad \lambda\,(e\,,\,O) \mapsto e\,,\,(\lambda\,(\mathsf{ok}\;j) \mapsto \mathbf{record}\ \{\ P \;=\; \mathsf{OptP}\;O\;(\mathsf{ok}\;j) \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad\quad ;\ i \;\;=\; (\mathit{hole}\;()\;(3))\}) \\
\quad ;\ \mathit{proofs-of-laws}\}
\end{array}
$$

We will give goal 3, i.e., the ornamental promotion isomorphisms, a new construction in the next section.

4.3 Reconstruction of the ornamental promotion and modularity isomorphisms

The morphism part of the functor $\mathit{RSem} \;:\; \mathit{Functor}\ \mathbb{O}\textsc{rn}\ \mathit{FRef}$ translates ornaments into refinements that use the optimised predicates, which are defined via parallel composition, so the pullback properties for parallel composition hold for the optimised predicates. The natural step to take, then, is to construct the ornamental promotion isomorphisms using the pullback properties — this we do in the proof of **??** below. Even more closely related are the modularity isomorphisms, which are about parallel composition and optimised predicates. They, too, can be constructed from the pullback properties for parallel composition, which is done in the proof of **??**.

We restate the ornamental promotion isomorphisms as the following theorem. For any ornament $O \;:\; \mathsf{Orn}\ e\ D\ E$ where $D \;:\; \mathsf{Desc}\ I$ and $E \;:\; \mathsf{Desc}\ J$, we have

$$\mu\ E\ j \;\cong\; \Sigma\langle\, x : \mu\ D\ (e\ j)\,\rangle\ \ \mathsf{OptP}\ O\ (\mathsf{ok}\ j)\ x$$

for all $j \;:\; J$. Since the optimised predicates $\mathsf{OptP}\ O$ are defined by parallel

composition of O and the singleton ornament $S = \mathit{singOrn}\ D$, the conclusion of the theorem expand to

$$\mu\ E\ j \cong \Sigma \langle\, x : \mu\ D\ (e\ j) \,\rangle\ \ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x)) \tag{4.1}$$

How do we derive these isomorphisms from the pullback properties for parallel composition? It turns out that the pullback property in *Fun* (**??**) can help.

First, observe that we have the following pullback square:

$$\begin{array}{ccc} \Sigma\ J\ (\mu\ E) & \xrightarrow{(\mathit{split}\ (e\ **\ \mathit{forget}\ O)\ (\mathit{singleton} \circ \mathit{forget}\ O \circ \mathsf{outr}))} & \Sigma\ (\Sigma\ I\ (\mu\ D))\ (\mu\ \lfloor S \rfloor) \\ {\scriptstyle \mathit{id}}\downarrow & \swarrow\ {\scriptstyle e\ **\ \mathit{forget}\ O} & \downarrow{\scriptstyle \mathsf{outl}\ **\ \mathit{forget}\ \lceil S \rceil} \\ \Sigma\ J\ (\mu\ E) & \xrightarrow[e\ **\ \mathit{forget}\ O]{} & \Sigma\ I\ (\mu\ D) \end{array} \tag{4.2}$$

If we view pullbacks as products of slices, since a singleton ornament does not add information to a datatype, the vertical slice on the right-hand side

$$s = \mathit{slice}\ (\Sigma\ (\Sigma\ I\ (\mu\ D))\ (\mu\ \lfloor S \rfloor))\ (\mathsf{outl}\ **\ \mathit{forget}\ \lceil S \rceil)$$

behaves like a "multiplicative unit": any (compatible) slice s' alone gives rise to a product of s and s'. As a consequence, we have the bottom-left type $\Sigma\ J\ (\mu\ E)$ as the vertex of the pullback. This pullback square is based on the same slices as the one in **??** with P substituted by $\lceil S \rceil$, so by **??** we obtain an isomorphism

$$\Sigma\ J\ (\mu\ E) \cong \Sigma\ (e \bowtie \mathsf{outl})\ (\mu\ \lfloor O \otimes \lceil S \rceil \rfloor) \tag{4.3}$$

To get from (4.3) to (4.1), we need to look more closely into the construction of (4.3). The right-to-left direction of (4.3) is obtained by applying the universal property of (4.2) to the square in **??** (with P substituted by $\lceil S \rceil$), so it is the unique mediating morphism m that makes the following diagram commute:

$$\begin{array}{ccccc} & & \Sigma\ (e \bowtie \mathsf{outl})\ (\mu\ \lfloor O \otimes \lceil S \rceil \rfloor) & & \\ & {\scriptstyle \mathsf{outl}\ **\ \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)} \swarrow & & \searrow {\scriptstyle \mathsf{outr}\ **\ \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P)} & \\ \Sigma\ J\ (\mu\ E) & & {\scriptstyle m}\ \vdots & & \Sigma\ (\Sigma\ I\ (\mu\ D))\ (\mu\ \lfloor S \rfloor) \\ & {\scriptstyle \mathit{id}} \nwarrow & \downarrow & \nearrow {\scriptstyle \langle e\ **\ \mathit{forget}\ O,\ \mathit{singleton} \circ \mathit{forget}\ O \circ \mathsf{outr} \rangle} & \\ & & \Sigma\ J\ (\mu\ E) & & \end{array}$$

From the left commuting triangle, we see that, extensionally, the morphism m is just $\mathsf{outl} \ast\ast \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$. This leads us to the following general lemma: If there is an isomorphism

$$\Sigma\, K\, X \cong \Sigma\, L\, Y$$

whose right-to-left direction is extensionally equal to some $f \ast\ast g$, then we have

$$X\, k \cong \Sigma \langle\, l : f^{-1}\, k \,\rangle\ Y\ (\mathit{und}\ l)$$

for all $k : K$. For a fixed $k : K$, an element of the form $(k\ ,\ x) : \Sigma\ K\ X$ must correspond, under the isomorphism, to some element $(l\ ,\ y) : \Sigma\ L\ Y$ such that $f\ l \equiv k$, so the set $X\ k$ corresponds to exactly the sum of the sets $Y\ l$ such that $f\ l \equiv k$. Specialising **??** for (4.3), we get

$$\mu\ E\ j \cong \Sigma \langle\, \mathit{jix} : \mathsf{outl}^{-1}\ j \,\rangle\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathit{und}\ \mathit{jix}) \tag{4.4}$$

for all $j : J$. Finally, observe that a canonical element of type $\mathsf{outl}^{-1}\ j$ must be of the form $\mathsf{ok}\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x))$ for some $x : \mu\ D\ (e\ j)$, so we perform a change of variables for the summation, turning the right-hand side of (4.4) into

$$\Sigma \langle\, x : \mu\ D\ (e\ j) \,\rangle\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x))$$

and arriving at (4.1).

There is a twist, however, due to Agda's intensionality: It is possible to formalise the above lemma and the change of variables individually and chain them together, but the resulting isomorphisms would have a very complicated definition due to suspended type casts. If we use them to construct the refinement family in the morphism part of *RSem*, it would be rather difficult to prove that the morphism part of *RSem* respects equivalence. We are thus forced to fuse all the above reasoning into one step to get a clean definition when we actually carry out this construction in Agda, but the idea is still essentially the same.

The other important family of isomorphisms we should consider is the modularity isomorphisms. Suppose that there are descriptions $D : \mathsf{Desc}\ I$, $E : \mathsf{Desc}\ J$ and $F : \mathsf{Desc}\ K$, and ornaments $O : \mathsf{Orn}\ e\ D\ E$, and $P : \mathsf{Orn}\ f\ D\ F$.

Then we have

$$\mathsf{OptP}\ \lceil O \otimes P \rceil\ (\mathsf{ok}\ (j\ ,\ k))\ x\ \cong\ \mathsf{OptP}\ O\ j\ x \times \mathsf{OptP}\ P\ k\ x$$

for all $i\ :\ I$, $j\ :\ e^{-1}\ i$, $k\ :\ f^{-1}\ i$, and $x\ :\ \mu\ D\ i$. The conclusion of the theorem expands to

$$\begin{aligned}&\mu\ \lfloor \lceil O \otimes P \rceil \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x)) \\ &\quad\cong\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) \times \mu\ \lfloor P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x))\end{aligned} \tag{4.5}$$

where again $S\ =\ \mathit{singOrn}\ D$. A quick observation is that they are componentwise isomorphisms between the two families of sets

$$M\ =\ \mu\ \lfloor \lceil O \otimes P \rceil \otimes \lceil S \rceil \rfloor$$

and

$$\begin{aligned}N\ =\ &\lambda\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x)) \mapsto \\ &\mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) \times \mu\ \lfloor P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x))\end{aligned}$$

both indexed by $\mathit{pull} \bowtie \mathsf{outl}$ where pull has type $e \bowtie f \to I$ and outl has type $\Sigma\ I\ X \to I$. This is just an isomorphism in Fam between $(\mathit{pull} \bowtie \mathit{proj}_1, M)$ and $(\mathit{pull} \bowtie \mathit{proj}_1, N)$ whose index part (i.e., the isomorphism obtained under the functor FamI) is identity. Thus we seek to prove that both $(\mathit{pull} \bowtie \mathit{proj}_1, M)$ and $(\mathit{pull} \bowtie \mathit{proj}_1, N)$ are vertices of pullbacks based on the same slices.

Let us look at $(\mathit{pull} \bowtie \mathit{proj}_1, N)$ first. For fixed i, j, k, and x, the set $N\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x))$ along with the cartesian projections is a product, which trivially extends to a pullback since there is a forgetful function from each of the two component sets to the *singleton* set $\mu\ \lfloor S \rfloor\ (i\ ,\ x)$, as shown in the following diagram:

$$\begin{array}{ccc} N\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x)) & \xrightarrow{\mathsf{outr}} & \lceil P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x)) \\ \Big\downarrow{\scriptstyle \mathsf{outl}} & \lrcorner & \Big\downarrow{\scriptstyle \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ P\ \lceil S \rceil)} \\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) & \xleftarrow[\mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ \lceil S \rceil)]{} & \mu\ \lfloor S \rfloor\ (i\ ,\ x) \end{array}$$

Note that this pullback square is possible because of the common x in the indices of the two component sets — otherwise they cannot project to the same

singleton set. Collecting all such pullback squares together, we get the following pullback square in *Fam*:

$$\begin{array}{ccc} \mathit{pull} \bowtie \mathsf{outl} \;,\; \overset{\mathsf{outr}}{\overbrace{N \rightarrow f}} \bowtie \mathsf{outl} \;,\; \mu\ \lfloor P \otimes \lceil S \rceil \rfloor \\ {\scriptstyle \_\,,\,\mathsf{outl}}\downarrow \lrcorner \qquad\qquad\qquad\qquad \downarrow{\scriptstyle \mathsf{outr}\,,\,\mathit{forget}\ (\mathit{diffOrn\text{-}r}\ P\ \lceil S \rceil)} \\ e \bowtie \mathsf{outl} \;,\; \mu\ \lfloor \underset{\mathsf{outr}\,,\,\mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ \lceil S \rceil)}{O \otimes \lceil S \rceil \rightarrow \Sigma}\ I\ (\mu\ D) \;,\; \mu\ \lfloor S \rfloor \end{array} \tag{4.6}$$

Next we prove that $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$ is also the vertex of a pullback based on the same slices as (4.6). This second pullback arises as a consequence of the following lemma. In any category, consider the objects X, Y, their product $X \Leftarrow X \boxtimes Y \Rightarrow Y$, and products of each of the three objects X, Y, and $X \boxtimes Y$ with an object Z. (All the projections are shown as solid arrows in the diagram below). Then $(X \boxtimes Y) \boxtimes Z$ is the vertex of a pullback of the two projections $X \boxtimes Z \Rightarrow Z$ and $Y \boxtimes Z \Rightarrow Z$.



We again intend to view a pullback as a product of slices, and instantiate **??** in *SliceCategory Fam* $(I\ ,\ \mu\ D)$, substituting all the objects by slices consisting of relevant ornamental forgetful functions in (4.5). The substitutions are as follows:

$$\begin{array}{rcl} X & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ O) \\ Y & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ P) \\ X \boxtimes Y & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ \lceil O \otimes P \rceil) \\ Z & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ \lceil S \rceil) \\ X \boxtimes Z & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ \lceil O \otimes \lceil S \rceil \rceil) \\ Y \boxtimes Z & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ \lceil P \otimes \lceil S \rceil \rceil) \\ (X \boxtimes Y) \boxtimes Z & \mapsto & \mathit{slice}\ \_\ (\_\ ,\ \mathit{forget}\ \lceil \lceil O \otimes P \rceil \otimes \lceil S \rceil \rceil) \end{array}$$

where $X \boxtimes Y$, $X \boxtimes Z$, $Y \boxtimes Z$, and $(X \boxtimes Y) \boxtimes Z$ indeed give rise to products in *SliceCategory Fam* $(I\ ,\ \mu\ D)$, i.e., pullbacks in *Fam*, by instantiating **??**. What we get out of this instantiation of the lemma is a pullback in *SliceCategory Fam* $(I\ ,\ \mu\ D)$

rather than *Fam*. This is easy to fix, since there is a forgetful functor from any *SliceCategory C B* to *C* whose object part is *Slice.T*, and it is pullback-preserving. We thus get a pullback in *Fam* which is based on the same slices as (4.6) and has vertex $(pull \bowtie \mathsf{outl}\ , M)$.

Having the two pullbacks, by **??** we get an isomorphism in *Fam* between $(pull \bowtie \mathsf{outl}\ , M)$ and $(pull \bowtie \mathsf{outl}\ , N)$, whose index part can be shown to be identity, so there are componentwise isomorphisms between M and N in *Fun*, arriving at (4.5).

```
record Slice C B : Set _ where
  constructor slice
  field
    T : Object
    s : T ==> B

record SliceMorphism C B (s t : Slice C B) : Set _ where
  constructor sliceMorphism
  field
    m : Slice.T s ==> Slice.T t
    triangle : Slice.s t ∆ m ≈ Slice.s s

SliceCategory C B : Category
SliceCategory C B =
  record
    { Object      = Slice C B
    ; Morphism    =
        λ s t ↦ record
                  { Carrier  = SliceMorphism C B s t
                  ; _≈_ = λ f g ↦ SliceMorphism.m f ≈
                                  SliceMorphism.m g
                  ; proofs-of-laws}
    ; proofs-of-laws}

record Span C L R : Set _ where
  constructor span
  field
    M : Object
    l : M ==> L
    r : M ==> R

record SpanMorphism C L R (s t : Span C L R) : Set _ where
  constructor spanMorphism
  field
    m : Span.M s ==> Span.M t
    triangle-l : Span.l t ∆ m ≈ Span.l s
    triangle-r : Span.r t ∆ m ≈ Span.r s

SpanCategory C L R : Category
SpanCategory C L R =
  record
    { Object      = Span C L R
    ; Morphism    =
        λ s t ↦ record
                  { Carrier  = SpanMorphism C L R s t
```



Figure 4.4 Categories (whose sets of objects and morphisms are listed below) and functors for the ornament–refinement framework.

Chapter 5

Relational algebraic ornaments

5.1 Relational program derivation in Agda and relational algebraic ornamentation

In this section, we first introduce and formalise some basic notions in relational program derivation Bird and de Moor [1997] by importing and generalising a small part of the AoPA library Mu et al. [2009]. We then introduce *relational algebraic ornamentation*, which acts as a bridge between the two worlds of internalist programming and relational program derivation. At the end of this section is an example about the *Fold Fusion Theorem* [Bird and de Moor, 1997, Section 6.2] and how the theorem translates to conversion functions between algebraically ornamented datatypes.

Basic definitions for relational program derivation. One common approach to program derivation is by algebraic transformations of functional programs: one begins with a specification in the form of a functional program that expresses straightforward but possibly inefficient computation, and transforms it into an extensionally equal but more efficient functional program by applying algebraic laws and theorems. Using functional programs as the specification language means that specifications are directly executable,

but the deterministic nature of functional programs can result in less flexible specifications. For example, when specifying an optimisation problem using a functional program that generates all feasible solutions and chooses an optimal one among them, the program would enforce a particular way of choosing the optimal solution, but such enforcement should not be part of the specification. To gain more flexibility, the specification language was later generalised to *relational programs*. With relational programs, we specify only the relationship between input and output without actually specifying a way to execute the programs, so specifications in the form of relational programs can be as flexible as possible. Though lacking a directly executable semantics, most relational programs can still be read computationally as potentially partial and nondeterministic mappings, so relational specifications largely remain computationally intuitive as functional specifications.

To emphasise the computational interpretation of relations, we will mainly model a relation between sets A and B as a function sending each element of A to a *subset* of B. We define subsets by

$$\wp : \mathsf{Set} \to \mathsf{Set}_1$$
$$(\mathit{Power}\ A) = A \to \mathsf{Set}$$

That is, a subset $s : (\mathit{Power}\ A)$ is a characteristic function that assigns a type to each element of A, and $a : A$ is considered to be a member of s if the type $s\ a : \mathsf{Set}$ is inhabited. We may regard $(\mathit{Power}\ A)$ as the type of computations that nondeterministically produce an element of A. A simple example is

$$\mathit{any} : \{A : \mathsf{Set}\} \to (\mathit{Power}\ A)$$
$$\mathit{any} = \mathit{const}\ \top$$

The subset $\mathit{any} : (\mathit{Power}\ A)$ associates the unit type $\top$ with every element of A. Since $\top$ is inhabited, *any* can produce any element of A. $\wp$ cannot be made into a conventional monad because it is not an endofunctor, but it still has a monadic structure Altenkirch et al. [2010]: *return* and $\_ >>= \_$ are defined as

$$\mathit{return} : \{A : \mathsf{Set}\} \to A \to (\mathit{Power}\ A)$$
$$\mathit{return} = \_ \equiv \_$$
$$\_ >>= \_ : \{A\ B : \mathsf{Set}\} \to (\mathit{Power}\ A) \to (A \to (\mathit{Power}\ B)) \to (\mathit{Power}\ B)$$

$$\_ >>= \_\ \{A\}\ s\ f\ =\ \lambda\ b \to \Sigma \langle\, a : A \,\rangle\ s\ a \times f\ a\ b$$

The subset $return\ a\ :\ (Power\ A)$ for some $a\ :\ A$ simplifies to $\lambda\ a' \to a\ \equiv\ a'$ (where $\_ \equiv \_$ is propositional equality), so a is the only member of the subset; if $s\ :\ (Power\ A)$ and $f\ :\ A \to (Power\ B)$, then the subset $s\ \gg\!=\ f\ :\ (Power\ B)$ is the union of all the subsets $f\ a\ :\ (Power\ B)$ where a ranges over the elements of A that belong to s, i.e., an element $b\ :\ B$ is a member of $s\ \gg\!=\ f$ exactly when there exists some $a\ :\ A$ belonging to s such that b is a member of $f\ a$.

We will mainly use relations between families of sets in this paper: if $X, Y\ :\ I \to \mathsf{Set}$ for some $I\ :\ \mathsf{Set}$, a relation from X to Y is defined as a family of relations from $X\ i$ to $Y\ i$ for every $i\ :\ I$.

$$\_ \rightsquigarrow \_\ :\ \{I\ :\ \mathsf{Set}\} \to (I \to \mathsf{Set}) \to (I \to \mathsf{Set}) \to \mathsf{Set}_1$$
$$X \rightsquigarrow Y\ =\ \forall\ \{i\} \to X\ i \to (Power\ (Y\ i))$$

We can use the subset combinators to define relations. For example, the following combinator *fun* lifts a family of functions into a family of relations.

$$fun\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ :\ I \to \mathsf{Set}\} \to (X \rightrightarrows Y) \to (X \rightsquigarrow Y)$$
$$fun\ f\ x\ =\ return\ (f\ x)$$

The identity relation is just the identity functions lifted to relations.

$$idR\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ :\ I \to \mathsf{Set}\} \to (X \rightsquigarrow X)$$
$$idR\ =\ fun\ id$$

Composition of relations is easily defined with $\_ >>= \_$: computing $R\ \cdot\ S$ on input x is first computing $S\ x$ and then feeding the result to R.

$$\_ffl\_\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ Z\ :\ I \to \mathsf{Set}\} \to$$
$$(Y \rightsquigarrow Z) \to (X \rightsquigarrow Y) \to (X \rightsquigarrow Z)$$
$$(R\ \cdot\ S)\ x\ =\ S\ x\ \gg\!=\ R$$

Or we may choose to define a relation pointwise, like

$$\_ \cap \_\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ :\ I \to \mathsf{Set}\} \to$$
$$(X \rightsquigarrow Y) \to (X \rightsquigarrow Y) \to (X \rightsquigarrow Y)$$
$$(R \cap S)\ x\ y\ =\ R\ x\ y \times S\ x\ y$$

This defines the meet of two relations. Unlike a function, which distinguishes between input and output, inherently a relation treats its domain and codomain

symmetrically. This is reflected by the presence of the following *converse* operator:

$$\begin{aligned}&\_^{\circ} : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\} \to (X \rightsquigarrow Y) \to (Y \rightsquigarrow X)\\&(R\ ^{\circ})\ y\ x\ =\ R\ x\ y\end{aligned}$$

A relation can thus be "run backwards" simply by taking its converse. The nondeterministic and bidirectional nature of relations makes them a powerful and concise language for specifications, as will be demonstrated in Section 5.3.

Laws and theorems in relational program derivation are formulated with *relational inclusion*

$$\begin{aligned}&\_ \subseteq \_ : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\}\ (R\ S : X \rightsquigarrow Y) \to \mathsf{Set}\\&R \subseteq S\ =\ \forall\ \{i\} \to (x : X\ i)\ (y : Y\ i) \to R\ x\ y \to S\ x\ y\end{aligned}$$

or equivalence of relations, which is defined as two-way inclusion:

$$\begin{aligned}&\_ \simeq \_ : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\}\ (R\ S : X \rightsquigarrow Y) \to \mathsf{Set}\\&R \simeq S\ =\ (R \subseteq S) \times (S \subseteq R)\end{aligned}$$

We will also need *relators*, i.e., monotonic functors on relations with respect to relational inclusion.

$$\begin{aligned}&\mathbb{R} : \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I)\ \{X\ Y : I \to \mathsf{Set}\} \to\\&\quad (X \rightsquigarrow Y) \to (\mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ Y)\end{aligned}$$

If $R : X \rightsquigarrow Y$, the relation $\mathbb{R}\ D\ R : \mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ Y$ applies R to the recursive positions of its input, leaving everything else intact. For example, if $D = \mathit{ListD}\ A$ (for some $A : \mathsf{Set}$), then $\mathbb{R}\ (\mathit{ListD}\ A)$ essentially specialises to

$$\begin{aligned}&\mathbb{R}\ (\mathit{ListD}\ A) : \{X\ Y : I \to \mathsf{Set}\} \to\\&\qquad (X \rightsquigarrow Y) \to (\mathbb{F}\ (\mathit{ListD}\ A)\ X \rightsquigarrow \mathbb{F}\ (\mathit{ListD}\ A)\ Y)\\&\mathbb{R}\ (\mathit{ListD}\ A)\ R\ (\mathit{nil} - \mathit{tag}\ \ , \blacksquare)\ \ \ \ =\ \mathit{return}\ (\mathit{nil} - \mathit{tag}\ , \blacksquare)\\&\mathbb{R}\ (\mathit{ListD}\ A)\ R\ (\mathit{cons} - \mathit{tag}\ , a\ , x)\ =\ R\ x\ \gg\!=\ \lambda\ y \to \mathit{return}\ (\mathit{cons} - \mathit{tag}\ , a\ , y)\end{aligned}$$

Among other properties, we can prove that $\mathbb{R}\ D$ preserves identity ($\mathbb{R}\ D\ \mathit{idR} \simeq \mathit{idR}$), composition ($\mathbb{R}\ D\ (R \cdot S) \simeq \mathbb{R}\ D\ R \cdot \mathbb{R}\ D\ S$), converse ($\mathbb{R}\ D\ (R\ ^{\circ}) \simeq (\mathbb{R}\ D\ R)\ ^{\circ}$), and is monotonic ($R \subseteq S$ implies $\mathbb{R}\ D\ R \subseteq \mathbb{R}\ D\ S$).

With relational inclusion, many concepts can be expressed in a surprisingly

concise way. For example, a relation R is a preorder if it is reflexive and transitive. In relational terms, these two conditions are expressed simply as $idR \subseteq R$ and $R \cdot R \subseteq R$, and are easily manipulable in calculations. Another important notion is *monotonic algebras* [Bird and de Moor, 1997, Section 7.2]: an algebra $S : \mathbb{F}\ D\ X \rightsquigarrow X$ is *monotonic* on $R : X \rightsquigarrow X$ (usually an ordering) if

$$S \cdot \mathbb{R}\ D\ R \subseteq R \cdot S$$

which says that if two input values to S have their recursive positions related by R and are otherwise equal, then the output values would still be related by R. In the context of optimisation problems, monotonicity can be used to capture the *principle of optimality*, as will be shown in Section 5.3.

Having defined relations as nondeterministic mappings, it is straightforward to port the datatype-generic *fold* to relations:

$$\begin{aligned} &foldR : \{I : \mathsf{Set}\}\ \{D : \mathsf{Desc}\ I\}\ \{X : I \to \mathsf{Set}\} \to \\ &\qquad (\mathbb{F}\ D\ X \rightsquigarrow X) \to (\mu\ D \rightsquigarrow X) \end{aligned}$$

The definition of *foldR* is obtained by rewriting the definition of *fold* with the subset combinators. For example, the relational fold on lists would essentially be

$$\begin{aligned} &foldR\ \{\top\}\ \{ListD\ A\} : \{X : \top \to \mathsf{Set}\} \to \\ &\qquad\qquad\qquad\qquad (\mathbb{F}\ (ListD\ A)\ X \rightsquigarrow X) \to \\ &\qquad\qquad\qquad\qquad (\mu\ (ListD\ A) \quad \rightsquigarrow X) \\ &(cata\ (R))\ [] \quad\quad = R\ (nil - tag\ ,\ \blacksquare) \\ &(cata\ (R))\ (a :: as) \ = \ (cata\ (R))\ as \ggg \lambda\ x \to R\ (cons - tag\ ,\ a\ ,\ x) \end{aligned}$$

The functional and relational fold operators are related by the following lemma:

$$\begin{aligned} &fun - preserves - fold : \\ &\quad \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I)\ \{X : I \to \mathsf{Set}\} \\ &\quad (f : \mathbb{F}\ D\ X \rightrightarrows X) \to fun\ (fold\ f) \simeq (cata\ (fun\ f)) \end{aligned}$$

Relational algebraic ornamentation. We now turn to relational algebraic ornamentation, the key construct that bridges internalist programming and

relational program derivation. Let $R : \mathbb{F}\ (ListD\ A)\ X \rightsquigarrow X$ (where $X : \top \to \mathsf{Set}$) be a relational algebra for lists. We can define a datatype of "algebraic lists" as

```
indexfirst data AlgList A R : X ▪ → Set where
  AlgList A R x accepts nil (rnil : R (nil − tag , ▪) x)
                      |     cons (a : A) (x′ : X ▪) (as : AlgList A R x′)
                                (rcons : R (cons − tag , a , x′) x)
```

There is an ornament from lists to algebraic lists which marks the fields *rnil*, x', and *rcons* in *AlgList* as additional and refines the index of the recursive position to x'. The promotion predicate for this ornament is

```
indexfirst data AlgListP A R : X ▪ → List A → Set where
  AlgListP A R x []        accepts nil (rnil : R (nil − tag , ▪) x)
  AlgListP A R x (a :: as) accepts cons (x′ : X ▪)
                                        (p : AlgListP A R x′ as)
                                        (rcons : R (cons − tag , a , x′) x)
```

A simple argument by induction shows that *AlgListP A R x as* is in fact isomorphic to $(cata\ (R))\ as\ x$ for any $as : \mathsf{List}\ A$ and $x : X\ ▪$. As a corollary, we have

$$AlgList\ A\ R\ x \ \cong\ \Sigma\langle\, as : \mathsf{List}\ A \,\rangle\ (cata\ (R))\ as\ x \tag{5.1}$$

for any $x : X\ ▪$ by (**??**). That is, an algebraic list is exactly a plain list and a proof that the list folds to x using the algebra R. The vector datatype is a special case of *AlgList* — to see that, define

```
length − alg : 𝔽 (ListD A) (const Nat) ⇉ const Nat
length − alg (nil − tag   , ▪)      = zero
length − alg (cons − tag , a , n)   = suc n
```

and take $R = fun\ length - alg$. From (5.1) we have the isomorphisms

$$\mathsf{Vec}\ A\ n \ \cong\ \Sigma\langle\, as : \mathsf{List}\ A \,\rangle\ (cata\ (fun\ length - alg))\ as\ n$$

for all $n : \mathsf{Nat}$, from which we can derive

$$\mathsf{Vec}\ A\ n \ \cong\ \Sigma\langle\, as : \mathsf{List}\ A \,\rangle\ length\ as \ \equiv\ n$$

by $fun - preserves - fold$, after defining $length \;=\; fold\ length - alg$.

The above can be generalised to all datatypes encoded by the Desc universe. Let $D \;:\; \mathsf{Desc}\ I$ be a description and $R \;:\; \mathbb{F}\ D\ X \rightsquigarrow X$ (where $X \;:\; I \to \mathsf{Set}$) an algebra. The (relational) *algebraic ornamentation* of D with R is an ornamental description

$$algOrn\ D\ R \;:\; \mathsf{OrnDesc}\ (\Sigma\ I\ X)\ \mathsf{outl}\ D$$

(where $\mathsf{outl} \;:\; \Sigma\ I\ X \to I$). Its definition is a slight generalisation of the one given by Dagand and McBride [Dagand and McBride, 2012b, supplementary code]. The promotion predicate for the ornament $\lceil\, algOrn\ D\ R \,\rceil$ is pointwise isomorphic to $(cata\ (R))$, i.e.,

$$PromP\ \lceil\, algOrn\ D\ R \,\rceil\ (\mathsf{ok}\ (i\ ,\ x))\ d \;\cong\; (cata\ (R))\ d\ x \tag{5.2}$$

for all $i \;:\; I$, $x \;:\; X\ i$, and $d \;:\; \mu\ D\ i$. As a corollary, we have the following isomorphisms

$$\mu\ \lfloor\, algOrn\ D\ R \,\rfloor\ (i\ ,\ x) \;\cong\; \Sigma \langle\, d : \mu\ D\ i \,\rangle\ \ (cata\ (R))\ d\ x \tag{5.3}$$

for all $i \;:\; I$ and $x \;:\; X\ i$ by (**??**). For example, taking $D \;=\; ListD\ A$, the type $AlgList\ A\ R\ x$ can be thought of as the high-level presentation of $\mu\ \lfloor\, algOrn\ (ListD\ A)\ R \,\rfloor\ (\blacksquare\ ,\ x)$. Algebraic ornamentation is a very convenient method for adding new indices to inductive families, and most importantly, it says precisely what the new indices mean. The method was demonstrated by McBride [2011] with a correct-by-construction compiler for a small language, and will be demonstrated again in Section 5.3.

Example: the Fold Fusion Theorem. As a first example of bridging internalist programming with relational program derivation through algebraic ornamentation, let us consider the *Fold Fusion Theorem* [Bird and de Moor, 1997, Section 6.2]: Let $D \;:\; \mathsf{Desc}\ I$ be a description, $R \;:\; X \rightsquigarrow Y$ a relation, and $S \;:\; \mathbb{F}\ D\ X \rightsquigarrow X$ and $T \;:\; \mathbb{F}\ D\ Y \rightsquigarrow Y$ be algebras. If R is a homomorphism from S to T, i.e.,

$$R \;\cdot\; S \simeq T \;\cdot\; \mathbb{R}\ D\ R$$

which is referred to as the *fusion condition*, then we have

$$R \cdot (cata\ (S)) \simeq (cata\ (T))$$

The above is, in fact, a corollary of two variations of Fold Fusion that replace relational equivalence in the statement of the theorem with relational inclusion. One of the variations is

$$R \cdot S \subseteq T \cdot \mathbb{R}\ D\ R \quad \text{implies} \quad R \cdot (cata\ (S)) \subseteq (cata\ (T))$$

This can be used with (5.3) to derive a conversion between algebraically ornamented datatypes:

$$\begin{aligned}
&algOrn - fusion - \subseteq D\ R\ S\ T\ : \\
&\quad R \cdot S \subseteq T \cdot \mathbb{R}\ D\ R \rightarrow \\
&\quad \{i : I\}\ (x : X\ i) \rightarrow \mu\ \lfloor algOrn\ D\ S \rfloor\ (i\ ,\ x) \rightarrow \\
&\quad (y : Y\ i) \rightarrow R\ x\ y \rightarrow \mu\ \lfloor algOrn\ D\ T \rfloor\ (i\ ,\ y)
\end{aligned}$$

The other variation of Fold Fusion simply reverses the direction of inclusion:

$$R \cdot S \supseteq T \cdot \mathbb{R}\ D\ R \quad \text{implies} \quad R \cdot (cata\ (S)) \supseteq (cata\ (T))$$

which translates to the conversion

$$\begin{aligned}
&algOrn - fusion - \supseteq D\ R\ S\ T\ : \\
&\quad R \cdot S \supseteq T \cdot \mathbb{R}\ D\ R \rightarrow \\
&\quad \{i : I\}\ (y : Y\ i) \rightarrow \mu\ \lfloor algOrn\ D\ T \rfloor\ (i\ ,\ y) \rightarrow \\
&\quad \Sigma \langle\, x : X\ i \,\rangle\ \ \mu\ \lfloor algOrn\ D\ S \rfloor\ (i\ ,\ x) \times R\ x\ y
\end{aligned}$$

For a simple example, suppose that we need a "bounded" vector datatype, i.e., lists indexed with an upper bound on their length. A quick thought might lead to this definition

$$\begin{aligned}
&BVec\ :\ \mathsf{Set} \rightarrow \mathsf{Nat} \rightarrow \mathsf{Set} \\
&BVec\ A\ m\ = \\
&\quad \mu\ \lfloor algOrn\ (ListD\ A)\ (geq\ \cdot\ fun\ length - alg) \rfloor\ (\blacksquare\ ,\ m)
\end{aligned}$$

where $geq\ =\ \lambda\ x\ y \rightarrow x \leqslant y\ :\ const\ \mathsf{Nat} \rightsquigarrow const\ \mathsf{Nat}$ maps a natural number x to any natural number that is at least x. The isomorphisms (5.3) specialise for *BVec* to

$$BVec\ A\ m\ \cong\ \Sigma \langle\, as : \mathsf{List}\ A \,\rangle\ \ (cata\ (geq\ \cdot\ fun\ length - alg))\ as\ m$$

But is *BVec* really the bounded vectors? Indeed it is, because we can deduce

$$geq \cdot (cata\ (fun\ length - alg)) \simeq (cata\ (geq \cdot fun\ length - alg))$$

by Fold Fusion (where $(cata\ (fun\ length - alg))$ is equivalent to $fun\ length$ by $fun - preserves - fold$). The fusion condition is

$$geq \cdot fun\ length - alg \simeq geq \cdot fun\ length - alg \cdot \mathbb{R}\ (ListD\ A)\ geq$$

The left-to-right inclusion is easily calculated as follows:

$$\begin{aligned}
& geq \cdot fun\ length - alg \\
\subseteq\ & \quad \{\ idR \text{ identity}\ \} \\
& geq \cdot fun\ length - alg \cdot idR \\
\subseteq\ & \quad \{\ \text{relator preserves identity}\ \} \\
& geq \cdot fun\ length - alg \cdot \mathbb{R}\ (ListD\ A)\ idR \\
\subseteq\ & \quad \{\ geq \text{ reflexive}\ \} \\
& geq \cdot fun\ length - alg \cdot \mathbb{R}\ (ListD\ A)\ geq
\end{aligned}$$

And from right to left:

$$\begin{aligned}
& geq \cdot fun\ length - alg \cdot \mathbb{R}\ (ListD\ A)\ geq \\
\subseteq\ & \quad \{fun\ length - alg \text{ monotonic on } geq\ \} \\
& geq \cdot geq \cdot fun\ length - alg \\
\subseteq\ & \quad \{\ geq \text{ transitive}\ \} \\
& geq \cdot fun\ length - alg
\end{aligned}$$

Note that these calculations are good illustrations of the power of relational calculation despite their simplicity — they are straightforward symbolic manipulations, hiding details like quantifier reasoning behind the scenes. As demonstrated by the AoPA library Mu et al. [2009], they can be faithfully formalised with preorder reasoning combinators in Agda and used to discharge the fusion conditions of $algOrn - fusion - \subseteq$ and $algOrn - fusion - \supseteq$. Hence we get two conversions, one of type

$$\mathsf{Vec}\ A\ n \to (n \leqslant m) \to BVec\ A\ m$$

which relaxes a vector of length n to a bounded vector whose length is bounded above by some m that is at least n, and the other of type

$$BVec\ A\ m \to \Sigma \langle\, n : \mathsf{Nat} \,\rangle\ \mathsf{Vec}\ A\ n \times (n \leqslant m)$$

which converts a bounded vector whose length is at most m to a vector of length precisely n and guarantees that n is at most m.

Theoretically, the conversions derived from Fold Fusion are actually more generally applicable than they seem, because *every ornament is an algebraic ornament up to isomorphism*. This we show next.

5.2 Completeness of relational algebraic ornaments

Consider the *AlgList* datatype in Section 5.1 again. The way it is refined relative to the plain list datatype looks canonical, in the sense that any variation of the list datatype can be programmed as a special case of *AlgList*: we can choose whatever index set we want by setting the carrier of the algebra R; and by carefully programming R, we can insert fields into the list datatype that add more information or put restriction on fields and indices. For example, if we want some new information in the nil case, we can program R such that $R\ (nil - tag\ ,\ \blacksquare)\ x$ contains a field requesting that information; if, in the cons case, we need the targeted index x, the head element a, and the index x' of the recursive position to be related in some way, we can program R such that $R\ (cons - tag\ ,\ a\ ,\ x')\ x$ expresses that relationship.

The above observation leads to the following general theorem: Let O : $\mathsf{Orn}\ e\ D\ E$ be an ornament from D : $\mathsf{Desc}\ I$ to E : $\mathsf{Desc}\ J$. There is a *classifying algebra* for O

$$clsAlg\ O\ :\ \mathbb{IF}\ D\ (\mathsf{InvImage}\ e) \rightsquigarrow \mathsf{InvImage}\ e$$

such that there are isomorphisms

$$\mu\ \lfloor algOrn\ D\ (clsAlg\ O) \rfloor\ (e\ j\ ,\ \mathsf{ok}\ j)\ \cong\ \mu\ E\ j$$

for all j : J. That is, the algebraic ornamentation of D using the classifying algebra derived from O produces a datatype isomorphic to $\mu\ E$, so intuitively the algebraic ornament has the same content as O. We may interpret this

theorem as saying that algebraic ornaments are "complete" for the ornament language: any relationship between datatypes that can be described by an ornament can be described up to isomorphism by an algebraic ornament.

The completeness theorem brings up a nice algebraic intuition about inductive families. Consider the ornament from lists to vectors, for example. This ornament specifies that the type List A is refined by the collection of types Vec A n for all n : Nat. A list, say $a :: b :: [] :$ List A, can be reconstructed as a vector by starting in the type Vec A zero as $[]$, jumping to the next type Vec A (suc zero) as $b :: []$, and finally landing in Vec A (suc (suc zero)) as $a :: b :: []$. The list is thus *classified* as having length 2, as computed by the fold function *length*, and the resulting vector is a fused representation of the list and the classification proof. In the case of vectors, this classification is total and deterministic: every list is classified under one and only one index. But in general, classifications can be partial and nondeterministic. For example, promoting a list to an ordered list is classifying the list under an index that is a lower bound of the list. The classification process checks at each jump whether the list is still ordered; this check can fail, so an unordered list would "disappear" midway through the classification. Also there can be more than one lower bound for an ordered list, so the list can end up being classified under any one of them. Algebraic ornamentation in its original functional form can only capture part of this intuition about classification, namely those classifications that are total and deterministic. By generalising algebraic ornamentation to accept relational algebras, bringing in partiality and nondeterminacy, this idea about classification is captured in its entirety — a classification is just a relational fold computing the index that classifies an element. All ornaments specify classifications, and thus can be transformed into algebraic ornaments.

For more examples, let us first look at the classifying algebra for the ornament from natural numbers to lists. The base functor for natural numbers is

$$\mathbb{F}\ \mathit{NatD} : (\top \to \mathsf{Set}) \to (\top \to \mathsf{Set})$$
$$\mathbb{F}\ \mathit{NatD}\ X\ \_ = \Sigma\ \mathit{LTag}\ (\lambda\ \{\mathit{nil} - \mathit{tag} \to \top; \mathit{cons} - \mathit{tag} \to X\ \blacksquare\})$$

And the classifying algebra for the ornament *NatD-ListD A* is essentially

$$\begin{aligned}
&\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A)\ :\ \mathbb{F}\ \mathit{NatD}\ (\mathsf{InvImage}\ !) \rightsquigarrow \mathsf{InvImage}\ !\\
&\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathit{nil} - \mathit{tag}\quad , _\quad)\ (\mathsf{ok}\ \blacksquare)\ =\ \top\\
&\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathit{cons} - \mathit{tag}\ , \mathsf{ok}\ t)\ (\mathsf{ok}\ \blacksquare)\ =\ A \times (t \equiv \blacksquare)
\end{aligned}$$

The result of folding a natural number n with this algebra is uninteresting, as it can only be $\mathsf{ok}\ \blacksquare$. The fold, however, requires an element of A for each successor node it encounters, so a proof that n goes through the fold consists of n elements of A. Another example is the ornament $OL\ =\ \lceil\, \mathit{OrdListOD}\ A\ \_\leqslant_A\_\,\rceil$ from lists to ordered lists, whose classifying algebra is essentially

$$\begin{aligned}
&\mathit{clsAlg}\ \mathit{OL}\ :\ \mathbb{F}\ (\mathit{ListD}\ A)\ (\mathsf{InvImage}\ !) \rightsquigarrow \mathsf{InvImage}\ !\\
&\mathit{clsAlg}\ \mathit{OL}\ (\mathit{nil} - \mathit{tag}\quad , _\qquad)\ (\mathsf{ok}\ b)\ =\ \top\\
&\mathit{clsAlg}\ \mathit{OL}\ (\mathit{cons} - \mathit{tag}\ , a\ , \mathsf{ok}\ b')\ (\mathsf{ok}\ b)\ =\ (b \leqslant_A a) \times (b' \equiv a)
\end{aligned}$$

In the nil case, the empty list can be mapped to any $\mathsf{ok}\ b$ because any $b\ :\ A$ is a lower bound of the empty list; in the cons case, where $a\ :\ A$ is the head and $\mathsf{ok}\ b'$ is a result of classifying the tail, i.e., $b'\ :\ A$ is a lower bound of the tail, the list can be mapped to $\mathsf{ok}\ b$ if $b\ :\ A$ is a lower bound of a and a is exactly b'.

Perhaps the most important consequence of the completeness theorem (in its present form) is that it provides a new perspective on the expressive power of ornaments and inductive families. We showed in a previous paper Ko and Gibbons [2013] that every ornament induces a promotion predicate and a corresponding family of isomorphisms (which is restated as (**??**) in **??**). But one question was untouched: can we determine (independently from ornaments) the range of predicates induced by ornaments? An answer to this question would tell us something about the expressive power of ornaments, and also about the expressive power of inductive families in general, since the inductive families we use are usually ornamentations of simpler algebraic datatypes from traditional functional programming. The completeness theorem offers such an answer: ornament-induced promotion predicates are exactly those expressible as relational folds (up to pointwise isomorphism). In other words, a predicate can be baked into a datatype by ornamentation if and only if it can be thought of as a nondeterministic classification of the elements of the datatype with a

relational fold. This is more a guideline than a precise criterion, though, as the closest work about characterisation of the expressive power of folds discusses only functional folds Gibbons et al. [2001] (however, we believe that those results generalise to relations too). But this does encourage us to think about ornamentation computationally and to design new datatypes with relational algebraic methods. We illustrate this point with a solution to the *minimum coin change problem* in the next section.

5.3 Example: the minimum coin change problem

Suppose that we have an unlimited number of 1-penny, 2-pence, and 5-pence coins, modelled by the following datatype:

$$\begin{array}{l}\textbf{data}\ \mathit{Coin}\ :\ \mathsf{Set}\ \textbf{where}\\ \quad \mathit{onep}\ \mathit{twop}\ \mathit{fivep}\ :\ \mathit{Coin}\end{array}$$

Given $n\ :\ \mathsf{Nat}$, the *minimum coin change problem* asks for the least number of coins that make up n pence. We can give a relational specification of the problem with the following operator:

$$\begin{array}{l}\mathit{min}\_ \cdot \Lambda\_\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ :\ I \to \mathsf{Set}\}\\ \qquad\qquad (R\ :\ Y \rightsquigarrow Y)\ (S\ :\ X \rightsquigarrow Y) \to (X \rightsquigarrow Y)\\ (\mathit{min}\ R\ \cdot \Lambda\ S)\ x\ y\ =\ S\ x\ y\ \times\ (\forall\ y' \to S\ x\ y' \to R\ y'\ y)\end{array}$$

An input $x\ :\ X\ i$ for some $i\ :\ I$ is mapped by $\mathit{min}\ R\ \cdot \Lambda\ S$ to $y\ :\ Y\ i$ if y is a possible result in $S\ x\ :\ (\mathit{Power}\ (Y\ i))$ and is the smallest such result under R, in the sense that any y' in $S\ x\ :\ (\mathit{Power}\ (Y\ i))$ must satisfy $R\ y'\ y$ (i.e., R maps larger inputs to smaller outputs). Intuitively, we can think of $\mathit{min}\ R\ \cdot \Lambda\ S$ as consisting of two steps: the first step $\Lambda\ S$ computes the set of all possible results yielded by S, and the second step $\mathit{min}\ R$ chooses a minimum result from that set (nondeterministically). We use bags of coins as the type of solutions, and represent them as decreasingly ordered lists indexed with an upper bound. (This is a deliberate choice to make the derivation work, but one would naturally turn to this representation having attempted to apply the

Greedy Theorem, which will be introduced shortly.) If we define the ordering on coins as

$$\begin{aligned}&\_ \leqslant C\_ \;:\; \mathit{Coin} \to \mathit{Coin} \to \mathsf{Set}\\ &c \leqslant C\; d \;=\; \mathit{value}\; c \leqslant \mathit{value}\; d\end{aligned}$$

where the values of the coins are defined by

$$\begin{aligned}&\mathit{value} \;:\; \mathit{Coin} \to \mathsf{Nat}\\ &\mathit{value}\; \mathit{onep} \;=\; 1\\ &\mathit{value}\; \mathit{twop} \;=\; 2\\ &\mathit{value}\; \mathit{fivep} \;=\; 5\end{aligned}$$

then the datatype of coin bags we use is

$$\begin{aligned}&\textbf{indexfirst data}\; \mathit{CoinBag} \;:\; \mathit{Coin} \to \mathsf{Set}\; \textbf{where}\\ &\quad \mathit{CoinBag}\; c\; \mathit{accepts}\; \mathsf{nil}\\ &\qquad\qquad \mid \quad \mathsf{cons}\; (d \;:\; \mathit{Coin})\; (\mathit{leq} \;:\; d \leqslant C\; c)\; (b \;:\; \mathit{CoinBag}\; d)\end{aligned}$$

Down at the universe level, the (ornamental) description of *CoinBag* (relative to List *Coin*) is simply that of OrdList *Coin* (*flip* $\_ \leqslant C\_$).

$$\begin{aligned}&\mathit{CoinBagOD} \;:\; \mathsf{OrnDesc}\; \mathit{Coin}\; !\; (\mathit{ListD}\; \mathit{Coin})\\ &\mathit{CoinBagOD} \;=\; \mathit{OrdListOD}\; \mathit{Coin}\; (\mathit{flip}\; \_ \leqslant C\_)\end{aligned}$$

$$\begin{aligned}&\mathit{CoinBagD} \;:\; \mathsf{Desc}\; \mathit{Coin}\\ &\mathit{CoinBagD} \;=\; \lfloor \mathit{CoinBagOD} \rfloor\end{aligned}$$

$$\begin{aligned}&\mathit{CoinBag} \;:\; \mathit{Coin} \to \mathsf{Set}\\ &\mathit{CoinBag} \;=\; \mu\; \mathit{CoinBagD}\end{aligned}$$

The base functor for *CoinBag* is

$$\begin{aligned}&\mathbb{F}\; \mathit{CoinBagD} \;:\; (\mathit{Coin} \to \mathsf{Set}) \to (\mathit{Coin} \to \mathsf{Set})\\ &\mathbb{F}\; \mathit{CoinBagD}\; X\; c \;=\\ &\quad \Sigma\; \mathit{LTag}\; (\lambda\; \{\mathit{nil} - \mathit{tag} \to \top; \mathit{cons} - \mathit{tag} \to \Sigma \langle\, d : \mathit{Coin} \,\rangle\; (d \leqslant C\; c) \times X\; d\})\end{aligned}$$

The total value of a coin bag is the sum of the values of the coins in the bag, which is computed by a (functional) fold:

$$\begin{aligned}&\mathit{total} - \mathit{value} - \mathit{alg} \;:\; \mathbb{F}\; \mathit{CoinBagD}\; (\mathit{const}\; \mathsf{Nat}) \rightrightarrows \mathit{const}\; \mathsf{Nat}\\ &\mathit{total} - \mathit{value} - \mathit{alg}\; (\mathit{nil} - \mathit{tag} \quad , \_ \qquad) \;=\; 0\end{aligned}$$

$$total - value - alg\ (cons - tag\ ,\ d\ ,\ \_\ ,\ n)\ =\ value\ d + n$$
$$total - value\ :\ CoinBag \rightrightarrows const\ \mathsf{Nat}$$
$$total - value\ =\ fold\ total - value - alg$$

and the number of coins in a coin bag is also computed by a fold:

$$size - alg\ :\ \mathbb{F}\ CoinBagD\ (const\ \mathsf{Nat}) \rightrightarrows const\ \mathsf{Nat}$$
$$size - alg\ (nil - tag\ \ ,\ \_\ \ \ \ \ \ )\ =\ 0$$
$$size - alg\ (cons - tag\ ,\ \_\ ,\ \_\ ,\ n)\ =\ 1 + n$$
$$size\ :\ CoinBag \rightrightarrows const\ \mathsf{Nat}$$
$$size\ =\ fold\ size - alg$$

The specification of the minimum coin change problem can now be written as

$$min - coin - change\ :\ const\ \mathsf{Nat} \rightsquigarrow CoinBag$$
$$min - coin - change\ =$$
$$\quad min\ (fun\ size^{\circ}\ \cdot\ leq\ \cdot\ fun\ size)\ \cdot \Lambda\ (fun\ total - value^{\circ})$$

where $leq\ =\ geq^{\circ}\ :\ const\ \mathsf{Nat} \rightsquigarrow const\ \mathsf{Nat}$ maps a natural number n to any natural number that is at most n. Intuitively, given an input $n\ :\ \mathsf{Nat}$, the relation $fun\ \ total - value^{\circ}$ computes an arbitrary coin bag whose total value is n, so $min - coin - change$ first computes the set of all such coin bags and then chooses from the set a coin bag whose size is smallest. Our goal, then, is to write a functional program $f\ :\ const\ \mathsf{Nat} \rightrightarrows CoinBag$ such that $fun\ f \subseteq min - coin - change$, and then $f\ fivep\ :\ \mathsf{Nat} \to CoinBag\ fivep$ would be a solution — note that the type $CoinBag\ fivep$ contains all coin bags, since $fivep$ is the largest denomination and hence a trivial upper bound on the content of bags. Of course, we may guess what f should look like, but its correctness proof is much harder. Can we construct the program and its correctness proof in a more manageable way?

The plan. In traditional relational program derivation, we would attempt to refine $min - coin - change$ to some simpler relational program and then to an executable functional program by applying algebraic laws and theorems. With algebraic ornamentation, however, there is a new possibility: if we can

derive that, for some algebra $R : \mathbb{F}\ \mathit{CoinBagD}\ (\mathit{const}\ \mathsf{Nat}) \rightsquigarrow \mathit{const}\ \mathsf{Nat}$,

$$(\mathit{cata}\ (R))^{\circ} \subseteq \mathit{min} - \mathit{coin} - \mathit{change} \tag{5.4}$$

then we can manufacture a new datatype

$$\begin{aligned}
&\mathit{GreedySolutionOD} : \mathsf{OrnDesc}\ (\mathit{Coin} \times \mathsf{Nat})\ \mathsf{outl}\ \mathit{CoinBagD} \\
&\mathit{GreedySolutionOD} = \mathit{algOrn}\ \mathit{CoinBagD}\ R \\
&\mathit{GreedySolution} : \mathit{Coin} \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathit{GreedySolution}\ c\ n = \mu\ \lfloor \mathit{GreedySolutionOD} \rfloor\ (c\ ,\ n)
\end{aligned}$$

and construct a function of type

$$\mathit{greedy} : (c : \mathit{Coin})\ (n : \mathsf{Nat}) \to \mathit{GreedySolution}\ c\ n$$

from which we can assemble a solution

$$\begin{aligned}
&\mathit{sol} : \mathsf{Nat} \to \mathit{CoinBag}\ \mathit{fivep} \\
&\mathit{sol} = \mathit{forget}\ \lceil \mathit{GreedySolutionOD} \rceil \circ \mathit{greedy}\ \mathit{fivep}
\end{aligned}$$

The program *sol* satisfies the specification because of the following argument: For any $c : \mathit{Coin}$ and $n : \mathsf{Nat}$, by (5.3) we have

$$\mathit{GreedySolution}\ c\ n \cong \Sigma \langle\, b : \mathit{CoinBag}\ c \,\rangle\ (\mathit{cata}\ (R))\ b\ n$$

In particular, since the first half of the left-to-right direction of the isomorphism is $\mathit{forget}\ \lceil \mathit{GreedySolutionOD} \rceil$, we have

$$(\mathit{cata}\ (R))\ (\mathit{forget}\ \lceil \mathit{GreedySolutionOD} \rceil\ g)\ n$$

for any $g : \mathit{GreedySolution}\ c\ n$. Substituting g by $\mathit{greedy}\ \mathit{fivep}\ n$, we get

$$(\mathit{cata}\ (R))\ (\mathit{sol}\ n)\ n$$

which implies, by (5.4),

$$\mathit{min} - \mathit{coin} - \mathit{change}\ n\ (\mathit{sol}\ n)$$

i.e., *sol* satisfies the specification. Thus all we need to do to solve the minimum coin change problem is (i) refine the specification $\mathit{min} - \mathit{coin} - \mathit{change}$ to the converse of a fold, i.e., find the algebra R in (5.4), and (ii) construct the internalist program *greedy*.

Refining the specification. The key to refining $min - coin - change$ to the converse of a fold lies in the following version of the *Greedy Theorem*, which is essentially a specialisation of Bird and de Moor's Theorem 10.1 Bird and de Moor [1997]: Let $D : \mathsf{Desc}\ I$ be a description, $R : \mu\ D \rightsquigarrow \mu\ D$ a preorder, and $S : \mathbb{F}\ D\ X \rightsquigarrow X$ an algebra. Consider the specification

$$min\ R\ \cdot \Lambda\ ((cata\ (S))^{\circ})$$

That is, given an input value $x : X\ i$ for some $i : I$, we choose a minimum under R among all those elements of $\mu\ D\ i$ that computes to x through $(cata\ (S))$. The Greedy Theorem states that, if the initial algebra $\alpha = fun\ \mathsf{con} : \mathbb{F}\ D\ (\mu\ D) \rightsquigarrow \mu\ D$ is monotonic on R (where $\mathsf{con} : \mathbb{F}\ D\ (\mu\ D) \rightrightarrows \mu\ D$ is the datatype-generic constructor), i.e.,

$$\alpha\ \cdot\ \mathbb{R}\ D\ R \subseteq R\ \cdot\ \alpha$$

and there is a relation (ordering) $Q : \mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ X$ such that the *greedy condition*

$$\alpha\ \cdot\ \mathbb{R}\ D\ ((cata\ (S))^{\circ})\ \cdot\ (Q \cap (S^{\circ}\ \cdot\ S))^{\circ} \subseteq R^{\circ}\ \cdot\ \alpha\ \cdot\ \mathbb{R}\ D\ ((cata\ (S))^{\circ})$$

is satisfied, then we have

$$(cata\ ((min\ Q\ \cdot \Lambda\ (S^{\circ}))^{\circ}))^{\circ} \subseteq min\ R\ \cdot \Lambda\ ((cata\ (S))^{\circ})$$

Here we offer an intuitive explanation of the Greedy Theorem, but the theorem admits an elegant calculational proof, which can be faithfully reprised in Agda. The monotonicity condition states that if $ds : \mathbb{F}\ D\ (\mu\ D)\ i$ for some $i : I$ is better than $ds' : \mathbb{F}\ D\ (\mu\ D)\ i$ under $\mathbb{R}\ D\ R$, i.e., ds and ds' are equal except that the recursive positions of ds are all better than the corresponding recursive positions of ds' under R, then $\mathsf{con}\ ds : \mu\ D\ i$ would be better than $\mathsf{con}\ ds' : \mu\ D\ i$ under R. This implies that, when solving the optimisation problem, better solutions to subproblems would lead to a better solution to the original problem, so the *principle of optimality* applies, i.e., to reach an optimal solution it suffices to find optimal solutions to subproblems, and we are entitled to use the converse of a fold to find optimal solutions recursively. The greedy condition further states that there is an ordering Q on the ways of decomposing the problem which has significant influence on the quality of solutions: Suppose

```
data CoinBag′View : {c : Coin} {n : Nat} {l : Nat} → CoinBag′ c n l → Set where
  empty     : {c : Coin} → CoinBag′View {c} {0} {0} bnil′
  oneponep  : {m l : Nat} {lep : onep ⩽ C onep} (b : CoinBag′ onep m l) → CoinBag′View {one
  oneptwop  : {m l : Nat} {lep : onep ⩽ C twop} (b : CoinBag′ onep m l) → CoinBag′View {two
  twoptwop  : {m l : Nat} {lep : twop ⩽ C twop} (b : CoinBag′ twop m l) → CoinBag′View {two
  onepfivep : {m l : Nat} {lep : onep ⩽ C fivep} (b : CoinBag′ onep m l) → CoinBag′View {five
  twopfivep : {m l : Nat} {lep : twop ⩽ C fivep} (b : CoinBag′ twop m l) → CoinBag′View {five
  fivepfivep : {m l : Nat} {lep : fivep ⩽ C fivep} (b : CoinBag′ fivep m l) → CoinBag′View {five
```

Figure 5.1 The view datatype on *CoinBag′*.

that there are two decompositions xs and $xs' : \mathbb{F}\ D\ X\ i$ of some problem $x : X\ i$ for some $i : I$, i.e., both xs and xs' are in $S\ ^{\circ}\ x : (Power\ (\mathbb{F}\ D\ X\ i))$, and assume that xs is better than xs' under Q. Then for any solution resulting from xs' (computed by $\alpha \cdot \mathbb{R}\ D\ ((cata\ (S))\ ^{\circ})$) there always exists a better solution resulting from xs, so ignoring xs' would only rule out worse solutions. The greedy condition thus guarantees that we will arrive at an optimal solution by always choosing the best decomposition, which is done by $min\ Q \cdot \Lambda\ (S\ ^{\circ}) : X \rightsquigarrow \mathbb{F}\ D\ X$.

Back to the coin changing problem: By $fun - preserves - fold$, the specification $min - coin - change$ is equivalent to

$$min\ (fun\ size\ ^{\circ} \cdot leq \cdot fun\ size) \cdot \Lambda\ ((cata\ (fun\ total - value - alg))\ ^{\circ})$$

which matches the form of the generic specification given in the Greedy Theorem, so we try to discharge the two conditions of the theorem. The monotonicity condition reduces to monotonicity of $fun\ size - alg$ on leq, and can be easily proved either by relational calculation or pointwise reasoning. As for the greedy condition, an obvious choice for Q is an ordering that leads us to choose the largest possible denomination, so we go for

```
Q : 𝔽 CoinBagD (const Nat) ⇝ 𝔽 CoinBagD (const Nat)
Q (nil − tag   , _    ) = return (nil − tag , ▪)
Q (cons − tag , d , _) =
```

$$greedy - lemma : (c\ d : Coin) \to c \leqslant C\ d \to (m\ n : \mathsf{Nat}) \to value\ c + m \equiv value\ d + n \to (l : \mathsf{Nat})\ (b : CoinBag'\ c\ m\ l) \to \Sigma\langle\, l' : \mathsf{Nat} \,\rangle\ CoinBag'\ d\ n\ l' \times (l' \leqslant l)$$

$$\begin{array}{llllllll}
greedy - lemma & c & d & c \leqslant d & m & n & eq\ l & b\ \mathbf{with}\ view - ordered - coin\ c\ d\ c \\
greedy - lemma & .onep & .onep & _ & .n & n & \mathsf{refl}\ l & b\ (vartype\ (CoinBag'\ onep\ n\ l)) \mid \\
greedy - lemma & .onep & .twop & _ & .(1+n) & n & \mathsf{refl}\ l & b \mid oneptwop\ \mathbf{with}\ view - CoinB \\
greedy - lemma & .onep & .twop & _ & .(1+n) & n & \mathsf{refl}\ .(1+l'')\ _ & \mid oneptwop \mid oneponep\ \{.n\}\ \{l \\
 & & & & & & & \\
greedy - lemma & .onep & .fivep & _ & .(4+n) & n & \mathsf{refl}\ l & b \mid onepfivep\ \mathbf{with}\ view - CoinB \\
greedy - lemma & .onep & .fivep & _ & .(4+n) & n & \mathsf{refl}\ ._ & ._ \mid onepfivep \mid oneponep\ \ b\ \mathbf{wit} \\
greedy - lemma & .onep & .fivep & _ & .(4+n) & n & \mathsf{refl}\ ._ & ._ \mid onepfivep \mid oneponep\ ._ \mid o \\
greedy - lemma & .onep & .fivep & _ & .(4+n) & n & \mathsf{refl}\ ._ & ._ \mid onepfivep \mid oneponep\ ._ \mid o \\
greedy - lemma & .onep & .fivep & _ & .(4+n) & n & \mathsf{refl}\ .(4+l'')\ _ & \mid onepfivep \mid oneponep\ ._ \mid o \\
 & & & & & & & \\
greedy - lemma & .twop & .twop & _ & .n & n & \mathsf{refl}\ l & b\ (vartype\ (CoinBag'\ twop\ n\ l)) \mid \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ l & b \mid twopfivep\ \mathbf{with}\ view - CoinB \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ ._ & ._ \mid twopfivep \mid oneptwop\ \ b\ \mathbf{wit} \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ ._ & ._ \mid twopfivep \mid oneptwop\ ._ \mid o \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ .(3+l'')\ _ & \mid twopfivep \mid oneptwop\ ._ \mid o \\
 & & & & & & & \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ ._ & ._ \mid twopfivep \mid twoptwop\ \ b\ \mathbf{wit} \\
greedy - lemma & .twop & .fivep & _ & .(3+n) & n & \mathsf{refl}\ .(2+l'')\ _ & \mid twopfivep \mid twoptwop\ ._ \mid o \\
 & & & & & & & \\
greedy - lemma & .twop & .fivep & _ & .(4+k) & .(1+k) & \mathsf{refl}\ .(2+l'')\ _ & \mid twopfivep \mid twoptwop\ ._ \mid t \\
 & & & & & & & \\
greedy - lemma & .fivep & .fivep & _ & .n & n & \mathsf{refl}\ l & b\ (vartype\ (CoinBag'\ fivep\ n\ l)) \mid
\end{array}$$

Figure 5.2 Cases of *greedy − lemma*, generated semi-automatically by Agda's interactive case-split mechanism. Shown in the (shaded) interaction points are their goal types, and the types of some pattern variables are shown in subscript beside them.

$$(\_ \leqslant C \_\ d) \gg\!= \lambda\, e \to any \gg\!= \lambda\, r \to return\ (cons - tag\ ,\ e\ ,\ r)$$

where, in the cons case, the output is required to be also a cons node, and the coin at its head position must be one that is no smaller than the coin d at the head position of the input. It is non-trivial to prove the greedy condition by relational calculation. Here we offer instead a brute-force yet conveniently expressed case analysis by dependent pattern matching, which also serves as an example of algebraic ornamentation. Define a new datatype $CoinBag'$: $Coin \to \mathsf{Nat} \to \mathsf{Nat} \to \mathsf{Set}$ by composing two algebraic ornaments on $CoinBagD$ in parallel:

$$\begin{aligned}
&CoinBag'OD\ :\ \mathsf{OrnDesc}\ (\mathsf{outl} \bowtie \mathsf{outl})\ pull\ CoinBagD\\
&CoinBag'OD\ =\ \lceil algOrn\ CoinBagD\ (fun\ total - value - alg) \rceil \otimes\\
&\qquad\qquad \lceil algOrn\ CoinBagD\ (fun\ size - alg) \rceil\\
&CoinBag'\ :\ Coin \to \mathsf{Nat} \to \mathsf{Nat} \to \mathsf{Set}\\
&CoinBag'\ c\ n\ l\ =\ \mu\ \lfloor CoinBag'OD \rfloor\ (\mathsf{ok}\ (c\ ,\ n)\ ,\ \mathsf{ok}\ (c\ ,\ l))
\end{aligned}$$

By (**??**), (5.2), and $fun - preserves - fold$, $CoinBag'$ is characterised by the isomorphisms

$$\begin{aligned}
CoinBag'\ c\ n\ l\ \cong\ \Sigma \langle\, b : CoinBag\ c \,\rangle\\
(total - value\ b\ \equiv\ n)\ \times\ (size\ b\ \equiv\ l)
\end{aligned} \tag{5.5}$$

for all c : $Coin$, n : Nat, and l : Nat. Hence a coin bag of type $CoinBag'\ c\ n\ l$ contains l coins that are no larger than c and sum up to n pence. We can give the following types to the two constructors of $CoinBag'$:

$$\begin{aligned}
&bnil'\ \ :\ \forall\ \{c\} \to CoinBag'\ c\ 0\ 0\\
&bcons'\ :\ \forall\ \{c\ n\ l\} \to (d\ :\ Coin) \to d \leqslant C\ c \to\\
&\qquad CoinBag'\ d\ n\ l \to CoinBag'\ c\ (value\ d + n)\ (1 + l)
\end{aligned}$$

The greedy condition then essentially reduces to this lemma:

$$\begin{aligned}
&greedy - lemma\ :\\
&\quad (c\ d\ :\ Coin) \to c \leqslant C\ d \to\\
&\quad (m\ n\ :\ \mathsf{Nat}) \to value\ c + m\ \equiv\ value\ d + n \to\\
&\quad (l\ :\ \mathsf{Nat})\ (b\ :\ CoinBag'\ c\ m\ l) \to\\
&\quad \Sigma \langle\, l' : \mathsf{Nat} \,\rangle\ CoinBag'\ d\ n\ l'\ \times\ (l' \leqslant l)
\end{aligned}$$

That is, given a problem (i.e., a value to be represented by coins), if $c : Coin$ is a choice of decomposition (i.e., the first coin used) no better than $d : Coin$ (recall that we prefer larger denominations), and $b : CoinBag'\ c\ m\ l$ is a solution of size l to the remaining subproblem m resulting from choosing c, then there is a solution to the remaining subproblem n resulting from choosing d whose size l' is no greater than l. We define two *views* McBride and McKinna [2004] — or "customised pattern matching" — to aid the analysis. The first view analyses a proof of $c \leqslant C\ d$ and exhausts all possibilities of c and d,

```
data CoinOrderedView : Coin → Coin → Set where
  oneponep  : CoinOrderedView onep onep
  oneptwop  : CoinOrderedView onep twop
  onepfivep : CoinOrderedView onep fivep
  twoptwop  : CoinOrderedView twop twop
  twopfivep : CoinOrderedView twop fivep
  fivepfivep : CoinOrderedView fivep fivep

view-ordered-coin :
  (c d : Coin) → c ⩽ C d → CoinOrderedView c d
```

where the covering function $view - ordered - coin$ is written by standard pattern matching on c and d. The second view analyses some $b : CoinBag'\ c\ n\ l$ and exhausts all possibilities of c, n, l, and the first coin in b (if any). The view datatype $CoinBag'View$ is shown in Figure 5.1, and the covering function

```
view-CoinBag' :
  ∀ {c n l} (b : CoinBag' c n l) → CoinBag'View b
```

is again written by standard pattern matching. Given these two views, $greedy - lemma$ can be split into eight cases by first exhausting all possibilities of c and d with $view - ordered - coin$ and then analysing the content of b with $view - CoinBag'$. Figure 5.2 shows the case-split tree generated semi-automatically by Agda; the detail is explained as follows:

- At goal 0 (and, similarly, goals 3 and 7), the input bag is $b : CoinBag'\ onep\ n\ l$, and we should produce a $CoinBag'\ onep\ n\ l'$ for some $l' :$ Nat such that $l' \leqslant l$. This is easy because b itself is a suitable bag.

- At goal 1 (and, similarly, goals 2, 4, and 5), the input bag is of type $CoinBag'\ onep\ (1 + n)\ l$, i.e., the coins in the bag are no larger than *onep* and the total value is $1 + n$. The bag must contain *onep* as its first coin; let the rest of the bag be $b\ :\ CoinBag'\ onep\ n\ l''$. At this point Agda can deduce that l must be $1 + l''$. Now we can return b as the result after the upper bound on its coins is relaxed from *onep* to *twop*, which is done by

$$\begin{aligned} &relax\ :\ \forall\ \{c\ n\ l\}\ (b\ :\ CoinBag'\ c\ n\ l) \rightarrow \\ &\qquad \forall\ \{d\} \rightarrow c \leqslant C\ d \rightarrow CoinBag'\ d\ n\ l \end{aligned}$$

- The remaining goal 6 is the most interesting one: The input bag has type $CoinBag'\ twop\ (3 + n)\ l$, which in this case contains two 2-pence coins, and the rest of the bag is $b\ :\ CoinBag'\ twop\ k\ l''$. Agda deduces that n must be $1 + k$ and l must be $2 + l''$. We thus need to add a penny to b to increase its total value to $1 + k$, which is done by

$$\begin{aligned} &add - penny\ : \\ &\quad \forall\ \{c\ n\ l\} \rightarrow CoinBag'\ c\ n\ l \rightarrow CoinBag'\ c\ (1 + n)\ (1 + l) \end{aligned}$$

  and relax the bound of $add - penny\ b$ from *twop* to *fivep*.

Throughout the proof, Agda is able to keep track of the total value and the size of bags and make deductions, so the case analysis is done with little overhead. The greedy condition can then be discharged by pointwise reasoning, using (5.5) to interface with $greedy - lemma$. We conclude that the Greedy Theorem is applicable, and obtain

$$(cata\ ((min\ Q\ \cdot \Lambda\ (fun\ total - value - alg\ ^{\circ}))\ ^{\circ}))\ ^{\circ} \subseteq min - coin - change$$

We have thus found the algebra

$$R\ =\ (min\ Q\ \cdot \Lambda\ (fun\ total - value - alg\ ^{\circ}))\ ^{\circ}$$

which will help us to construct the final internalist program.

Constructing the internalist program. As planned, we synthesise a new datatype by ornamenting *CoinBag* using the algebra R:

$$
\begin{aligned}
&\mathit{GreedySolutionOD} \;:\; \mathsf{OrnDesc}\;(\mathit{Coin} \times \mathsf{Nat})\;\mathsf{outl}\;\mathit{CoinBagD}\\
&\mathit{GreedySolutionOD} \;=\; \mathit{algOrn}\;\mathit{CoinBagD}\;R
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{GreedySolution} \;:\; \mathit{Coin} \to \mathsf{Nat} \to \mathsf{Set}\\
&\mathit{GreedySolution}\;c\;n \;=\; \mu\;\lfloor \mathit{GreedySolutionOD} \rfloor\;(c\;,\;n)
\end{aligned}
$$

The two constructors of *GreedySolution* can be given the following types:

$$
\begin{aligned}
&\mathit{gnil} \;:\; \forall\;\{c\;n\} \to\\
&\qquad \mathit{total} - \mathit{value} - \mathit{alg}\;(\mathit{nil} - \mathit{tag}\;,\;\blacksquare) \;\equiv\; n \to\\
&\qquad (\forall\;\mathit{ns} \to \mathit{total} - \mathit{value} - \mathit{alg}\;\mathit{ns} \;\equiv\; n \to Q\;\mathit{ns}\;(\mathit{nil} - \mathit{tag}\;,\;\blacksquare)) \to\\
&\qquad \mathit{GreedySolution}\;c\;n
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{gcons} \;:\\
&\quad \forall\;\{c\;n\} \to (d \;:\; \mathit{Coin})\;(d \leqslant c \;:\; d \leqslant C\;c) \to\\
&\quad \forall\;\{n'\} \to \mathit{total} - \mathit{value} - \mathit{alg}\;(\mathit{cons} - \mathit{tag}\;,\;d\;,\;d \leqslant c\;,\;n') \;\equiv\; n \to\\
&\quad (\forall\;\mathit{ns} \to \mathit{total} - \mathit{value} - \mathit{alg}\;\mathit{ns} \;\equiv\; n \to Q\;\mathit{ns}\;(\mathit{cons} - \mathit{tag}\;,\;d\;,\;d \leqslant c\;,\;n')) \to\\
&\quad \mathit{GreedySolution}\;d\;n' \to \mathit{GreedySolution}\;c\;n
\end{aligned}
$$

Before we proceed to construct the internalist program

$$\mathit{greedy} \;:\; (c \;:\; \mathit{Coin})\;(n \;:\; \mathsf{Nat}) \to \mathit{GreedySolution}\;c\;n$$

let us first simplify the two constructors of *GreedySolution*. Each of the two constructors has two additional proof obligations coming from the algebra R: For *gnil*, since $\mathit{total} - \mathit{value} - \mathit{alg}\;(\mathit{nil} - \mathit{tag}, \blacksquare)$ reduces to 0, we may just specialise n to 0 and discharge the equality proof obligation. For the second proof obligation, ns is necessarily $(\mathit{nil} - \mathit{tag}\;,\;\blacksquare)$ if $\mathit{total} - \mathit{value} - \mathit{alg}\;\mathit{ns} \;\equiv\; 0$, and indeed Q maps $(\mathit{nil} - \mathit{tag}\;,\;\blacksquare)$ to $(\mathit{nil} - \mathit{tag}\;,\;\blacksquare)$, so the second proof obligation can be discharged as well. We thus obtain a simpler constructor defined using *gnil*:

$$\mathit{gnil}' \;:\; \forall\;\{c\} \to \mathit{GreedySolution}\;c\;0$$

For *gcons*, again since $\mathit{total} - \mathit{value} - \mathit{alg}\;(\mathit{cons} - \mathit{tag}\;,\;d\;,\;d \leqslant c\;,\;n')$ reduces to $\mathit{value}\;d\;+\;n'$, we may just specialise n to $\mathit{value}\;d\;+\;n'$ and discharge the equality proof obligation. For the second proof obligation, any ns that satisfies $\mathit{total} - \mathit{value} - \mathit{alg}\;\mathit{ns} \;\equiv\; \mathit{value}\;d + n'$ must be of the form $(\mathit{cons} - \mathit{tag}\;,\;e\;,\;e \leqslant c\;,\;m')$ for some $e \;:\; \mathit{Coin}$, $e \leqslant c \;:\; e \leqslant C\;c$, and $m' \;:\; \mathsf{Nat}$ since the right-hand side $\mathit{value}\;d + n'$ is nonzero, and Q maps ns to $(\mathit{cons} - \mathit{tag}\;,\;d\;,\;d \leqslant c\;,\;n')$ if $e \leqslant C\;d$,

so d should be the largest "usable" coin if this proof obligation is to be discharged. We say that $d : \mathit{Coin}$ is *usable* with respect to some $c : \mathit{Coin}$ and $n : \mathsf{Nat}$ if d is bounded above by c and can be part of a solution to the problem for n pence:

$$
\begin{aligned}
&\mathit{UsableCoin} : \mathsf{Nat} \to \mathit{Coin} \to \mathit{Coin} \to \mathsf{Set} \\
&\mathit{UsableCoin}\ n\ c\ d\ = \\
&\quad (d \leqslant C\ c) \times (\Sigma \langle\, n' : \mathsf{Nat} \,\rangle\ \mathit{value}\ d + n' \equiv n)
\end{aligned}
$$

Now we can define a new constructor using *gcons*:

$$
\begin{aligned}
&\mathit{gcons}' : \\
&\quad \forall\ \{c\} \to (d : \mathit{Coin}) \to d \leqslant C\ c \to \\
&\quad \forall\ \{n'\} \to \\
&\quad ((e : \mathit{Coin}) \to \mathit{UsableCoin}\ (\mathit{value}\ d + n')\ c\ e \to e \leqslant C\ d) \to \\
&\quad \mathit{GreedySolution}\ d\ n' \to \mathit{GreedySolution}\ c\ (\mathit{value}\ d + n')
\end{aligned}
$$

which requires that d is the largest usable coin with respect to c and $\mathit{value}\ d + n'$. We are thus directed to implement a function $\mathit{maximum-coin}$ that computes the largest usable coin with respect to any $c : \mathit{Coin}$ and nonzero $n : \mathsf{Nat}$,

$$
\begin{aligned}
&\mathit{maximum-coin} : \\
&\quad (c : \mathit{Coin})\ (n : \mathsf{Nat}) \to n > 0 \to \\
&\quad \Sigma \langle\, d : \mathit{Coin} \,\rangle\ \mathit{UsableCoin}\ n\ c\ d \times \\
&\qquad ((e : \mathit{Coin}) \to \mathit{UsableCoin}\ n\ c\ e \to e \leqslant C\ d)
\end{aligned}
$$

which takes some theorem proving but is overall a typical Agda exercise in dealing with natural numbers and ordering. Now we can implement the greedy algorithm as the internalist program

$$
\begin{aligned}
&\mathit{greedy} : (c : \mathit{Coin})\ (n : \mathsf{Nat}) \to \mathit{GreedySolution}\ c\ n \\
&\mathit{greedy}\ c\ n\ = <\!-\mathit{rec}\ P\ f\ n\ c \\
&\quad \textbf{where} \\
&\qquad P : \mathsf{Nat} \to \mathsf{Set} \\
&\qquad P\ n\ = (c : \mathit{Coin}) \to \mathit{GreedySolution}\ c\ n \\
&\qquad f : (n : \mathsf{Nat}) \to ((n' : \mathsf{Nat}) \to n' < n \to P\ n') \to P\ n \\
&\qquad f\ n \qquad\qquad \mathit{rec}\ c\ \textbf{with}\ \mathit{compare-with-zero}\ n \\
&\qquad f\ .0 \qquad\qquad \mathit{rec}\ c\ \mid\ \mathit{is-zero}\ =\ \mathit{gnil}'
\end{aligned}
$$

$$
\begin{array}{lll}
f\ \ n & rec\ c & \mid\ above - zero\ n > z \\
 & & \quad \textbf{with}\ maximum - coin\ c\ n\ n > z \\
f\ .(value\ d + n')\ rec\ c & & \mid\ above - zero\ n > z \\
 & & \mid\ d\ ,\ (d \leqslant c\ ,\ n'\ ,\ \mathsf{refl})\ ,\ guc\ = \\
 & & \quad gcons'\ d\ d \leqslant c\ guc\ (rec\ n'\ (hole\ ()\ (8))\ d)
\end{array}
$$

where the combinator

$$
\begin{array}{l}
< -rec\ :\ (P\ :\ \mathsf{Nat} \to \mathsf{Set}) \to \\
\qquad ((n\ :\ \mathsf{Nat}) \to ((n'\ :\ \mathsf{Nat}) \to n' < n \to P\ n') \to P\ n) \to \\
\qquad (n\ :\ \mathsf{Nat}) \to P\ n
\end{array}
$$

is for well-founded recursion on $\_ < \_$, and the function

$$compare - with - zero\ :\ (n\ :\ \mathsf{Nat}) \to ZeroView\ n$$

is a covering function for the view

$$
\begin{array}{l}
\textbf{data}\ ZeroView\ :\ \mathsf{Nat} \to \mathsf{Set}\ \textbf{where} \\
\quad is - zero\ \ \ \ \ \ :\ ZeroView\ 0 \\
\quad above - zero\ :\ \{n\ :\ \mathsf{Nat}\} \to n > 0 \to ZeroView\ n
\end{array}
$$

At goal 8, Agda deduces that n is $value\ d + n'$ and demands that we prove $n' < value\ d + n'$ in order to make the recursive call, which is easily discharged since $value\ d > 0$.

related work: Atkey et al. [2012]

Chapter 6

Categorical equivalence of ornaments and relational algebras

algebras corresponding to singleton ornaments and ornaments for optimised predicates; banana-split law corresponding to parallel composition; optimised predicates for functional algebraic ornaments amount to equality

Chapter 7

Conclusion

type computation — easy one like upgrades, swaps and more intricate one relying on universe construction

7.1 Future work

functor-level abstraction

Bibliography

Thorsten Altenkirch, James Chapman, and Tarmo Uustalu [2010]. Monads need not be endofunctors. In *Foundations of Software Science and Computational Structures*, volume 6014 of *Lecture Notes in Computer Science*, pages 297–311. Springer-Verlag. doi:10.1007/978-3-642-12032-9_21. ↰ page 88

Thorsten Altenkirch and Conor McBride [2003]. Generic programming within dependently typed programming. In *IFIP TC2/WG2.1 Working Conference on Generic Programming*, pages 1–20. Kluwer, B.V. doi:10.1007/978-0-387-35672-3_1. ↰ page 3

Thorsten Altenkirch, Conor McBride, and James McKinna [2005]. Why dependent types matter. Available at `http://www.cs.nott.ac.uk/~txa/publ/ydtm.pdf`. ↰ page 2

Thorsten Altenkirch and Peter Morris [2009]. Indexed containers. In *Logic in Computer Science*, LICS'09, pages 277–285. IEEE. doi:10.1109/LICS.2009.33. ↰ page 10

Robert Atkey, Patricia Johann, and Neil Ghani [2012]. Refining inductive types. *Logical Methods in Computer Science*, 8(2:09). doi:10.2168/LMCS-8(2:9)2012. ↰ pages 111 and 119

Gilles Barthe, Venanzio Capretta, and Olivier Pons [2003]. Setoids in type theory. *Journal of Functional Programming*, 13(2):261–293. doi:10.1017/S0956796802004501. ↰ page 65

Jean-Philippe Bernardy and Moulin Guilhem [2013]. Type theory in color.

In *International Conference on Functional Programming*, ICFP'13, pages 61–72. ACM. doi:10.1145/2500365.2500577. ↰ pages 17 and 62

Richard Bird and Oege de Moor [1997]. *Algebra of Programming*. Prentice-Hall. ↰ pages 87, 91, 93, and 103

Errett Bishop and Douglas Bridges [1985]. *Constructive Analysis*. Springer-Verlag. ↰ page 2

Ana Bove and Peter Dybjer [2009]. Dependent types at work. In *Language Engineering and Rigorous Software Development*, volume 5520 of *Lecture Notes in Computer Science*, pages 57–99. Springer-Verlag. doi:10.1007/978-3-642-03153-3_2. ↰ page 2

Edwin Brady, Conor McBride, and James McKinna [2004]. Inductive families need not store their indices. In *Types for Proofs and Programs*, volume 3085 of *Lecture Notes in Computer Science*, pages 115–129. Springer-Verlag. doi:10.1007/978-3-540-24849-1_8. ↰ page 5

James Chapman, Pierre-Évariste Dagand, Conor McBride, and Peter Morris [2010]. The gentle art of levitation. In *International Conference on Functional Programming*, ICFP'10, pages 3–14. ACM. doi:10.1145/1863543.1863547. ↰ pages 3 and 33

Pierre-Évariste Dagand and Conor McBride [2012a]. Elaborating inductive definitions. arXiv:1210.6390. ↰ page 9

Pierre-Évariste Dagand and Conor McBride [2012b]. Transporting functions across ornaments. In *International Conference on Functional Programming*, ICFP'12, pages 103–114. ACM. doi:10.1145/2364527.2364544. ↰ pages 3, 18, 21, 32, 93, and 118

Peter Dybjer [1998]. A general formulation of simultaneous inductive-recursive definitions in type theory. *Journal of Symbolic Logic*, 65(2):525–549. doi:10.2307/2586554. ↰ pages 3 and 118

Jeremy Gibbons, Graham Hutton, and Thorsten Altenkirch [2001]. When is a function a fold or an unfold? *Electronic Notes in Theoretical Computer Science*, 44(1):146–160. doi:10.1016/S1571-0661(04)80906-X. ↰ page 99

Hsiang-Shang Ko and Jeremy Gibbons [2013]. Modularising inductive families. *Progress in Informatics*, 10:65–88. doi:10.2201/NiiPi.2013.10.5. ↰ pages 3, 4, and 98

Per Martin-Löf [1984]. *Intuitionistic Type Theory*. Bibliopolis, Napoli. ↰ page 2

Conor McBride [1999]. *Dependently Typed Functional Programs and their Proofs*. Ph.D. thesis, University of Edinburgh. ↰ page 68

Conor McBride [2004]. Epigram: Practical programming with dependent types. In *Advanced Functional Programming*, volume 3622 of *Lecture Notes in Computer Science*, pages 130–170. Springer-Verlag. doi:10.1007/11546382_3. ↰ page 2

Conor McBride [2011]. Ornamental algebras, algebraic ornaments. To appear in *Journal of Functional Programming*. ↰ pages 9, 32, and 93

Conor McBride and James McKinna [2004]. The view from the left. *Journal of Functional Programming*, 14(1):69–111. doi:10.1017/S0956796803004829. ↰ page 107

Stefan Monnier and David Haguenauer [2010]. Singleton types here, singleton types there, singleton types everywhere. In *Programming Languages meets Program Verification*, PLPV'10, pages 1–8. ACM. doi:10.1145/1707790.1707792. ↰ page 31

Shin-Cheng Mu, Hsiang-Shang Ko, and Patrik Jansson [2009]. Algebra of Programming in Agda: Dependent types for relational program derivation. *Journal of Functional Programming*, 19(5):545–579. doi:10.1017/S0956796809007345. ↰ pages 68, 87, and 95

Ulf Norell [2007]. *Towards a practical programming language based on dependent type theory*. Ph.D. thesis, Chalmers University of Technology. ↰ page 2

Ulf Norell [2009]. Dependently typed programming in Agda. In *Advanced Functional Programming*, volume 5832 of *Lecture Notes in Computer Science*, pages 230–266. Springer-Verlag. doi:10.1007/978-3-642-04652-0_5. ↰ page 2

Chris Okasaki [1999]. *Purely functional data structures*. Cambridge University Press. ↰ pages 47, 49, 56, 57, and 61

Wouter Swierstra [2008]. Data types à la carte. *Journal of Functional Programming*, 18(4):423–436. doi:10.1017/S0956796808006758. ↰ page 33

Todo list