Analysis and synthesis of inductive families

Hsiang-Shang Ko

15 February 2014

Contents

1 Introduction **1**

2 From intuitionistic type theory to dependently typed programming **2**

2.1 Propositions as types 3

2.2 Elimination and pattern matching 7

2.2.1 Pattern matching and interactive development 9

2.2.2 Pattern matching on intermediate computation 11

2.3 Equality 14

2.4 Universes and datatype-generic programming 19

2.4.1 High-level introduction to index-first datatypes 20

2.4.2 Universe construction 22

2.5 Externalism and internalism 29

3 Refinements and ornaments **38**

3.1 Refinements 39

3.1.1 Refinements between individual types 39

3.1.2 Upgrades 43

3.1.3 Refinement families 47

3.2 Ornaments 49
3.2.1 Universe construction 50
3.2.2 Ornamental descriptions 54
3.2.3 Parallel composition of ornaments 60
3.3 Refinement semantics of ornaments 66
3.3.1 Optimised predicates 66
3.3.2 Predicate swapping for parallel composition 70
3.4 Examples 73
3.4.1 Insertion into a list 74
3.4.2 Binomial heaps 74
3.4.3 Leftist heaps 82
3.5 Discussion 90

4 Categorical organisation of the ornament–refinement framework 91
4.1 Categories and functors 93
4.2 Pullback properties of parallel composition 104
4.3 Consequences 111
4.3.1 The ornamental conversion isomorphisms 111
4.3.2 The modularity isomorphisms 113
4.4 Discussion 116

5 Relational algebraic ornaments 117
5.1 Relational programming in Agda 117
5.2 Definition of relational algebraic ornaments 123

5.3 Examples 125
5.3.1 The Fold Fusion Theorem 125
5.3.2 The Streaming Theorem for list metamorphisms 128
5.3.3 The minimum coin change problem 134
5.4 Discussion 147

6 Categorical equivalence of ornaments and relational algebras 148
6.1 Ornaments and horizontal transformations 151
6.2 Ornaments and relational algebras 151
6.3 Consequences 151
6.3.1 Parallel composition and the banana-split law 151
6.3.2 Ornamental algebraic ornaments 151
6.4 Discussion 151

7 Conclusion 152
7.1 Future work 152

Chapter 1

Introduction

program correctness by construction (from specifications to programs); type theory (unification of logic and computation and/vs types as classification/specification); new direction of program derivation, while inheriting problems

"datatypes" for inductive families

We start with an introduction to Martin-Löf's intuitionistic type theory [Martin-Löf, 1975, 1984b; Nordström et al., 1990] and dependently typed programming [Altenkirch et al., 2005; McBride, 2004] using the Agda language [Norell, 2007, 2009; Bove and Dybjer, 2009]. Intuitionistic type theory was developed by Martin-Löf to serve as a foundation of intuitionistic mathematics, like Bishop's renowned work on constructive analysis [Bishop and Bridges, 1985]. While originated from intuitionistic type theory, dependently typed programming is more concerned with mechanisation and practicalities, and is influenced by the program-correctness-by-construction movement. It has thus departed from the mathematical traditions considerably, and deviations can be found from syntactic presentations to the underlying philosophy.

Chapter 2

From intuitionistic type theory to dependently typed programming

This chapter serves three purposes.

- The general theme of the thesis is set up by describing the propositions-as-types-principle (Section 2.1) and its influence on program construction (Section 2.5).
- It is explained that, while we adopt Agda as the expository language, we make sure that, at least in principle, every Agda program in this thesis can be translated down to well-studied aspects of type theory — no mysterious Agda-specific semantics is used. Specifically, we briefly discuss foundational aspects of pattern matching (Section 2.2) and equality (Section 2.3).
- Agda-specific syntax, notational conventions, and basic constructions used throughout the thesis are also introduced. In particular, a universe for "index-first" inductive families is constructed (Section 2.4), which is the basis of essentially all later constructions.

2.1 Propositions as types

Mathematics is all about mental constructions, that is, the intuitive grasp and manipulation of mental objects, the intuitionists say [Heyting, 1971; Dummett, 2000]. Take the natural numbers as an example. We have a distinct idea of how natural numbers are built: start from an origin 0, and form its successor 1, and then the successor of 1, which is 2, and so on. In other words, it is in our nature to be able to count, and counting is just the way the natural numbers are constructed. This construction then gives a specification of when we can immediately (i.e., directly intuitively) recognise a natural number, namely when it is 0 or a successor of some other natural number, and this specification of immediately recognisable forms is one of the conditions of forming the **set** of natural numbers in Martin-Löf Type Theory. In symbols, we are justified by our intuition to have the **formation rule**

$$\frac{}{\mathsf{Nat} \ : \ \mathsf{Set}}$$

saying that we can conclude (below the line) that Nat is a set from no assumptions (above the line), and the two **introduction rules**

$$\frac{}{\mathsf{zero} \ : \ \mathsf{Nat}} \qquad \frac{n \ : \ \mathsf{Nat}}{\mathsf{suc}\ n \ : \ \mathsf{Nat}}$$

specifying the **canonical inhabitants** of Nat, i.e., those inhabitants that are immediately recognisable as belonging to Nat, namely zero and suc n whenever n is an inhabitant of Nat. There are natural numbers which are not in canonical form (like 10^{10}) but instead encode an effective method for computing a canonical inhabitant. We accept them as **non-canonical inhabitants** of Nat, as long as they compute to a canonical form so we can see that they are indeed natural numbers. Thus, to form a set, we should be able to recognise its inhabitants, either directly or indirectly, as bearing a certain form and thus belonging to the set, so the inhabitants of the set are intuitively clear to us as a certain kind of mental constructions.

What is more characteristic of intuitionism is that the intuitionistic interpretation of propositions — in particular the logical constants/connectives — follows the same line of thought as the specification of the set of natural numbers. A proposition is an expression of its truth condition, and since intuitionistic truth follows from proofs, a proposition is clearly specified exactly when what constitutes a proof of it is determined [Martin-Löf, 1987]. What is a proof of a proposition, then? It is a piece of mental construction such that, upon inspection, the truth of the proposition is immediately recognised. For a simple example, in type theory we can formulate the formation rule for conjunctions

$$\frac{A \ : \ \mathsf{Set} \qquad B \ : \ \mathsf{Set}}{A \wedge B \ : \ \mathsf{Set}}$$

and the introduction rule

$$\frac{a \ : \ A \qquad b \ : \ B}{(a \ , b) \ : \ A \wedge B}$$

saying that an immediately acceptable proof (canonical inhabitant) of $A \wedge B$ is a pair of a proof (inhabitant) of A and a proof (inhabitant) of B. Any other (non-canonical) way of proving a conjunction must effectively yield a proof in the form of a pair. The relationship between a proposition and its proofs is thus exactly the same as the one between a set and its inhabitants — the proofs must be effectively recognisable as proving the proposition. Hence, in type theory, the notion of propositions and proofs is subsumed by the notion of sets and inhabitants. This is called the **propositions-as-types principle**, which reflects the observation that proofs are nothing but a certain kind of mental constructions.

Notice that the notion of "effective methods" — or computation — was presumed when the notion of sets was introduced, and at some point we need to concretely specify an effective method. Since the description of every set includes an effective way to construct its canonical inhabitants, it is possible to express an effective method that mimics the construction of an inhabitant by saying that the computation has the same structure as how the inhabitant is

constructed, and the computation is guaranteed to terminate since the structure of the inhabitant is finitary. For a typical example, let us look again at the natural numbers. Suppose that we have a **family of sets** $P : \mathsf{Nat} \to \mathsf{Set}$ indexed by inhabitants of Nat. (Since we only aim to present a casual sketch of type theory, we take the liberty of using Agda functions (including Set-computing ones like P above) in places where terms under contexts should have been used.) If we have an inhabitant z of $P\ \mathsf{zero}$ and a method s that, for any $n : \mathsf{Nat}$, transforms an inhabitant of $P\ n$ to an inhabitant of $P\ (\mathsf{suc}\ n)$, then we can compute an inhabitant of $P\ n$ for any given n by essentially the same counting process with which we construct n, but the counting now starts from z instead of zero and proceeds with s instead of suc. For instance, if a proof of $P\ 2$ is required, we can simply apply s to z twice, just like we apply suc to zero twice to form 2, so the computation was guided by the structure of 2. This explanation justifies the following **elimination rule**

$$\frac{P : \mathsf{Nat} \to \mathsf{Set} \qquad z : P\ \mathsf{zero} \qquad s : (n : \mathsf{Nat}) \to P\ n \to P\ (\mathsf{suc}\ n) \qquad n : \mathsf{Nat}}{\mathsf{Nat\text{-}elim}\ P\ z\ s\ n : P\ n}$$

(The type of s illustrates Agda's syntax for dependent functions — the value n of the first argument is referred to in the types of the second argument and the result.) The symbol Nat-elim symbolises the method described above, which, given P, z, and s, transforms every natural number n into an inhabitant of the set $P\ n$. The actual computation performed by Nat-elim is stated as two **computation rules** in the form of equality judgements (see Section 2.3):

$$\frac{P : \mathsf{Nat} \to \mathsf{Set} \qquad z : P\ \mathsf{zero} \qquad s : (n : \mathsf{Nat}) \to P\ n \to P\ (\mathsf{suc}\ n)}{\mathsf{Nat\text{-}elim}\ P\ z\ s\ \mathsf{zero} = z \in P\ \mathsf{zero}}$$

$$\frac{P : \mathsf{Nat} \to \mathsf{Set} \qquad z : P\ \mathsf{zero} \qquad s : (n : \mathsf{Nat}) \to P\ n \to P\ (\mathsf{suc}\ n) \qquad n : \mathsf{Nat}}{\mathsf{Nat\text{-}elim}\ P\ z\ s\ (\mathsf{suc}\ n) = s\ n\ (\mathsf{Nat\text{-}elim}\ P\ z\ s\ n) \in P\ (\mathsf{suc}\ n)}$$

From the logic perspective, predicates on Nat are a special case of Nat-indexed families of sets like P; Nat-elim then delivers the induction principle for natural numbers, as it produces a proof of $P\ n$ for every $n : \mathsf{Nat}$ if the base case z and the inductive case s can be proved. In general, the propositions-as-types principle treats logical entities as ordinary mathematical objects; the logic hence

inherits the computational meaning of intuitionistic mathematics and becomes constructive.

By enabling the interplay of various sets governed by rules like the above ones, type theory is capable of formalising various mental constructions we manipulate in mathematics in a fully computational way, making it a powerful programming language. As Martin-Löf [1984a] noted: "If programming is understood [...] as the design of the methods of computation [...], then it no longer seems possible to distinguish the discipline of programming from constructive mathematics". Indeed, sets are easily comparable with inductive datatypes in functional programming — a formation rule names a datatype, the associated introduction rules list the constructors of the datatype, and the associated elimination rule and computation rules define a precisely typed version of primitive recursion on the datatype. Consequently, we identify "sets" with "types" in this thesis, and regard them as interchangeable terms.

The uniform treatment of programs and proofs in type theory reveals new possibilities regarding proofs of program correctness. Traditional mathematical theories employ a standalone logic language for talking about some postulated objects. For example, Peano arithmetic is set up by postulating axioms about natural numbers in the language of first-order logic. Inside the postulated system of natural numbers, there is no knowledge of logic formulas or proofs (except via exotic encodings) — logic is at a higher level than the objects they are used to talk about. Programming systems based on such principle (e.g., Hoare logic) then need to have a meta-level logic language to reason about properties of programs. In **dependently typed** languages based on type theory, however, the two traditional levels are coherently integrated into one, so programs can be naturally constructed along with their correctness proofs. For example, the proposition $\forall\, a : A.\ \exists\, b : B.\ R\ a\ b$ is interpreted as the type of a function taking $a : A$ to a pair consisting of $b : B$ and a proof of the proposition $R\ a\ b$, so the output of type B is guaranteed to be related to the input of type A by R. Checking of proof validity reduces to typechecking, and correctness proofs coexist with programs, as opposed to being separately presented at a meta-level.

The propositions-as-types principle, however, can lead to a more intimate form of program correctness by construction by blurring the distinction between programs and proofs even further; this form of program correctness — called **internalism** — is introduced in Section 2.5, which opens the central topic studied by this thesis. Before that, we make a transition from type theory to practical programming in Agda, starting with its pattern matching notation.

2.2 Elimination and pattern matching

The formation rules and introduction rules for sets in type theory directly translate into inductive datatype declarations in functional programming. For example, the set of natural numbers is translated into Agda as an inductive datatype with two constructors, with their full types displayed:

```
data Nat : Set where
  zero : Nat
  suc  : Nat → Nat
```

In type theory, computations on inductive datatypes are specified using eliminators like Nat-elim, whose style corresponds to **recursion schemes** [Meijer et al., 1991] in functional programming. One reason for making elimination as the only option is that programs in type theory are demanded to terminate — a consequence of the requirement that an inhabitant should be effectively recognisable as belonging to a set — and using eliminators throughout is a straightforward way of enforcing termination. On the other hand, in functional programming, the **pattern matching** notation is widely used for defining programs on (inductive) datatypes (see, e.g., Hudak et al. [2007, Section 5]) in addition to recursion schemes. Pattern matching is vital to the clarity of functional programs because it not only allows a function to be intuitively defined by equations suggesting how the function computes, but also clearly conveys the programming strategy of splitting a problem into sub-problems by case analysis.

When it comes to dependently typed programming, the situation becomes

more complicated due to the presence of **inductive families** [Dybjer, 1994], i.e., simultaneously inductively defined families of sets, like the following:

```
data _⩽ℕ_ (m : Nat) : Nat → Set where
  refl : m ⩽ℕ m
  step : (n : Nat) → m ⩽ℕ n → m ⩽ℕ suc n
```

(An Agda name with underscores (like $\_\leqslant_{\mathsf{N}}\_$) can be applied to arguments either normally (like "$\_\leqslant_{\mathsf{N}}\_\ m\ n$") or by substituting the arguments for the underscores with proper spacing (like "$m \leqslant_{\mathsf{N}} n$").) Reading the declaration logically, the types of the two constructors refl and step give the two inference rules for establishing that one natural number is less than or equal to another. More generally, we read the declaration as a datatype parametrised by $m : \mathsf{Nat}$ (as signified by its appearance right next to **data** $\_\leqslant_{\mathsf{N}}\_$) and indexed by Nat. For any $m : \mathsf{Nat}$, the type family $\_\leqslant_{\mathsf{N}}\_\ m : \mathsf{Nat} \to \mathsf{Set}$ as a whole is inductively populated: we have an inhabitant refl in the set $(\_\leqslant_{\mathsf{N}}\_\ m)\ m$, and whenever we have an inhabitant $p : (\_\leqslant_{\mathsf{N}}\_\ m)\ n$ for some $n : \mathsf{Nat}$, we can make a larger inhabitant step $n\ p$ in another set $(\_\leqslant_{\mathsf{N}}\_\ m)\ (\mathsf{suc}\ n)$ in the family. (From now on we usually refer to inductive families simply as datatypes, especially when emphasising their use in programming and de-emphasising the distinction between them and non-indexed datatypes like Nat.)

With inductive families, splitting a problem into sub-problems by case analysis in dependently typed programming often leads to nontrivial refinement of the goal type and the context, and such refinement can be tricky to handle with eliminators. Admittedly, in terms of expressive power, pattern matching and elimination are basically equivalent, as eliminators can be easily defined by dependent pattern matching, and conversely, it has been shown that dependent pattern matching can be reduced to elimination if **uniqueness of identity proofs** — or, equivalently, the **K axiom** [Streicher, 1993] — is assumed [McBride, 1999; Goguen et al., 2006]. (See Section 2.3 for more on uniqueness of identity proofs.) Nevertheless, there is a significant notational advantage of recovering pattern matching in dependently typed programming, especially with the support of an interactive development environment. Below we look at an example of interactively constructing a program with pattern

matching in Agda, whose design was inspired by Epigram [McBride, 2004; McBride and McKinna, 2004].

2.2.1 Pattern matching and interactive development

Suppose that we are asked to prove that $\_\leqslant_{\mathsf{N}}\_$ is transitive, i.e., to construct the program

$$\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z$$

(The "telescopic" quantification "$(x\ y\ z \;:\; \mathsf{Nat}) \to$" is a shorthand for "$(x \;:\; \mathsf{Nat})\ (y \;:\; \mathsf{Nat})\ (z \;:\; \mathsf{Nat}) \to$", which, in turn, is a shorthand for "$(x \;:\; \mathsf{Nat}) \to (y \;:\; \mathsf{Nat}) \to (z \;:\; \mathsf{Nat}) \to$".) We define *trans* interactively by first putting pattern variables for the arguments on the left of the defining equation and then leaving an "interaction point" — also called a "goal" — on the right, which is numbered 0. Agda then tells us that a term of type $x \leqslant_{\mathsf{N}} z$ is expected (shown in the goal).

$$\begin{aligned}&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z\\ &\mathit{trans}\ x\ y\ z\ p\ q \;=\; \{\, x \leqslant_{\mathsf{N}} z \,\}_0\end{aligned}$$

We instruct Agda to perform case analysis on q, and there are two cases: refl and $\mathsf{step}\ w\ r$ where r has type $y \leqslant_{\mathsf{N}} w$. The original Goal 0 is split into two sub-goals, and unification is triggered for each sub-goal.

$$\begin{aligned}&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z\\ &\mathit{trans}\ x\ .z\ \ z\qquad p^{\,x \leqslant_{\mathsf{N}} z}\ \mathsf{refl}\qquad\qquad\quad = \{\, x \leqslant_{\mathsf{N}} z \,\}_1\\ &\mathit{trans}\ x\ \ y\ .(\mathsf{suc}\ w)\ p^{\,x \leqslant_{\mathsf{N}} y}\ (\mathsf{step}\ w\ r^{\,y \leqslant_{\mathsf{N}} w}) \;=\; \{\, x \leqslant_{\mathsf{N}} \mathsf{suc}\ w \,\}_2\end{aligned}$$

In Goal 1, the type of refl demands that y be unified with z, and hence the pattern variable y is replaced with a "dot pattern" $.z$ indicating that the value of y is determined by unification to be z. Therefore, upon enquiry, Agda tells us that the type of p in the context — which was originally $x \leqslant_{\mathsf{N}} y$ — is now $x \leqslant_{\mathsf{N}} z$ (shown in superscript next to p, which is not part of the Agda program but only appears when interacting with Agda). Similarly for Goal 2, z is unified with $\mathsf{suc}\ w$ and the goal type is rewritten accordingly. We see that the case analysis has led to two sub-problems with different goal types and

contexts, where Goal 1 is easily solvable as there is a term in the context with the right type, namely p.

$$
\begin{aligned}
&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z \\
&\mathit{trans}\ x\ .z\ \ z \qquad p \qquad \mathsf{refl} \qquad\qquad = p \\
&\mathit{trans}\ x\ \ y\ .(\mathsf{suc}\ w)\ p\ ^{x \leqslant_{\mathsf{N}} y}\ (\mathsf{step}\ w\ r\ ^{y \leqslant_{\mathsf{N}} w}) \;=\; \{\, x \leqslant_{\mathsf{N}} \mathsf{suc}\ w \,\}_2
\end{aligned}
$$

The second goal type $x \leqslant_{\mathsf{N}} \mathsf{suc}\ w$ looks like the conclusion in the type of the term $\mathsf{step}\ w \;:\; x \leqslant_{\mathsf{N}} w \to x \leqslant_{\mathsf{N}} \mathsf{suc}\ w$, so we use this term to reduce Goal 2 to Goal 3, which now requires a term of type $x \leqslant_{\mathsf{N}} w$.

$$
\begin{aligned}
&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z \\
&\mathit{trans}\ x\ .z\ \ z \qquad p \qquad \mathsf{refl} \qquad\qquad = p \\
&\mathit{trans}\ x\ \ y\ .(\mathsf{suc}\ w)\ p\ ^{x \leqslant_{\mathsf{N}} y}\ (\mathsf{step}\ w\ r\ ^{y \leqslant_{\mathsf{N}} w}) \;=\; \mathsf{step}\ w\ \{\, x \leqslant_{\mathsf{N}} w \,\}_3
\end{aligned}
$$

Now we see that the induction hypothesis term $\mathit{trans}\ x\ y\ w\ p\ r \;:\; x \leqslant_{\mathsf{N}} w$ has the right type. Filling the term into Goal 3 completes the program.

$$
\begin{aligned}
&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z \\
&\mathit{trans}\ x\ .z\ \ z \qquad p\ \mathsf{refl} \qquad\quad = p \\
&\mathit{trans}\ x\ \ y\ .(\mathsf{suc}\ w)\ p\ (\mathsf{step}\ w\ r) \;=\; \mathsf{step}\ w\ (\mathit{trans}\ x\ y\ w\ p\ r)
\end{aligned}
$$

In contrast, if we stick to the default elimination approach in type theory, we would use the eliminator

$$
\begin{aligned}
\leqslant_{\mathsf{N}}\text{-}\mathsf{elim} \;:\; &(m \;:\; \mathsf{Nat})\ (P \;:\; (n \;:\; \mathsf{Nat}) \to m \leqslant_{\mathsf{N}} n \to \mathsf{Set}) \to \\
&((t \;:\; m \leqslant_{\mathsf{N}} m) \to P\ m\ t) \to \\
&((n \;:\; \mathsf{Nat})\ (t \;:\; m \leqslant_{\mathsf{N}} n) \to P\ n\ t \to P\ (\mathsf{suc}\ n)\ (\mathsf{step}\ n\ t)) \to \\
&(n \;:\; \mathsf{Nat})\ (t \;:\; m \leqslant_{\mathsf{N}} n) \to P\ n\ t
\end{aligned}
$$

and write

$$
\begin{aligned}
&\mathit{trans} \;:\; (x\ y\ z \;:\; \mathsf{Nat}) \to x \leqslant_{\mathsf{N}} y \to y \leqslant_{\mathsf{N}} z \to x \leqslant_{\mathsf{N}} z \\
&\mathit{trans}\ x\ y\ z\ p\ q \;=\; \leqslant_{\mathsf{N}}\text{-}\mathsf{elim}\ y\ (\lambda\, y'\ \_ \mapsto x \leqslant_{\mathsf{N}} y \to x \leqslant_{\mathsf{N}} y') \\
&\qquad\qquad (\lambda\ \_\ p' \mapsto p')\ (\lambda\, w\ r\ ih\ p' \mapsto \mathsf{step}\ w\ (ih\ p'))\ z\ q\ p
\end{aligned}
$$

We are forced to write the program in continuation passing style, where the two continuations correspond to the two clauses in the pattern matching version and likewise have more specific goal types, and the relevant context (p in this case) must be explicitly passed into the continuations in order to be refined

to a more specific type. Even with interactive support, the eliminator version is inherently harder to write and understand, especially when complicated dependent types are involved. If a function definition requires more than one level of elimination, then the advantage of using pattern matching over using eliminators becomes even more apparent.

2.2.2 Pattern matching on intermediate computation

It is often the case that we need to perform pattern matching not only on an argument but also on some intermediate computation. In simply typed languages, this is usually achieved by "case expressions", a special case being if-then-else expressions for booleans. But again, pattern matching on intermediate computation can make refinements to the goal type and the context in dependently typed languages, so case expressions — being more like eliminators — become less convenient. McBride and McKinna [2004] thus proposed **with-matching**, which generalises pattern guards [Peyton Jones, 1997] and shifts pattern matching on intermediate computations from the right of an equation to the left, thereby granting them equal status with pattern matching on arguments, in particular the power to refine contexts and goal types.

Example (*insertion into a list*)**.** To demonstrate the syntax of **with**-matching, we give a simple example of writing the function inserting an element into a list as used in, e.g., insertion sort. (More precisely, the element is inserted at the rightmost position to the left of which all elements are strictly smaller.) First we define the usual list datatype:

$$
\begin{array}{l}
\textbf{data}\ \mathsf{List}\ (A\ :\ \mathsf{Set})\ :\ \mathsf{Set}\ \textbf{where} \\
\quad [\,] \quad\ :\ \mathsf{List}\ A \\
\quad \_::\_\ :\ A \to \mathsf{List}\ A \to \mathsf{List}\ A
\end{array}
$$

and (throughout this thesis) let $Val\ :\ \mathsf{Set}$ be equipped with a decidable total ordering, i.e., there is a relation

$$\_\leqslant\_\ :\ Val \to Val \to \mathsf{Set}$$

with the following operations:

$$
\begin{aligned}
&\leqslant\text{-}\mathit{refl} &&: \{x : \mathit{Val}\} \to x \leqslant x \\
&\leqslant\text{-}\mathit{trans} &&: \{x\ y\ z : \mathit{Val}\} \to x \leqslant y \to y \leqslant z \to x \leqslant z \\
&\_\leqslant_?\_ &&: (x\ y : \mathit{Val}) \to \mathsf{Dec}\ (x \leqslant y) \\
&\not\leqslant\text{-}\mathit{invert} &&: \{x\ y : \mathit{Val}\} \to \neg\ (x \leqslant y) \to y \leqslant x
\end{aligned}
$$

where Dec is the following datatype witnessing whether a set is inhabited or not:

$$
\begin{aligned}
&\textbf{data}\ \mathsf{Dec}\ (A : \mathsf{Set}) : \mathsf{Set}\ \textbf{where} \\
&\quad \mathsf{yes} : \quad A \to \mathsf{Dec}\ A \\
&\quad \mathsf{no}\ : \neg\ A \to \mathsf{Dec}\ A \quad \text{-- } \neg\ A\ =\ A \to \bot\text{, where } \bot \text{ is the empty set}
\end{aligned}
$$

(Quantifications like $\{x : \mathit{Val}\}$ are implicit arguments to a function, which can be omitted when applying the function, and Agda would try to infer them.) The insertion function is then written as

$$
\begin{aligned}
&\mathit{insert}\ :\ \mathit{Val} \to \mathsf{List}\ \mathit{Val} \to \mathsf{List}\ \mathit{Val} \\
&\mathit{insert}\ y\ [\,]\ =\ y :: [\,] \\
&\mathit{insert}\ y\ (x :: \mathit{xs})\ \textbf{with}\ y \leqslant_?\ x \\
&\mathit{insert}\ y\ (x :: \mathit{xs})\ \mid\ \mathsf{yes}\ \_\ =\ y :: x :: \mathit{xs} \\
&\mathit{insert}\ y\ (x :: \mathit{xs})\ \mid\ \mathsf{no}\ \_\ =\ x :: \mathit{insert}\ y\ \mathit{xs}
\end{aligned}
$$

The result of the intermediate computation $y \leqslant_?\ x\ :\ \mathsf{Dec}\ (y \leqslant x)$ is matched against yes and no on the left-hand side of the last two equations, just like we are performing pattern matching on a new argument. (In fact, Agda implements **with**-matching exactly by synthesising an auxiliary function with an additional argument [Norell, 2007, Section 2.3].) The witnesses carried by yes and no are ignored in this case (whose names are suppressed by underscores), but in general these can be proofs that are further matched and change the context and the goal type. (Admittedly, this is a trivial example with regard to the full power of **with**-matching, but *insert* will be used in later examples in this thesis.) □

An important application of **with**-matching is McBride and McKinna's adaptation [2004] of Wadler's **views** [1987] — or "customised pattern matching" — for dependently typed programming. Suppose that we wish to implement a snoc-list view for cons-lists, i.e., to say that a list is either empty or has

the form $ys +\!\!+ (y :: [])$ (where $\_+\!\!+\_$ is list append, a definition of which is shown in Section 2.4.2). We would define the following view type

```
data SnocView {A : Set} : List A → Set where
  nil  : SnocView []
  snoc : (ys : List A) (y : A) → SnocView (ys ++ (y :: []))
```

and write a **covering function** for the view:

```
snocView : {A : Set} (xs : List A) → SnocView xs
snocView []                    = nil
snocView (x :: xs)             with snocView xs
snocView (x :: .[])            | nil        = snoc [] x
snocView (x :: .(ys ++ (y :: [])))  | snoc ys y  = snoc (x :: ys) y
```

Note that the type of *snocView* ensures that every list is covered under one constructor of SnocView. Also, this is a nontrivial example of **with**-matching, because performing pattern matching on the result of *snocView xs* refines *xs* in the context to either [] or $ys +\!\!+ (y :: [])$, and the refinement is propogated to the goal type. Now, for example, the function *init* which removes the last element (if any) in a list can be implemented simply as

```
init : {A : Set} → List A → List A
init xs              with snocView xs
init .[]             | nil        = []
init .(ys ++ (y :: [])) | snoc ys y  = ys
```

Views are not enough for dependently typed programming, though, because views offer only customised case analyses but not terminating recursion. McBride and McKinna [2004] proposed a general mechanism for invoking any programmer-defined eliminator using the pattern matching syntax, so the programmer can choose whichever recursive problem-splitting strategy s/he needs and express it conveniently with pattern matching. This mechanism is implemented in Epigram and greatly improves readability of dependently typed programs. Specifically, Epigram syntax makes it clear which and in what order eliminators are invoked, so programs are easily guaranteed to be based on elimination — and thus are terminating — while remaining readable.

Agda's design does not emphasise reducibility to elimination, but almost all the recursive programs in this thesis are written with this in mind, so termination is evident even without understanding Agda's termination checker in detail. Also, although Agda provides pattern matching only for datatype constructors, all but one of the recursive programs in this thesis use default structural induction (without need of programmer-defined elimination), so Agda syntax suffices.

2.3 Equality

In logic, the **intension** of a concept is its defining description, while the **extension** of the concept is the range of objects it refers to. Two concepts can differ intensionally yet agree extensionally when they use different ways to describe the same range of objects. Classical mathematics cares about extensions only (e.g., the axiom of extensionality in set theory defines that two sets are equal exactly when they have the same inhabitants, regardless of how they are described), whereas in intuitionistic mathematics, objects are given to us as mental constructions, which are inherently intensional descriptions. As a consequence, the fundamental equality for intuitionistic mathematics is intensional [Dummett, 2000, Section 1.2], since we can only compare the intensional descriptions given to us, and furthermore it might be impossible to effectively recognise whether two intensionally different constructions are extensionally the same. For example, functions describing different computational procedures are distinguished even when they always map the same input to the same output, as it is well known that pointwise equality of functions is undecidable. We can, of course, still talk about extensional equalities in intuitionistic mathematics, but they are just treated as ordinary propositions.

The fundamental equality is formulated in type theory as **judgemental equality**, a meta-level notion for determining whether two types match in typechecking, which also involves determining whether two terms match because types can contain terms. (The computation rules for Nat-elim in Section 2.1 are

examples of judgemental equalities between terms.) If we take the position of intuitionistic mathematics seriously, judgemental equality would be chosen to be the intensional, syntactic equality — also called **definitional equality** — which can be implemented by reducing two types/terms to normal forms and checking whether the normal forms match. The resulting type theory is called an **intensional type theory**, whose characteristic feature is decidable typechecking (enabling effective recognition of set membership) due to decidability of judgemental equality. Agda, in particular, is intensional in this sense.

Judgemental equality — being a meta-level notion — is not an entity inside the theory. To state equality between two terms as a proposition and have proof for that proposition inside the theory, we need **propositional equality**, which can be defined in Agda by the following inductive family:

$$\begin{aligned}&\textbf{data}\ \_{\equiv}\_\ \{A\ :\ \mathsf{Set}\}\ (x\ :\ A)\ :\ A \to \mathsf{Set}\ \textbf{where}\\ &\quad \mathsf{refl}\ :\ x\ \equiv\ x\end{aligned}$$

The canonical way to prove an equality proposition $x \equiv y$ is refl, which is permitted when x and y are judgementally equal. Under contexts, however, it is possible to prove that two judgementally different terms are propositionally equal. For example, the following "catamorphic" identity function on natural numbers

$$\begin{aligned}&id'\ :\ \mathsf{Nat} \to \mathsf{Nat}\\ &id'\ \mathsf{zero}\ \ \ \ =\ \mathsf{zero}\\ &id'\ (\mathsf{suc}\ n)\ =\ \mathsf{suc}\ (id'\ n)\end{aligned}$$

can be shown to be pointwise equal to the polymorphic identity function

$$\begin{aligned}&id\ :\ \{A\ :\ \mathsf{Set}\} \to A \to A\\ &id\ x\ =\ x\end{aligned}$$

That is, given $n\ :\ \mathsf{Nat}$ in the context, even though the two open terms $id\ n$ (which is definitionally just n) and $id'\ n$ are judgementally different, we can still prove $id\ n \equiv id'\ n$ by induction (elimination) on n, whose two cases instantiate n to a more specific form and make computation on $id'\ n$ happen. It might be said that propositional equality — in this most basic, inductive form — is "delayed" judgemental equality as a proposition: the judgementally different

terms $id\ n$ and $id'\ n$ would compute to the same canonical term — and hence become judgementally equal — after substituting a canonical natural number for n, turning them into closed terms and allowing the computation to complete. Formally, this is stated as the "reflection principle": two propositionally equal closed terms are judgementally equal (see, e.g., Luo [1994, Section 5.1.3] and Streicher [1993, Section 1.1]).

A propositional equality satisfying the reflection principle — e.g., the Agda one — can sometimes be too discriminating. For example, in category theory (which we use in Chapters 4 and 6), a universal function (i.e., a universal morphism in the category of sets and total functions) is unique up to extensional (i.e., pointwise) equality, but pointwise propositionally equal functions are usually not propositionally equal themselves. (If, under the empty context, two pointwise propositionally equal functions are propositionally equal themselves, then by the reflection principle they are also judgementally equal, but the two functions can well have different intensions.) While we can explicitly work with pointwise equality on functions, not being able to identify extensionally equal functions propositionally means that we do not automatically get basic properties like

$$cong\ :\ \{A\ B\ :\ \mathsf{Set}\}\ (f\ :\ A \to B)\ \{x\ y\ :\ A\} \to x \equiv y \to f\ x \equiv f\ y$$

i.e., a function maps equal arguments to equal results, and

$$subst\ :\ \{A\ :\ \mathsf{Set}\}\ (P\ :\ A \to \mathsf{Set})\ \{x\ y\ :\ A\} \to x \equiv y \to P\ x \to P\ y$$

i.e., inhabitants in an indexed set can be transported to another set with an equal index, for extensionally equal functions. We would need to prove such properties for every entity like f and P on a case-by-case basis, which quickly becomes tedious.

Several foundational modifications to intensional type theory have been proposed to obtain a more liberal notion of equality. While this thesis sticks to Agda's intensional approach and does not adopt any of the alternatives, it is interesting to reflect on the relationship between these alternatives and our development.

- A simple yet radical approach is to add the **equality reflection rule** to the

theory, injecting propositional equality back into judgemental equality.

$$\frac{x : A \qquad y : A \qquad eq : x \equiv y}{x = y \in A}$$

Extensionally equal functions become judgementally equal (and thus propositionally equal) in such a theory: Suppose that f and g are functions of type $A \to B$ and we have a proof

$$fgeq : (x : A) \to f\ x \equiv g\ x$$

Then, judgementally,

$$\begin{aligned}
& f \\
= \quad & \{\, \eta\text{-expansion} \,\} \\
& \lambda\, x \mapsto f\ x \\
= \quad & \{\, \text{equality reflection} — f\ x = g\ x \in B \text{ since } fgeq\ x : f\ x \equiv g\ x \,\} \\
& \lambda\, x \mapsto g\ x \\
= \quad & \{\, \eta\text{-contraction} \,\} \\
& g \qquad \in A \to B
\end{aligned}$$

The judgemental equality is thus able to identify pointwise equal functions and becomes extensional, and such a theory is called an **extensional type theory** (see, e.g., Nordström et al. [1990, Section 8.2]). In extensional type theory, combinators like *subst* become unnecessary, since having a proof of $x \equiv y$ means that x and y are identified judgementally and hence are regarded as the same during typechecking, so $P\ x$ and $P\ y$ are simply the same type — no explicit transportation is needed. Consequently, programs become very lightweight, with all the equality justifications moved to typing derivations at the meta-level. The downside is that typechecking in extensional type theory is undecidable, because whenever there is possibility that the equality reflection rule is needed, the typechecker would have to somehow determine whether there is a suitable equality proof, for which there is no effective procedure. This is not a big problem for proof assistants like NUPRL [Constable et al., 1985], in which the programmer instructs the proof assistant to construct typing derivations and can supply the right

proof when using the equality reflection rule. (NUPRL, in fact, simply identifies judgemental equality and propositional equality and does not have the equality reflection rule explicitly.) But for programming languages like ΩMEGA [Sheard and Linger, 2007], equality reflection does present a problem, since the programmer constructs a term only, and the typing derivation has to be constructed by the typechecker, which then has to search for proofs. ΩMEGA can take hints from the programmer so the proof search is more likely to succeed, but the fundamental problem is that justification of program correctness now relies on the proof searching algorithm and is tied to the implementation detail of a specific programming system. Since the focus of this thesis is on dependently typed programming, extensional type theory is not a satisfactory foundation.

- Altenkirch et al. [2007] proposed a variant of intensional type theory called **observational type theory**, which defines propositional equality to be an extensional one — in particular, propositional equality on functions is pointwise equality — but retains computational behaviour (strong normalisation and canonicity) and decidable typechecking of intensional type theory. We step away from observational type theory merely for a practical reason: the theory is not implemented natively in any programming system yet. While it might be possible to model observational equality in AGDA to some extent and then construct the universes of descriptions (Section 2.4) and ornaments (Section 3.2) inside the observational model, developing and programming within the nested model would be too complex to be worthwhile.
- A new direction is being pursued by **homotopy type theory** [The Univalent Foundations Program, 2013], which gives propositional equality a higher-dimensional homotopic interpretation and broaden its scope with the univalence axiom and higher inductive types, but the computational meaning of the theory remains an open problem. Its investigations into the higher dimensional structure of propositional equality might eventually lead to a systematic treatment of equality in dependently typed programming. For this thesis, however, we confine ourselves to a basic setting in which we freely invoke **uniqueness of identity proofs**, which is definable in AGDA by

pattern matching:

$$UIP : \{A : \mathsf{Set}\}\ \{x\ y : A\}\ (p\ q : x \equiv y) \to p \equiv q$$
$$UIP\ \mathsf{refl}\ \mathsf{refl} = \mathsf{refl}$$

Our identification of types and sets is thus consistent with the terminology of homotopy type theory, as types on which identity proofs are unique (and hence lack higher dimensional structure) are indeed termed "sets" by homotopy type theorists [The Univalent Foundations Program, 2013, Section 3.1].

With only Agda's intensional equality, we have to explicitly work with equality-like propositions (like pointwise equality on functions) and manage them with the help of **setoids** [Barthe et al., 2003] in this thesis — Chapters 4 and 6, specifically. We will discuss to what extent the intensional approach works at the end of these chapters.

2.4 Universes and datatype-generic programming

Martin-Löf [1984b] introduced the notion of **universes** to support "large elimination", i.e., arbitrary computation of sets, which is necessary for proving the fourth Peano axiom that zero is not the successor of any natural number [Smith, 1988]. A universe (à la Tarski) is a set of codes for sets, which is equipped with a decoding function translating codes to sets. Large elimination is then computing an inhabitant in the universe (via ordinary elimination like Nat-elim) and decoding the result to a set. Another important purpose of universes is to support quantification over sets while precluding paradoxical formation of self-referential sets — Agda, for example, has a universe hierarchy (Set, Set_1, Set_2, ...) for this purpose. From the programming perspective, however, the most interesting case is when the universe is supplied with an elimination rule [Nordström et al., 1990, Section 14.2]: allowing computation on universes turns out to correspond to **datatype-generic programming** [Gibbons, 2007a], whose idea is that the "shapes" of datatypes can be encoded so programs can be defined in terms of these shapes. Universes can encode such shapes, and since universes are just ordinary sets, datatype-generic

programming becomes just ordinary programming in dependently typed languages [Altenkirch and McBride, 2003].

In this thesis we not only use but also need to compute a class of inductive families which we call **index-first datatypes** [Chapman et al., 2010; Dagand and McBride, 2012b], and hence need to construct a universe for them. Before that, we give a high-level introduction to these datatypes first.

2.4.1 High-level introduction to index-first datatypes

In Agda, an inductive family is declared by listing all possible constructors and their types, all ending with one of the types in that inductive family. This conveys the idea that the index in the type of an inhabitant is synthesised in a bottom-up fashion following the construction of the inhabitant. For example, consider the following datatype of **vectors**, i.e., length-indexed lists:

```
data Vec (A : Set) : Nat → Set where
  []    : Vec A zero
  _::_ : A → {n : Nat} → Vec A n → Vec A (suc n)
```

The cons constructor _::_ takes a vector at some index n and constructs a vector at suc n — the final index is computed bottom-up from the index of the sub-vector. This approach can yield redundant representation, though — the cons constructor for vectors has to store the index of the sub-vector, so the representation of a vector would be cluttered with all the intermediate lengths. If we switch to the opposite perspective, determining top-down from the targeted index what constructors should be supplied, then the representation can usually be significantly cleaned up — for a vector, if the index of its type is known to be suc n for some n, then we know that its top-level constructor can only be cons and the index of the sub-vector must be n. To reflect this important reversal of logical order, Dagand and McBride [2012b] proposed a new notation for index-first datatype declarations, in which we first list all possible patterns of (the indices of) the types in the inductive family, and then specify for each pattern which constructors it offers. Below we follow Ko and Gibbons's slightly

more AGDA-like adaptation of the notation [2013].

Index-first declarations of simple datatypes look almost like Haskell data declarations. For example, natural numbers are declared by

```
indexfirst data Nat : Set where
  Nat ∋ zero
      | suc (n : Nat)
```

We use the keyword **indexfirst** to explicitly mark the declaration as an index-first one, distinguishing it from an AGDA datatype declaration. The only possible pattern of the datatype is Nat, which offers two constructors zero and suc, the latter taking a recursive argument named n. We declare lists similarly, this time with a uniform parameter A : Set:

```
indexfirst data List (A : Set) : Set where
  List A ∋ []
         | _::_ (a : A) (as : List A)
```

The declaration of vectors is more interesting, fully exploiting the power of index-first datatypes:

```
indexfirst data Vec (A : Set) : Nat → Set where
  Vec A zero    ∋ []
  Vec A (suc n) ∋ _::_ (a : A) (as : Vec A n)
```

Vec A is a family of types indexed by Nat, and we do pattern matching on the index, splitting the datatype into two cases Vec A zero and Vec A (suc n) for some n : Nat. The first case only offers the nil constructor [], and the second case only offers the cons constructor _::_. Because the form of the index restricts constructor choice, the recursive structure of a vector as : Vec A n must follow that of n, i.e., the number of cons nodes in as must match the number of successor nodes in n. We can also declare the bottom-up vector datatype in index-first style:

```
indexfirst data Vec′ (A : Set) : Nat → Set where
  Vec′ A n ∋ nil (neq : n ≡ zero)
           | cons (a : A) {m : Nat}
                  (as : Vec′ A m) (meq : n ≡ suc m)
```

Besides the field m storing the length of the tail, two more fields *neq* and *meq* are inserted, demanding explicit equality proofs about the indices. When a vector of type $\mathsf{Vec}'\ A\ n$ is demanded, we are "free" to choose between nil or cons regardless of the index n; however, because of the equality constraints, we are indirectly forced into a particular choice.

Remark (*detagging*)**.** The transformation from bottom-up vectors to top-down vectors is exactly what Brady et al.'s **detagging** optimisation [2004] does. With index-first datatypes, however, detagged representations are available directly, rather than arising from a compiler optimisation. □

2.4.2 Universe construction

Now we proceed to construct a universe for index-first datatypes. An inductive family of type $I \to \mathsf{Set}$ is constructed by taking the least fixed point of a base endofunctor on $I \to \mathsf{Set}$. For example, to get index-first vectors, we would define a base functor (parametrised by $A : \mathsf{Set}$)

$$
\begin{aligned}
&\mathit{VecF}\ A\ :\ (\mathsf{Nat} \to \mathsf{Set}) \to (\mathsf{Nat} \to \mathsf{Set}) \\
&\mathit{VecF}\ A\ X\ \mathsf{zero} \quad\ = \top \\
&\mathit{VecF}\ A\ X\ (\mathsf{suc}\ n)\ =\ A \times X\ n \quad \text{-- } \_\times\_ \text{ is cartesian product}
\end{aligned}
$$

and take its least fixed point. ($\top$ is a singleton set whose only inhabitant is "$\blacksquare$".) If we flip the order of the arguments of $\mathit{VecF}\ A$:

$$
\begin{aligned}
&\mathit{VecF}'\ A\ :\ \mathsf{Nat} \to (\mathsf{Nat} \to \mathsf{Set}) \to \mathsf{Set} \\
&\mathit{VecF}'\ A\ \mathsf{zero} \quad\ = \lambda\ X \to \top \\
&\mathit{VecF}'\ A\ (\mathsf{suc}\ n)\ =\ \lambda\ X \to A \times X\ n
\end{aligned}
$$

we see that $\mathit{VecF}'\ A$ consists of two different "responses" to the index request, each of type $(\mathsf{Nat} \to \mathsf{Set}) \to \mathsf{Set}$. It suffices to construct for such responses a universe

$$\textbf{data}\ \mathsf{RDesc}\ (I\ :\ \mathsf{Set})\ :\ \mathsf{Set}_1$$

with a decoding function specifying its semantics:

$$[\![\_]\!]\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{RDesc}\ I \to (I \to \mathsf{Set}) \to \mathsf{Set}$$

Inhabitants of RDesc I will be called **response descriptions**. A function of type $I \to \mathsf{RDesc}\ I$, then, can be decoded to an endofunctor on $I \to \mathsf{Set}$, so the type $I \to \mathsf{RDesc}\ I$ acts as a universe for index-first datatypes. We hence define

$$\begin{aligned}&\mathsf{Desc} : \mathsf{Set} \to \mathsf{Set}_1\\&\mathsf{Desc}\ I\ =\ I \to \mathsf{RDesc}\ I\end{aligned}$$

with decoding function

$$\begin{aligned}&\mathbb{F} : \{I : \mathsf{Set}\} \to \mathsf{Desc}\ I \to (I \to \mathsf{Set}) \to (I \to \mathsf{Set})\\&\mathbb{F}\ D\ X\ i\ =\ [\![\ D\ i\ ]\!]\ X\end{aligned}$$

Inhabitants of type Desc I will be called **datatype descriptions**, or **descriptions** for short. Actual datatypes are manufactured from descriptions by the least fixed point operator:

$$\begin{aligned}&\textbf{data}\ \mu\ \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I) : I \to \mathsf{Set}\ \textbf{where}\\&\quad \mathsf{con} : \mathbb{F}\ D\ (\mu\ D) \rightrightarrows \mu\ D\end{aligned}$$

where $\_\rightrightarrows\_$ is defined by

$$\begin{aligned}&\_\rightrightarrows\_ : \{I : \mathsf{Set}\} \to (I \to \mathsf{Set}) \to (I \to \mathsf{Set}) \to \mathsf{Set}\\&\_\rightrightarrows\_\ \{I\}\ X\ Y\ =\ \{i : I\} \to X\ i \to Y\ i\end{aligned}$$

(The implicit argument I is explicitly displayed in curly braces on the left-hand side of the equation since we need to refer to I on the right-hand side.)

Remark (*presentation of universes and their decoding*)**.** We always present universes (e.g., RDesc) along with their decoding (e.g., $[\![\_]\!]$ for RDesc) to emphasise the meaning of the codes, even when the decoding is not logically tied to the codes (cf. Martin-Löf's universe [1984b], which is inductive-recursive [Dybjer, 1998] and must present the universe and its decoding simultaneously). □

Notation (*dependent pairs and* Agda *records*)**.** Cartesian product is a special case of **Σ-types**, also known as **dependent pairs**, which are defined in Agda as a record:

$$\begin{aligned}&\textbf{record}\ \Sigma\ (A : \mathsf{Set})\ (X : A \to \mathsf{Set}) : \mathsf{Set}\ \textbf{where}\\&\quad \textbf{constructor}\ \_,\_\\&\quad \textbf{field}\end{aligned}$$

```
    outl : A
    outr : X outl
infixr 4 _,_
open Σ
syntax Σ A (λ a → T) = Σ[ a : A ] T
```

An inhabitant $p : \Sigma\ A\ X$ is a pair where the type of the second component depends on the first component — a cartesian product $A \times B$ is thus a special case, which is defined as $\Sigma\ A\ (\lambda\ \_ \mapsto\ B)$. The **constructor** declaration gives rise to a constructor function

```
_,_ : {A : Set} {X : A → Set} → (a : A) → X a → Σ A X
```

which associates to the right when used as an infix operator because of the **infixr** statement below, and can be used in pattern matching. The two field declarations give rise to two projection functions, qualified by "Σ.":

```
Σ.outl : {A : Set} {X : A → Set} → Σ A X → A
Σ.outr : {A : Set} {X : A → Set} → (p : Σ A X) → X (Σ.outl p)
```

We can drop the qualifications and refer to them simply as outl and outr due to the **open** statement. Finally, we can treat Σ as a binder and write, e.g., $\Sigma\ A\ X$ as $\Sigma[\, a : A \,]\ X\ a$, due to the **syntax** statement. □

We now define the datatype of response descriptions — which determines the syntax available for defining base functors — and its decoding function:

```
data RDesc (I : Set) : Set₁ where
  v : (is : List I) → RDesc I
  σ : (S : Set) (D : S → RDesc I) → RDesc I
⟦_⟧ : {I : Set} → RDesc I → (I → Set) → Set
⟦ v is   ⟧ X = ℙ is X   -- see below
⟦ σ S D ⟧ X = Σ[ s : S ] ⟦ D s ⟧ X
```

The operator ℙ computes the product of a finite number of types in a type family, whose indices are given in a list:

```
ℙ : {I : Set} → List I → (I → Set) → Set
```

$$\begin{aligned}\mathbb{P}\ [\,]\ \ X &= \top\\ \mathbb{P}\ (i :: is)\ X &= X\,i \times \mathbb{P}\ is\ X\end{aligned}$$

Thus, in a response, given $X : I \to \mathsf{Set}$, we are allowed to form dependent sums (by σ) and the product of a finite number of types in X (via v, suggesting variable positions in the base functor).

Convention. We informally refer to the index part of a σ as a **field** of the datatype. Like Σ, we sometimes regard σ as a binder and write $\sigma[\,s : S\,]\ D\ s$ for $\sigma\ S\ (\lambda s \mapsto D\ s)$. □

Example (*natural numbers*)**.** The datatype of natural numbers is considered to be an inductive family trivially indexed by $\top$, so the declaration of Nat corresponds to an inhabitant of $\mathsf{Desc}\ \top$.

$$\begin{aligned}&\textbf{data}\ \mathsf{ListTag} : \mathsf{Set}\ \textbf{where}\\ &\quad \text{`}\mathsf{nil} \ : \mathsf{ListTag}\\ &\quad \text{`}\mathsf{cons} : \mathsf{ListTag}\\ &\mathit{NatD} : \mathsf{Desc}\ \top\\ &\mathit{NatD}\ \blacksquare = \sigma\ \mathsf{ListTag}\ \lambda\ \{\ \text{`}\mathsf{nil} \mapsto \mathsf{v}\ [\,]\\ &\qquad\qquad\qquad\qquad ;\ \text{`}\mathsf{cons} \mapsto \mathsf{v}\ (\blacksquare :: [\,])\ \}\end{aligned}$$

The index request is necessarily $\blacksquare$, and we respond with a field of type $\mathsf{ListTag}$ representing the constructor choices. A pattern-matching lambda function follows, which computes the trailing responses to the two possible values ‘nil and ‘cons for the field: if the field receives ‘nil, then we are constructing zero, which takes no recursive values, so we write $\mathsf{v}\ [\,]$ to end this branch; if the $\mathsf{ListTag}$ field receives ‘cons, then we are constructing a successor, which takes a recursive value at index $\blacksquare$, so we write $\mathsf{v}\ (\blacksquare :: [\,])$. □

Example (*lists*)**.** The datatype of lists is parametrised by the element type. We represent parametrised descriptions simply as functions producing descriptions, so the declaration of lists corresponds to a function taking element types to descriptions.

$$\begin{aligned}&\mathit{ListD} : \mathsf{Set} \to \mathsf{Desc}\ \top\\ &\mathit{ListD}\ A\ \blacksquare = \sigma\ \mathsf{ListTag}\ \lambda\ \{\ \text{`}\mathsf{nil} \mapsto \mathsf{v}\ [\,]\end{aligned}$$

$$; \text{`cons} \mapsto \sigma[\_ : A] \ \mathsf{v} \ (\blacksquare :: []) \ \}$$

ListD A is the same as *NatD* except that, in the ‘cons case, we use σ to insert a field of type *A* for storing an element. □

Example (*vectors*)**.** The datatype of vectors is parametrised by the element type and (nontrivially) indexed by Nat, so the declaration of vectors corresponds to

$$\begin{aligned}
&\mathit{VecD} : \mathsf{Set} \to \mathsf{Desc\ Nat} \\
&\mathit{VecD}\ A\ \mathsf{zero} \quad = \mathsf{v}\ [] \\
&\mathit{VecD}\ A\ (\mathsf{suc}\ n) \ = \ \sigma[_ : A]\ \mathsf{v}\ (n :: [])
\end{aligned}$$

which is directly comparable to the index-first base functor *VecF′* at the beginning of this section. □

There is no problem defining functions on the encoded datatypes except that it has to be done with the raw representation. For example, list append is defined by

$$\begin{aligned}
&_+\!\!+_ : \mu\ (\mathit{ListD}\ A)\ \blacksquare \to \mu\ (\mathit{ListD}\ A)\ \blacksquare \to \mu\ (\mathit{ListD}\ A)\ \blacksquare \\
&\mathsf{con}\ (\text{`nil} \quad , \qquad \blacksquare) +\!\!+ \mathit{bs} \ = \ \mathit{bs} \\
&\mathsf{con}\ (\text{`cons} , a , \mathit{as} , \blacksquare) +\!\!+ \mathit{bs} \ = \ \mathsf{con}\ (\text{`cons} , a , \mathit{as} +\!\!+ \mathit{bs} , \blacksquare)
\end{aligned}$$

To improve readability, we define the following higher-level terms:

$$\begin{aligned}
&\mathsf{List} : \mathsf{Set} \to \mathsf{Set} \\
&\mathsf{List}\ A \ = \ \mu\ (\mathit{ListD}\ A)\ \blacksquare \\
&[] : \{A : \mathsf{Set}\} \to \mathsf{List}\ A \\
&[] \ = \ \mathsf{con}\ (\text{`nil} , \blacksquare) \\
&_::_ : \{A : \mathsf{Set}\} \to A \to \mathsf{List}\ A \to \mathsf{List}\ A \\
&a :: \mathit{as} \ = \ \mathsf{con}\ (\text{`cons} , a , \mathit{as} , \blacksquare)
\end{aligned}$$

List append can then be rewritten in the usual form (assuming that the terms [] and _::_ can be used in pattern matching, which is not actually allowed in Agda, though):

$$\begin{aligned}
&_+\!\!+_ : \mathsf{List}\ A \to \mathsf{List}\ A \to \mathsf{List}\ A \\
&[] \qquad +\!\!+ \mathit{bs} \ = \ \mathit{bs}
\end{aligned}$$

```
mutual
  fold : {I : Set} {D : Desc I} {X : I → Set} → (𝕀 D X ⇉ X) → (μ D ⇉ X)
  fold {I} {D} f {i} (con ds) = f (mapFold D (D i) f ds)

  mapFold : {I : Set} (D : Desc I) (D′ : RDesc I) →
            {X : I → Set} → (𝕀 D X ⇉ X) → ⟦ D′ ⟧ (μ D) → ⟦ D′ ⟧ X
  mapFold D (v [])       f ▪        = ▪
  mapFold D (v (i :: is)) f (d , ds) = fold f d , mapFold D (v is) f ds
  mapFold D (σ S D′)     f (s , ds) = s , mapFold D (D′ s) f ds
```

Figure 2.1 Definition of the datatype-generic *fold* operator.

```
(a :: as) ++ bs = a :: (as ++ bs)
```

Later on, when an encoded datatype is defined, we almost always supply a corresponding index-first datatype declaration immediately afterwards, which is thought of as giving definitions of higher-level terms for type and data constructors — the terms List, [], and _::_ above, for example, can be considered to be defined by the index-first declaration of lists given in Section 2.4.1. Index-first declarations will only be regarded in this thesis as informal hints at how encoded datatypes are presented at a higher level; we do not give a formal treatment of the elaboration process from index-first declarations to corresponding descriptions and definitions of higher-level terms. (One such treatment was given by Dagand and McBride [2012a].)

Direct function definitions by pattern matching work fine for individual datatypes, but when we need to define operations and to state properties for all the datatypes encoded by the universe, it is necessary to have a generic *fold* operator parametrised by descriptions:

```
fold : {I : Set} {D : Desc I} {X : I → Set} → (𝕀 D X ⇉ X) → (μ D ⇉ X)
```

There is also a generic *induction* operator, which can be used to prove generic propositions about all encoded datatypes and subsumes *fold*, but *fold* is much easier to use when the full power of *induction* is not required. The implemen-

tations of both operators are adapted for our two-level universe from those in McBride's original work [2011]. We look at the implementation of the *fold* operator only, which is shown in Figure 2.1. As McBride, we would have wished to define *fold* by

$$\begin{aligned}
&\mathit{fold} \;:\; \{I \;:\; \mathsf{Set}\}\;\{D \;:\; \mathsf{Desc}\;I\}\;\{X \;:\; I \to \mathsf{Set}\} \to (\mathbb{F}\;D\;X \rightrightarrows X) \to (\mu\;D \rightrightarrows X)\\
&\mathit{fold}\;\{I\}\;\{D\}\;f\;\{i\}\;(\mathsf{con}\;\mathit{ds}) \;=\; f\;(\mathit{mapRD}\;(D\;i)\;(\mathit{fold}\;f)\;\mathit{ds})
\end{aligned}$$

where the functorial mapping *mapRD* on response structures is defined by

$$\begin{aligned}
&\mathit{mapRD} \;:\; \{I \;:\; \mathsf{Set}\}\;(D \;:\; \mathsf{RDesc}\;I) \to\\
&\qquad\qquad \{X\;Y \;:\; I \to \mathsf{Set}\}\;(g \;:\; X \rightrightarrows Y) \to [\![\; D \;]\!]\;X \to [\![\; D \;]\!]\;Y\\
&\mathit{mapRD}\;(\mathsf{v}\;[\,])\qquad g\;\blacksquare \qquad\quad = \;\blacksquare\\
&\mathit{mapRD}\;(\mathsf{v}\;(i :: \mathit{is}))\;g\;(x\;,\;\mathit{xs}) \;=\; g\;x\;,\;\mathit{mapRD}\;(\mathsf{v}\;\mathit{is})\;g\;\mathit{xs}\\
&\mathit{mapRD}\;(\sigma\;S\;D)\quad g\;(s\;,\;\mathit{xs}) \;=\; s\;,\;\mathit{mapRD}\;(D\;s)\;g\;\mathit{xs}
\end{aligned}$$

Agda does not see that this definition of *fold* is terminating, however, since the termination checker does not expand the definition of *mapRD* to see that *fold f* is applied to structurally smaller arguments. To make termination obvious to Agda, we instead define *fold* mutually recursively with *mapFold*, which is *mapRD* specialised by fixing its argument g to *fold f*.

It is helpful to form a two-dimensional image of our datatype manufacturing scheme: We manufacture a datatype by first defining a base functor, and then recursively duplicating the functorial structure by taking its least fixed point. The shape of the base functor can be imagined to stretch horizontally, whereas the recursive structure generated by the least fixed point grows vertically. This image works directly when the recursive structure is linear, like lists. (Otherwise one resorts to the abstraction of functor composition.) For example, we can typeset a list two-dimensionally like

```
con (‘cons , a ,
con (‘cons , b ,
con (‘nil   ,
  ▪) , ▪) , ▪)
```

Ignoring the last line of trailing ▪'s, things following con on each line — including constructor tags and list elements — are shaped by the base functor of

lists, whereas the con nodes, aligned vertically, are generated by the least fixed point.

Remark (*first-order vs higher-order representation*)**.** The functorial structures generated by descriptions are strongly reminiscent of **indexed containers** [Altenkirch and Morris, 2009]; this will be explored and exploited in Chapter 6. For now, it is enough to mention that we choose to stick to a first-order datatype manufacturing scheme, i.e., the datatypes we manufacture with descriptions use finite product types rather than dependent function types for branching, but it is easy to switch to a higher-order representation that is even closer to indexed containers (allowing infinite branching) by storing in v a collection of I-indices indexed by an arbitrary set S:

$$\mathsf{v} \,:\, (S \,:\, \mathsf{Set})\ (f \,:\, S \to I) \to \mathsf{RDesc}\ I$$

whose semantics is defined in terms of dependent functions:

$$[\![\ \mathsf{v}\ S\ f\ ]\!]\ X \;=\; (s \,:\, S) \to X\ (f\ s)$$

The reason for choosing to stick to first-order representation is simply to obtain a simpler equality for the manufactured datatypes (Agda's default equality would suffice); the examples of manufactured datatypes in this thesis are all finitely branching and do not require the power of higher-order representation anyway. This choice, however, does complicate some subsequent datatype-generic definitions (e.g., ornaments in Chapter 3). It would probably be helpful to think of the parts involving v and $\mathbb{P}$ in these definitions as specialisations of higher-order representations to first-order ones. □

2.5 Externalism and internalism

The use of "such that" to describe objects that have certain properties is universal in mathematics. If the objects in question have type A, then objects with certain properties form a subset of A, and using "such that" to describe such objects means that the subset is formed by specifying a suitable predicate on A. In type theory, this can be modelled by Σ-types of dependent pairs.

In a type $\Sigma\ A\ P$, when A is interpreted as a ground set and P as a predicate on A, an inhabitant of $\Sigma\ A\ P$ is an inhabitant a of A paired with a proof that $P\ a$ holds. When programs are the objects we reason about, this style naturally suggests a distinction between programs and proofs: programs are written in the first place, and proofs are conducted afterwards with reference to existing programs and do not interfere with their execution. This conception underlies many developments in type theory and theorem proving. For example: Luo [1994] consistently argued that proofs should not be identified with programs, one of the reasons being that logic should be regarded as independent from the objects being reasoned about. A type theory of subsets was given by Nordström et al. [1990] to suppress the second component — i.e., the proof part — of Σ-types. The proof assistant Coq [Bertot and Castéran, 2004] uses a type-theoretic foundation [Coquand and Huet, 1988; Coquand and Paulin-Mohring, 1990] which distinguishes programs and proofs, and proofs are written in a tactic-based language, differently from programs; it is also famous for the ability to extract executable portions of proof scripts into programs [Paulin-Mohring, 1989; Letouzey, 2003], as used in the CompCert project developing a verified C compiler [Leroy, 2009], for example.

On the other hand, having unified programs and proofs in type theory (Section 2.1), it seems a pity if the unification is not exploited to a deeper level. In Dijkstra's proposal for program correctness by construction, proofs and programs should be conceived "hand in hand", and the unification of programs and proofs brings us unprecedentedly closer to this ideal, since we can start thinking about programs that also serve as correctness proofs themselves. The dependently typed programming community has been exploring the use of inductive families not only for defining predicates on data (like $\_\leqslant_{\mathsf{N}}\_$) but also for representing data with embedded constraints (like Vec). Programs manipulating such datatypes would also deal with the embedded constraints and are thus correct by construction. Vector append is a classic (albeit somewhat trivial) example: Defining addition on natural numbers as

$$\begin{array}{ll} \_+\_ : \mathsf{Nat} \to \mathsf{Nat} \to \mathsf{Nat} & \\ \mathsf{zero} \quad + n = n & \end{array}$$

$$(\mathsf{suc}\ m) + n \ = \ \mathsf{suc}\ (m + n)$$

vector append is then defined by

$$\begin{aligned}
&_{+\!\!+}_ \ : \ \{A \ : \ \mathsf{Set}\}\ \{m\ n \ : \ \mathsf{Nat}\} \to \mathsf{Vec}\ A\ m \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (m + n) \\
&[\,] \qquad {+\!\!+}\ ys \ = \ ys \\
&(x :: xs) \ {+\!\!+}\ ys \ = \ x :: (xs \ {+\!\!+}\ ys)
\end{aligned}$$

The program for vector append looks exactly like the one for list append except for the more informative type, which makes it possible for the program to meet its specification — the length of the result of append should be the sum of the lengths of the two input lists — by construction. For list append, whose type uses the plain list datatype, we need to separately produce the following proof:

$$\begin{aligned}
&\mathit{append\text{-}length} \ : \\
&\quad \{A \ : \ \mathsf{Set}\}\ (xs\ ys \ : \ \mathsf{List}\ A) \to \mathit{length}\ (xs \ {+\!\!+}\ ys) \ \equiv \ \mathit{length}\ xs + \mathit{length}\ ys \\
&\mathit{append\text{-}length}\ [\,] \qquad ys \ = \ \mathsf{refl} \\
&\mathit{append\text{-}length}\ (x :: xs)\ ys \ = \ \mathit{cong}\ \mathsf{suc}\ (\mathit{append\text{-}length}\ xs\ ys)
\end{aligned}$$

But by switching to the vector datatype, the proof dissolves into the typing of the program and needs no separate handling. This is possible because list append and the proof *append-length* share the same structure; consequently, by careful type design, the vector append program alone is able to carry both of them simultaneously. Ko and Gibbons [2011] proposed to call this programming style **internalism**, suggesting that proofs are internalised in programs, while the traditional proving-after-programming style is called **externalism**. Externalism is necessary when we wish to show that an existing program satisfies additional properties, especially when the proofs are complicated and do not follow the structure of the program. On the other hand, writing a (simply typed) program and an externalist proof following the structure of the program is like stating the same thing twice: even though the programmer knows the meaning of the program, s/he has to first state the meaningless symbol manipulation aspect and then explain its meaning via a separate proof, doubling the effort. In contrast, internalism is programming with informative types so as to give precise description of meaningful computations, so explanations via separate proofs become unnecessary. As McBride [2004] aptly put it, internal-

ism makes programs and their explanations via proofs "not merely coexist but coincide". In addition, by encoding meanings in types, semantic considerations are (at least partially) reduced to syntax and can be aided mechanically — Agda's interactive development environment (Section 2.2.1) is one form of such aid. With interactive development, internalist types not only passively rule out nonsensical programs, but can actively provide helpful information to guide program construction. We will see several examples of such "type-directed programming" in this thesis (notably in Section 5.3).

Historical remark. The distinction between internalism and externalism at least goes back to simply typed λ-calculus á la Church and Curry — a Church-style λ-term always appears with its type and they together denote a value of the type, whereas Curry-style λ-terms are untyped and all reside in a common domain, and a term is later stated to have additional properties via typing. Reynolds [2000] called the two styles as giving "intrinsic" and "extrinsic" semantics to a language, and established relationship between the two kinds of semantics via a logical relations theorem. Our emphasis is on a more practical aspect about how intrinsic and extrinsic semantics can affect the attitude and approach to program construction. □

Internalism comes with its own problems, however. For one: internalist type design is difficult, yet there is almost no effective guideline or discipline for such design. Carelessly designed internalist types can lead to less natural programs. A simple example is when we switch the order of the arguments of the addition in the type of vector append,

$$
\begin{aligned}
&\_{+\!\!+}\_ : \{A : \mathsf{Set}\}\ \{m\ n : \mathsf{Nat}\} \to \mathsf{Vec}\ A\ m \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (n + m)\\
&[]\ \ {+\!\!+}\ ys\ =\ \mathit{subst}\ (\mathsf{Vec}\ A)\ \{\, n \equiv n + \mathsf{zero}\,\}_0\ ys\\
&(x :: xs)\ {+\!\!+}\ ys\ =\ \mathit{subst}\ (\mathsf{Vec}\ A)\ \{\, \mathsf{suc}\ (n + m) \equiv n + \mathsf{suc}\ m\,\}_1\ (x :: (xs\ {+\!\!+}\ ys))
\end{aligned}
$$

we would be forced to perform two type-casts that could have been avoided. Not only does the program look ugly, but this can also cause a cascade effect when we need to write other programs whose types depend on this program (e.g., externalist proofs about it), which would have to deal with the type-casts. McBride [2012] gave a more interesting example: to prove the polynomial test-

ing principle (two polynomial functions of degree n are pointwise equal if they agree at $n+1$ different points), McBride started with a datatype of encodings of polynomial functions indexed by degree but, after trying to program with the datatype, quickly found out that the datatype should instead be indexed by an arbitrary upper bound of the degree so a relaxed form of the polynomial testing principle can be naturally programmed (two polynomial functions of degree **at most** n are pointwise equal if they agree at $n+1$ different points). Scalability is another issue, especially when the only tool we have is the primitive language of datatype declarations: writing a datatype declaration with a sophisticated property internalised is comparable to programming a sophisticated algorithm in assembly language, and understanding the meaning of a complicated datatype declaration takes thorough reading and some inductive guessing and reasoning. In short, the complexity of internalist types has made type design a nontrivial programming problem, and we are in serious lack of type-level programming support.

Internalist library design also poses a problem: Since internalism requires differently indexed versions of the same data structure, an internalist library should provide more or less the same set of operations for all possible variants of the data structure. Without a way to manage these formally unrelated datatypes and operations modularly, an ad hoc library would need to duplicate the same structure and logic for all the variants and becomes hard to expand. For example, suppose that we have constructed a library for lists that include vectors, ordered lists, and ordered vectors, and now wish to add a new flavour of lists, say, association lists indexed with the list of keys. Operations need to be reimplemented not only for such key-indexed lists, but also for key-indexed vectors, ordered key-indexed lists, and ordered key-indexed vectors, even though key-indexing is the only new feature. An ideal structure for such a library would be having a separate module for each of the properties about length, ordering, and key-indexing. These modules can be developed independently, and there would be a way to assemble components in these modules at will — for example, ordered vectors and related operations would be synthesised from the components in the modules about length and ordering. This

ideal library structure calls for some form of composability of internalist datatypes and operations.

Composability has never been a problem for externalism, however. In an externalist list library, we would have only one basic list datatype and several predicates on lists about length, ordering, key-indexing, etc. Lists are "promoted" to vectors, ordered lists, or ordered vectors by simply pairing the list datatype with the length predicate, the ordering predicate, or the pointwise conjunction of the two predicates, respectively. Common operations are implemented for basic lists only, and their properties regarding length or ordering are proved independently and invoked when needed. Can we somehow introduce this beneficial composability to internalism as well? The answer is yes, because there are isomorphisms between externalist and internalist datatypes to be exploited.

To illustrate, let us go through a small case study about upgrading the *insert* function on lists (Section 2.2.2) for vectors, ordered lists, and ordered vectors. The externalist would define vectors as a Σ-type,

$$\begin{aligned}&\mathit{ExtVec} \;:\; \mathsf{Set} \to \mathsf{Nat} \to \mathsf{Set}\\ &\mathit{ExtVec}\ A\ n \;=\; \Sigma[\, xs : \mathsf{List}\ A \,]\ \mathit{length}\ xs \;\equiv\; n\end{aligned}$$

prove that *insert* increases the length of a list by one,

$$\begin{aligned}&\mathit{insert\text{-}length} \;:\; (y \;:\; \mathit{Val})\ \{xs \;:\; \mathsf{List}\ \mathit{Val}\}\ \{n \;:\; \mathsf{Nat}\} \to\\ &\qquad\qquad \mathit{length}\ xs \;\equiv\; n \to \mathit{length}\ (\mathit{insert}\ y\ xs) \;\equiv\; \mathsf{suc}\ n\end{aligned}$$

and define insertion on vectors as

$$\begin{aligned}&\mathit{insert}_{EV} \;:\; \mathit{Val} \to \{n \;:\; \mathsf{Nat}\} \to \mathit{ExtVec}\ \mathit{Val}\ n \to \mathit{ExtVec}\ \mathit{Val}\ (\mathsf{suc}\ n)\\ &\mathit{insert}_{EV}\ y\ (xs\ ,\ len) \;=\; \mathit{insert}\ y\ xs\ ,\ \mathit{insert\text{-}length}\ y\ len\end{aligned}$$

which processes the list and the length proof by *insert* and *insert-length* respectively. Similarly for ordered lists (indexed with a lower bound), the externalist would use the Σ-type

$$\begin{aligned}&\mathit{ExtOrdList} \;:\; \mathit{Val} \to \mathsf{Set}\\ &\mathit{ExtOrdList}\ b \;=\; \Sigma[\, xs : \mathsf{List}\ \mathit{Val} \,]\ \mathsf{Ordered}\ b\ xs\end{aligned}$$

where the Ordered predicate is defined by

$$\begin{array}{l}\textbf{indexfirst data}\ \mathsf{Ordered}\ :\ \mathit{Val} \to \mathsf{List}\ \mathit{Val} \to \mathsf{Set}\ \textbf{where}\\ \quad \mathsf{Ordered}\ b\ [\,] \quad\ \ni\ \mathsf{nil}\\ \quad \mathsf{Ordered}\ b\ (x :: xs)\ \ni\ \mathsf{cons}\ (\mathit{leq}\ :\ b \leqslant x)\ (\mathit{ord}\ :\ \mathsf{Ordered}\ x\ xs)\end{array}$$

Insertion on ordered lists is then

$$\begin{array}{l}\mathit{insert}_{EO}\ :\ (y\ :\ \mathit{Val})\ \{b\ :\ \mathit{Val}\} \to \mathit{ExtOrdList}\ b \to\\ \qquad\qquad \{b'\ :\ \mathit{Val}\} \to b' \leqslant y \to b' \leqslant b \to \mathit{ExtOrdList}\ b'\\ \mathit{insert}_{EO}\ y\ (xs\ ,\ ord)\ b'{\leqslant}y\ b'{\leqslant}b\ =\ \mathit{insert}\ y\ xs\ ,\ \mathit{insert\text{-}ordered}\ y\ ord\ b'{\leqslant}y\ b'{\leqslant}b\end{array}$$

where *insert-ordered* proves that *insert* preserves ordering:

$$\begin{array}{l}\mathit{insert\text{-}ordered}\ :\ (y\ :\ \mathit{Val})\ \{xs\ :\ \mathsf{List}\ \mathit{Val}\}\ \{b\ :\ \mathit{Val}\} \to \mathsf{Ordered}\ b\ xs \to\\ \qquad\qquad \{b'\ :\ \mathit{Val}\} \to b' \leqslant y \to b' \leqslant b \to \mathsf{Ordered}\ b'\ (\mathit{insert}\ y\ xs)\end{array}$$

Now the externalist has arrived at a modular list library (albeit a tiny one), which contains

- a basic module consisting of the basic list datatype and insertion on basic lists, and
- two independent upgrading modules about length and ordering, each consisting of a predicate on lists and a related proof about insertion.

It is easy to mix all three modules and get ordered vectors and insertion on them. The Σ-type uses the pointwise conjunction of the two predicates,

$$\begin{array}{l}\mathit{ExtOrdVec}\ :\ \mathit{Val} \to \mathsf{Nat} \to \mathsf{Set}\\ \mathit{ExtOrdVec}\ b\ n\ =\ \Sigma[\,xs : \mathsf{List}\ \mathit{Val}\,]\ \mathsf{Ordered}\ b\ xs \times \mathit{length}\ xs\ \equiv\ n\end{array}$$

and insertion simply uses *insert-ordered* and *insert-length* to process the two proofs bundled with a list:

$$\begin{array}{l}\mathit{insert}_{EOV}\ :\ (y\ :\ \mathit{Val})\ \{b\ :\ \mathit{Val}\}\ \{n\ :\ \mathsf{Nat}\} \to \mathit{ExtOrdVec}\ b\ n \to\\ \qquad\qquad \{b'\ :\ \mathit{Val}\} \to b' \leqslant y \to b' \leqslant b \to \mathit{ExtOrdVec}\ b'\ (\mathsf{suc}\ n)\\ \mathit{insert}_{EOV}\ y\ (xs\ ,\ ord\ ,\ len)\ b'{\leqslant}y\ b'{\leqslant}b\ =\ \mathit{insert}\ y\ xs\ ,\\ \qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad \mathit{insert\text{-}ordered}\ y\ ord\ b'{\leqslant}y\ b'{\leqslant}b\ ,\\ \qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad \mathit{insert\text{-}length}\ y\ len\end{array}$$

This is the kind of library we are looking for, except that the types are all externalist. The externalist and internalist types are not unrelated, however. For

example, internalist and externalist vectors are related by the representation-converting isomorphisms:

$$\mathit{Vec\text{-}iso}\ A\ :\ (n\ :\ \mathsf{Nat}) \to \mathsf{Vec}\ A\ n\ \cong\ \mathit{ExtVec}\ A\ n$$

Fixing $n\ :\ \mathsf{Nat}$, the left-to-right direction of the isomorphism

$$\mathsf{Iso}.\mathit{to}\ (\mathit{Vec\text{-}iso}\ A\ n)\ :\ \mathsf{Vec}\ A\ n \to \Sigma[\,xs\ :\ \mathsf{List}\ A\,]\ \mathit{length}\ xs\ \equiv\ n$$

computes the underlying list of a vector and a proof that the list has length n, and the right-to-left direction

$$\mathsf{Iso}.\mathit{from}\ (\mathit{Vec\text{-}iso}\ A\ n)\ :\ (\Sigma[\,xs\ :\ \mathsf{List}\ A\,]\ \mathit{length}\ xs\ \equiv\ n) \to \mathsf{Vec}\ A\ n$$

promotes a list to a vector when there is a proof that the list has length n. (The definition of $\_\cong\_$, which is actually an instance of a record type Iso, appears in Section 4.1.) To get insertion on internalist vectors, we convert the input vector to its externalist representation, make insert_{EV} do the work, and convert the result back to the internalist representation; more formally, the operation

$$\mathit{insert}_V\ :\ \mathit{Val} \to \{n\ :\ \mathsf{Nat}\} \to \mathsf{Vec}\ \mathit{Val}\ n \to \mathsf{Vec}\ \mathit{Val}\ (\mathsf{suc}\ n)$$

is defined by the commutative diagram:

$$\begin{array}{ccc}
\mathsf{Vec}\ \mathit{Val}\ n & \xrightarrow{\quad \mathit{insert}_V\ y \quad} & \mathsf{Vec}\ \mathit{Val}\ (\mathsf{suc}\ n) \\
\Big\downarrow{\scriptstyle \mathsf{Iso}.\mathit{to}\ (\mathit{Vec\text{-}iso}\ \mathit{Val}\ n)} & & \Big\uparrow{\scriptstyle \mathsf{Iso}.\mathit{from}\ (\mathit{Vec\text{-}iso}\ \mathit{Val}\ (\mathsf{suc}\ n))} \\
\Sigma[\,xs : \mathsf{List}\ \mathit{Val}\,]\ \ \mathit{length}\ xs\ \equiv\ n & \xrightarrow[\mathit{insert\text{-}length}\ y]{\mathit{insert}\ y\ *} & \Sigma[\,xs : \mathsf{List}\ \mathit{Val}\,]\ \ \mathit{length}\ xs\ \equiv\ \mathsf{suc}\ n
\end{array}$$

Similarly, we can get insertion for internalist ordered lists and ordered vectors from the externalist library by suitable conversion isomorphisms of the same form as *Vec-iso*. It is due to these conversion isomorphisms between internalist and externalist representations that we can **analyse** internalist datatypes into externalist components, which can then be modularly processed. This analysis of internalist datatypes and its application to modular library structuring is explored in Chapter 3 (in particular, the insertion example is resolved in Section 3.4.1).

The interconnecting isomorphisms between internalism and externalism also shed some light on supporting internalist type design. The **synthetic** direction of the interconnection goes from basic types and predicates to internalist types. It is conceivable that, for externalist predicates of some particular form, we can manufacture corresponding internalist types on the other side of the interconnection. The externalist side of the interconnection is usually kept non-dependently typed, so it is possible to use existing non-dependently typed calculi to derive suitable externalist predicates from specifications, which are then used to manufacture datatypes on the internalist side for type-directed programming. Chapter 5 presents one such approach, using relational calculus [Bird and de Moor, 1997] as a design language for internalist datatypes. Rather than improvising internalist types and hoping that they will work, we write specifications in the form of relational programs, which are amenable to algebraic transformation and can be much more concise and readable than the language of datatype declarations, making it easier to arrive at helpful and comprehensible internalist types.

To sum up: While internalism offers type-directed program construction and reduces the burden of producing correctness proofs, additional or more complicated properties of existing programs can only be established by externalism, which naturally gives rise to a modular organisation. Both internalism and externalism are here to stay, since the two styles serve different purposes and have their own pros and cons. Exploiting their interconnection can lead to interesting programming patterns, which the rest of this thesis explores.

Chapter 3

Refinements and ornaments

This chapter begins our exploration of the interconnection between internalism and externalism by looking at **the analytic direction**, i.e., the decomposition of sophisticated types into basic types and predicates on them. More specifically, we assume that the more sophisticated type and the basic type are known and there is straightforward evidence that they are related, and derive from the evidence an externalist predicate and a conversion isomorphism. As discussed in Section 2.5, one purpose of such decomposition is for internalist datatypes and operations to take a round trip to the externalist world so as to harvest composability there. The abstractions and constructions facilitating externalist composition of internalist datatypes and operations are developed in this chapter as follows:

- Conversion isomorphisms between internalist and externalist datatypes are axiomatised as **refinements** (Section 3.1).
- Refinements are coordinated by **upgrades** (Section 3.1.2) to enable switching from externalist function types to internalist ones, so simply typed operations satisfying suitable properties can be upgraded to have more sophisticated types.
- A class of refinements can be mechanically derived (Section 3.3) by marking differences between datatype descriptions with an adapted version of

McBride's **ornaments** [2011] (Section 3.2), which relate datatype descriptions that are vertically the same but horizontally different. (For example, the datatypes of vectors and lists can be related by an ornament.)

- Ornaments can be composed in parallel (Section 3.2.3), producing new composite internalist datatypes which correspond to pointwise conjunction of externalist predicates (Section 3.3.2). (For example, ordered vectors can be produced by composing the ornament from lists to ordered lists and the ornament from lists to vectors in parallel.)

3.1 Refinements

3.1.1 Refinements between individual types

A **refinement** from a basic type X to a more informative type Y is a **promotion predicate** $P : X \to \mathsf{Set}$ and a **conversion isomorphism** $i : Y \cong \Sigma\, X\, P$.

```
record Refinement (X Y : Set) : Set₁ where
  field
    P : X → Set
    i : Y ≅ Σ X P
  forget : Y → X
  forget = outl ∘ Iso.to i
```

Refinements are not guaranteed to be interesting in general. For example, Y can be chosen to be $\Sigma\ X\ P$ and the conversion isomorphism simply the identity. Most of the time, however, we will only be interested in refinements from basic types to their more informative — often internalist — variants. The conversion isomorphism tells us that the inhabitants of Y exactly correspond to the inhabitants of X bundled with more information, i.e., proofs that the promotion predicate P is satisfied. Computationally, any inhabitant of Y can be decomposed (by $\mathsf{Iso}.to\ i$) into an underlying value $x : X$ and a proof that x satisfies the promotion predicate P (which we will call a **promotion proof** for x), and conversely, if $x : X$ satisfies P, then it can be promoted (by $\mathsf{Iso}.from\ i$)

to an inhabitant of Y.

Example (*refinement from lists to ordered lists*)**.** Suppose A : Set is equipped with an ordering $\_\leqslant_A\_$. Fixing b : A, there is a refinement from List A to OrdList A $\_\leqslant_A\_$ b whose promotion predicate is Ordered A $\_\leqslant_A\_$ b, since we have an isomorphism of type

$$\mathsf{OrdList}\ A\ \_\leqslant_A\_\ b\ \cong\ \Sigma\ (\mathsf{List}\ A)\ (\mathsf{Ordered}\ A\ \_\leqslant_A\_\ b)$$

An ordered list of type OrdList A $\_\leqslant_A\_$ b can be decomposed into a list *as* : List A and a proof of type Ordered A $\_\leqslant_A\_$ b *as* that the list *as* is ordered and bounded below by b; conversely, a list satisfying Ordered A $\_\leqslant_A\_$ b can be promoted to an ordered list of type OrdList A $\_\leqslant_A\_$ b. □

XD

Example (*refinement from natural numbers to lists*)**.** Let A : Set. We have a refinement from Nat to List A

$$\mathit{Nat\text{-}List}\ A\ :\ \mathsf{Refinement}\ \mathsf{Nat}\ (\mathsf{List}\ A)$$

for which Vec A serves as the promotion predicate — there is a conversion isomorphism of type

$$\mathsf{List}\ A\ \cong\ \Sigma\ \mathsf{Nat}\ (\mathsf{Vec}\ A)$$

whose decomposing direction computes from a list its length and a vector containing the same elements. We might say that a natural number n : Nat is an incomplete list — the list elements are missing from the successor nodes of n. To promote n to a List A, we need to supply a vector of type Vec A n, i.e., n elements of type A. This example helps to emphasise that the notion of refinements is **proof-relevant**: An underlying value can have more than one promotion proofs, and consequently the more informative type in a refinement can have more elements than the basic type does. Thus it is more helpful to think that a type is more refined in the sense of being more informative rather than being a subset. □

In a refinement r, we denote the forgetful computation of underlying values — i.e., outl ∘ Iso.*to* (Refinement.*i* r) — as Refinement.*forget* r. The forgetful function is actually the core of a refinement, which is justified by the following facts:

- The forgetful function determines a refinement extensionally — if the forgetful functions of two refinements are extensionally equal, then their promotion predicates are pointwise isomorphic:

$$\begin{aligned}&\mathit{forget\text{-}iso} : \{X\ Y : \mathsf{Set}\}\ (r\ s : \mathsf{Refinement}\ X\ Y) \to \\ &\qquad (\mathsf{Refinement}.\mathit{forget}\ r \doteq \mathsf{Refinement}.\mathit{forget}\ s) \to \\ &\qquad (x : X) \to \mathsf{Refinement}.P\ r\ x \cong \mathsf{Refinement}.P\ s\ x\end{aligned}$$

- From any function f, we can construct a **canonical refinement** which uses a simplistic promotion predicate and has f as its forgetful function:

$$\begin{aligned}&\mathit{canonRef} : \{X\ Y : \mathsf{Set}\} \to (Y \to X) \to \mathsf{Refinement}\ X\ Y \\ &\mathit{canonRef}\ \{X\}\ \{Y\}\ f = \mathbf{record} \\ &\quad \{\ P = \lambda x \mapsto \Sigma[\,y : Y\,]\ f\ y \equiv x \\ &\quad ;\ i = \mathbf{record}\ \{\ \mathit{to} \quad = f \vartriangle (\mathit{id} \vartriangle (\lambda y \mapsto \mathsf{refl})) \\ &\qquad\qquad\ ;\ \mathit{from} = \mathsf{outl} \circ \mathsf{outr} \\ &\qquad\qquad\ ;\ \text{proofs of laws}\ \}\ \}\end{aligned}$$

  We call $\lambda x \mapsto \Sigma[\,y : Y\,]\ f\ y \equiv x$ the **canonical promotion predicate**, which says that, to promote $x : X$ to type Y, we are required to supply a complete $y : Y$ and prove that its underlying value is x.

- For any refinement $r : \mathsf{Refinement}\ X\ Y$, its forgetful function is exactly that of *canonRef* $(\mathsf{Refinement}.\mathit{forget}\ r)$, so from *forget-iso* we can prove that a promotion predicate is always pointwise isomorphic to the canonical promotion predicate:

$$\begin{aligned}&\mathit{coherence} : \{X\ Y : \mathsf{Set}\}\ (r : \mathsf{Refinement}\ X\ Y) \to \\ &\qquad (x : X) \to \mathsf{Refinement}.P\ r\ x \\ &\qquad\qquad\qquad \cong \Sigma[\,y : Y\,]\ \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x \\ &\mathit{coherence}\ r\ x = \mathit{forget\text{-}iso}\ r\ (\mathit{canonRef}\ (\mathsf{Refinement}.\mathit{forget}\ r))\ (\lambda y \mapsto \mathsf{refl})\end{aligned}$$

  This is closely related to an alternative "coherence-based" definition of refinements, which will shortly be discussed.

The refinement mechanism's purpose of being is thus to express intensional (representational) optimisations of the canonical promotion predicate, such that it is possible work on just the residual information of the more refined

type that is not present in the basic type.

Example (*promoting lists to ordered lists*)**.** Consider the refinement from lists to ordered lists using Ordered as its promotion predicate. A promotion proof of type Ordered $A\ \_{\leqslant_A}\_\ b\ as$ for the list as consists of only the inequality proofs necessary for ensuring that as is ordered and bounded below by b. Thus, to promote a list to an ordered list, we only need to supply the inequality proofs without providing the list elements again. □

Coherence-based definition of refinements

There is an alternative definition of refinements which, instead of the conversion isomorphism, postulates the forgetful computation and characterises the promotion predicate in term of it:

$$
\begin{array}{l}
\textbf{record}\ \mathsf{Refinement}'\ (X\ Y\ :\ \mathsf{Set})\ :\ \mathsf{Set}_1\ \textbf{where} \\
\quad \textbf{field} \\
\qquad P\ \ \ \ \ \ : X \to \mathsf{Set} \\
\qquad \mathit{forget}\ : Y \to X \\
\qquad p\ \ \ \ \ \ : (x\ :\ X) \to P\ x\ \cong\ \Sigma[\, y : Y \,]\ \mathit{forget}\ y\ \equiv\ x
\end{array}
$$

We say that $x\ :\ X$ and $y\ :\ Y$ are **in coherence** when $\mathit{forget}\ y\ \equiv\ x$, i.e., when x underlies y. The two definitions of refinements are equivalent. Of particular importance is the direction from Refinement to Refinement′:

$$
\begin{array}{l}
\mathit{toRefinement}'\ :\ \{X\ Y\ :\ \mathsf{Set}\} \to \mathsf{Refinement}\ X\ Y \to \mathsf{Refinement}'\ X\ Y \\
\mathit{toRefinement}'\ r\ =\ \textbf{record}\ \{\ P\ \ \ \ \ \ =\ \mathsf{Refinement}.P\ r \\
\qquad\qquad\qquad\qquad\qquad\quad ;\ \mathit{forget}\ =\ \mathsf{Refinement}.\mathit{forget}\ r \\
\qquad\qquad\qquad\qquad\qquad\quad ;\ p\ \ \ \ \ \ =\ \mathit{coherence}\ r\ \}
\end{array}
$$

We prefer the definition of refinements in terms of conversion isomorphisms because it is more concise and directly applicable to function upgrading. The coherence-based definition, however, is easier to generalise for function types, as we will see below.

3.1.2 Upgrades

Refinements are less useful when we move on to function types: the requirement that a conversion isomorphism exists between related function types is too strong, even when we have extensional equality for functions so isomorphisms between function types make more sense. For example, it is not — and should not be — possible to have a refinement from the function type $\mathsf{Nat} \to \mathsf{Nat}$ to the function type $\mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$, despite that the component types Nat and List Nat are related by a refinement: If such a refinement existed, we would be able to extract from any function $f : \mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$ an "underlying" function of type $\mathsf{Nat} \to \mathsf{Nat}$ which has roughly the same behaviour as f. However, the behaviour of a function taking a list may depend essentially on the list elements, which is not available to a function taking only a natural number. For example, a function of type $\mathsf{List}\ \mathsf{Nat} \to \mathsf{List}\ \mathsf{Nat}$ might compute the sum s of the input list and emit a list of length s whose elements are all zero. We cannot hope to write a function of type $\mathsf{Nat} \to \mathsf{Nat}$ that reproduces the corresponding behaviour on natural numbers.

It is only the decomposing direction of refinements that causes problem in the case of function types, however; the promoting direction is perfectly valid for function types. For example, to promote the function

$$
\begin{aligned}
&\mathit{double} : \mathsf{Nat} \to \mathsf{Nat} \\
&\mathit{double}\ \mathsf{zero} \quad\ = \mathsf{zero} \\
&\mathit{double}\ (\mathsf{suc}\ n) \ = \mathsf{suc}\ (\mathsf{suc}\ (\mathit{double}\ n))
\end{aligned}
$$

to a function of type $\mathsf{List}\ A \to \mathsf{List}\ A$ for some fixed $A : \mathsf{Set}$, we can use

$$Q = \lambda f \mapsto (n : \mathsf{Nat}) \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (\mathit{double}\ n)$$

as the promotion predicate: Consider the refinement from Nat to List A. Given a promotion proof of type $Q\ \mathit{double}$, say

Explain the meaning of this (scoping).

$$
\begin{aligned}
&\mathit{duplicate}' : (n : \mathsf{Nat}) \to \mathsf{Vec}\ A\ n \to \mathsf{Vec}\ A\ (\mathit{double}\ n) \\
&\mathit{duplicate}'\ \mathsf{zero}\quad [\,] \qquad\ = [\,] \\
&\mathit{duplicate}'\ (\mathsf{suc}\ n)\ (x :: xs) \ = x :: x :: \mathit{duplicate}'\ n\ xs
\end{aligned}
$$

we can synthesise a function $\mathit{duplicate} : \mathsf{List}\ A \to \mathsf{List}\ A$ by

definition of $*$

$$duplicate \;=\; \mathsf{Iso}.from\ i \circ (double * duplicate'\ \_) \circ \mathsf{Iso}.to\ i$$

i.e., we decompose the input list into the underlying natural number and a vector of elements, process the two parts separately with *double* and $duplicate'$, and finally combine the results back to a list. The relationship between the promoted function *duplicate* and the underlying function *double* is characterised by the coherence property [Dagand and McBride, 2012b]

definition of pointwise equality

$$double \circ length \;\doteq\; length \circ duplicate$$

or as a commutative diagram:



which states that *duplicate* preserves length as computed by *double*, or in more generic terms, processes the recursive structure (i.e., nil and cons nodes) of its input in the same way as *double* does.

We thus define **upgrades** to capture the promoting direction and the coherence property abstractly. An upgrade from $X : \mathsf{Set}$ to $Y : \mathsf{Set}$ is a promotion predicate $P : X \to \mathsf{Set}$, a coherence property $C : X \to Y \to \mathsf{Set}$ relating basic elements of type X and promoted elements of type Y, an upgrading (promoting) operation $u : (x : X) \to P\ x \to Y$, and a coherence proof $c : (x : X)\ (p : P\ x) \to C\ x\ (u\ x\ p)$ saying that the result of promoting a basic element $x : X$ must be in coherence with x.

record Upgrade $(X\ Y : \mathsf{Set}) : \mathsf{Set}_1$ **where**
  field
    $P : X \to \mathsf{Set}$
    $C : X \to Y \to \mathsf{Set}$
    $u : (x : X) \to P\ x \to Y$
    $c : (x : X)\ (p : P\ x) \to C\ x\ (u\ x\ p)$

Like refinements, arbitrary upgrades are not guaranteed to be interesting, but we will only use the upgrades synthesised by the combinators we define below specifically for deriving coherence properties and upgrading operations for function types from refinements between component types.

Upgrades from refinements

As we said, upgrades amount to only the promoting direction of refinements. This is most obvious when we look at the coherence-based refinements, of which upgrades are a direct generalisation: we get from Refinement′ to Upgrade by abstracting the notion of coherence and weakening the isomorphism to only the left-to-right computation. Any coherence-based refinement can thus be weakened to an upgrade,

$$
\begin{aligned}
&\mathit{toUpgrade'} : \{X\ Y : \mathsf{Set}\} \to \mathsf{Refinement'}\ X\ Y \to \mathsf{Upgrade}\ X\ Y \\
&\mathit{toUpgrade'}\ r = \mathbf{record}\ \{\ P = \mathsf{Refinement'}.P\ r \\
&\qquad\qquad\qquad\qquad ;\ C = \lambda\, x\ y \mapsto \mathsf{Refinement'}.\mathit{forget}\ r\ y \equiv x \\
&\qquad\qquad\qquad\qquad ;\ u = \lambda\, x \mapsto \mathsf{outl} \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement'}.p\ r\ x) \\
&\qquad\qquad\qquad\qquad ;\ c = \lambda\, x \mapsto \mathsf{outr} \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement'}.p\ r\ x)\ \}
\end{aligned}
$$

and consequently any refinement gives rise to an upgrade.

$$
\begin{aligned}
&\mathit{toUpgrade} : \{X\ Y : \mathsf{Set}\} \to \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ X\ Y \\
&\mathit{toUpgrade} = \mathit{toUpgrade'} \circ \mathit{toRefinement'}
\end{aligned}
$$

Composition of upgrades

The most representative combinator for upgrades is the following one for synthesising upgrades between function types:

$$
\begin{aligned}
&\_\rightharpoonup\_ : \{X\ Y\ Z\ W : \mathsf{Set}\} \to \\
&\qquad \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ Z\ W \to \mathsf{Upgrade}\ (X \to Z)\ (Y \to W)
\end{aligned}
$$

Note that there should be a refinement between the source types X and Y, rather than just an upgrade. (As a consequence, we can produce upgrades

between curried multi-argument function types but not between higher-order function types.) This is because, as we see in the *double–duplicate* example, we need the ability to decompose the source type Y.

Let r : Refinement X Y and s : Upgrade Z W. The upgrading operation takes a function $f : X \to Z$ and combines it with a promotion proof to get a function $g : Y \to W$, which should transform underlying values in coherence with f. That is, as g takes $y : Y$ to $g\ y : W$ at the more informative level, correspondingly at the underlying level the value Refinement.*forget* $r\ y : X$ underlying y should be taken by f to a value in coherence with $g\ y$. We thus define the statement "g is in coherence with f" as

$$(x : X)\ (y : Y) \to \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x \to \mathsf{Upgrade}.C\ s\ (f\ x)\ (g\ y)$$

As for the type of promotion proofs, since we already know that the underlying values are transformed by f, the missing information is only how the residual parts are transformed — that is, we need to know for any $x : X$ how a promotion proof for x is transformed to a promotion proof for $f\ x$. The type of promotion proofs for f is thus

$$(x : X) \to \mathsf{Refinement}.P\ r\ x \to \mathsf{Upgrade}.P\ s\ (f\ x)$$

Having determined the coherence property and the promotion predicate, it is then easy to construct the upgrading operation and the coherence proof. In particular, following the *double–duplicate* example, the upgrading operation breaks an input $y : Y$ into its underlying value $x = \mathsf{Refinement}.\mathit{forget}\ r\ y : X$ and a promotion proof for x, computes a promotion proof q for $f\ x : Z$ using the given promotion proof for f, and upgrades $f\ x$ to an inhabitant of type W using q. To sum up, the complete definition of $\_\rightharpoonup\_$ is

$$\begin{aligned}
&_\rightharpoonup_ : \{X\ Y\ Z\ W : \mathsf{Set}\} \to \\
&\qquad \mathsf{Refinement}\ X\ Y \to \mathsf{Upgrade}\ Z\ W \to \mathsf{Upgrade}\ (X \to Z)\ (Y \to W) \\
&r \rightharpoonup s = \mathbf{record} \\
&\quad \{ P = \lambda f \mapsto (x : X) \to \mathsf{Refinement}.P\ r\ x \to \mathsf{Upgrade}.P\ s\ (f\ x) \\
&\quad ; C = \lambda f\ g \mapsto (x : X)\ (y : Y) \to \\
&\qquad\qquad \mathsf{Refinement}.\mathit{forget}\ r\ y \equiv x \to \mathsf{Upgrade}.C\ s\ (f\ x)\ (g\ y) \\
&\quad ; u = \lambda f\ h \mapsto \mathsf{Upgrade}.u\ s\ _ \circ \mathit{uncurry}\ h \circ \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement}.i\ r)
\end{aligned}$$

$$; c \;=\; \lambda\,\{\, f\ h\ \mathit{inferred}\ y\ \mathsf{refl} \;\mapsto\; \mathbf{let}\ (x\ ,\ p) \;=\; \mathsf{Iso}.\mathit{to}\ (\mathsf{Refinement}.i\ r)\ y \\ \mathbf{in}\ \ \mathsf{Upgrade}.c\ s\ (f\ x)\ (h\ x\ p)\ \}\ \}$$

Example (*upgrade from* Nat → Nat *to* List A → List A)**.** Using the _⇀_ combinator on the refinement

$$r \;=\; \mathit{Nat\text{-}List}\ A \;:\; \mathsf{Refinement}\ \mathsf{Nat}\ (\mathsf{List}\ A)$$

and the upgrade derived from r, we get an upgrade

$$u \;=\; r \rightharpoonup \mathit{toUpgrade}\ r \;:\; \mathsf{Upgrade}\ (\mathsf{Nat} \to \mathsf{Nat})\ (\mathsf{List}\ A \to \mathsf{List}\ A)$$

The type $\mathsf{Upgrade}.P\ u\ \mathit{double}$ is exactly the type of $\mathit{duplicate}'$, and the type $\mathsf{Upgrade}.C\ u\ \mathit{double}\ \mathit{duplicate}$ is exactly the coherence property satisfied by *double* and *duplicate*. □

Comparison (*functional ornaments*)**.**

Dagand and McBride [2012b], origin of coherence property, no need to construct a universe (open for easy extension)

We can define more combinators for upgrades, like the ones in Figure 3.1. □

3.1.3 Refinement families

When we move on to consider refinements between indexed families of types, refinement relationship exists not only between the member types but also between the index sets: a type family $X : I \to \mathsf{Set}$ is refined by another type family $Y : J \to \mathsf{Set}$ when

- at the index level, there is a refinement r from I to J, and
- at the member type level, there is a refinement from $X\ i$ to $Y\ j$ whenever $i : I$ underlies $j : J$, i.e., $\mathsf{Refinement}.\mathit{forget}\ r\ j \equiv i$.

In short, each type $X\ i$ is refined by a collection of types in Y, the underlying values of their indices all being i. We will not exploit the full refinement structure on indices, though, so in the actual definition of **refinement families**

```
  -- the upgraded function type has an extra argument
new : {X : Set} (I : Set) {Y : I → Set} →
      (∀ i → Upgrade X (Y i)) → Upgrade X ((i : I) → Y i)
new I u = record { P = λ x ↦ ∀ i → Upgrade.P (u i) x
                 ; C = λ x y ↦ ∀ i → Upgrade.C (u i) x (y i)
                 ; u = λ x p i ↦ Upgrade.u (u i) x (p i)
                 ; c = λ x p i ↦ Upgrade.c (u i) x (p i) }

syntax new I (λ i ↦ u) = ∀⁺[ i : I ] u

  -- implicit version of new
new′ : {X : Set} (I : Set) {Y : I → Set} →
       (∀ i → Upgrade X (Y i)) → Upgrade X ({i : I} → Y i)
new′ I u = record { P = λ x ↦ ∀ {i} → Upgrade.P (u i) x
                  ; C = λ x y ↦ ∀ {i} → Upgrade.C (u i) x (y {i})
                  ; u = λ x p {i} ↦ Upgrade.u (u i) x (p {i})
                  ; c = λ x p {i} ↦ Upgrade.c (u i) x (p {i}) }

syntax new′ I (λ i ↦ u) = ∀⁺[[ i : I ]] u

  -- the underlying and the upgraded function types have a common argument
fixed : (I : Set) {X : I → Set} {Y : I → Set} →
        (∀ i → Upgrade (X i) (Y i)) → Upgrade ((i : I) → X i) ((i : I) → Y i)
fixed I u = record { P = λ f ↦ ∀ i → Upgrade.P (u i) (f i)
                   ; C = λ f g ↦ ∀ i → Upgrade.C (u i) (f i) (g i)
                   ; u = λ f h i ↦ Upgrade.u (u i) (f i) (h i)
                   ; c = λ f h i ↦ Upgrade.c (u i) (f i) (h i) }

syntax fixed I (λ i ↦ u) = ∀[ i : I ] u

  -- implicit version of fixed
fixed′ : (I : Set) {X : I → Set} {Y : I → Set} →
         (∀ i → Upgrade (X i) (Y i)) → Upgrade ({i : I} → X i) ({i : I} → Y i)
fixed′ I u = record { P = λ f ↦ ∀ {i} → Upgrade.P (u i) (f {i})
                    ; C = λ f g ↦ ∀ {i} → Upgrade.C (u i) (f {i}) (g {i})
                    ; u = λ f h {i} ↦ Upgrade.u (u i) (f {i}) (h {i})
                    ; c = λ f h {i} ↦ Upgrade.c (u i) (f {i}) (h {i}) }

syntax fixed′ I (λ i ↦ u) = ∀[[ i : I ]] u
```

Figure 3.1 More combinators for upgrades.

below, the index-level refinement degenerates into just the forgetful function.

$$\mathsf{FRefinement} : \{I\ J : \mathsf{Set}\}\ (e : J \to I)\ (X : I \to \mathsf{Set})\ (Y : J \to \mathsf{Set}) \to \mathsf{Set}_1$$
$$\mathsf{FRefinement}\ \{I\}\ e\ X\ Y\ =\ \{i : I\}\ (j : e^{-1}\ i) \to \mathsf{Refinement}\ (X\ i)\ (Y\ (\mathit{und}\ j))$$

Example (*refinement family from ordered lists to ordered vectors*)**.** The datatype $\mathsf{OrdList}\ A\ \_\leqslant_A\_ : A \to \mathsf{Set}$ is a family of types into which ordered lists are classified according to their lower bound. For each type of ordered lists having a particular lower bound, we can further classify them by their length, yielding $\mathsf{OrdVec}\ A\ \_\leqslant_A\_ : A \to \mathsf{Nat} \to \mathsf{Set}$. This further classification is captured as a refinement family of type

$$\mathsf{FRefinement}\ \mathsf{outl}\ (\mathsf{OrdList}\ A\ \_\leqslant_A\_)\ (\mathit{uncurry}\ (\mathsf{OrdVec}\ A\ \_\leqslant_A\_))$$

which consists of refinements from $\mathsf{OrdList}\ A\ \_\leqslant_A\_\ b$ to $\mathsf{OrdVec}\ A\ \_\leqslant_A\_\ b\ n$ for all $b : A$ and $n : \mathsf{Nat}$. □

relationship with ornaments

3.2 Ornaments

One possible way to establish relationships between datatypes is to write conversion functions. Conversions that involve only modifications of horizontal structures like copying, projecting away, or assigning default values to fields, however, may instead be stated at the level of datatype declarations, i.e., in terms of natural transformations between base functors. For example, a list is a natural number whose successor nodes are decorated with elements, and to convert a list to its length, we simply discard those elements. The essential information in this conversion is just that the elements associated with cons nodes should be discarded, which is described by the following natural transformation between the two base functors $\mathbb{F}\ (\mathit{ListD}\ A)$ and $\mathbb{F}\ \mathit{NatD}$:

$$\mathit{erase} : \{A : \mathsf{Set}\}\ \{X : \top \to \mathsf{Set}\} \to \mathbb{F}\ (\mathit{ListD}\ A)\ X \rightrightarrows \mathbb{F}\ \mathit{NatD}\ X$$
$$\mathit{erase}\ (\text{`nil}\ ,\ \blacksquare) = \text{`nil}\ ,\ \blacksquare \quad \text{-- `nil copied}$$
$$\mathit{erase}\ (\text{`cons}\ ,\ a\ ,\ x\ ,\ \blacksquare) = \text{`cons}\ ,\ x\ ,\ \blacksquare \quad \text{-- `cons copied, } a \text{ discarded,}$$

-- and x retained

The transformation can then be lifted to work on the least fixed points.

$$\begin{aligned}&\mathit{length} : \{A : \mathsf{Set}\} \to \mu\ (\mathit{ListD}\ A) \rightrightarrows \mu\ \mathit{NatD}\\ &\mathit{length}\ \{A\} = \mathit{fold}\ (\mathsf{con} \circ \mathit{erase}\ \{A\}\ \{\mu\ \mathit{NatD}\})\end{aligned}$$

Our goal in this section is to construct a universe for such horizontal natural transformations between the base functors arising as decodings of descriptions. The inhabitants of this universe are called **ornaments**. By encoding the relationship between datatype descriptions as a universe, whose inhabitants are analysable syntactic objects, we will not only be able to derive conversion functions between datatypes, but even compute new datatypes that are related to old ones in prescribed ways, which is something we cannot achieve if we simply write the conversion functions directly.

3.2.1 Universe construction

The definition of ornaments has the same two-level structure as that of datatype descriptions. We have an upper-level datatype Orn of ornaments

$$\begin{aligned}&\mathsf{Orn} : \{I\ J : \mathsf{Set}\}\ (e : J \to I)\ (D : \mathsf{Desc}\ I)\ (E : \mathsf{Desc}\ J) \to \mathsf{Set}_1\\ &\mathsf{Orn}\ e\ D\ E = \{i : I\}\ (j : e^{-1}\ i) \to \mathsf{ROrn}\ e\ (D\ i)\ (E\ (\mathit{und}\ j))\end{aligned}$$

which is defined in terms of a lower-level datatype ROrn of **response ornaments**, while ROrn contains the actual encoding of horizontal transformations and is decoded by the function *erase*:

$$\begin{aligned}&\mathbf{data}\ \mathsf{ROrn}\ \{I\ J : \mathsf{Set}\}\ (e : J \to I) : \mathsf{RDesc}\ I \to \mathsf{RDesc}\ J \to \mathsf{Set}_1\\ &\mathit{erase} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{RDesc}\ I\}\ \{E : \mathsf{RDesc}\ J\} \to\\ &\qquad \mathsf{ROrn}\ e\ D\ E \to \{X : I \to \mathsf{Set}\} \to [\![\ E\ ]\!]\ (X \circ e) \to [\![\ D\ ]\!]\ X\end{aligned}$$

The datatype Orn is parametrised by an erasure function $e : J \to I$ on the index sets and relates two datatype descriptions $D : \mathsf{Desc}\ I$ and $E : \mathsf{Desc}\ J$ such that from any ornament $O : \mathsf{Orn}\ e\ D\ E$ we can derive a forgetful map:

$$\mathit{forget}\ O : \mu\ E \rightrightarrows \mu\ D \circ e$$

By design, this forgetful map necessarily preserves the recursive structure of its input. In terms of the two-dimensional metaphor mentioned towards the end of Section 2.4.2, an ornament describes only how the horizontal shapes change, and the forgetful map — which is a *fold* — simply applies the changes to each vertical level; it never alters the vertical structure. For example, the *length* function discards elements associated with cons nodes, shrinking the list horizontally to a natural number, but keeps the vertical structure (i.e., the con nodes) intact. Look more closely: Given $y : \mu\ E\ j$, we should transform it into an inhabitant of type $\mu\ D\ (e\ j)$. Deconstructing y into con ys where $ys : [\![\ E\ j\ ]\!]\ (\mu\ E)$ and assuming that the $(\mu\ E)$–inhabitants at the recursive positions of ys have been inductively transformed into $(\mu\ D \circ e)$–inhabitants, we horizontally modify the resulting structure of type $[\![\ E\ j\ ]\!]\ (\mu\ D \circ e)$ to one of type $[\![\ D\ (e\ j)\ ]\!]\ (\mu\ D)$, which can then be wrapped by con to an inhabitant of type $\mu\ D\ (e\ j)$. The above steps are performed by the **ornamental algebra** induced by O:

$$
\begin{aligned}
&\mathit{ornAlg} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{Desc}\ I\}\ \{E : \mathsf{Desc}\ J\} \\
&\qquad (O : \mathsf{Orn}\ e\ D\ E) \to \mathbb{F}\ E\ (\mu\ D \circ e) \rightrightarrows \mu\ D \circ e \\
&\mathit{ornAlg}\ O\ \{j\} \ = \ \mathsf{con} \circ \mathit{erase}\ (O\ (\mathsf{ok}\ j))
\end{aligned}
$$

where the horizontal modification — a transformation from $[\![\ E\ j\ ]\!]\ (X \circ e)$ to $[\![\ D\ (e\ j)\ ]\!]\ X$, natural in X — is decoded by *erase* from a response ornament relating $D\ (e\ j)$ and $E\ j$. The forgetful function is then defined by

$$\mathit{forget}\ O \ = \ \mathit{fold}\ (\mathit{ornAlg}\ O)$$

Hence an ornament of type Orn $e\ D\ E$ contains, for each index request j, a response ornament of type ROrn $e\ (D\ (e\ j))\ (E\ j)$ to cope with all possible horizontal structures that can occur in a $(\mu\ E)$–inhabitant. The definition of Orn given above is a restatement of this in an intensionally more flexible form.

connection to refinement families

Now we look at the definitions of ROrn and *erase*, followed by explanations of the four cases.

```
data ROrn {I J : Set} (e : J → I) : RDesc I → RDesc J → Set₁ where
  v : {js : List J} {is : List I} (eqs : 𝔼 e js is) → ROrn e (v is) (v js)
  σ : (S : Set) {D : S → RDesc I} {E : S → RDesc J}
```

$$
\begin{aligned}
&\quad (O : (s : S) \to \mathsf{ROrn}\ e\ (D\ s)\ (E\ s)) \to \mathsf{ROrn}\ e\ (\sigma\ S\ D)\ (\sigma\ S\ E)\\
\Delta :\ &(T : \mathsf{Set})\ \{D : \mathsf{RDesc}\ I\}\ \{E : T \to \mathsf{RDesc}\ J\}\\
&\quad (O : (t : T) \to \mathsf{ROrn}\ e\ D\ (E\ t)) \to \mathsf{ROrn}\ e\ D\ (\sigma\ T\ E)\\
\nabla :\ &\{S : \mathsf{Set}\}\ (s : S)\ \{D : S \to \mathsf{RDesc}\ I\}\ \{E : \mathsf{RDesc}\ J\}\\
&\quad (O : \mathsf{ROrn}\ e\ (D\ s)\ E) \to \mathsf{ROrn}\ e\ (\sigma\ S\ D)\ E
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{erase} : \{I\ J : \mathsf{Set}\}\ \{e : J \to I\}\ \{D : \mathsf{RDesc}\ I\}\ \{E : \mathsf{RDesc}\ J\} \to\\
&\quad \mathsf{ROrn}\ e\ D\ E \to \{X : I \to \mathsf{Set}\} \to [\![\ E\ ]\!]\ (X \circ e) \to [\![\ D\ ]\!]\ X\\
&\mathit{erase}\ (\mathsf{v}\ [\,]\qquad\quad)\ \blacksquare\qquad\quad = \blacksquare\\
&\mathit{erase}\ (\mathsf{v}\ (\mathsf{refl} :: \mathit{eqs}))\ (x\ ,\ \mathit{xs}) = x\ ,\ \mathit{erase}\ (\mathsf{v}\ \mathit{eqs})\ \mathit{xs}\quad \text{-- } x \text{ retained}\\
&\mathit{erase}\ (\sigma\ S\ O)\qquad (s\ ,\ \mathit{xs}) = s\ ,\ \mathit{erase}\ (O\ s)\quad \mathit{xs}\quad \text{-- } s \text{ copied}\\
&\mathit{erase}\ (\Delta\ T\ O)\qquad (t\ ,\ \mathit{xs}) = \quad \mathit{erase}\ (O\ t)\quad \mathit{xs}\quad \text{-- } t \text{ discarded}\\
&\mathit{erase}\ (\nabla\ s\ O)\qquad\quad \mathit{xs} = s\ ,\ \mathit{erase}\ O\qquad \mathit{xs}\quad \text{-- } s \text{ inserted}
\end{aligned}
$$

The first two cases v and σ of ROrn relate response descriptions that have the same top-level constructor, and the transformations decoded from them preserve horizontal structure.

- The v case of ROrn states that a response description v *js* refines another response description v *is*, i.e., when $[\![\ \mathsf{v}\ \mathit{js}\ ]\!]\ (X \circ e)$ can be transformed into $[\![\ \mathsf{v}\ \mathit{is}\ ]\!]\ X$. The source type $[\![\ \mathsf{v}\ \mathit{js}\ ]\!]\ (X \circ e)$ expands to a product of types of the form $X\ (e\ j)$ for some $j : J$ and the target type $[\![\ \mathsf{v}\ \mathit{is}\ ]\!]\ X$ to a product of types of the form $X\ i$ for some $i : I$. There are no horizontal contents and thus no horizontal modifications to make, and the input values should be preserved. We thus demand that *js* and *is* have the same number of elements and the corresponding pairs of indices $e\ j$ and i are equal; that is, we demand a proof of $\mathit{map}\ e\ \mathit{js} \equiv \mathit{is}$ (where *map* is the usual functorial mapping on lists). To make it easier to analyse a proof of $\mathit{map}\ e\ \mathit{js} \equiv \mathit{is}$ in the v case of *erase*, we instead define the proposition inductively as $\mathbb{E}\ e\ \mathit{js}\ \mathit{is}$, where the datatype $\mathbb{E}$ is defined by

$$
\begin{aligned}
&\mathbf{data}\ \mathbb{E}\ \{I\ J : \mathsf{Set}\}\ (e : J \to I) : \mathsf{List}\ J \to \mathsf{List}\ I \to \mathsf{Set}\ \mathbf{where}\\
&\quad [\,] \quad\ : \mathbb{E}\ e\ [\,]\ [\,]\\
&\quad \_::\_ : \{j : J\}\ \{i : I\}\ (\mathit{eq} : e\ j \equiv i) \to\\
&\qquad \{\mathit{js} : \mathsf{List}\ J\}\ \{\mathit{is} : \mathsf{List}\ I\}\ (\mathit{eqs} : \mathbb{E}\ e\ \mathit{js}\ \mathit{is}) \to \mathbb{E}\ e\ (j :: \mathit{js})\ (i :: \mathit{is})
\end{aligned}
$$

- The σ case of ROrn states that σ S E refines σ S D, i.e., that both response descriptions start with the same field of type S. The intended semantics — the σ case of *erase* — is to preserve (copy) the value of this field. To be able to transform the rest of the input structure, we should demand that, for any value $s : S$ of the field, the remaining response description $E\ s$ refines the other remaining response description $D\ s$.

The other two cases Δ and ∇ of ROrn deal with mismatching fields in the two response descriptions being related and prompt *erase* to perform nontrivial horizontal transformations.

- The Δ case of ROrn states that σ T E refines D, the former having an additional field of type T whose value is not retained — the Δ case of *erase* discards the value of this field. We still need to transform the rest of the input structure, so the Δ constructor demands that, for every possible value $t : T$ of the field, the response description D is refined by the remaining response description $E\ t$.
- Conversely, the ∇ case of ROrn states that E refines σ S D, the latter having an additional field of type S. The value of this field needs to be restored by the ∇ case of *erase*, so the ∇ constructor demands a default value $s : S$ for the field. To be able to continue with the transformation, the ∇ constructor also demands that the response description E refines the remaining response description $D\ s$.

Convention. Again we regard Δ as a binder and write $\Delta[t : T]\ O\ t$ for $\Delta\ T\ (\lambda t \mapsto O\ t)$. Also, even though ∇ is not a binder, we write $\nabla[s]\ O$ for $\nabla\ s\ O$ to save the parentheses around O when O is a complex expression. □

Example (*ornament from natural numbers to lists*)**.** For any A : Set, there is an ornament from the description *NatD* of natural numbers to the description *ListD A* of lists:

```
NatD-ListD A : Orn ! NatD (ListD A)
NatD-ListD A (ok ▪) = σ ListTag λ { 'nil  ↦ v []
                                  ; 'cons ↦ Δ[ _ : A ] v (refl :: []) }
```

There is only one response ornament in *NatD-ListD A* since the datatype of

lists is trivially indexed. The constructor tag is preserved (σ ListTag), and, in the cons case, the list element field is marked as additional by Δ. Consequently, the forgetful function

$$\mathit{forget}\ (\mathit{NatD\text{-}ListD}\ A)\ \{\blacksquare\}\ :\ \mathsf{List}\ A \to \mathsf{Nat}$$

discards all list elements from a list and returns its underlying natural number, i.e., its length. □

Example (*ornament from lists to vectors*)**.** Again for any A : Set, there is an ornament from the description *ListD A* of lists to the description *VecD A* of vectors:

$$\begin{aligned}
&\mathit{ListD\text{-}VecD}\ A\ :\ \mathsf{Orn}\ !\ (\mathit{ListD}\ A)\ (\mathit{VecD}\ A)\\
&\mathit{ListD\text{-}VecD}\ A\ (\mathsf{ok}\ \mathsf{zero}\quad)\ =\ \nabla[\,\text{`}\mathsf{nil}\,]\quad \mathsf{v}\ []\\
&\mathit{ListD\text{-}VecD}\ A\ (\mathsf{ok}\ (\mathsf{suc}\ n))\ =\ \nabla[\,\text{`}\mathsf{cons}\,]\ \sigma[\,_ : A\,]\ \mathsf{v}\ (\mathsf{refl} :: [])\ \}
\end{aligned}$$

The response ornaments are indexed by Nat, since Nat is the index set of the datatype of vectors. We do pattern matching on the index request, resulting in two cases. In both cases, the constructor tag field exists for lists but not for vectors (since the constructor choice for vectors is determined from the index), so ∇ is used to insert the appropriate tag; in the suc case, the list element field is preserved by σ. Consequently, the forgetful function

$$\mathit{forget}\ (\mathit{ListD\text{-}VecD}\ A)\ :\ \{n\ :\ \mathsf{Nat}\} \to \mathsf{Vec}\ A\ n \to \mathsf{List}\ A$$

computes the underlying list of a vector. □

Remark (*vertical invariance of ornamental relationship*)**.** It is worth emphasising again that ornaments encode only horizontal transformations, so datatypes related by ornaments necessarily have the same recursion patterns (as enforced by the v constructor) — ornamental relationship exists between list-like datatypes but not between lists and binary trees, for example. □

3.2.2 Ornamental descriptions

There is apparent similarity between, e.g., the description *ListD A* and the ornament *NatD-ListD A*, which is typical: frequently we define a new description

```
data ROrnDesc {I : Set} (J : Set) (e : J → I) : RDesc I → Set₁ where
  v : {is : List I} (js : ℙ is (InvImage e)) → ROrnDesc J e (v is)
  σ : (S : Set) {D : S → RDesc I}
      (OD : (s : S) → ROrnDesc J e (D s)) → ROrnDesc J e (σ S D)
  Δ : (T : Set) {D : RDesc I} (OD : T → ROrnDesc J e D) → ROrnDesc J e D
  ∇ : {S : Set} (s : S) {D : S → RDesc I}
      (OD : ROrnDesc J e (D s)) → ROrnDesc J e (σ S D)

und-ℙ : {I J : Set} {e : J → I} (is : List I) → ℙ is (InvImage e) → List J
und-ℙ []       ▪       = []
und-ℙ (i :: is) (j , js) = und j :: und-ℙ is js

toRDesc : {I J : Set} {e : J → I} {D : RDesc I} → ROrnDesc J e D → RDesc J
toRDesc (v {is} js) = v (und-ℙ is js)
toRDesc (σ S OD)    = σ[ s : S ] toRDesc (OD s)
toRDesc (Δ T OD)    = σ[ t : T ] toRDesc (OD t)
toRDesc (∇ s OD)    = toRDesc OD

toEq-ℙ : {I J : Set} {e : J → I} (is : List I) (js : ℙ is (InvImage e)) → 𝔼 e (und-ℙ is js) is
toEq-ℙ []       ▪       = []
toEq-ℙ (i :: is) (j , js) = toEq j :: toEq-ℙ is js

toROrn : {I J : Set} {e : J → I} {D : RDesc I} →
         (OD : ROrnDesc J e D) → ROrn e D (toRDesc OD)
toROrn (v js)    = v (toEq-ℙ _ js)
toROrn (σ S OD)  = σ[ s : S ] toROrn (OD s)
toROrn (Δ T OD)  = Δ[ t : T ] toROrn (OD t)
toROrn (∇ s OD)  = ∇[ s ] (toROrn OD)

OrnDesc : {I : Set} (J : Set) (e : J → I) (D : Desc I) → Set₁
OrnDesc J e D = {i : I} (j : e ⁻¹ i) → ROrnDesc J e (D i)

⌊_⌋ : {I J : Set} {e : J → I} {D : Desc I} → OrnDesc J e D → Desc J
⌊ OD ⌋ j = toRDesc (OD (ok j))

⌈_⌉ : {I J : Set} {e : J → I} {D : Desc I} (OD : OrnDesc J e D) → Orn e D ⌊ OD ⌋
⌈ OD ⌉ (ok j) = toROrn (OD (ok j))
```

Figure 3.2 Definitions for ornamental descriptions.

(e.g. *ListD A*), intending it to be a more refined version of an existing one (e.g., *NatD*), and then immediately write an ornament from the latter to the former (e.g., *NatD-ListD A*). The syntactic structures of the new description and of the ornament are essentially the same, however, so the effort is duplicated. It would be more efficient if we could use the existing description as a template and just write a "relative description" specifying how to "patch" the template, and afterwards from this "relative description" extract a new description and an ornament from the template to the new description.

Ornamental descriptions are designed for this purpose; the definitions are shown in Figure 3.2 and closely follow the definitions for ornaments, having a upper-level type OrnDesc of ornamental descriptions which refers to a lower-level datatype ROrnDesc of response ornamental descriptions. An ornamental description looks like an annotated description, on which we can use a greater variety of constructors to mark differences from the template description. We think of an ornamental description

$$OD \;:\; \mathsf{OrnDesc}\ J\ e\ D$$

as simultaneously denoting a new description of type $\mathsf{Desc}\ J$ and an ornament from the template description D to the new description, and use floor and ceiling brackets $\lfloor \_ \rfloor$ and $\lceil \_ \rceil$ to resolve ambiguity: the new description is

$$\lfloor OD \rfloor \;:\; \mathsf{Desc}\ J$$

and the ornament is

$$\lceil OD \rceil \;:\; \mathsf{Orn}\ e\ D\ \lfloor OD \rfloor$$

Example (*ordered lists as an ornamentation of lists*)**.** Given $A \;:\; \mathsf{Set}$ with an ordering relation $\_\leqslant_{A}\_ \;:\; A \to A \to \mathsf{Set}$, we can define ordered lists on A by an ornamental description, using the description of lists as the template:

$$\begin{aligned}
&\mathit{OrdListOD}\ A\ _\leqslant_{A}_ \;:\; \mathsf{OrnDesc}\ A\ !\ (\mathit{ListD}\ A)\\
&\mathit{OrdListOD}\ A\ _\leqslant_{A}_\ (\mathsf{ok}\ b)\ =\\
&\quad \sigma\ \mathsf{ListTag}\ \lambda\ \{\ \text{`nil} \quad\ \mapsto\ \mathsf{v}\ \blacksquare\\
&\qquad\qquad ;\ \text{`cons} \quad\ \mapsto\ \sigma[\,a : A\,]\ \Delta[\,\mathit{leq} : b \leqslant_{A} a\,]\ \mathsf{v}\ (a\ ,\ \blacksquare)\ \}\\
&\textbf{indexfirst data}\ \mathsf{OrdList}\ A\ _\leqslant_{A}_ \;:\; A \to \mathsf{Set}\ \textbf{where}
\end{aligned}$$

$$\begin{array}{l}\mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ b\ \ni\ \mathsf{nil}\\ \qquad\qquad\qquad\qquad\ \mid\ \mathsf{cons}\ (a\ :\ A)\ (\mathit{leq}\ :\ b \leqslant_A a)\ (\mathit{as}\ :\ \mathsf{OrdList}\ A\ \_{\leqslant_A}\_\ a)\end{array}$$

If we read $\mathit{OrdListOD}\ A\ \_{\leqslant_A}\_$ as an annotated description, we can think of the *leq* field as being marked as additional (relative to the description of lists) by using Δ rather than σ. To decode $\mathit{OrdListOD}\ A\ \_{\leqslant_A}\_$ to an ordinary description of ordered lists, we write

$$\lfloor \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rfloor\ :\ \mathsf{Desc}\ A$$

and

$$\lceil \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rceil\ :\ \mathsf{Orn}\ !\ (\mathit{ListD}\ A)\ \lfloor \mathit{OrdListOD}\ A\ \_{\leqslant_A}\_ \rfloor$$

is an ornament from lists to ordered lists. □

Example (*singleton ornamentation*)**.** Consider the following **singleton datatype** for lists:

$$\begin{array}{l}\textbf{indexfirst data}\ \mathsf{ListS}\ A\ :\ \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\ \quad \mathsf{ListS}\ A\ [\,]\ \ni\ \mathsf{nil}\\ \quad \mathsf{ListS}\ A\ (x :: xs)\ \ni\ \mathsf{cons}\ (s\ :\ \mathsf{ListS}\ A\ xs)\end{array}$$

For each type $\mathsf{ListS}\ A\ xs$, there is exactly one (canonical) inhabitant (hence the name "singleton datatype" [Monnier and Haguenauer, 2010]), which has the same vertical structure as *xs* and is devoid of any horizontal contents. We can encode the datatype as an ornamental description relative to $\mathit{ListD}\ A$:

$$\begin{array}{l}\mathit{ListSOD}\ :\ (A\ :\ \mathsf{Set}) \to \mathsf{OrnDesc}\ (\mathsf{List}\ A)\ !\ (\mathit{ListD}\ A)\\ \mathit{ListSOD}\ A\ (\mathsf{ok}\ [\,]\qquad)\ =\ \nabla[\,\text{`}\mathsf{nil}\,]\ \mathsf{v}\ \blacksquare\\ \mathit{ListSOD}\ A\ (\mathsf{ok}\ (x :: xs))\ =\ \nabla[\,\text{`}\mathsf{cons}\,]\ \nabla[\,x\,]\ \mathsf{v}\ (\mathsf{ok}\ xs\ ,\ \blacksquare)\end{array}$$

which does pattern matching on the index request, in each case restricts the constructor choice to the one matched against, and in the cons case deletes the element field and sets the index of the recursive position to be the value of the tail in the pattern. In general, we can define a parametrised ornamental description

$$\mathit{singletonOD}\ :\ \{I\ :\ \mathsf{Set}\}\ (D\ :\ \mathsf{Desc}\ I) \to \mathsf{OrnDesc}\ (\Sigma\ I\ (\mu\ D))\ \mathsf{outl}\ D$$

called the **singleton ornamental description**, which delivers a singleton datatype as an ornamentation of any datatype. The complete definition is

$$
\begin{aligned}
&\mathit{erode} : \{I : \mathsf{Set}\}\ (D : \mathsf{RDesc}\ I)\ \{J : I \to \mathsf{Set}\} \to \\
&\qquad [\![\ D\ ]\!]\ J \to \mathsf{ROrnDesc}\ (\Sigma\ I\ J)\ \mathsf{outl}\ D \\
&\mathit{erode}\ (\mathsf{v}\ \mathit{is})\quad \mathit{js} \quad = \mathsf{v}\ (\mathbb{P}\text{-}\mathit{map}\ (\lambda\ \{i\}\ j \mapsto \mathsf{ok}\ (i\ ,\ j))\ \mathit{is}\ \mathit{js}) \\
&\mathit{erode}\ (\sigma\ S\ D)\ (s\ ,\ \mathit{js}) = \nabla[\,s\,]\ \mathit{erode}\ (D\ s)\ \mathit{js}
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{singletonOD} : \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I) \to \mathsf{OrnDesc}\ (\Sigma\ I\ (\mu\ D))\ \mathsf{outl}\ D \\
&\mathit{singletonOD}\ D\ (\mathsf{ok}\ (i\ ,\ \mathsf{con}\ \mathit{ds})) = \mathit{erode}\ (D\ i)\ \mathit{ds}
\end{aligned}
$$

where

$$
\begin{aligned}
&\mathbb{P}\text{-}\mathit{map} : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\} \to (X \rightrightarrows Y) \to \\
&\qquad (\mathit{is} : \mathsf{List}\ I) \to \mathbb{P}\ \mathit{is}\ X \to \mathbb{P}\ \mathit{is}\ Y \\
&\mathbb{P}\text{-}\mathit{map}\ f\ [\,]\qquad \blacksquare \qquad = \blacksquare \\
&\mathbb{P}\text{-}\mathit{map}\ f\ (i :: \mathit{is})\ (x\ ,\ \mathit{xs}) = f\ x\ ,\ \mathbb{P}\text{-}\mathit{map}\ f\ \mathit{is}\ \mathit{xs}
\end{aligned}
$$

Note that *erode* deletes all fields (i.e., horizontal contents), drawing default values from the index request, and retains only the vertical structure. We will see in Section 3.3 that singleton ornamentation plays a key role in the ornament–refinement framework. □

Remark (*ornaments as relations*)**.** We define ornaments as relations between descriptions (indexed with an erasure function), whereas the original ornaments [McBride, 2011; Dagand and McBride, 2012b] are rebranded as ornamental descriptions. One obvious advantage of relational ornaments is that they can arise between existing descriptions, whereas ornamental descriptions always produce (definitionally) new descriptions at the more informative end. A consequence is that there can be multiple ornaments between a pair of descriptions. For example, consider the following description of a datatype consisting of two fields of the same type:

$$
\begin{aligned}
&\mathit{SquareD} : (A : \mathsf{Set}) \to \mathsf{Desc}\ \top \\
&\mathit{SquareD}\ A\ \blacksquare = \sigma[\,\_ : A\,]\ \sigma[\,\_ : A\,]\ \mathsf{v}\ [\,]
\end{aligned}
$$

Between *SquareD A* and itself, we have the identity ornament

$$\lambda\ \{\ \blacksquare \mapsto\ \sigma[\,\_ : A\,]\ \sigma[\,\_ : A\,]\ \mathsf{v}\ [\,]\ \}$$

and the "swapping" ornament

$$\lambda\ \{\ \blacksquare \mapsto\ \Delta[\,x : A\,]\ \Delta[\,y : A\,]\ \nabla[\,y\,]\ \nabla[\,x\,]\ \mathsf{v}\ [\,]\ \}$$

whose forgetful function swaps the two fields.

The other advantage of relational ornaments is that they allow new datatypes to arise at the less informative end. For example, **coproduct of signatures** as used in, e.g., data types à la carte [Swierstra, 2008], can be implemented naturally with relational ornaments but not with ornamental descriptions. In more detail: Consider (a simplistic version of) **tagged descriptions** [Chapman et al., 2010], which are descriptions that, for any index request, always respond with a constructor field first. A tagged description with index set $I : \mathsf{Set}$ thus consists of a family of types $C : I \to \mathsf{Set}$, where each $C\ i$ is the set of constructor tags for the index request $i : I$, and a family of subsequent response descriptions for each constructor tag.

$$\begin{aligned}&\mathsf{TDesc} : \mathsf{Set} \to \mathsf{Set}_1\\&\mathsf{TDesc}\ I\ =\ \Sigma[\,C : I \to \mathsf{Set}\,]\ ((i\ :\ I) \to C\ i \to \mathsf{RDesc}\ I)\end{aligned}$$

Tagged descriptions are decoded to ordinary descriptions by

$$\begin{aligned}&\lfloor\_\rfloor_T\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{TDesc}\ I \to \mathsf{Desc}\ I\\&\lfloor C\ ,\ D\rfloor_T\ i\ =\ \sigma\ (C\ i)\ (D\ i)\end{aligned}$$

We can then define binary coproduct of tagged descriptions, which sums the corresponding constructor fields, as follows:

coproduct-related definitions

$$\begin{aligned}&\_\oplus\_\ :\ \{I\ :\ \mathsf{Set}\} \to \mathsf{TDesc}\ I \to \mathsf{TDesc}\ I \to \mathsf{TDesc}\ I\\&(C\ ,\ D) \oplus (C'\ ,\ D')\ =\ (\lambda\, i \mapsto C\ i + C'\ i)\ ,\ (\lambda\, i \mapsto D\ i \triangledown D'\ i)\end{aligned}$$

Now given two tagged descriptions $tD\ =\ (C\ ,\ D)$ and $tD'\ =\ (C'\ ,\ D')$ of type $\mathsf{TDesc}\ I$, there are two ornaments from $\lfloor tD \oplus tD'\rfloor_T$ to $\lfloor tD\rfloor_T$ and $\lfloor tD'\rfloor_T$

$$\begin{aligned}&\mathit{inlOrn}\ :\ \mathsf{Orn}\ \mathit{id}\ \lfloor tD \oplus tD'\rfloor_T\ \lfloor tD\rfloor_T\\&\mathit{inlOrn}\ (\mathsf{ok}\ i)\ =\ \Delta[\,c\ :C\ i\,]\ \ \nabla[\,\mathsf{inl}\ c\,]\ \ \mathit{idOrn}\ (D\ i\ c)\\&\mathit{inrOrn}\ :\ \mathsf{Orn}\ \mathit{id}\ \lfloor tD \oplus tD'\rfloor_T\ \lfloor tD'\rfloor_T\\&\mathit{inrOrn}\ (\mathsf{ok}\ i)\ =\ \Delta[\,c' : C'\ i\,]\ \nabla[\,\mathsf{inr}\ c'\,]\ \mathit{idOrn}\ (D'\ i\ c')\end{aligned}$$

whose forgetful functions perform suitable injection of constructor tags. Note that the synthesised new description $\lfloor\ tD \oplus tD'\ \rfloor_T$ is at the less informative end of *inlOrn* and *inrOrn*. (This, of course, is not a complete implementation of data types à la carte and requires more engineering for practical use.) □

3.2.3 Parallel composition of ornaments

intro — analysis for composability

The generic scenario is illustrated below: Chapter 4



Given three descriptions D : Desc I, E : Desc J, and F : Desc K and two ornaments O : Orn e D E and P : Orn e D F independently specifying how D is refined to E and F, we can compute an ornamental description

$$O \otimes P \ : \ \mathsf{OrnDesc}\ (e \bowtie f)\ \mathit{pull}\ D$$

Intuitively, since both O and P encode modifications to the same base description D, we can commit all modifications encoded by O and P to D to get a new description $\lfloor O \otimes P \rfloor$, and encode all these modifications in one ornament $\lceil O \otimes P \rceil$. (This merging of two sets of modifications is best characterised by a category-theoretic pullback, which we defer until Chapter 4.) The forgetful function of the ornament $\lceil O \otimes P \rceil$ removes all modifications, taking $\mu \lfloor O \otimes P \rfloor$ all the way back to the base datatype $\mu\ D$; there are also two **difference ornaments**

$$\begin{array}{ll} \mathit{diffOrn\text{-}l}\ O\ P \ : \ \mathsf{Orn}\ \mathsf{outl}\ E\ \lfloor O \otimes P \rfloor & \text{-- left difference ornament} \\ \mathit{diffOrn\text{-}r}\ O\ P \ : \ \mathsf{Orn}\ \mathsf{outr}\ F\ \lfloor O \otimes P \rfloor & \text{-- right difference ornament} \end{array}$$

which give rise to "less forgetful" functions taking $\mu \lfloor O \otimes P \rfloor$ to $\mu\ E$ and $\mu\ F$, such that both

$$\mathit{forget}\ O \circ \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$$

and

$$\mathit{forget}\ P \circ \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P)$$

are extensionally equal to $\mathit{forget}\ \lceil O \otimes P \rceil$.

Example (*ordered vectors*)**.** Consider the two ornaments $\lceil \mathit{OrdListOD}\ A\ \_\leqslant_A\_ \rceil$ from lists to ordered lists and *ListD-VecD A* from lists to vectors. Composing them in parallel gives us an ornamental description from which we can decode (i) a new datatype of ordered vectors

$$\begin{aligned}
&\mathsf{OrdVec}\ A\ _\leqslant_A_\ :\ A \to \mathsf{Nat} \to \mathsf{Set}\\
&\mathsf{OrdVec}\ A\ _\leqslant_A_\ b\ n\ =\\
&\quad \mu\ \lfloor \lceil \mathit{OrdListOD}\ A\ _\leqslant_A_ \rceil \otimes \mathit{ListD\text{-}VecD}\ A \rfloor\ (\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ n))\\
&\textbf{indexfirst data}\ \mathsf{OrdVec}\ A\ _\leqslant_A_\ :\ A \to \mathsf{Nat} \to \mathsf{Set}\ \textbf{where}\\
&\quad \mathsf{OrdVec}\ A\ _\leqslant_A_\ b\ \mathsf{zero} \quad \ni\ \mathsf{nil}\\
&\quad \mathsf{OrdVec}\ A\ _\leqslant_A_\ b\ (\mathsf{suc}\ n)\ \ni\ \mathsf{cons}\ (a\ :\ A)\ (\mathit{leq}\ :\ b \leqslant_A a)\\
&\qquad\qquad\qquad\qquad\qquad\qquad (\mathit{as}\ :\ \mathsf{OrdVec}\ A\ _\leqslant_A_\ a\ n)
\end{aligned}$$

and (ii) an ornament whose forgetful function converts ordered vectors to plain lists, retaining the list elements. The forgetful functions of the difference ornaments convert ordered vectors to ordered lists and vectors, removing only length and ordering information respectively. □

The complete definitions for parallel composition are shown in Figure 3.3. The core definition is *pcROD*, which analyses and merges the modifications encoded by two response ornaments into a response ornamental description at the level of individual fields. Below are some representative cases of *pcROD*.

- When both response ornaments use σ, both of them preserve the same field in the base description — no modification is made. Consequently, the field is preserved in the resulting response ornamental description as well.

  $$\mathit{pcROD}\ (\sigma\ S\ O)\ (\sigma\ .S\ P)\ =\ \sigma[\,s : S\,]\ \mathit{pcROD}\ (O\ s)\ (P\ s)$$

- When one of the response ornaments uses Δ to mark the addition of a new field, that field would be added into the resulting response ornamental description, like in

```
pc-𝔼 : {I J K : Set} {e : J → I} {f : K → I} →
       {is : List I} {js : List J} {ks : List K} →
       𝔼 e js is → 𝔼 f ks is → ℙ is (InvImage pull)
pc-𝔼              []          []          = ▪
pc-𝔼 {e := e} {f} (eeq :: eeqs) (feq :: feqs) = ok (fromEq e eeq , fromEq f feq) ,
                                                pc-𝔼 eeqs feqs

mutual
  pcROD : {I J K : Set} {e : J → I} {f : K → I}
          {D : RDesc I} {E : RDesc J} {F : RDesc K} →
          ROrn e D E → ROrn f D F → ROrnDesc (e ⋈ f) pull D
  pcROD (v eeqs) (v feqs)   = v (pc-𝔼 eeqs feqs)
  pcROD (v eeqs) (Δ T P)    = Δ[ t : T ] pcROD (v eeqs) (P t)
  pcROD (σ S O) (σ .S P)    = σ[ s : S ] pcROD (O s)    (P s)
  pcROD (σ f O) (Δ T P)     = Δ[ t : T ] pcROD (σ f O) (P t)
  pcROD (σ S O) (∇ s P)     = ∇[ s ]     pcROD (O s)    P
  pcROD (Δ T O) P           = Δ[ t : T ] pcROD (O t)    P
  pcROD (∇ s O) (σ S P)     = ∇[ s ]     pcROD O        (P s)
  pcROD (∇ s O) (Δ T P)     = Δ[ t : T ] pcROD (∇ s O) (P t)
  pcROD (∇ s O) (∇ s′ P)    = Δ (s ≡ s′) (pcROD-double∇ O P)

  pcROD-double∇ :
    {I J K S : Set} {e : J → I} {f : K → I}
    {D : S → RDesc I} {E : RDesc J} {F : RDesc K} {s s′ : S} →
    ROrn e (D s) E → ROrn f (D s′) F →
    s ≡ s′ → ROrnDesc (e ⋈ f) pull (σ S D)
  pcROD-double∇ {s := s} O P refl = ∇[ s ] pcROD O P

_⊗_ : {I J K : Set} {e : J → I} {f : K → I}
      {D : Desc I} {E : Desc J} {F : Desc K} →
      Orn e D E → Orn f D F → OrnDesc (e ⋈ f) pull D
(O ⊗ P) (ok (j , k)) = pcROD (O j) (P k)
```

Figure 3.3 Definitions for parallel composition of ornaments.

$$\mathit{pcROD}\ (\Delta\ T\ O)\ P\ =\ \Delta[\,t : T\,]\ \mathit{pcROD}\ (O\ t)\ P$$

- If one of the response ornaments retains a field by σ and the other deletes it by ∇, the only modification to the field is deletion, and thus the field is deleted in the resulting response ornamental description, like in

  $$\mathit{pcROD}\ (\sigma\ S\ O)\ (\nabla\ s\ P)\ =\ \nabla[\,s\,]\ \mathit{pcROD}\ (O\ s)\ P$$

- The most interesting case is when both response ornaments encode deletion: we would add an equality field demanding that the default values supplied in the two response ornaments be equal,

  $$\mathit{pcROD}\ (\nabla\ s\ O)\ (\nabla\ s'\ P)\ =\ \Delta\ (s\ \equiv\ s')\ (\mathit{pcROD\text{-}double}\nabla\ O\ P)$$

  and then *pcROD-double*∇ puts the deletion into the resulting response ornamental description after matching the proof of the equality field with refl.

  $$\mathit{pcROD\text{-}double}\nabla\ \{s := s\}\ O\ P\ \mathsf{refl}\ =\ \nabla[\,s\,]\ \mathit{pcROD}\ O\ P$$

  It might seem bizarre that two deletions results in a new field (and a deletion), but consider this informally described scenario: A field $\sigma\ S$ in the base response description is refined by two independent response ornaments

  $$\Delta[\,t : T\,]\quad \nabla[\,g\ t\,]$$

  and

  $$\Delta[\,u : U\,]\ \nabla[\,h\ u\,]$$

  That is, instead of S-values, the response descriptions at the more informative end of the two response ornaments use T- and U-values at this position, which are erased to their underlying S-value by $g\ :\ T \to S$ and $h\ :\ U \to S$ respectively. Composing the two response ornaments in parallel, we get

  $$\Delta[\,t : T\,]\ \Delta[\,u : U\,]\ \Delta[\,\_ : g\ t\ \equiv\ h\ u\,]\ \nabla[\,g\ t\,]$$

  where the added equality field completes the construction of a set-theoretic pullback of g and h. Here indeed we need a pullback: When we have an actual value for the field $\sigma\ S$, which gets refined to values of types T and U, the generic way to mix the two refining values is to store them both, as a product. If we wish to retrieve the underlying value of type S, we can either

extract the value of type T and apply g to it or extract the value of type U and apply h to it, and through either path we should get the same underlying value. So the product should really be a pullback to ensure this.

Chapter 4

Example (*ornamental description of ordered vectors*)**.** Composing the ornaments $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ and *ListD-VecD A* in parallel yields the following ornamental description relative to *ListD A*:

$$\begin{aligned}\lambda\,\{\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ \mathsf{zero}\quad))\ \mapsto\ \nabla[\,\text{`}\mathsf{nil}\,]\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\mathsf{suc}\ n)))\ \mapsto\ \nabla[\,\text{`}\mathsf{cons}\,]\ \sigma[\,a : A\,]\\ &\qquad\qquad\qquad\qquad\Delta[\,\_ : b \leqslant_A a\,]\ \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ a\ ,\ \mathsf{ok}\ n)\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates modifications from $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ and darker box from *ListD-VecD A*. □

Finally, the definitions for left difference ornament are shown in Figure 3.4. Left difference ornament has the same structure as parallel composition, but records only modifications from the right-hand side ornament. For example, the case

$$\mathit{diffROrn\text{-}l}\ (\sigma\ S\ O)\ (\nabla\ s\ P)\ =\ \nabla[\,s\,]\ \mathit{diffROrn\text{-}l}\ (O\ s)\ P$$

is the same as the corresponding case of *pcROD*, since the deletion comes from the right-hand side response ornament, whereas the case

$$\mathit{diffROrn\text{-}l}\ (\Delta\ T\ O)\ P\ =\ \sigma[\,t : T\,]\ \mathit{diffROrn\text{-}l}\ (O\ t)\ P$$

produces σ (a preservation) rather than Δ (a modification) as in the corresponding case of *pcROD*, since the addition comes from the left-hand side response ornament. We can then see that the composition of the forgetful functions

$$\mathit{forget}\ O \circ \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$$

is indeed extensionally equal to $\mathit{forget}\ \lceil\ O \otimes P\ \rceil$, since $\mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$ removes modifications encoded in the right-hand side ornament and then *forget O* removes modifications encoded in the left-hand side ornament. Right difference ornament is defined analogously and is omitted from the presentation.

```
diff-𝔼-l : {I J K : Set} {e : J → I} {f : K → I} →
           {is : List I} {js : List J} {ks : List K} →
           (eeqs : 𝔼 e js is) (feqs : 𝔼 f ks is) → 𝔼 outl (und-ℙ is (pc-𝔼 eeqs feqs)) js
diff-𝔼-l          []          []          = []
diff-𝔼-l {e := e} (eeq :: eeqs) (feq :: feqs) = und-fromEq e eeq :: diff-𝔼-l eeqs feqs

mutual
  diffROrn-l :
    {I J K : Set} {e : J → I} {f : K → I} →
    {D : RDesc I} {E : RDesc J} {F : RDesc K} →
    (O : ROrn e D E) (P : ROrn f D F) → ROrn outl E (toRDesc (pcROD O P))
  diffROrn-l (v eeqs) (v feqs)   = v (diff-𝔼-l eeqs feqs)
  diffROrn-l (v eeqs) (Δ T P)    = Δ[ t : T ] diffROrn-l (v eeqs) (P t)
  diffROrn-l (σ S O) (σ .S P)    = σ[ s : S ] diffROrn-l (O s)    (P s)
  diffROrn-l (σ S O) (Δ T P)     = Δ[ t : T ] diffROrn-l (σ S O) (P t)
  diffROrn-l (σ S O) (∇ s P)     = ∇[ s ]     diffROrn-l (O s)    P
  diffROrn-l (Δ T O) P           = σ[ t : T ] diffROrn-l (O t)    P
  diffROrn-l (∇ s O) (σ S P)     =            diffROrn-l O        (P s)
  diffROrn-l (∇ s O) (Δ T P)     = Δ[ t : T ] diffROrn-l (∇ s O) (P t)
  diffROrn-l (∇ s O) (∇ s′ P)    = Δ (s ≡ s′) (diffROrn-l-double∇ O P)

  diffROrn-l-double∇ :
    {I J K S : Set} {e : J → I} {f : K → I} →
    {D : S → RDesc I} {E : RDesc J} {F : RDesc K} {s s′ : S} →
    (O : ROrn e (D s) E) (P : ROrn f (D s′) F) (eq : s ≡ s′) →
    ROrn outl E (toRDesc (pcROD-double∇ O P eq))
  diffROrn-l-double∇ O P refl = diffROrn-l O P

diffOrn-l : {I J K : Set} {e : J → I} {f : K → I} →
            {D : Desc I} {E : Desc J} {F : Desc K} →
            (O : Orn e D E) (P : Orn f D F) → Orn outl E ⌊ O ⊗ P ⌋
diffOrn-l O P (ok (j , k)) = diffROrn-l (O j) (P k)
```

Figure 3.4 Definitions for left difference ornament.

3.3 Refinement semantics of ornaments

Every ornament $O : \mathsf{Orn}\ e\ D\ E$ induces a refinement family from $\mu\ D$ to $\mu\ E$. That is, we can construct a function

$$\begin{aligned} RSem\ :\ &\{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to \\ &\mathsf{Orn}\ e\ D\ E \to \mathsf{FRefinement}\ e\ (\mu\ D)\ (\mu\ E) \end{aligned}$$

which is called the **refinement semantics** of ornaments.

intro

3.3.1 Optimised predicates

Our most important task for now is to construct a promotion predicate

$$\begin{aligned} \mathsf{OptP}\ :\ &\{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to \\ &(O\ :\ \mathsf{Orn}\ e\ D\ E)\ \{i\ :\ I\}\ (j\ :\ e^{-1}\ i)\ (x\ :\ \mu\ D\ i) \to \mathsf{Set} \end{aligned}$$

which is called the **optimised predicate** for the ornament O. Given $x : \mu\ D\ i$, a proof of type $\mathsf{OptP}\ O\ j\ x$ contains the necessary information for complementing x and forming an inhabitant y of type $\mu\ E\ (und\ j)$ with the same recursive structure — the proof is the "horizontal" difference between y and x, speaking in terms of the two-dimensional metaphor. Such a proof should have the same vertical structure as x, and, at each recursive node, store horizontally only those data marked as modified by the ornament. For example, if we are promoting the natural number

Optimised in what sense?

$$\begin{aligned} two\ =\ &\mathsf{con}\ (\text{`cons}\ , \\ &\mathsf{con}\ (\text{`cons}\ , \\ &\mathsf{con}\ (\text{`nil}\quad , \\ &\quad \blacksquare)\ , \blacksquare)\ , \blacksquare)\ :\ \mu\ NatD\ \blacksquare \end{aligned}$$

to a list, an optimised promotion proof would look like

$$\begin{aligned} p\ =\ &\mathsf{con}\ (a\ , \\ &\mathsf{con}\ (a'\ , \end{aligned}$$

$$\begin{array}{l}\mathsf{con}\ (\\ \quad \blacksquare)\ ,\ \blacksquare)\ ,\ \blacksquare)\ :\ \mathsf{OptP}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathsf{ok}\ \blacksquare)\ \mathit{two}\end{array}$$

where a and a' are some elements of type A, so we get a list by zipping together *two* and r node by node:

$$\begin{array}{l}\mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a\ \ ,\\ \mathsf{con}\ (\text{`}\mathsf{cons}\ ,\ a'\ ,\\ \mathsf{con}\ (\text{`}\mathsf{nil}\ \ \ ,\\ \quad \blacksquare)\ ,\ \blacksquare)\ ,\ \blacksquare)\ :\ \mu\ (\mathit{ListD}\ A)\ \blacksquare\end{array}$$

Note that p contains only values of the field marked as additional by Δ in the ornament *NatD-ListD A*. The constructor tags are essential for determining the recursive structure of p, but instead of being stored in p, they are derived from *two*, which is part of the index of the type of p. In general, here is how we compute an ornamental description for such proofs, using D as the template: we incorporate the modifications made by O, and delete the fields that already exist in D, whose default values are derived in the index-first fashion from the inhabitant being promoted, which appears in the index of the type of a proof. The deletion is independent of O and can be performed by the singleton ornament for D (Section 3.2.2), so the desired ornamental description is produced by the parallel composition of O and $\lceil \mathit{singletonOD}\ D \rceil$:

$$\begin{array}{l}\mathit{OptPOD}\ :\ \{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to \\ \qquad\qquad \mathsf{Orn}\ e\ D\ E \to \mathsf{OrnDesc}\ (e \bowtie \mathsf{outl})\ \mathit{pull}\ D \\ \mathit{OptPOD}\ \{D := D\}\ O\ =\ O \otimes \lceil \mathit{singletonOD}\ D \rceil\end{array}$$

where outl has type $\Sigma\ I\ (\mu\ D) \to I$. The optimised predicate, then, is the least fixed point of the description.

$$\begin{array}{l}\mathsf{OptP}\ :\ \{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to \\ \qquad (O\ :\ \mathsf{Orn}\ e\ D\ E)\ \{i\ :\ I\}\ (j\ :\ e^{-1}\ i)\ (x\ :\ \mu\ D\ i) \to \mathsf{Set} \\ \mathsf{OptP}\ O\ \{i\}\ j\ d\ =\ \mu\ \lfloor \mathit{OptPOD}\ O \rfloor\ (j\ ,\ (\mathsf{ok}\ (i\ ,\ d)))\end{array}$$

Example (*index-first vectors as an optimised predicate*)**.** The optimised predicate for the ornament *NatD-ListD A* from natural numbers to lists is the datatype of index-first vectors. Expanding the definition of the ornamental description

OptPOD (*NatD-ListD A*) relative to *NatD*:

$$\begin{aligned}\lambda\,\{\ &(\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ \mathsf{zero}\ )))\ \mapsto\ \nabla[\,\text{`nil}\,]\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ \mathsf{suc}\ n)))\ \mapsto\ \nabla[\,\text{`cons}\,]\ \Delta[\,\_ : A\,]\\ &\qquad\qquad\qquad\qquad\qquad\qquad \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ \blacksquare\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ n))\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates contributions from the ornament *NatD-ListD A* and darker box from the singleton ornament $\lceil$*singletonOD NatD*$\rceil$, we see that the ornamental description indeed yields the datatype of index-first vectors (albeit indexed by a more heavily packaged datatype of natural numbers). □

Example (*predicate characterising ordered lists*)**.** The optimised predicate for the ornament $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ from lists to ordered lists is given by the ornamental description *OptPOD* $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ relative to *ListD A*, which expands to

$$\begin{aligned}\lambda\,\{\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ []\quad\ )))\ \mapsto\ \nabla[\,\text{`nil}\,]\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ b\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ a :: as)))\ \mapsto\ \nabla[\,\text{`cons}\,]\ \nabla[\,a\,]\ \Delta[\,leq : b \leqslant_A a\,]\\ &\qquad\qquad\qquad\qquad\qquad\qquad \mathsf{v}\ (\mathsf{ok}\ (\mathsf{ok}\ a\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ as))\ ,\ \blacksquare)\ \}\end{aligned}$$

where lighter box indicates contributions from $\lceil$ *OrdListOD A* $\_\leqslant_A\_$ $\rceil$ and darker box from $\lceil$*singletonOD* (*ListD A*)$\rceil$.

$$\begin{aligned}&\textbf{indexfirst data}\ \mathsf{Ordered}\ A\ \_\leqslant_A\_\ :\ A \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\ &\quad \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ []\qquad\ \ni\ \mathsf{nil}\\ &\quad \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ (a :: as)\ \ni\ \mathsf{cons}\ (leq\ :\ b \leqslant_A a)\ (o\ :\ \mathsf{Ordered}\ A\ \_\leqslant_A\_\ a\ as)\end{aligned}$$

Since a proof of $\mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ as$ consists of exactly the inequality proofs necessary for ensuring that *as* is ordered and bounded below by *b*, its representation is optimised, justifying the name "optimised predicate". □

Example (*inductive length predicate on lists*)**.** The optimised predicate for the ornament *ListD-VecD A* from lists to vectors is produced by the ornamental description *OptPOD* (*ListD-VecD A*) relative to *ListD A*:

$$\begin{aligned}\lambda\,\{\ &(\mathsf{ok}\ (\mathsf{ok}\ \mathsf{zero}\quad\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ []\quad\ )))\ \mapsto\ \Delta[\,\_ : \text{`nil}\ \equiv\ \text{`nil}\,]\ \nabla[\,\text{`nil}\,]\ \mathsf{v}\ \blacksquare\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ \mathsf{zero}\quad\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ a :: as)))\ \mapsto\ \Delta\ (\text{`nil}\ \equiv\ \text{`cons})\ \lambda\ ()\\ ;\ &(\mathsf{ok}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ []\quad\ )))\ \mapsto\ \Delta\ (\text{`cons}\ \equiv\ \text{`nil})\ \lambda\ ()\end{aligned}$$

$$; \ (\mathsf{ok}\ (\mathsf{ok}\ (\mathsf{suc}\ n)\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ a :: as))) \ \mapsto\ \Delta[\ \_ : \text{`}\mathsf{cons} \ \equiv\ \text{`}\mathsf{cons}\ ]\ \nabla[\ \text{`}\mathsf{cons}\ ]$$
$$\nabla[\ a\ ]\ \mathsf{v}\ (\mathsf{ok}\ (\ \mathsf{ok}\ n\ ,\ \mathsf{ok}\ (\blacksquare\ ,\ as)\ )\ ,\ \blacksquare)\ \}$$

where lighter box indicates contributions from *ListD-VecD A* and darker box from $\lceil\ singletonOD\ (ListD\ A)\ \rceil$. Both ornaments perform pattern matching and accordingly restrict constructor choices by ∇, so the resulting four cases all start with an equality field demanding that the constructor choices specified by the two ornaments are equal.

- In the first and last cases, where the specified constructor choices match, the equality proof obligation can be successfully discharged and the response ornamental description can continue after installing the constructor choice by ∇;
- in the middle two cases, where the specified constructor choices mismatch, the equality is obviously unprovable and the rest of the response ornamental description is (extensionally) the empty function $\lambda\ ()$.

Thus, in effect, the ornamental description produces the following inductive length predicate on lists:

$$\begin{array}{l}
\textbf{indexfirst data}\ \mathsf{Length}\ A\ :\ \mathsf{Nat} \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where} \\
\quad \mathsf{Length}\ A\ \mathsf{zero}\ \quad []\ \quad\ni\ \mathsf{nil} \\
\quad \mathsf{Length}\ A\ \mathsf{zero}\ \quad (a :: as)\ \ni \\
\quad \mathsf{Length}\ A\ (\mathsf{suc}\ n)\ []\ \quad\ni \\
\quad \mathsf{Length}\ A\ (\mathsf{suc}\ n)\ (a :: as)\ \ni\ \mathsf{cons}\ (l\ :\ \mathsf{Length}\ A\ n\ as)
\end{array}$$

where $\ni$ indicates that a case is uninhabited. $\square$

We have thus determined the promotion predicate used by the refinement semantics of ornaments to be the optimised predicate:

$$\begin{array}{l}
RSem\ :\ \{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\} \to \\
\qquad \mathsf{Orn}\ e\ D\ E \to \mathsf{FRefinement}\ e\ (\mu\ D)\ (\mu\ E) \\
RSem\ O\ j\ =\ \textbf{record}\ \{\ P\ =\ \mathsf{OptP}\ O\ j \\
\qquad\qquad\qquad\qquad ;\ i\ \ =\ ornConvIso\ O\ j\ \}
\end{array}$$

We call *ornConvIso* the **ornamental conversion isomorphisms**, whose type is

$$ornConvIso\ :$$

$\{I\ J\ :\ \mathsf{Set}\}\ \{e\ :\ J \to I\}\ \{D\ :\ \mathsf{Desc}\ I\}\ \{E\ :\ \mathsf{Desc}\ J\}\ (O\ :\ \mathsf{Orn}\ e\ D\ E) \to$
$\{i\ :\ I\}\ (j\ :\ e^{-1}\ i) \to \mu\ E\ (und\ j)\ \cong\ \Sigma[\, x\ :\ \mu\ D\ i\,]\ \mathsf{OptP}\ O\ j\ x$

The construction of *ornConvIso* will be deferred until Chapter 4.

3.3.2 Predicate swapping for parallel composition

An ornament describes differences between two datatypes, and the optimised predicate for the ornament is the datatype of differences between inhabitants of the two datatypes. To promote an inhabitant from the less informative end to the more informative end of the ornament using its refinement semantics, we give a proof that the object satisfies the optimised predicate for the ornament. If, however, the ornament is a parallel composition, say $\lceil\, O \otimes P\,\rceil$, then the differences recorded in the ornament are simply collected from the component ornaments O and P. Consequently, it should suffice to give separate proofs that the inhabitant satisfies the optimised predicates for O and P, instead of a proof that it satisfies the monolithic optimised predicate induced by $\lceil O \otimes P \rceil$. We are thus led to prove that the optimised predicate for $\lceil\, O \otimes P\,\rceil$ amounts to the pointwise conjunction of the optimised predicates for O and P. More precisely: if $O\ :\ \mathsf{Orn}\ e\ D\ E$ and $P\ :\ \mathsf{Orn}\ f\ D\ F$ where $D\ :\ \mathsf{Desc}\ I$, $E\ :\ \mathsf{Desc}\ J$, and $F\ :\ \mathsf{Desc}\ K$, then we expect the existence of the **modularity isomorphisms**

$$\mathsf{OptP}\ \lceil O \otimes P \rceil\ (\mathsf{ok}\ (j\ ,\ k))\ x\ \cong\ \mathsf{OptP}\ O\ j\ x \times \mathsf{OptP}\ P\ k\ x$$

for all $i\ :\ I, j\ :\ e^{-1}\ i, k\ :\ f^{-1}\ i$, and $x\ :\ \mu\ D\ i$.

Example (*promotion predicate from lists to ordered vectors*)**.** The optimised predicate for the ornament $\lceil\, \lceil\, \mathit{OrdListOD}\ A\ \_\leqslant_A\_\,\rceil \otimes \mathit{ListD\text{-}VecD}\ A\,\rceil$ from lists to ordered vectors is

indexfirst data $\mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ :\ A \to \mathsf{Nat} \to \mathsf{List}\ A \to \mathsf{Set}$ **where**
$\mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ b\ \mathsf{zero}\quad [\,]\quad \ni\ \mathsf{nil}$
$\mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ b\ \mathsf{zero}\quad (a :: as)\quad \not\ni$
$\mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ b\ (\mathsf{suc}\ n)\ [\,]\quad \not\ni$
$\mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ b\ (\mathsf{suc}\ n)\ (a :: as)$
$\quad \ni\ \mathsf{cons}\ (leq\ :\ b \leqslant_A a)\ (ol\ :\ \mathsf{OrderedLength}\ A\ \_\leqslant_A\_\ a\ n\ as)$

which is monolithic and inflexible. We can avoid using this predicate directly by exploiting the modularity isomorphisms

$$\mathsf{OrderedLength}\ A\ \_{\leqslant_A}\_\ b\ n\ as\ \cong\ \mathsf{Ordered}\ A\ \_{\leqslant_A}\_\ b\ as\ \times\ \mathsf{Length}\ A\ n\ as$$

for all $b : A$, $n : \mathsf{Nat}$, and $as : \mathsf{List}\ A$ — to promote a list to an ordered vector, we can prove that it satisfies Ordered and Length instead of OrderedLength. Promotion proofs from lists to ordered vectors can thus be divided into ordering and length aspects and carried out separately. □

Along with the ornamental conversion isomorphisms, the construction of the modularity isomorphisms will be deferred until Chapter 4. Here we deal with a practical issue regarding composition of modularity isomorphisms: for example, to get pointwise isomorphisms between the optimised predicate for $\lceil O \otimes \lceil P \otimes Q \rceil \rceil$ and the pointwise conjunction of the optimised predicates for O, P, and Q, we need to instantiate the modularity isomorphisms twice and compose the results appropriately, a procedure which quickly becomes tedious. What we need is an auxiliary mechanism that helps with organising computation of composite predicates and isomorphisms following the parallel compositional structure of ornaments, in the same spirit as the upgrade mechanism (Section 3.1.2) helping with organising computation of coherence properties and proofs following the syntactic structure of function types.

We thus define the following auxiliary datatype Swap, parametrised with a refinement whose promotion predicate is to be swapped for a new one:

$$\begin{array}{l}\textbf{record}\ \mathsf{Swap}\ \{X\ Y : \mathsf{Set}\}\ (r : \mathsf{Refinement}\ X\ Y) : \mathsf{Set}_1\ \textbf{where}\\ \quad \textbf{field}\\ \quad\quad Q : X \to \mathsf{Set}\\ \quad\quad i\ \ : (x : X) \to \mathsf{Refinement}.P\ r\ x\ \cong\ Q\ x\end{array}$$

An inhabitant of Swap r consists of a new promotion predicate for r and a proof that the new predicate is pointwise isomorphic to the original one in r. The actual swapping is done by the function

$$\begin{array}{l}toRefinement : \{X\ Y : \mathsf{Set}\}\ \{r : \mathsf{Refinement}\ X\ Y\} \to \mathsf{Swap}\ r \to \mathsf{Refinement}\ X\ Y\\ toRefinement\ s\ =\ \textbf{record}\ \{\ P\ =\ \mathsf{Swap}.Q\ s\\ \qquad\qquad\qquad\qquad\quad ;\ i\ \ =\ \{\ \}_0\ \}\end{array}$$

where Goal 0 is the new conversion isomorphism

$$Y \cong \Sigma\, X\, (\mathsf{Refinement}.P\; r) \cong \Sigma\, X\, (\mathsf{Swap}.Q\; s)$$

constructed by using transitivity and product of isomorphisms to compose $\mathsf{Refinement}.i\; r$ and $\mathsf{Swap}.i\; s$. We can then define the datatype FSwap of "swap families" in the usual way:

$$\begin{aligned}&\mathsf{FSwap} : \{I\, J : \mathsf{Set}\}\, \{e : J \to I\}\, \{X : I \to \mathsf{Set}\}\, \{Y : J \to \mathsf{Set}\} \to \\ &\qquad (rs : \mathsf{FRefinement}\; e\; X\; Y) \to \mathsf{Set}_1 \\ &\mathsf{FSwap}\; rs = \{i : I\}\, (j : e^{-1}\; i) \to \mathsf{Swap}\; (rs\; j)\end{aligned}$$

and provide the following combinator on swap families, which says that if there are alternative promotion predicates for the refinement semantics of O and P, then the pointwise conjunction of the two predicates is an alternative promotion predicate for the refinement semantics of $\lceil O \otimes P \rceil$:

$$\begin{aligned}&\otimes\text{-}FSwap : \{I\, J\, K : \mathsf{Set}\}\, \{e : J \to I\}\, \{f : K \to I\} \to \\ &\qquad \{D : \mathsf{Desc}\; I\}\, \{E : \mathsf{Desc}\; J\}\, \{F : \mathsf{Desc}\; K\} \to \\ &\qquad (O : \mathsf{Orn}\; e\; D\; E)\, (P : \mathsf{Orn}\; f\; D\; F) \to \\ &\qquad \mathsf{FSwap}\; (RSem\; O) \to \mathsf{FSwap}\; (RSem\; P) \to \mathsf{FSwap}\; (RSem\; \lceil O \otimes P \rceil) \\ &\otimes\text{-}FSwap\; O\; P\; ss\; ts\; (\mathsf{ok}\; (j\, , k)) = \\ &\quad \mathbf{record}\; \{\, Q = \lambda\, x \mapsto \mathsf{Swap}.Q\; (ss\; j)\; x \times \mathsf{Swap}.Q\; (ts\; k)\; x \\ &\qquad\quad ;\, i = \lambda\, x \mapsto \{\,\}_1\, \}\end{aligned}$$

Goal 1 is straightforwardly discharged by composing the modularity isomorphisms and the isomorphisms in *ss* and *ts*:

$$\begin{aligned}\mathsf{OptP}\; \lceil O \otimes P \rceil\; (\mathsf{ok}\; (j\, , k))\; x &\cong \mathsf{OptP}\; O\; j\; x \times \mathsf{OptP}\; P\; k\; x \\ &\cong \mathsf{Swap}.Q\; (ss\; j)\; x \times \mathsf{Swap}.Q\; (ts\; k)\; x\end{aligned}$$

Example (*modular promotion predicate for the parallel composition of three ornaments*)**.** To use the pointwise conjunction of the optimised predicates for ornaments O, P, and Q as an alternative promotion predicate for $\lceil O \otimes \lceil P \otimes Q \rceil \rceil$, we use the swap family

$$\otimes\text{-}FSwap\; O\; \lceil P \otimes Q \rceil\; id\text{-}FSwap\; (\otimes\text{-}FSwap\; P\; Q\; id\text{-}FSwap\; id\text{-}FSwap)$$

where

$$\mathit{id\text{-}FSwap} : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\}\ \{rs : \mathsf{FRefinement}\ X\ Y\} \to \mathsf{FSwap}\ rs$$

simply retains the original promotion predicate in rs. □

Example (*swapping the promotion predicate from lists to ordered vectors*)**.** The swap family

$$\otimes\text{-}\mathit{FSwap}\ \lceil \mathit{OrdListOD}\ A\ \_\leqslant_A\_ \rceil\ (\mathit{ListD\text{-}VecD}\ A)\ \mathit{id\text{-}FSwap}\ (\mathit{Length\text{-}FSwap}\ A)$$

yields a refinement family from lists to ordered vectors using

$$\lambda\, b\ n\ as \mapsto \mathsf{Ordered}\ A\ \_\leqslant_A\_\ b\ as \times \mathit{length}\ as \equiv n$$

as the promotion predicate, where

$$\mathit{Length\text{-}FSwap}\ A : \mathsf{FSwap}\ (\mathit{RSem}\ (\mathit{ListD\text{-}VecD}\ A))$$

swaps $\mathsf{Length}\ A$ for $\lambda\, n\ as \mapsto \mathit{length}\ as \equiv n$. □

3.4 Examples

To demonstrate the use of the ornament–refinement framework, we first resolve the library structuring problem about insertion into a list and then look at two dependently typed heap data structures adapted from Okasaki's work on purely functional data structures [1999]. Of the latter two examples,

- the first one about **binomial heaps** shows that Okasaki's idea of **numerical representations** can be elegantly captured by ornaments and the coherence properties computed with upgrades, and
- the second one about **leftist heaps** demonstrates the power of parallel composition of ornaments by treating heap ordering and leftist balancing properties modularly.

Postulate operations on *Val* like $\_\leqslant?\_$, $\leqslant$*-refl*, $\leqslant$*-trans*, and $\not\leqslant$*-invert* in Chapter 2.



Figure 3.5 **Left:** a binomial heap of size 3 consisting of two binomial trees storing elements x, y, and z. **Right:** the result of inserting an element w into the heap. (Note that the digits of the underlying binary numbers are ordered with the least significant digit first.)

3.4.1 Insertion into a list

3.4.2 Binomial heaps

We are all familiar with the idea of **positional number systems**, in which we represent numbers as a list of digits. Each position in a list of digits is associated with a weight, and the interpretation of the list is the weighted sum of the digits. (For example, the weights used for binary numbers are powers of 2.) Some container data structures and associated operations strongly resemble positional representations of natural numbers and associated operations. For example, a **binomial heap** (illustrated in Figure 3.5) can be thought of as a binary number in which every 1-digit stores a **binomial tree** — the actual place for storing elements — whose size is exactly the weight of the digit. The number of elements stored in a binomial heap is therefore exactly the value of the underlying binary number. Inserting a new element into a binomial heap is analogous to incrementing a binary number, with carrying corresponding to combining smaller binomial trees into larger ones. Okasaki thus proposed to design container data structures by analogy with positional representations of natural numbers, and called such data structures **numerical representations**.



Figure 3.6 **Left:** inductive definition of binomial trees. **Right:** decomposition of binomial trees of ranks 1 to 3.

Using an ornament, it is easy to express the relationship between a numerically represented container datatype (e.g., binomial heaps) and its underlying numeric datatype (e.g., binary numbers). But the ability to express the relationship alone is not too surprising. What is more interesting is that the ornament can give rise to upgrades such that

- the coherence properties of the upgrades semantically characterise the resemblance between container operations and corresponding numeric operations, and
- the promotion predicates give the precise types of the container operations that guarantee such resemblance.

We use insertion into a binomial heap as an example, which is presented in detail below.

Binomial trees

The basic building blocks of binomial heaps are **binomial trees**, in which elements are stored. Binomial trees are defined inductively on their **rank**, which is a natural number (see Figure 3.6):

- a binomial tree of rank 0 is a single node storing an element of type *Val*, and
- a binomial tree of rank $1 + r$ consists of two binomial trees of rank r, with

one attached under the other's root node.

From this definition we can readily deduce that a binomial tree of rank r has 2^r elements. To actually define binomial trees as a datatype, however, an alternative view is more useful: a binomial tree of rank r is constructed by attaching binomial trees of ranks 0 to $r-1$ under a root node. (Figure 3.6 shows how binomial trees of ranks 1 to 3 can be decomposed according to this view.) We thus define the datatype $\mathsf{BTree} : \mathsf{Nat} \to \mathsf{Set}$ — which is indexed with the rank of binomial trees — as follows: for any rank $r : \mathsf{Nat}$, the type $\mathsf{BTree}\ r$ has a field of type *Val* — which is the root node — and r recursive positions indexed from $r-1$ down to 0. This is directly encoded as a description:

$$
\begin{aligned}
&\mathit{BTreeD} : \mathsf{Desc\ Nat} \\
&\mathit{BTreeD}\ r = \sigma[\_ : \mathit{Val}]\ \mathsf{v}\ (\mathit{descend}\ r) \\
&\mathsf{BTree} : \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{BTree} = \mu\ \mathit{BTreeD}
\end{aligned}
$$

where *descend* r is a list from $r-1$ down to 0:

$$
\begin{aligned}
&\mathit{descend} : \mathsf{Nat} \to \mathsf{List\ Nat} \\
&\mathit{descend}\ \mathsf{zero} = [] \\
&\mathit{descend}\ (\mathsf{suc}\ n) = n :: \mathit{descend}\ n
\end{aligned}
$$

Note that, in *BTreeD*, we are exploiting the full computational power of Desc, computing the list of recursive indices from the index request. Due to this, it is tricky to wrap up *BTreeD* as an index-first datatype declaration, so we will skip this step and work directly with the raw representation, which looks reasonably intuitive anyway: a binomial tree of type $\mathsf{BTree}\ r$ is of the form $\mathsf{con}\ (x\ ,\ ts)$ where $x : \mathit{Val}$ is the root element and $ts : \mathbb{P}\ (\mathit{descend}\ r)\ \mathsf{BTree}$ is a series of sub-trees.

The most important operation on binomial trees is combining two smaller binomial trees of the same rank into a larger one, which corresponds to carrying in positional arithmetic. Given two binomial trees of the same rank r, one can be *attach*ed under the root of the other, forming a single binomial tree of rank $1 + r$ — this is exactly the inductive definition of binomial trees.

```
attach : {r : Nat} → BTree r → BTree r → BTree (suc r)
attach t (con (y , us)) = con (y , t , us)
```

For use in binomial heaps, though, we should ensure that elements in binomial trees are in **heap order**, i.e., the root of any binomial tree (including sub-trees) is the minimum element in the tree. This is achieved by comparing the roots of two binomial trees before deciding which one is to be attached to the other:

```
link : {r : Nat} → BTree r → BTree r → BTree (suc r)
link t u with root t ⩽? root u
link t u | yes _ = attach u t
link t u | no  _ = attach t u
```

where *root* extracts the root element of a binomial tree:

```
root : {r : Nat} → BTree r → Val
root (con (x , ts)) = x
```

If we always build binomial trees of positive rank by *link*, then the elements in any binomial tree we build would be in heap order. This is a crucial assumption in binomial heaps (which is not essential to our development, though).

From binary numbers to binomial heaps

The datatype Bin : Set of binary numbers is just a specialised datatype of lists of binary digits:

```
data BinTag : Set where 'nil 'zero 'one : BinTag

BinD : Desc ⊤
BinD ▪ = σ BinTag λ { 'nil  ↦ v []
                    ; 'zero ↦ v (▪ :: [])
                    ; 'one  ↦ v (▪ :: []) }

indexfirst data Bin : Set where
  Bin ∋ nil
      | zero (b : Bin)
      | one  (b : Bin)
```

The intended interpretation of binary numbers is given by

$$
\begin{aligned}
&\mathit{toNat} \;:\; \mathsf{Bin} \to \mathsf{Nat} \\
&\mathit{toNat}\ \mathsf{nil} &&= 0 \\
&\mathit{toNat}\ (\mathsf{zero}\ b) &&= 0 + 2 * \mathit{toNat}\ b \\
&\mathit{toNat}\ (\mathsf{one}\ \ b) &&= 1 + 2 * \mathit{toNat}\ b
\end{aligned}
$$

That is, the list of digits of a binary number of type Bin starts from the least significant digit, and the i-th digit (counting from 0) has weight 2^i. We refer to the position of a digit as its rank, i.e., the i-th digit is said to have rank i.

As stated in the beginning, binomial heaps are binary numbers whose 1-digits are decorated with binomial trees of matching rank, which can be expressed straightforwardly as an ornamentation of binary numbers. To ensure that the binomial trees in binomial heaps have the right rank, the datatype BHeap : Nat → Set is indexed with a "starting rank": if a binomial heap of type BHeap r is nonempty (i.e., not nil), then its first digit has rank r (and stores a binomial tree of rank r when the digit is one), and the rest of the heap is indexed with $1 + r$.

```
BHeapOD : OrnDesc Nat ! BinD
BHeapOD (ok r) = σ BinTag λ { 'nil  ↦ v ▪
                            ; 'zero ↦ v (ok (suc r) , ▪)
                            ; 'one  ↦ Δ[ t : BTree r ] v (ok (suc r) , ▪) }

indexfirst data BHeap : Nat → Set where
  BHeap r ∋ nil
          | zero (h : BHeap (suc r))
          | one  (t : BTree r) (h : BHeap (suc r))
```

In applications, we would use binomial heaps of type BHeap 0, which encompasses binomial heaps of all sizes.

Increment and insertion, in coherence

Increment of binary numbers is defined by

$$
\begin{array}{l}
\mathit{incr} \;:\; \mathsf{Bin} \to \mathsf{Bin} \\
\mathit{incr}\ \mathsf{nil} \quad\;\; = \; \mathsf{one}\ \mathsf{nil} \\
\mathit{incr}\ (\mathsf{zero}\ b) \; = \; \mathsf{one}\ b \\
\mathit{incr}\ (\mathsf{one}\ \ b) \; = \; \mathsf{zero}\ (\mathit{incr}\ b)
\end{array}
$$

The corresponding operation on binomial heaps is insertion of a binomial tree into a binomial heap (of matching rank), whose direct implementation is

$$
\begin{array}{l}
\mathit{insT} \;:\; \{r \;:\; \mathsf{Nat}\} \to \mathsf{BTree}\ r \to \mathsf{BHeap}\ r \to \mathsf{BHeap}\ r \\
\mathit{insT}\ t\ \mathsf{nil} \quad\;\; = \; \mathsf{one}\ t\ \mathsf{nil} \\
\mathit{insT}\ t\ (\mathsf{zero}\ \ h) \; = \; \mathsf{one}\ t\ h \\
\mathit{insT}\ t\ (\mathsf{one}\ u\ h) \; = \; \mathsf{zero}\ (\mathit{insT}\ (\mathit{link}\ t\ u)\ h)
\end{array}
$$

Conceptually, *incr* puts a 1-digit into the least significant position of a binary number, triggering a series of carries, i.e., summing 1-digits of smaller ranks into 1-digits of larger ranks; *insT* follows the pattern of *incr*, but since 1-digits now have to store a binomial tree of matching rank, *insT* takes an additional binomial tree as input and *link*s binomial trees of smaller ranks into binomial trees of larger ranks whenever carrying happens. Having defined *insT*, inserting a single element into a binomial heap of type $\mathsf{BHeap}\ 0$ is then inserting, by *insT*, a rank-0 binomial tree (i.e., a single node) storing the element into the heap.

$$
\begin{array}{l}
\mathit{insert} \;:\; \mathit{Val} \to \mathsf{BHeap}\ 0 \to \mathsf{BHeap}\ 0 \\
\mathit{insert}\ x \;=\; \mathit{insT}\ (\mathsf{con}\ (x\ ,\ \blacksquare))
\end{array}
$$

It is apparent that the program structure of *insT* strongly resembles that of *incr* — they manipulate the list-of-binary-digits structure in the same way. But can we characterise the resemblance semantically? It turns out that the coherence property of the following upgrade from the type of *incr* to that of *insT* is an appropriate answer:

$$
\begin{array}{l}
\mathit{upg} \;:\; \mathsf{Upgrade}\ (\mathsf{Bin} \to \mathsf{Bin})\ (\{r \;:\; \mathsf{Nat}\} \to \mathsf{BTree}\ r \to \mathsf{BHeap}\ r \to \mathsf{BHeap}\ r) \\
\mathit{upg} \;=\; \forall^{+}[[\, r : \mathsf{Nat}\,]]\ \forall^{+}[\, \_ : \mathsf{BTree}\ r\,] \\
\qquad\quad \mathbf{let}\ \mathit{ref} \;:\; \mathsf{Refinement}\ \mathsf{Bin}\ (\mathsf{BHeap}\ r) \\
\qquad\qquad\quad \mathit{ref} \;=\; \mathit{RSem}\ \lceil \mathit{BHeapOD} \rceil\ (\mathsf{ok}\ r) \\
\qquad\quad \mathbf{in}\ \ \mathit{ref} \rightharpoonup \mathit{toUpgrade}\ \mathit{ref}
\end{array}
$$

The upgrade *upg* says that, compared to the type of *incr*, the type of *insT* has two new arguments — the implicit argument r : Nat and the explicit argument of type BTree r — and that the two occurrences of BHeap r in the type of *insT* refine the corresponding occurrences of Bin in the type of *incr* using the refinement semantics of the ornament $\lceil BHeapOD \rceil$ (ok r) from Bin to BHeap r. The type Upgrade.*C upg incr insT* (which states that *incr* and *insT* are in coherence with respect to *upg*) expands to

$\{r : \mathsf{Nat}\}\ (t : \mathsf{BTree}\ r)\ (b : \mathsf{Bin})\ (h : \mathsf{BHeap}\ r) \rightarrow$
$toBin\ h \equiv b \rightarrow toBin\ (insT\ t\ h) \equiv incr\ b$

where *toBin* extracts the underlying binary number of a binomial heap:

$toBin : \{r : \mathsf{Nat}\} \rightarrow \mathsf{BHeap}\ r \rightarrow \mathsf{Bin}$
$toBin = forget\ \lceil BHeapOD \rceil$

That is, given a binomial heap h : BHeap r whose underlying binary number is b : Bin, after inserting a binomial tree into h by *insT*, the underlying binary number of the result is $incr\ b$. This says exactly that *insT* manipulates the underlying binary number in the same way as *incr* does.

We have seen that the coherence property of *upg* is appropriate for characterising the resemblance of *incr* and *insT*; proving that it holds for *incr* and *insT* is a separate matter, though. We can, however, avoid doing the implementation of insertion and the coherence proof separately: instead of implementing *insT* directly, we can implement insertion with a more precise type in the first place such that, from this more precisely typed version, we can derive *insT* that satisfies the coherence property automatically. The above process is fully supported by the mechanism of upgrades. Specifically, the more precise type for insertion is given by the promotion predicate of *upg* (applied to *incr*), the more precisely typed version of insertion acts as a promotion proof of *incr* (with respect to *upg*), and the promotion gives us *insT*, accompanied by a proof that *insT* is in coherence with *incr*.

Let BHeap′ be the optimised predicate for the ornament from Bin to BHeap r:

$\mathsf{BHeap}' : \mathsf{Nat} \rightarrow \mathsf{Bin} \rightarrow \mathsf{Set}$
$\mathsf{BHeap}'\ r\ b = \mathsf{OptP}\ \lceil BHeapOD \rceil\ (\mathsf{ok}\ r)\ b$

```
indexfirst data BHeap′ : Nat → Bin → Set where
  BHeap′ r nil       ∋ nil
  BHeap′ r (zero b)  ∋ zero (h : BHeap′ (suc r) b)
  BHeap′ r (one  b)  ∋ one  (t : BTree r) (h : BHeap′ (suc r) b)
```

Here a more helpful interpretation is that BHeap' is a datatype of binomial heaps additionally indexed with the underlying binary number. The type $\mathsf{Upgrade}.P\ upg\ incr$ of promotion proofs for *incr* then expands to

$$\{r : \mathsf{Nat}\} \to \mathsf{BTree}\ r \to (b : \mathsf{Bin}) \to \mathsf{BHeap}'\ r\ b \to \mathsf{BHeap}'\ r\ (incr\ b)$$

A function of this type is explicitly required to transform the underlying binary number structure of its input in the same way as *incr* does. Insertion can now be implemented as

```
insT′ : {r : Nat} → BTree r → (b : Bin) → BHeap′ r b → BHeap′ r (incr b)
insT′ t nil       nil       = one t nil
insT′ t (zero b) (zero  h)  = one t h
insT′ t (one  b) (one u h)  = zero (insT′ (link t u) h)
```

which is very much the same as the original *insT*. It is interesting to note that all the constructor choices for binomial heaps in *insT′* are actually completely determined by the types. This fact is easier to observe if we desugar *insT′* to the raw representation:

```
insT′ : {r : Nat} → BTree r → (b : Bin) → BHeap′ r b → BHeap′ r (incr b)
insT′ t (con (‘nil  ,    ▪)) (con          ▪ ) = con (t , con ▪          , ▪)
insT′ t (con (‘zero , b , ▪)) (con (    h , ▪)) = con (t , h            , ▪)
insT′ t (con (‘one  , b , ▪)) (con (u , h , ▪)) = con (   insT′ (link t u) b h , ▪)
```

in which no constructor tags for binomial heaps are present. This means that the types would instruct which constructors to use when programming *insT′*, establishing the coherence property by construction. Finally, since *insT′* is a promotion proof for *incr*, we can invoke the upgrading operation of *upg* and get *insT*:

```
insT : {r : Nat} → BTree r → BHeap r → BHeap r
insT = Upgrade.u upg incr insT′
```

which is automatically in coherence with *incr*:

```
incr-insT-coherence : {r : Nat} (t : BTree r) (b : Bin) (h : BHeap r) →
                      toBin h ≡ b → toBin (insT t h) ≡ incr b
incr-insT-coherence = Upgrade.c upg incr insT′
```

Summary

We define Bin, *incr*, and then BHeap as an ornamentation of Bin, describe in *upg* how the type of *insT* is an upgraded version of the type of *incr*, and implement *insT*′, whose type is supplied by *upg*. We can then derive *insT*, the coherence property of *insT* with respect to *incr*, and its proof, all automatically by *upg*. Compared to Okasaki's implementation, besides rank-indexing, which elegantly transfers the management of rank-related invariants to the type system, the extra work is only the straightforward markings of the differences between Bin and BHeap (in *BHeapOD*) and between the type of *incr* and that of *insT* (in *upg*). The reward is huge in comparison: we get a coherence property that precisely characterises the structural behaviour of insertion with respect to increment, and an enriched function type that guides the implementation of insertion such that the coherence property is satisfied by construction. From straightforward markings to nontrivial types and programs — this clearly demonstrates the power of the ornament–refinement framework.

3.4.3 Leftist heaps

Our second example is about treating the ordering and balancing properties of **leftist heaps** modularly. In Okasaki's words:

> Leftist heaps [...] are heap-ordered binary trees that satisfy the **leftist property**: the rank of any left child is at least as large as the rank of its right sibling. The rank of a node is defined to be the length of its **right spine** (i.e., the rightmost path from the node in question to an empty node).



Figure 3.7 Datatypes about leftist heaps and their ornamental relationships.

From this passage we can immediately analyse the concept of leftist heaps into three: leftist heaps (i) are binary trees that (ii) are heap-ordered and (iii) satisfy the leftist property. This suggests that there is a basic datatype of binary trees together with two ornamentations, one expressing heap ordering and the other the leftist property. The datatype of leftist heaps is then synthesised by composing the two ornamentations in parallel. All the datatypes about leftist heaps and their ornamental relationships are shown in Figure 3.7.

Datatypes leading to leftist heaps

The basic datatype Tree : Set of "skeletal" binary trees, which consist of empty nodes and internal nodes not storing any elements, is defined by

```
data TreeTag : Set where ‘nil ‘node : TreeTag

TreeD : Desc ⊤
TreeD ▪ = σ TreeTag λ { ‘nil  ↦ v []
                      ; ‘node ↦ v (▪ :: ▪ :: []) }
```

```
indexfirst data Tree : Set where
  Tree ∋ nil
       | node (t : Tree) (u : Tree)
```

Leftist trees — skeletal binary trees satisfying the leftist property — are then an ornamented version of Tree. The datatype LTree : Nat → Set of leftist trees is indexed with the rank of the root of the trees. The constructor choices can be determined from the rank: the only node that can have rank zero is the empty node nil; otherwise, when the rank of a node is non-zero, it must be an internal node constructed by the node constructor, which enforces the leftist property.

```
LTreeOD : OrnDesc Nat ! TreeD
LTreeOD (ok zero   ) = ∇[ 'nil ] v ▪
LTreeOD (ok (suc r)) = ∇[ 'node ] Δ[ l : Nat ] Δ[ r⩽l : r ⩽ l ] v (ok l , ok r , ▪)
indexfirst data LTree : Nat → Set where
  Tree zero    ∋ nil
  Tree (suc r) ∋ node {l : Nat} (r⩽l : r ⩽ l) (t : Tree l) (u : Tree r)
```

Independently, **heap-ordered trees** are also an ornamented version of Tree. The datatype Heap : *Val* → Set of heap-ordered trees can be regarded as a generalisation of ordered lists: in a heap-ordered tree, every path from the root to an empty node is an ordered list.

```
HeapOD : OrnDesc Val ! TreeD
HeapOD (ok b) =
  σ TreeTag λ { 'nil  ↦ v ▪
              ; 'node ↦ Δ[ x : Val ] Δ[ b⩽x : b ⩽ x ] v (ok x , ok x , ▪) }
indexfirst data Heap : Val → Set where
  Heap b ∋ nil
         | node (x : Val) (b⩽x : b ⩽ x) (t : Heap x) (u : Heap x)
```

Composing the two ornaments in parallel gives us exactly the datatype of leftist heaps.

```
LHeapOD : OrnDesc (! ⋈ !) pull TreeD
LHeapOD = ⌈HeapOD⌉ ⊗ ⌈LTreeOD⌉
```

```
indexfirst data LHeap : Val → Nat → Set where
  LHeap b zero     ∋ nil
  LHeap b (suc r)  ∋ node (x : Val) (b⩽x : b ⩽ x)
                          {l : Nat} (r⩽l : r ⩽ l) (t : Heap x l) (u : Heap x r)
```

Operations on leftist heaps

The analysis of leftist heaps as the parallel composition of the two ornamentations allows us to talk about heap ordering and the leftist property independently. For example, a useful operation on heap-ordered trees is relaxing the lower bound. It can be regarded as an upgraded version of the identity function on Tree, since it leaves the tree structure intact, changing only the ordering information. With the help of the optimised predicate for $\lceil HeapOD \rceil$,

```
Heap′ : Val → Set
Heap′ b = OptP ⌈HeapOD⌉ (ok b)

indexfirst data Heap′ : Val → Tree → Set where
  Heap′ b nil        ∋ nil
  Heap′ b (node t u) ∋ node (x : Val) (b⩽x : b ⩽ x)
                            (t′ : Heap x t) (u′ : Heap x u)
```

we can give the type of bound-relaxing in predicate form, stating explicitly in the type that the underlying tree structure is unchanged:

```
relax : {b b′ : Val} → b′ ⩽ b → {t : Tree} → Heap′ b t → Heap′ b′ t
relax b′⩽b {nil}      nil                = nil
relax b′⩽b {node _ _} (node x b⩽x t u) = node x (⩽-trans b′⩽b b⩽x) t u
```

Since the identity function on LTree can also be seen as an upgraded version of the identity function on Tree, we can combine *relax* and the predicate form of the identity function on LTree to get bound-relaxing on leftist heaps, which modifies only the heap-ordering portion of a leftist heap:

```
lhrelax : {b b′ : Val} → b′ ⩽ b → {r : Nat} → LHeap b r → LHeap b′ r
lhrelax {b} {b′} b′⩽b {r} = Upgrade.u upg id (λ _ ↦ relax b′⩽b ∗ id)
```

```
  where
    ref : (b'' : Val) → Refinement Tree (LHeap b'' r)
    ref b'' = toRefinement
                (⊗-FSwap ⌈HeapOD⌉ ⌈LTreeOD⌉ id-FSwap id-FSwap
                  (ok (ok b'' , ok r)))
    upg : Upgrade (Tree → Tree) (LHeap b r → LHeap b' r)
    upg = ref b ⇀ toUpgrade (ref b')
```

In general, non-modifying heap operations do not depend on the leftist property and can be implemented for heap-ordered trees and later lifted to work with leftist heaps, relieving us of the unnecessary work of dealing with the leftist property when it is simply to be ignored. For another example, converting a leftist heap to a list of its elements has nothing to do with the leftist property. In fact, it even has nothing to do with heap ordering, but only with the internal labelling. Hence we can define the **internally labelled trees** as an ornamentation of skeletal binary trees:

```
ITreeOD : Set → OrnDesc ⊤ ! TreeD
ITreeOD A ▪ = σ TreeTag λ { 'nil  ↦ v ▪
                          ; 'node ↦ Δ[ _ : A ] v (ok tt, ok tt, ▪) }

indexfirst data ITree (A : Set) : Set where
  ITree A ∋ nil
          | node (x : A) (t : ITree A) (u : ITree A)
```

on which we can do preorder traversal:

```
preorder : {A : Set} → ITree A → List A
preorder nil        = []
preorder (node x t u) = x :: preorder t ++ preorder u
```

We have an ornament from internally labelled trees to heap-ordered trees:

```
ITreeD-HeapD : Orn ! ⌊ITreeOD Val⌋ ⌊HeapOD⌋
ITreeD-HeapD (ok b) =
  σ TreeTag λ { 'nil  ↦ v []
              ; 'node ↦ σ[ x : Val ] Δ[ _ : b ⩽ x ] v (refl :: refl :: []) }
```

So, to get a list of the elements of a leftist heap (whose first element is the minimum one), we convert the leftist heap to an internally labelled tree and then invoke *preorder*.

use an ornament-parametrised upgrade

$$\begin{aligned}&\mathit{toList} : \{b : \mathit{Val}\}\ \{r : \mathsf{Nat}\} \to \mathsf{LHeap}\ b\ r \to \mathsf{List}\ \mathit{Val}\\&\mathit{toList} = \mathit{preorder} \circ \mathit{forget}\ (\mathit{ITreeD\text{-}HeapD} \odot \mathit{diffOrn\text{-}l}\ \lceil \mathit{HeapOD} \rceil\ \lceil \mathit{LTreeOD} \rceil)\end{aligned}$$

For modifying operations, however, we need to consider both heap ordering and the leftist property at the same time, so we should program directly with the composite datatype of leftist heaps. For example, a key operation is merging two heaps:

$$\begin{aligned}\mathit{merge} : {} & \{b_0 : \mathit{Val}\}\ \{r_0 : \mathsf{Nat}\} \to \mathsf{LHeap}\ b_0\ r_0 \to\\ & \{b_1 : \mathit{Val}\}\ \{r_1 : \mathsf{Nat}\} \to \mathsf{LHeap}\ b_1\ r_1 \to\\ & \{b : \mathit{Val}\} \to b \leqslant b_0 \to b \leqslant b_1 \to \Sigma[\, r : \mathsf{Nat}\,]\ \mathsf{LHeap}\ b\ r\end{aligned}$$

with which we can easily implement insertion of a new element (by merging with a singleton heap) and deletion of the minimum element (by deleting the root and merging the two sub-heaps). The definition of *merge* is shown in Figure 3.8. It is a more precisely typed version of Okasaki's implementation, split into two mutually recursive functions to make it clear to Agda's termination checker that we are doing two-level induction on the ranks of the two input heaps. When one of the ranks is zero, meaning that the corresponding heap is nil, we simply return the other heap (whose bound is suitably relaxed) as the result. When both ranks are nonzero, meaning that both heaps are nonempty, we compare the roots of the two heaps and recursively merge the heap with the larger root into the right branch of the heap with the smaller root. The recursion is structural because the rank of the right branch of a nonempty heap is strictly smaller. There is a catch, however: the rank of the new right subheap resulting from the recursive merging might be larger than that of the left sub-heap, violating the leftist property, so there is a helper function *makeT* that swaps the sub-heaps when necessary.

```
makeT : (x : Nat) → {r₀ : Nat} (h₀ : LHeap x r₀) →
                    {r₁ : Nat} (h₁ : LHeap x r₁) → Σ[ r : Nat ] LHeap x r
makeT x {r₀} h₀ {r₁} h₁ with r₀ ⩽? r₁
makeT x {r₀} h₀ {r₁} h₁ | yes r₀⩽r₁ = suc r₀ , node x ⩽-refl r₀⩽r₁          h₁ h₀
makeT x {r₀} h₀ {r₁} h₁ | no  r₀⩰r₁ = suc r₁ , node x ⩽-refl (⩰-invert r₀⩰r₁) h₀ h₁

mutual

  merge : {b₀ : Val} {r₀ : Nat} → LHeap b₀ r₀ →
          {b₁ : Val} {r₁ : Nat} → LHeap b₁ r₁ →
          {b  : Val} → b ⩽ b₀ → b ⩽ b₁ → Σ[ r : Nat ] LHeap b r
  merge {b₀} {zero  } nil h₁ b⩽b₀ b⩽b₁ = _ , lhrelax b⩽b₁ h₁
  merge {b₀} {suc r₀} h₀ h₁ b⩽b₀ b⩽b₁ = merge′ h₀ h₁ b⩽b₀ b⩽b₁

  merge′ : {b₀ : Val} {r₀ : Nat} → LHeap b₀ (suc r₀) →
           {b₁ : Val} {r₁ : Nat} → LHeap b₁ r₁       →
           {b  : Val} → b ⩽ b₀ → b ⩽ b₁ → Σ[ r : Nat ] LHeap b r
  merge′ h₀                  {b₁} {zero  } nil                                  b⩽b₀ b⩽b₁ = _ , lhrelax b⩽b₀ h₀
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁) b⩽b₀ b⩽b₁ with x₀ ⩽? x₁
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁) b⩽b₀ b⩽b₁ | yes x₀⩽x₁ =
    _ , lhrelax (⩽-trans b⩽b₀ b₀⩽x₀) (outr (makeT x₀ t₀ (outr (merge u₀ (node x₁ x₀⩽x₁ r₁⩽l₁ t₁ u₁) ⩽-refl ⩽-refl))))
  merge′ (node x₀ b₀⩽x₀ r₀⩽l₀ t₀ u₀) {b₁} {suc r₁} (node x₁ b₁⩽x₁ r₁⩽l₁ t₁ u₁) b⩽b₀ b⩽b₁ | no  x₀⩰x₁ =
    _ , lhrelax (⩽-trans b⩽b₁ b₁⩽x₁) (outr (makeT x₁ t₁ (outr (merge′ (node x₀ (⩰-invert x₀⩰x₁) r₀⩽l₀ t₀ u₀) u₁ ⩽-refl ⩽-refl))))
```

Figure 3.8 Merging two leftist heaps. Proof terms about ordering are coloured grey to aid comprehension (taking inspiration from — but not really employing — Bernardy and Guilhem's type theory in colour [2013]).

Weight-biased leftist heaps

Another advantage of separating the leftist property and heap ordering is that we can swap the leftist property for another balancing property. The non-modifying operations, previously defined for heap-ordered trees, can be upgraded to work with the new balanced heap datatype in the same way, while the modifying operations are reimplemented with respect to the new balancing property. For example, the leftist property requires that the **rank** of the left sub-tree is at least that of the right one; we can replace "rank" with "size" in its statement and get the **weight-biased leftist property**. This is again codified as an ornamentation of skeletal binary trees:

$$
\begin{aligned}
&\mathit{WLTreeOD} \;:\; \mathsf{OrnDesc}\ \mathsf{Nat}\ !\ \mathit{TreeD}\\
&\mathit{WLTreeOD}\ (\mathsf{ok}\ \mathsf{zero}\quad) \;=\; \nabla[\,\text{`}\mathsf{nil}\,]\ \mathsf{v}\ \blacksquare\\
&\mathit{WLTreeOD}\ (\mathsf{ok}\ (\mathsf{suc}\ n)) \;=\; \nabla[\,\text{`}\mathsf{node}\,]\ \Delta[\,l : \mathsf{Nat}\,]\ \Delta[\,r : \mathsf{Nat}\,]\\
&\qquad\qquad\qquad\qquad \Delta[\,\_ : r \leqslant l\,]\ \Delta[\,\_ : n \equiv l + r\,]\ \mathsf{v}\ (\mathsf{ok}\ l\ ,\ \mathsf{ok}\ r\ ,\ \blacksquare)
\end{aligned}
$$

$$
\begin{aligned}
&\textbf{indexfirst data}\ \mathsf{WLTree}\ :\ \mathsf{Nat} \to \mathsf{Set}\ \textbf{where}\\
&\quad \mathsf{WLTree}\ \mathsf{zero}\quad \ni\ \mathsf{nil}\\
&\quad \mathsf{WLTree}\ (\mathsf{suc}\ n)\ \ni\ \mathsf{node}\ \{l\ :\ \mathsf{Nat}\}\ \{r\ :\ \mathsf{Nat}\}\\
&\qquad\qquad\qquad\qquad (r{\leqslant}l\ :\ r \leqslant l)\ (n{\equiv}l{+}r\ :\ n \equiv l + r)\\
&\qquad\qquad\qquad\qquad (t\ :\ \mathsf{WLTree}\ l)\ (u\ :\ \mathsf{WLTree}\ r)
\end{aligned}
$$

which can be composed in parallel with the heap-ordering ornament $\lceil \mathit{HeapOD} \rceil$ and gives us weight-biased leftist heaps.

$$
\begin{aligned}
&\mathit{WLHeapD}\ :\ \mathsf{Desc}\ (!\bowtie !)\\
&\mathit{WLHeapD}\ =\ \lfloor \lceil \mathit{HeapOD} \rceil \otimes \lceil \mathit{WLTreeOD} \rceil \rfloor
\end{aligned}
$$

$$
\begin{aligned}
&\textbf{indexfirst data}\ \mathsf{WLHeap}\ :\ \mathit{Val} \to \mathsf{Nat} \to \mathsf{Set}\ \textbf{where}\\
&\quad \mathsf{WLHeap}\ b\ \mathsf{zero}\quad \ni\ \mathsf{nil}\\
&\quad \mathsf{WLHeap}\ b\ (\mathsf{suc}\ n)\ \ni\ \mathsf{node}\ (x\ :\ \mathit{Val})\ (b{\leqslant}x\ :\ b \leqslant x)\\
&\qquad\qquad\qquad\qquad \{l\ :\ \mathsf{Nat}\}\ \{r\ :\ \mathsf{Nat}\}\\
&\qquad\qquad\qquad\qquad (r{\leqslant}l\ :\ r \leqslant l)\ (n{\equiv}l{+}r\ :\ n \equiv l + r)\\
&\qquad\qquad\qquad\qquad (t\ :\ \mathsf{WLHeap}\ x\ l)\ (u\ :\ \mathsf{WLHeap}\ x\ r)
\end{aligned}
$$

The weight-biased leftist property makes it possible to reimplement merg-

ing in a single, top-down pass rather than two passes: With the original rank-biased leftist property, recursive calls to *merge* are determined top-down by comparing root elements, and the helper function *makeT* swaps a recursively computed sub-heap with the other sub-heap if the rank of the former is larger; the rank of a recursively computed sub-heap, however, is not known before a recursive call returns (which is reflected by the existential quantification of the rank index in the result type of *merge*), so during the whole merging process *makeT* does the swapping in a second bottom-up pass. On the other hand, with the weight-biased leftist property, the merging operation has type

$$\begin{aligned} \mathit{wmerge} \;:\; & \{b_0 \;:\; \mathit{Val}\}\; \{n_0 \;:\; \mathsf{Nat}\} \to \mathsf{WLHeap}\; b_0\; n_0 \to \\ & \{b_1 \;:\; \mathit{Val}\}\; \{n_1 \;:\; \mathsf{Nat}\} \to \mathsf{WLHeap}\; b_1\; n_1 \to \\ & \{b \;\;:\; \mathit{Val}\} \to b \leqslant b_0 \to b \leqslant b_1 \to \mathsf{WLHeap}\; b\; (n_0 + n_1) \end{aligned}$$

The implementation of *wmerge* is largely similar to *merge* and is omitted here. For *wmerge*, however, the weight of a recursively computed sub-heap is known before the recursive merging is actually performed (so the weight index can be given explicitly in the result type of *wmerge*). The counterpart of *makeT* can thus determine before a recursive call whether to do the swapping or not, and the whole merging process requires only one top-down pass.

Do we need a summary here?

3.5 Discussion

summary of the three-level architecture of ornaments, refinements, and upgrades; bundle; why ornaments?; functor-level computation and recursion schemes; compare with Bernardy and Guilhem [2013]

Chapter 4

Categorical organisation of the ornament–refinement framework

Chapter 3 left some obvious holes in the theory of ornaments. For instance:

- When it comes to composition of ornaments, the following **sequential composition** is probably the first that comes to mind (rather than parallel composition), which is evidence that the ornamental relation is transitive:

  $$\begin{aligned}&\_\odot\_ : \{I\ J\ K : \mathsf{Set}\}\ \{e : J \to I\}\ \{f : K \to J\} \to \\ &\quad \{D : \mathsf{Desc}\ I\}\ \{E : \mathsf{Desc}\ J\}\ \{F : \mathsf{Desc}\ K\} \to \\ &\quad \mathsf{Orn}\ e\ D\ E \to \mathsf{Orn}\ f\ E\ F \to \mathsf{Orn}\ (e \circ f)\ D\ F\end{aligned}$$

  -- definition in Figure 4.4

  Correspondingly, we expect that

  $$\mathit{forget}\ (O \odot P) \qquad \text{and} \qquad \mathit{forget}\ O \circ \mathit{forget}\ P$$

  are extensionally equal. That is, the sequential compositional structure of ornaments corresponds to the compositional structure of forgetful functions. We wish to state such correspondences in concise terms.

- While parallel composition of ornaments has a sensible definition (Section 3.2.3), it is defined by case analysis at the microscopic level of individual fields. Such a microscopic definition is difficult to comprehend, and so are any subsequent definitions and proofs. It is desirable to have a macroscopic char-

acterisation of parallel composition, so the nature of parallel composition is immediately clear, and subsequent definitions and proofs can be done in a more abstract manner.

- The ornamental conversion isomorphisms (Section 3.3.1) and the modularity isomorphisms (Section 3.3.2) were left unimplemented. Both sets of isomorphisms are about the optimised predicates (Section 3.3.1), which are defined in terms of parallel composition with singleton ornaments (Section 3.2.2). We thus expect that the existence of these isomorphisms can be explained in terms of properties of parallel composition and singleton ornaments.

A lightweight organisation of the ornament–refinement framework in basic category theory [Mac Lane, 1998] can help to fill up all these holes. In more detail:

- Categories and functors are abstractions for compositional structures and structure-preserving maps between them. Facts about translations between ornaments, refinements, and functions can thus be neatly organised under the categorical language (Section 4.1).
- Parallel composition merges two compatible ornaments and does nothing more; in other words, it computes the least informative ornament that contains the information of both ornaments. Characterisation of such **universal constructions** is a speciality of category theory; in our case, parallel composition can be shown to satisfy some **pullback properties** (Section 4.2).
- Universal constructions are unique up to isomorphism, so it is convenient to establish isomorphisms about universal constructions. The pullback properties of parallel composition can thus help to construct the ornamental conversion isomorphisms (Section 4.3.1) and the modularity isomorphisms (Section 4.3.2).



Figure 4.1 Categories and functors for the ornament–refinement framework.

4.1 Categories and functors

A first approximation of a category is a (directed multi-) **graph**, which consists of a set of objects (nodes) and a collection of sets of morphisms (edges) indexed with their source and target objects:

```
record Graph {l m : Level} : Set (suc (l ⊔ m)) where
  field
    Object : Set l
    _⇒_    : Object → Object → Set m
```

For example, the underlying graph of the category 𝔽un of (small) sets and (total) functions is just

```
𝔽un-graph : Graph
𝔽un-graph = record { Object = Set
                   ; _⇒_    = λ A B ↦ A → B }
```

A category is a graph whose morphisms are equipped with a monoid-like compositional structure — there is a morphism composition operator of type

$$\_\cdot\_ : \{X\ Y\ Z : \mathit{Object}\} \to (Y \Rightarrow Z) \to (X \Rightarrow Y) \to (X \Rightarrow Z)$$

which has identities and is associative.

Syntactic remark (*universe polymorphism*)**.** Many definitions in this chapter

(like Graph above) employ Agda's universe polymorphism [Harper and Pollack, 1991], so the definitions can be instantiated at suitable levels of the Set hierarchy as needed. (For example, the type of 𝔽un-*graph* is implicitly instantiated as Graph $\{1\}$ $\{1\}$, since both Set and any $A \to B$ (where A, B : Set) are of type Set_1.) We will give the first few universe-polymorphic definitions with full detail about the levels, but will later suppress the syntactic noise wherever possible. □

Before we move on to the definition of categories, though, special attention must be paid to equality on morphisms, which is usually coarser than definitional equality — in 𝔽un, for example, it is necessary to identify functions up to extensional equality (so uniqueness of morphisms in universal properties would make sense). One ad hoc way to achieve this in Agda's intensional setting is to use **setoids** [Barthe et al., 2003] — sets with an explicitly specified equivalence relation — to represent sets of morphisms. The type of setoids can be defined as a record which contains a carrier set, an equivalence relation on the set, and the three laws for the equivalence relation:

```
record Setoid {c d : Level} : Set (suc (c ⊔ d)) where
  field
    Carrier : Set c
    _≈_     : Carrier → Carrier → Set d
    refl    : {x : Carrier} → x ≈ x
    sym     : {x y : Carrier} → x ≈ y → y ≈ x
    trans   : {x y z : Carrier} → x ≈ y → y ≈ z → x ≈ z
```

For example, we can define a setoid of functions that uses extensional equality:

```
FunSetoid : Set → Set → Setoid
FunSetoid A B = record { Carrier = A → B
                       ; _≈_     = _≐_
                       ; proofs of laws }
```

Proofs of the three laws are omitted from the presentation.

Similarly, we can define the type of categories as a record containing a set of objects, a collection of **setoids** of morphisms indexed by source and target

```
record Category {l m n : Level} : Set (suc (l ⊔ m ⊔ n)) where
  field
    Object      : Set l
    Morphism    : Object → Object → Setoid {m} {n}
  _⇒_ : Object → Object → Set m
  X ⇒ Y = Setoid.Carrier (Morphism X Y)
  _≈_ : {X Y : Object} → (X ⇒ Y) → (X ⇒ Y) → Set n
  _≈_ {X} {Y} = Setoid._≈_ (Morphism X Y)
  field
    _·_   : {X Y Z : Object} → (Y ⇒ Z) → (X ⇒ Y) → (X ⇒ Z)
    id    : {X : Object} → (X ⇒ X)
    id-l  : {X Y : Object} (f : X ⇒ Y) →
            id · f ≈ f
    id-r  : {X Y : Object} (f : X ⇒ Y) →
            f · id ≈ f
    assoc : {X Y Z W : Object} (f : Z ⇒ W) (g : Y ⇒ Z) (h : X ⇒ Y) →
            (f · g) · h ≈ f · (g · h)
    cong-l : {X Y Z : Object} {f g : Y ⇒ Z} (h : X ⇒ Y) →
             f ≈ g → f · h ≈ g · h
    cong-r : {X Y Z : Object} (h : Y ⇒ Z) {f g : X ⇒ Y} →
             f ≈ g → h · f ≈ h · g
```

Figure 4.2 Definition of categories.

objects, the composition operator on morphisms, the identity morphisms, and the identity and associativity laws for composition. The definition is shown in Figure 4.2. Two notations are introduced to improve readability: $X \Rightarrow Y$ is defined to be the carrier set of the setoid of morphisms from X to Y, and $f \approx g$ is defined to be the equivalence between the morphisms f and g as specified by the setoid to which f and g belong. The last two laws *cong-l* and *cong-r* require morphism composition to preserve the equivalence on morphisms; they are given in this form to work better with the equational reasoning combinators commonly used in Agda (see, for example, the AoPA library [Mu et al., 2009]).

Now we can define the category $\mathbb{F}\textsc{un}$ of sets and functions as

$$
\begin{array}{l}
\mathbb{F}\textsc{un} : \mathsf{Category} \\
\mathbb{F}\textsc{un} = \textbf{record} \{ \mathit{Object} \quad = \mathsf{Set} \\
\qquad\qquad\quad ; \mathit{Morphism} = \mathit{FunSetoid} \\
\qquad\qquad\quad ; \_\cdot\_ = \_\circ\_ \\
\qquad\qquad\quad ; \mathit{id} \quad = \lambda\, x \mapsto x \\
\qquad\qquad\quad ; \text{proofs of laws} \}
\end{array}
$$

Another important category that we will make use of is $\mathbb{F}\textsc{am}$, the category of indexed families of sets and indexed families of functions, which is useful for talking about componentwise structures. An object in $\mathbb{F}\textsc{am}$ has type $\Sigma[\, I : \mathsf{Set} \,]\ I \to \mathsf{Set}$, i.e., it is a set I and a family of sets indexed by I; a morphism from $(J\ ,\ Y)$ to $(I\ ,\ X)$ is a function $e : J \to I$ and a family of functions from $Y\ j$ to $X\ (e\ j)$ for each $j : J$.

$$
\begin{array}{l}
\mathbb{F}\textsc{am} : \mathsf{Category} \\
\mathbb{F}\textsc{am} = \textbf{record} \\
\quad \{ \mathit{Object} \quad = \Sigma[\, I : \mathsf{Set} \,]\ I \to \mathsf{Set} \\
\quad ; \mathit{Morphism} = \lambda\, (J\ ,\ Y)\ (I\ ,\ X) \mapsto \textbf{record} \\
\qquad\qquad\qquad \{ \mathit{Carrier} = \Sigma[\, e : J \to I \,]\ Y \rightrightarrows (X \circ e) \\
\qquad\qquad\qquad ; \_\approx\_ \quad = \lambda\, (e\ ,\ u)\ (e'\ ,\ u') \mapsto \\
\qquad\qquad\qquad\qquad\qquad\qquad (e \doteq e') \times ((j : J) \to u\ \{j\} \mathrel{\dot{\cong}} u'\ \{j\}) \\
\qquad\qquad\qquad ; \text{proofs of laws} \} \\
\quad ; \_\cdot\_ = \lambda\, (e\ ,\ u)\ (f\ ,\ v) \mapsto (e \circ f)\ ,\ (\lambda\, \{k\} \mapsto u\ \{f\ k\} \circ v\ \{k\})
\end{array}
$$

```
record Functor {l m n l′ m′ n′ : Level}
  (C : Category {l} {m} {n}) (D : Category {l′} {m′} {n′}) :
  Set (l ⊔ m ⊔ n ⊔ l′ ⊔ m′ ⊔ n′) where
  field
    object      : Object C → Object D
    morphism    : {X Y : Object C} → X ⇒C Y → object X ⇒D object Y
    equiv-preserving : {X Y : Object C} {f g : X ⇒C Y} →
                       f ≈C g → morphism f ≈D morphism g
    id-preserving    : {X : Object C} →
                       morphism (id C {X}) ≈D id D {object X}
    comp-preserving  : {X Y Z : Object C} (f : Y ⇒C Z) (g : X ⇒C Y) →
                       morphism (f ·C g) ≈D (morphism f ·D morphism g)
```

Figure 4.3 Definition of functors. Subscripts are used to indicate to which category an operator belongs.

```
  ; id    = (λ x ↦ x) , (λ {i} x ↦ x)
  ; proofs of laws }
```

Note that the equivalence on morphisms is defined to be componentwise extensional equality, which is formulated with the help of McBride's "John Major" heterogeneous equality $\_\cong\_$ [McBride, 1999] — the equivalence $\_\dot{\cong}\_$ is defined by $g \mathrel{\dot{\cong}} h = \forall\, x \to g\, x \cong h\, x$. (Given $y : Y\, j$ for some $j : J$, the types of $u\ \{j\}\ y$ and $u'\ \{j\}\ y$ are not definitionally equal but only provably equal, so it is necessary to employ heterogeneous equality.)

Categories are graphs with a compositional structure, and **functors** are transformations between categories that preserve the compositional structure. The definition of functors is shown in Figure 4.3: a functor consists of two mappings, one on objects and the other on morphisms, where the morphism part preserves all structures on morphisms, including equivalence, identity, and composition. For example, we have two functors from $\mathbb{F}\textsc{am}$ to $\mathbb{F}\textsc{un}$, one summing components together

$$\begin{aligned}
&\textsc{Com} : \mathsf{Functor}\ \mathbb{F}\textsc{am}\ \mathbb{F}\textsc{un} \quad \text{-- the comprehension functor}\\
&\textsc{Com} = \mathbf{record}\ \{\ \mathit{object} \quad = \lambda\,(I\ ,\ X) \mapsto \Sigma\ I\ X\\
&\qquad\qquad\ ;\ \mathit{morphism} = \lambda\,(e\ ,\ u\ ) \mapsto e * u\\
&\qquad\qquad\ ;\ \text{proofs of laws}\ \}
\end{aligned}$$

and the other extracting the index part.

$$\begin{aligned}
&\textsc{FamF} : \mathsf{Functor}\ \mathbb{F}\textsc{am}\ \mathbb{F}\textsc{un} \quad \text{-- the family fibration functor}\\
&\textsc{FamF} = \mathbf{record}\ \{\ \mathit{object} \quad = \lambda\,(I\ ,\ X) \mapsto I\\
&\qquad\qquad\ ;\ \mathit{morphism} = \lambda\,(e\ ,\ u\ ) \mapsto e\\
&\qquad\qquad\ ;\ \text{proofs of laws}\ \}
\end{aligned}$$

The functor laws should be proved for both functors alongside their object and morphism maps. In particular, we need to prove that the morphism part preserves equivalence: for Com, this means we need to prove, for all $e : J \to I$, $u : Y \rightrightarrows (X \circ e)$ and $f : J \to I$, $v : Y \rightrightarrows (X \circ f)$, that

$$(e \doteq f) \times ((j : J) \to u\ \{j\} \mathrel{\dot{\cong}} v\ \{j\}) \to (e * u \doteq f * v)$$

and for FamF we need to prove

$$(e \doteq f) \times ((j : J) \to u\ \{j\} \mathrel{\dot{\cong}} v\ \{j\}) \to (e \doteq f)$$

both of which can be easily discharged.

Categories for refinements and ornaments

Some constructions in Chapter 3 can now be organised under several categories and functors. For a start, we already saw that refinements are interesting only because of their intensional contents; this is reflected in an isomorphism of categories between the category $\mathbb{F}\textsc{am}$ and the category $\mathbb{F}\textsc{ref}$ of type families and refinement families (i.e., there are two functors back and forth inverse to each other). An object in $\mathbb{F}\textsc{ref}$ is an indexed family of sets as in $\mathbb{F}\textsc{am}$, and a morphism from $(J\ ,\ Y)$ to $(I\ ,\ X)$ consists of a function $e : J \to I$ on the indices and a refinement family of type $\mathsf{FRefinement}\ e\ X\ Y$. As for the equivalence on morphisms, it suffices to use extensional equality on the index functions and componentwise extensional equality on refinement families, where extensional

equality on refinements means extensional equality on their forgetful functions (extracted by Refinement.*forget*), which we have shown in Section 3.1.1 to be the core of refinements. Formally:

$$
\begin{array}{l}
\mathbb{F}\textsc{ref} : \mathsf{Category} \\
\mathbb{F}\textsc{ref} = \mathbf{record} \\
\quad \{\ \mathit{Object} \quad = \Sigma[\, I : \mathsf{Set} \,]\ I \to \mathsf{Set} \\
\quad ;\ \mathit{Morphism} = \lambda\,(J\,,\,Y)\ (I\,,\,X) \mapsto \mathbf{record} \\
\qquad\qquad \{\ \mathit{Carrier} = \Sigma[\, e : J \to I \,]\ \mathsf{FRefinement}\ e\ X\ Y \\
\qquad\qquad ;\ \_{\approx}\_ = \lambda\,(e\,,\,rs)\ (e'\,,\,rs') \mapsto \\
\qquad\qquad\qquad\qquad (e \doteq e') \times \\
\qquad\qquad\qquad\qquad ((j : J) \to \mathsf{Refinement}.\mathit{forget}\ (rs\ \ (\mathsf{ok}\ j)) \mathrel{\dot{\approx}} \\
\qquad\qquad\qquad\qquad\qquad\qquad \mathsf{Refinement}.\mathit{forget}\ (rs'\ (\mathsf{ok}\ j))) \\
\qquad\qquad ;\ \text{proofs of laws}\ \} \\
\quad ;\ \text{proofs of laws}\ \}
\end{array}
$$

Note that a refinement family from $X : I \to \mathsf{Set}$ to $Y : J \to \mathsf{Set}$ is deliberately cast as a morphism in the opposite direction from $(J\,,\,Y)$ to $(I\,,\,X)$; think of this as suggesting the direction of the forgetful functions of refinements. $\mathbb{F}\textsc{ref}$ is no more powerful than $\mathbb{F}\textsc{am}$ since $\mathbb{F}\textsc{ref}$ ignores the intensional contents of refinements by using an extensional equality, and consequently there are two functors between $\mathbb{F}\textsc{ref}$ and $\mathbb{F}\textsc{am}$ that are inverse to each other, forming an isomorphism of categories:

- We have a forgetful functor FRefF : Functor $\mathbb{F}\textsc{ref}$ $\mathbb{F}\textsc{am}$ which is identity on objects and componentwise Refinement.*forget* on morphisms (which preserves equivalence automatically):

  $$
  \begin{array}{l}
  \textsc{RefF} : \mathsf{Functor}\ \mathbb{F}\textsc{ref}\ \mathbb{F}\textsc{am} \\
  \textsc{RefF} = \mathbf{record} \\
  \quad \{\ \mathit{object} \quad = \mathit{id} \\
  \quad ;\ \mathit{morphism} = \lambda\,(e\,,\,rs) \mapsto e\,,\,(\lambda\, j \mapsto \mathsf{Refinement}.\mathit{forget}\ (rs\ (\mathsf{ok}\ j))) \\
  \quad ;\ \text{proofs of laws}\ \}
  \end{array}
  $$

  Note that FRefF remains a familiar covariant functor rather than a contravariant one because of our choice of morphism direction.

- Conversely, there is a functor FRefC : Functor 𝔽am 𝔽ref whose object part is identity and whose morphism part is componentwise *canonRef*:

  ```
  FRefC : Functor 𝔽am 𝔽ref
  FRefC = record
    { object   = id
    ; morphism = λ (e , u) ↦ e , (λ (ok j) ↦ canonRef (u {j}))
    ; proofs of laws }
  ```

The two functors FRefF and FRefC are inverse to each other by definition.

There is another category 𝕆rn, which has objects of type $\Sigma[\, I : \mathsf{Set}\,]\ \mathsf{Desc}\ I$, i.e., descriptions paired with index sets, and morphisms from $(J\ ,\ E)$ to $(I\ ,\ D)$ of type $\Sigma[\, e\ :\ J\ \to\ I\,]\ \ \mathsf{Orn}\ e\ D\ E$, i.e., ornaments paired with index erasure functions. To complete the definition of 𝕆rn:

- We need to devise an equivalence on ornaments

  ```
  OrnEq : {I J : Set} {e f : J → I} {D : Desc I} {E : Desc J} →
          Orn e D E → Orn f D E → Set
  ```

  such that it implies extensional equality of e and f and that of ornamental forgetful functions:

  ```
  OrnEq-forget : {I J : Set} {e f : J → I} {D : Desc I} {E : Desc J} →
                 (O : Orn e D E) (P : Orn f D E) → OrnEq O P →
                 (e ≐ f) × ((j : J) → forget O {j} ≅̃ forget P {j})
  ```

  The actual definition of *OrnEq* is deferred until Chapter 6.

- Morphism composition is sequential composition $\_\odot\_$, which merges two successive batches of modifications in a straightforward way. The definition is shown in Figure 4.4. There is also a family of **identity ornaments**:

  ```
  idOrn : {I : Set} (D : Desc I) → Orn id D D
  idOrn {I} D (ok i) = idROrn (D i)
    where
      𝔼-refl : (is : List I) → 𝔼 id is is
      𝔼-refl []       = []
      𝔼-refl (i :: is) = refl :: 𝔼-refl is
  ```

```
𝔼-trans : {I J K : Set} {e : J → I} {f : K → J} →
          {is : List I} {js : List J} {ks : List K} →
          𝔼 e js is → 𝔼 f ks js → 𝔼 (e ∘ f) ks is
𝔼-trans          []          []          = []
𝔼-trans {e := e} (eeq :: eeqs) (feq :: feqs) = trans (cong e feq) eeq ::
                                               𝔼-trans eeqs feqs

scROrn : {I J K : Set} {e : J → I} {f : K → J} →
         {D : RDesc I} {E : RDesc J} {F : RDesc K} →
         ROrn e D E → ROrn f E F → ROrn (e ∘ f) D F
scROrn (v eeqs) (v feqs)   = v (𝔼-trans eeqs feqs)
scROrn (v eeqs) (Δ T P)    = Δ[ t : T ]  scROrn (v eeqs) (P t)
scROrn (σ S O)  (σ .S P)   = σ[ s : S ]  scROrn (O s)    (P s)
scROrn (σ S O)  (Δ T P)    = Δ[ t : T ]  scROrn (σ S O)  (P t)
scROrn (σ S O)  (∇ s P)    = ∇[ s ]      scROrn (O s)    P
scROrn (Δ T O)  (σ .T P)   = Δ[ t : T ]  scROrn (O t)    (P t)
scROrn (Δ T O)  (Δ U P)    = Δ[ u : U ]  scROrn (Δ T O)  (P u)
scROrn (Δ T O)  (∇ t P)    =             scROrn (O t)    P
scROrn (∇ s O)  P          = ∇[ s ]      scROrn O        P

_⊙_ : {I J K : Set} {e : J → I} {f : K → J} →
      {D : Desc I} {E : Desc J} {F : Desc K} →
      Orn e D E → Orn f E F → Orn (e ∘ f) D F
_⊙_ {f := f} O P (ok k) = scROrn (O (ok (f k))) (P (ok k))
```

Figure 4.4 Definitions for sequential composition of ornaments.

$$
\begin{aligned}
&\mathit{idROrn} \;:\; (E \;:\; \mathsf{RDesc}\; I) \to \mathsf{ROrn}\; \mathit{id}\; E\; E\\
&\mathit{idROrn}\; (\mathsf{v}\; \mathit{is}) \;\;\;\; = \; \mathsf{v}\; (\mathbb{E}\text{-}\mathit{refl}\; \mathit{is})\\
&\mathit{idROrn}\; (\sigma\; S\; E) \;=\; \sigma[\, s : S\,]\;\; \mathit{idROrn}\; (E\; s)
\end{aligned}
$$

which simply use σ and v everywhere to express that a description is identical to itself. Unsurprisingly, the identity ornaments serve as identity of sequential composition.

To summarise:

$$
\begin{aligned}
&\mathbb{O}\mathrm{RN} \;:\; \mathsf{Category}\\
&\mathbb{O}\mathrm{RN} \;=\; \mathbf{record}\\
&\quad \{\; \mathit{Object} \;\;\;\; = \; \Sigma[\, I : \mathsf{Set}\,]\;\; \mathsf{Desc}\; I\\
&\quad ;\; \mathit{Morphism} \;=\; \lambda\, (J\;,\; E)\; (I\;,\; D) \;\mapsto\; \mathbf{record}\\
&\qquad\qquad \{\; \mathit{Carrier} \;=\; \Sigma[\, e : J \to I\,]\;\; \mathsf{Orn}\; e\; D\; E\\
&\qquad\qquad ;\; \_{\approx}\_ \;\;\;\; = \; \lambda\, (e\;,\; O)\; (f\;,\; P) \;\mapsto\; \mathit{OrnEq}\; O\; P\\
&\qquad\qquad ;\; \text{proofs of laws}\;\}\\
&\quad ;\; \_{\cdot}\_ \;=\; \lambda\, (e\;,\; O)\; (f\;,\; P) \;\mapsto\; (e \circ f)\;,\; (O \odot P)\\
&\quad ;\; \mathit{id} \;\;\;\; = \; \lambda\, \{I\;,\; D\} \;\mapsto\; \mathit{id}\;,\; \mathit{idOrn}\; D\\
&\quad ;\; \text{proofs of laws}\;\}
\end{aligned}
$$

A functor $\textsc{Ind} \;:\; \mathsf{Functor}\; \mathbb{O}\mathrm{RN}\; \mathbb{F}\mathrm{AM}$ can then be constructed, which gives the ordinary semantics of descriptions and ornaments: the object part of Ind decodes a description $(I\;,\; D)$ to its least fixed point $(I\;,\; \mu\; D)$, and the morphism part translates an ornament $(e\;,\; O)$ to the forgetful function $(e\;,\; \mathit{forget}\; O)$, the latter preserving equivalence by virtue of *OrnEq-forget*.

$$
\begin{aligned}
&\textsc{Ind} \;:\; \mathsf{Functor}\; \mathbb{O}\mathrm{RN}\; \mathbb{F}\mathrm{AM}\\
&\textsc{Ind} \;=\; \mathbf{record}\; \{\; \mathit{object} \;\;\;\; = \; \lambda\, (I\;,\; D) \;\mapsto\; I\;,\; \mu\; D\\
&\qquad\qquad\quad ;\; \mathit{morphism} \;=\; \lambda\, (e\;,\; O) \;\mapsto\; e\;,\mathit{forget}\; O\\
&\qquad\qquad\quad ;\; \text{proofs of laws}\;\}
\end{aligned}
$$

To translate Orn to Fref, i.e., datatype declarations to refinements, a naive way is to use the composite functor

$$\mathbb{O}\mathrm{RN} \xrightarrow{\;\textsc{Ind}\;} \mathbb{F}\mathrm{AM} \xrightarrow{\;\textsc{FRefC}\;} \mathbb{F}\mathrm{REF}$$

The resulting refinements would then use the canonical promotion predicates. However, the whole point of incorporating ℂORN in the framework is that we can construct an alternative functor RSEM directly from ℂORN to 𝕀REF. The functor RSEM is extensionally equal to the above composite functor, but intensionally very different. While its object part still takes the least fixed point of a description, its morphism part is the refinement semantics of ornaments given in Section 3.3, whose promotion predicates are the optimised predicates and have a more efficient representation.

```
RSem : Functor ℂOrn 𝕀Fam
RSem = record { object    = λ (I , D) ↦ I , μ D
              ; morphism  = λ (e , O) ↦ e , RSem O
              ; proofs of laws }
```

Categorical isomorphisms

So far switching to the categorical language offers no obvious benefits.

Define the type of isomorphisms between two objects X and Y in C as

```
record Iso C X Y : Set _ where
  field
    to      : X ⇒ Y
    from    : Y ⇒ X
    from-to-inverse : from · to ≈ id
    to-from-inverse : to · from ≈ id
```

(The relation $\_\cong\_$ is formally defined as Iso 𝕀FUN.)

functors preserve isomorphisms (a quick demonstration of preorder reasoning); TBC

4.2 Pullback properties of parallel composition

One of the great advantages of category theory is the ability to formulate the idea of **universal constructions** generically and concisely, which we will use to give parallel composition a useful macroscopic characterisation. An intuitive way to understand the idea of a universal construction is to think of it as a "best" solution to some specification. More precisely, the specification is represented as a category whose objects are all possible solutions and whose morphisms are evidence of how the solutions "compare" with each other, and a "best" solution is a **terminal object** in this category, meaning that it is "evidently better" than all objects in the category. For the actual definition: an object in a category C is **terminal** when it satisfies the **universal property** that for every object X' there is a unique morphism from X' to X, i.e., the setoid *Morphism* X' X has a unique inhabitant:

name scoping

```
Terminal C : Object → Set _
Terminal C X = (X' : Object) → Singleton (Morphism X' X)
```

where *Singleton* is defined by

```
Singleton : (S : Setoid) → Set _
Singleton S = Setoid.Carrier S × ((x y : Setoid.Carrier S) → x ≈S y)
```

The uniqueness condition ensures that terminal objects are unique up to (a unique) isomorphism — that is, if two objects are both terminal in C, then there is an isomorphism between them:

```
terminal-iso C : (X Y : Object) → Terminal C X → Terminal C Y → Iso C X Y
terminal-iso C X Y tX tY =
  let f : X ⇒ Y
      f = outl (tY X)
      g : Y ⇒ X
      g = outl (tX Y)
  in record { to      = f
            ; from    = g
            ; from-to-inverse = outr (tX X) (g · f) id
```

$$; \textit{to-from-inverse} \;=\; \mathsf{outr}\ (tY\ Y)\ (f \cdot g)\ \textit{id}\ \}$$

Thus, to prove that two constructions are isomorphic, one way would be to prove that they are universal in the same sense, i.e., they are both terminal objects in the same category. This is the main method we use to construct the ornamental conversion isomorphisms in Section 4.3.1 and the modularity isomorphisms in Section 4.3.2, both involving parallel composition. The goal of the rest of this section, then, is to find suitable universal properties that characterise parallel composition, preparing for Sections 4.3.1 and 4.3.2.

As said earlier, parallel composition computes the least informative ornament that contains the information of two compatible ornaments, and this is exactly a categorical **product**. Below we construct the definition of categorical products step by step. Let C be a category and L, R two objects in C. A **span** over L and R is defined by

```
record Span C L R : Set _ where
  constructor span
  field
    M : Object
    l : M ⇒ L
    r : M ⇒ R
```

or diagrammatically:

$$L \xleftarrow{\ l\ } M \xrightarrow{\ r\ } R$$

If we interpret a morphism $X \Rightarrow Y$ as evidence that X is more informative than Y, then a span over L and R is essentially an object which is more informative than both L and R. Spans over the same objects can be "compared": define a morphism between two spans by

```
record SpanMorphism C L R (s s' : Span C L R) : Set _ where
  constructor spanMorphism
  field
    m : Span.M s ⇒ Span.M s'
    triangle-l : Span.l s' · m ≈ Span.l s
    triangle-r : Span.r s' · m ≈ Span.r s
```

or diagrammatically (abbreviating $\mathsf{Span}.l\ s'$ to l' and so forth):



where the two triangles are required to commute. Thus a span s is more informative than another span s' when $\mathsf{Span}.M\ s$ is more informative than $\mathsf{Span}.M\ s'$ and the morphisms factorise appropriately. We can then form a category of spans over L and R:

diagram commutativity not yet defined

morphism equivalence and proof irrelevance

```
SpanCategory C L R : Category
SpanCategory C L R = record
  { Object      = Span C L R
  ; Morphism    =
      λ s s' ↦ record
        { Carrier  = SpanMorphism C L R s s'
        ; _≈_      = λ f g ↦ SpanMorphism.m f ≈ SpanMorphism.m g
        ; proofs of laws }
  ; proofs of laws }
```

and a product of L and R is a terminal object in this category:

```
Product C L R : Span C L R → Set _
Product C L R = Terminal (SpanCategory C L R)
```

In particular, a product of L and R contains the least informative object that is more informative than both L and R.

product diagram; “morphism relevance”?

We thus aim to characterise parallel composition as a product of two compatible ornaments. This means that ornaments should be the objects of some category, but so far we only know that ornaments are morphisms of the category Orn. We are thus directed to construct a category whose objects are morphisms in an ambient category C, so when we use Orn as the ambient category, parallel composition can be characterised as a product in the derived category. Such a category is in general a **comma category** [Mac Lane, 1998, § II.6], whose

objects are morphisms with arbitrary source and target objects, but here we should restrict ourselves to a special case called a **slice category**, since we seek to form products of only compatible ornaments (whose less informative end coincide) rather than arbitrary ones. A slice category is parametrised with an ambient category C and an object B in C, and has

- objects: all the morphisms in C with target B,

```
record Slice C B : Set _ where
  constructor slice
  field
    T : Object
    s : T ⇒ B
```

and

- morphisms: mediating morphisms giving rise to commutative triangles,

```
record SliceMorphism C B (s s′ : Slice C B) : Set _ where
  constructor sliceMorphism
  field
    m : Slice.T s ⇒ Slice.T s′
    triangle : Slice.s s′ · m ≈ Slice.s s
```

or diagrammatically:

$$\text{objects } \begin{array}{c} T \\ {\scriptstyle s}\downarrow \\ B \end{array} \qquad \text{and} \qquad \text{morphisms } \begin{array}{ccc} T & \xrightarrow{m} & T' \\ & {\scriptstyle s}\searrow \quad \swarrow{\scriptstyle s'} & \\ & B & \end{array}$$

The definitions above are assembled into the definition of slice categories in much the same way as span categories:

```
SliceCategory C B : Category
SliceCategory C B = record
  { Object      = Slice C B
  ; Morphism    =
      λ s s′ ↦ record
```

```
      { Carrier = SliceMorphism C B s s′
      ; _≈_ = λ f g ↦ SliceMorphism.m f ≈ SliceMorphism.m g
      ; proofs of laws }
  ; proofs of laws }
```

Objects in a slice category are thus morphisms with a common target, and when the ambient category is Orn, they are exactly the compatible ornaments that can be composed in parallel.

We have arrived at the characterisation of parallel composition as a product in a slice category on top of Orn. The composite term "product in a slice category" has become a multi-layered concept and can be confusing; to facilitate comprehension, we give several new definitions that can sometimes deliver better intuition. Let C be an ambient category and X an object in C. We refer to spans over two slices f, g : Slice C X alternatively as **squares** over f and g:

```
Square C f g : Set _
Square C f g = Span (SliceCategory C X) f g
```

since diagrammatically a square looks like



which is the same as



In a square q, we will refer to the object Slice.T (Span.M q), i.e., the node W in the diagrams above, as the **vertex** of q:

```
vertex : Square C f g → Object
vertex = Slice.T ∘ Span.M
```

A product of f and g is alternatively referred to as a **pullback** of f and g; that is, it is a square over f and g satisfying

```
Pullback C f g : Square C f g → Set _
Pullback C f g = Product (SliceCategory C X) f g
```

Equivalently, if we define the **square category** over f and g as

```
SquareCategory C f g : Category
SquareCategory C f g = SpanCategory (SliceCategory C X) f g
```

then a pullback of f and g is a terminal object in the square category over f and g — indeed, *Product* (*SliceCategory C X*) f g is definitionally equal to *Terminal* (*SquareCategory C f g*). This means that, by *terminal-iso*, there is an isomorphism between any two pullbacks p and q of the same slices f and g:

diagram of pullback

```
Iso (SquareCategory C f g) p q
```

Subsequently, since there is a forgetful functor from *SquareCategory C f g* to C whose object part is *vertex*, and functors preserve isomorphisms, we also have an isomorphism

```
Iso C (vertex p) (vertex q)
```
(4.1)

which is what we actually use in Sections 4.3.1 and 4.3.2.

pullback diagram; pullback preservation

We are now ready to state precisely the pullback properties for parallel composition that we make use of later. We could attempt to establish that, for any two ornaments O : Orn e D E and P : Orn f D F where D : Desc I, E : Desc J, and F : Desc K, the following square in Orn is a pullback:



(4.2)

The Agda term for this square is

```
pc-square : Square Orn (slice (J , E) (e , O)) (slice (K , F) (f , P))
pc-square = span (slice (e ⋈ f , ⌊O ⊗ P⌋) (pull, ⌈O ⊗ P⌉))
                 (sliceMorphism (outl , diffOrn-l O P) { }0)
                 (sliceMorphism (outr , diffOrn-r O P) { }1)
```

where Goal 0 has type *OrnEq* (*O* ⊙ *diffOrn-l O P*) ⌈*O* ⊗ *P*⌉ and Goal 1 has type

OrnEq $(P \odot \mathit{diffOrn\text{-}r}\ O\ P)\ \lceil O \otimes P \rceil$, both of which can be discharged. Comparing the commutative diagram (4.2) and the Agda term *pc-square*, it should be obvious how concise the categorical language can be — the commutative diagram expresses the structure of the Agda term in a clean and visually intuitive way. Since terms like *pc-square* can be reconstructed from commutative diagrams and the categorical definitions, from now on we will present commutative diagrams as representations of the corresponding Agda terms and omit the latter. The pullback property of (4.2) is not too useful by itself, though: Orn is a quite restricted category, so a universal property established in Orn has limited applicability. Instead, we are more interested in the pullback property of the image of (4.2) under Ind in Fam:

$$\begin{array}{ccc} & & \mathsf{outr}\ ,\mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P) \\ e \bowtie f\ ,\ \mu\ \lfloor O \otimes P \rfloor & \longrightarrow & K\ ,\ \mu\ F \\ \Big\downarrow{\scriptstyle \mathsf{outl}\ ,\mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)} & \lrcorner\ \searrow^{\mathit{pull},\mathit{forget}\ \lceil O \otimes P \rceil} & \Big\downarrow{\scriptstyle f\ ,\mathit{forget}\ P} \\ J\ ,\ \mu\ E & \xrightarrow[e\ ,\mathit{forget}\ O]{} & I\ ,\ \mu\ D \end{array} \tag{4.3}$$

We assert that the above square is a pullback by marking its vertex with "⌟". The proof of its universal property boils down to, very roughly speaking, datatype-generic construction of an inverse to

$$\mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P) \vartriangle \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P)$$

which involves tricky manipulation of equality proofs but is achievable. After the pullback square (4.3) is established in Fam, since the functor Com is pullback-preserving, we also get a pullback square in Fun:

$$\begin{array}{ccc} & & \mathsf{outr} * \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ P) \\ \Sigma\ (e \bowtie f)\ (\mu\ \lfloor O \otimes P \rfloor) & \longrightarrow & \Sigma\ K\ (\mu\ F) \\ \Big\downarrow{\scriptstyle \mathsf{outl} * \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)} & \lrcorner\ \searrow^{\mathit{pull} * \mathit{forget}\ \lceil O \otimes P \rceil} & \Big\downarrow{\scriptstyle f * \mathit{forget}\ P} \\ \Sigma\ J\ (\mu\ E) & \xrightarrow[e * \mathit{forget}\ O]{} & \Sigma\ I\ (\mu\ D) \end{array} \tag{4.4}$$

mention commutativity, associativity; should probably refactor the pullback square in FAM into the pullback square in ORN and pullback preservation of IND

4.3 Consequences

4.3.1 The ornamental conversion isomorphisms

We restate the ornamental conversion isomorphisms as follows: for any ornament O : Orn e D E where D : Desc I and E : Desc J, we have

$$\mu\ E\ j\ \cong\ \Sigma[\,x : \mu\ D\ (e\ j)\,]\ \mathsf{OptP}\ O\ (\mathsf{ok}\ j)\ x$$

for all j : J. Since the optimised predicates OptP O are defined by parallel composition of O and the singleton ornament S = *singletonOD* D, the isomorphism expands to

$$\mu\ E\ j\ \cong\ \Sigma[\,x : \mu\ D\ (e\ j)\,]\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x)) \tag{4.5}$$

How do we derive this from the pullback properties for parallel composition? It turns out that the pullback property in FUN (4.4) can help.

- First, observe that we have the following pullback square:

$$\begin{array}{ccc} & (e * \mathit{forget}\ O)\ \triangle\ (\mathit{singleton} \circ \mathit{forget}\ O \circ \mathsf{outr}) & \\ \Sigma\ J\ (\mu\ E) & \longrightarrow & \Sigma\ (\Sigma\ I\ (\mu\ D))\ (\mu\ \lfloor S \rfloor) \\ \Big\downarrow{\scriptstyle \mathit{id}} & \lrcorner \quad \searrow^{e * \mathit{forget}\ O} & \Big\downarrow{\scriptstyle \mathsf{outl} * \mathit{forget}\ \lceil S \rceil} \\ \Sigma\ J\ (\mu\ E) & \xrightarrow[e * \mathit{forget}\ O]{} & \Sigma\ I\ (\mu\ D) \end{array} \tag{4.6}$$

  Viewing pullbacks as products of slices, since a singleton ornament does not add information to a datatype, the vertical slice on the right-hand side

$$s\ =\ \mathsf{slice}\ (\Sigma\ (\Sigma\ I\ (\mu\ D))\ (\mu\ \lfloor S \rfloor))\ (\mathsf{outl} * \mathit{forget}\ \lceil S \rceil)$$

  behaves like a "multiplicative unit": any (compatible) slice s' alone gives rise to a product of s and s'. As a consequence, we have the bottom-left type

$\Sigma\ J\ (\mu\ E)$ as the vertex of the pullback. This pullback square is over the same slices as the pullback square (4.4) with P substituted by $\lceil S \rceil$, so by (4.1) we obtain an isomorphism

$$\Sigma\ J\ (\mu\ E)\ \cong\ \Sigma\ (e \bowtie \mathsf{outl})\ (\mu\ \lfloor O \otimes \lceil S \rceil \rfloor) \tag{4.7}$$

- To get from (4.7) to (4.5), we need to look more closely into the construction of (4.7). The right-to-left direction of (4.7) is obtained by applying the universal property of (4.6) to the square (4.4) (with P substituted by $\lceil S \rceil$), so it is the unique mediating morphism m that makes the following diagram commute:



From the left commuting triangle, we see that, extensionally, the morphism m is just $\mathsf{outl} * \mathit{forget}\ (\mathit{diffOrn\text{-}l}\ O\ P)$.

- This leads us to the following general lemma: if there is an isomorphism

$$\Sigma\ K\ X\ \cong\ \Sigma\ L\ Y$$

whose right-to-left direction is extensionally equal to some $f * g$, then we have

$$X\ k\ \cong\ \Sigma[\, l : f^{-1}\ k\,]\ Y\ (\mathit{und}\ l)$$

for all $k : K$. For a justification: fixing $k : K$, an element of the form $(k\ ,\ x) : \Sigma\ K\ X$ must correspond, under the given isomorphism, to some element $(l\ ,\ y) : \Sigma\ L\ Y$ such that $f\ l \equiv k$, so the set $X\ k$ corresponds to exactly the sum of the sets $Y\ l$ such that $f\ l \equiv k$.

- Specialising the lemma above for (4.7), we get

$$\mu\ E\ j\ \cong\ \Sigma[\, \mathit{jix} : \mathsf{outl}^{-1}\ j\,]\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathit{und}\ \mathit{jix}) \tag{4.8}$$

for all $j : J$. Finally, observe that a canonical element of type $\mathsf{outl}^{-1}\ j$ must be of the form $\mathsf{ok}\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x))$ for some $x : \mu\ D\ (e\ j)$, so we perform a change of variables for the summation, turning the right-hand side of (4.8) into

$$\Sigma[\, x : \mu\ D\ (e\ j)\,]\ \ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ j\ ,\ \mathsf{ok}\ (e\ j\ ,\ x))$$

and arriving at (4.5).

Formalisation detail. There is a twist when it comes to formalisation of the proof in Agda, however, due to Agda's intensionality: It is possible to formalise the lemma and the change of variables individually and chain them together, but the resulting isomorphisms would have a very complicated definition due to suspended type casts. If we use them to construct the refinement family in the morphism part of RSem, it would be rather difficult to prove that the morphism part of RSem preserves equivalence. We are thus forced to fuse all the above reasoning into one step to get a clean Agda definition such that RSem preserves equivalence automatically, but the idea is still essentially the same. □

4.3.2 The modularity isomorphisms

The other important family of isomorphisms we should construct from the pullback properties of parallel composition is the modularity isomorphisms, which is restated as follows: Suppose that there are descriptions $D : \mathsf{Desc}\ I$, $E : \mathsf{Desc}\ J$ and $F : \mathsf{Desc}\ K$, and ornaments $O : \mathsf{Orn}\ e\ D\ E$, and $P : \mathsf{Orn}\ f\ D\ F$. Then we have

$$\mathsf{OptP}\ \lceil O \otimes P \rceil\ (\mathsf{ok}\ (j\ ,\ k))\ x\ \cong\ \mathsf{OptP}\ O\ j\ x \times \mathsf{OptP}\ P\ k\ x$$

for all $i : I$, $j : e^{-1}\ i$, $k : f^{-1}\ i$, and $x : \mu\ D\ i$. The isomorphism expands to

$$\begin{aligned}&\mu\ \lfloor \lceil O \otimes P \rceil \otimes \lceil S \rceil \rfloor\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x))\\ &\quad\cong\ \mu\ \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) \times \mu\ \lfloor P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x))\end{aligned} \tag{4.9}$$

where again $S = \mathit{singletonOD}\ D$. A quick observation is that they are componentwise isomorphisms between the two families of sets

$$M \;=\; \mu\, \lfloor \lceil O \otimes P \rceil \otimes \lceil S \rceil \rfloor$$

and

$$\begin{aligned} N \;=\; & \lambda\, (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x)) \mapsto \\ & \quad \mu\, \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) \times \mu\, \lfloor P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x)) \end{aligned}$$

both indexed by $\mathit{pull} \bowtie \mathsf{outl}$ where pull has type $e \bowtie f \to I$ and outl has type $\Sigma\ I\ X \to I$. This is just an isomorphism in $\mathbb{F}\textsc{am}$ between $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$ and $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ N)$ whose index part (i.e., the isomorphism obtained under the functor FamF) is identity. Thus we seek to prove that both $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$ and $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ N)$ are vertices of pullbacks of the same slices.

- We look at $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ N)$ first. For fixed i, j, k, and x, the set

  $$N\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x))$$

  along with the cartesian projections is a product, which trivially extends to a pullback since there is a forgetful function from each of the two component sets to the **singleton** set $\mu\, \lfloor S \rfloor\ (i\ ,\ x)$, as shown in the following diagram:

$$\begin{array}{ccc} N\ (\mathsf{ok}\ (j\ ,\ k)\ ,\ \mathsf{ok}\ (i\ ,\ x)) & \xrightarrow{\mathsf{outr}} & \mu\, \lfloor P \otimes \lceil S \rceil \rfloor\ (k\ ,\ \mathsf{ok}\ (i\ ,\ x)) \\ \Big\downarrow{\scriptstyle \mathsf{outl}} & \lrcorner & \Big\downarrow{\scriptstyle \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ P\ \lceil S \rceil)} \\ \mu\, \lfloor O \otimes \lceil S \rceil \rfloor\ (j\ ,\ \mathsf{ok}\ (i\ ,\ x)) & \xrightarrow[\mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ \lceil S \rceil)]{} & \mu\, \lfloor S \rfloor\ (i\ ,\ x) \end{array}$$

  Note that this pullback square is possible because of the common x in the indices of the two component sets — otherwise they cannot project to the same singleton set. Collecting all such pullback squares together, we get the following pullback square in $\mathbb{F}\textsc{am}$:

$$\begin{array}{ccc} \mathit{pull} \bowtie \mathsf{outl}\ ,\ N & \xrightarrow{\_\ ,\ \mathsf{outr}} & f \bowtie \mathsf{outl}\ ,\ \mu\, \lfloor P \otimes \lceil S \rceil \rfloor \\ \Big\downarrow{\scriptstyle \_\ ,\ \mathsf{outl}} & \lrcorner & \Big\downarrow{\scriptstyle \mathsf{outr}\ ,\ \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ P\ \lceil S \rceil)} \\ e \bowtie \mathsf{outl}\ ,\ \mu\, \lfloor O \otimes \lceil S \rceil \rfloor & \xrightarrow[\mathsf{outr}\ ,\ \mathit{forget}\ (\mathit{diffOrn\text{-}r}\ O\ \lceil S \rceil)]{} & \Sigma\ I\ (\mu\ D)\ ,\ \mu\, \lfloor S \rfloor \end{array} \tag{4.10}$$

- Next we prove that $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$ is also the vertex of a pullback of the same slices as (4.10). This second pullback arises as a consequence of the following lemma (illustrated in the diagram below): In any category, consider the objects X, Y, their product $X \Leftarrow X \boxtimes Y \Rightarrow Y$, and products of each of the three objects X, Y, and $X \boxtimes Y$ with an object Z. (All the projections are shown as solid arrows in the diagram.) Then $(X \boxtimes Y) \boxtimes Z$ is the vertex of a pullback of the two projections $X \boxtimes Z \Rightarrow Z$ and $Y \boxtimes Z \Rightarrow Z$.



We again intend to view a pullback as a product of slices, and instantiate the lemma in *SliceCategory* $\mathbb{F}\textsc{am}\ (I\ ,\ \mu\ D)$, substituting all the objects by slices consisting of relevant ornamental forgetful functions in (4.9). The substitutions are as follows:

$$\begin{aligned}
X &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ O) \\
Y &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ P) \\
X \boxtimes Y &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ \lceil O \otimes P \rceil) \\
Z &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ \lceil S \rceil) \\
X \boxtimes Z &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ \lceil O \otimes \lceil S \rceil \rceil) \\
Y \boxtimes Z &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ \lceil P \otimes \lceil S \rceil \rceil) \\
(X \boxtimes Y) \boxtimes Z &\mapsto \mathsf{slice}\ _\ (_\ , \mathit{forget}\ \lceil \lceil O \otimes P \rceil \otimes \lceil S \rceil \rceil)
\end{aligned}$$

where $X \boxtimes Y$, $X \boxtimes Z$, $Y \boxtimes Z$, and $(X \boxtimes Y) \boxtimes Z$ indeed give rise to products in *SliceCategory* $\mathbb{F}\textsc{am}\ (I\ ,\ \mu\ D)$, i.e., pullbacks in $\mathbb{F}\textsc{am}$, by instantiating (4.3). What we get out of this instantiation of the lemma is a pullback in *SliceCategory* $\mathbb{F}\textsc{am}\ (I\ ,\ \mu\ D)$ rather than $\mathbb{F}\textsc{am}$. This is easy to fix, since there is a forgetful functor from any *SliceCategory* $C\ B$ to C whose object part is $\mathsf{Slice}.T$, and it is pullback-preserving. We thus get a pullback in $\mathbb{F}\textsc{am}$ of the same slices as (4.10) whose vertex is $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$.

Having the two pullbacks, by (4.1) we get an isomorphism in $\mathbb{F}\textsc{am}$ between $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ M)$ and $(\mathit{pull} \bowtie \mathsf{outl}\ ,\ N)$, whose index part can be shown to be identity, so there are componentwise isomorphisms between M and N in $\mathbb{F}\textsc{un}$, arriving at (4.9).

4.4 Discussion

elimination of arbitrariness of type-theoretic constructions; functor-level abstraction; compare with purely categorical approach

Chapter 5

Relational algebraic ornaments

the synthetic direction of the conversion isomorphism; emphasis no longer only on program derivation (relational calculus) but also on relational specifications

5.1 Relational programming in Agda

intro needs revision to de-emphasise program derivation a bit

One common approach to program derivation is by algebraic transformations of functional programs: one begins with a specification in the form of a functional program that expresses straightforward but possibly inefficient computation, and transforms it into an extensionally equal but more efficient functional program by applying algebraic laws and theorems. Using functional programs as the specification language means that specifications are directly executable, but the deterministic nature of functional programs can result in less flexible specifications. For example, when specifying an optimisation problem using a functional program that generates all feasible solutions and chooses an optimal one among them, the program would enforce a particular way of choosing the optimal solution, but such enforcement should not be part of the specification. To gain more flexibility, the specification language

was later generalised to **relational programs**. With relational programs, we specify only the relationship between input and output without actually specifying a way to execute the programs, so specifications in the form of relational programs can be as flexible as possible. Though lacking a directly executable semantics, most relational programs can still be read computationally as potentially partial and nondeterministic mappings, so relational specifications largely remain computationally intuitive as functional specifications.

To emphasise the computational interpretation of relations, we will mainly model a relation between sets A and B as a function sending each element of A to a **subset** of B. We define subsets by

$$\begin{aligned}&\mathscr{P} \;:\; \mathsf{Set} \to \mathsf{Set}_1\\&\mathscr{P}A \;=\; A \to \mathsf{Set}\end{aligned}$$

That is, a subset $s \;:\; \mathscr{P}A$ is a characteristic function that assigns a type to each element of A, and $a \;:\; A$ is considered to be a member of s if the type $s\ a \;:\; \mathsf{Set}$ is inhabited. We may regard $\mathscr{P}A$ as the type of computations that nondeterministically produce an element of A. A simple example is

$$\begin{aligned}&\mathit{any} \;:\; \{A \;:\; \mathsf{Set}\} \to \mathscr{P}A\\&\mathit{any} \;=\; \mathit{const}\ \top\end{aligned}$$

The subset $\mathit{any} \;:\; \mathscr{P}A$ associates the unit type $\top$ with every element of A. Since $\top$ is inhabited, *any* can produce any element of A. While $\mathscr{P}$ cannot be made into a conventional monad [Moggi, 1991; Wadler, 1992] because it is not an endofunctor, it can still be equipped with the usual monadic programming combinators, giving rise to a **relative monad** [Altenkirch et al., 2010]:

- The monadic unit is defined as

  $$\begin{aligned}&\mathit{return} \;:\; \{A \;:\; \mathsf{Set}\} \to A \to \mathscr{P}A\\&\mathit{return} \;=\; \_{\equiv}\_\end{aligned}$$

  The subset $\mathit{return}\ a \;:\; \mathscr{P}A$ for some $a \;:\; A$ simplifies to $\lambda\, a' \mapsto a \equiv a'$, so a is the only member of the subset.

- The monadic bind is defined as

  $$\_{\gg\!=}\_ \;:\; \{A\ B \;:\; \mathsf{Set}\} \to \mathscr{P}A \to (A \to \mathscr{P}B) \to \mathscr{P}B$$

$$\_\gg\!=\_\ \{A\}\ s\ f\ =\ \lambda\, b \mapsto \Sigma[\, a : A\,]\ s\ a \times f\ a\ b$$

If $s : \mathscr{P} A$ and $f : A \to \mathscr{P} B$, then the subset $s \gg\!= f : \mathscr{P} B$ is the disjoint union of all the subsets $f\ a : \mathscr{P} B$ where a ranges over the elements of A that belong to s; that is, an element $b : B$ is a member of $s \gg\!= f$ exactly when there exists some $a : A$ belonging to s such that b is a member of $f\ a$.

It is easy to show that the two combinators satisfy the (relative) monad laws up to pointwise isomorphism, whose proofs we omit from the presentation. On top of *return* and $\_\gg\!=\_$, the functorial map of $\mathscr{P}$ is defined as

$$\begin{aligned}&\_\langle\$\rangle\ :\ \{A\ B\ :\ \mathsf{Set}\} \to (A \to B) \to \mathscr{P} A \to \mathscr{P} B\\ &f\ \langle\$\rangle\ s\ =\ s \gg\!= \lambda\, a \mapsto \mathit{return}\ (f\ a)\end{aligned}$$

and we also define a two-argument version for convenience:

$$\begin{aligned}&\_\langle\$\rangle^2\ :\ \{A\ B\ C\ :\ \mathsf{Set}\} \to (A \to B \to C) \to \mathscr{P} A \to \mathscr{P} B \to \mathscr{P} C\\ &f\ \langle\$\rangle^2\ s\ t\ =\ s \gg\!= \lambda\, a \mapsto t \gg\!= \lambda\, b \mapsto \mathit{return}\ (f\ a\ b)\end{aligned}$$

The notation is a reference to applicative functors [McBride and Paterson, 2008], allowing us to think of functorial maps of $\mathscr{P}$ as applications of pure functions to effectful arguments.

We will mainly use families of relations between families of sets:

$$\begin{aligned}&\_\rightsquigarrow\_\ :\ \{I\ :\ \mathsf{Set}\} \to (I \to \mathsf{Set}) \to (I \to \mathsf{Set}) \to \mathsf{Set}_1\\ &X \rightsquigarrow Y\ =\ \forall\ \{i\} \to X\ i \to \mathscr{P}\,(Y\ i)\end{aligned}$$

which is the usual generalisation of $\_\rightrightarrows\_$ to allow nondeterminacy. Here we define several relational operators that we will need.

relations vs families of relations

- Since functions are deterministic relations, we have the following combinator *fun* that lifts functions to relations using *return*.

$$\begin{aligned}&\mathit{fun}\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ Y\ :\ I \to \mathsf{Set}\} \to (X \rightrightarrows Y) \to (X \rightsquigarrow Y)\\ &\mathit{fun}\ f\ x\ =\ \mathit{return}\ (f\ x)\end{aligned}$$

- The identity relation is just the identity function lifted by *fun*.

$$\begin{aligned}&\mathit{idR}\ :\ \{I\ :\ \mathsf{Set}\}\ \{X\ :\ I \to \mathsf{Set}\} \to (X \rightsquigarrow X)\\ &\mathit{idR}\ =\ \mathit{fun}\ \mathit{id}\end{aligned}$$

```
mapR : {I : Set} (D : RDesc I) {X Y : I → Set} →
       (X ⇝ Y) → ⟦ D ⟧ X → 𝒫(⟦ D ⟧ Y)
mapR (ν [])       R ∎       = return ∎
mapR (ν (i :: is)) R (x , xs) = _,_ <$>² (R x) (mapR (ν is) R xs)
mapR (σ S D)      R (s , xs) = (_,_ s) <$> (mapR (D s) R xs)
ℝ : {I : Set} (D : Desc I) {X Y : I → Set} → (X ⇝ Y) → (𝔽 D X ⇝ 𝔽 D Y)
ℝ D R {i} = mapR (D i) R
```

Figure 5.1 Definition for relators.

- Composition of relations is easily defined with $\_\gg\!=\_$: computing $R \bullet S$ on input x is first computing $S\ x$ and then feeding the result to R.

  ```
  _•_ : {I : Set} {X Y Z : I → Set} → (Y ⇝ Z) → (X ⇝ Y) → (X ⇝ Z)
  (R • S) x = S x >>= R
  ```

- Some relations do not carry obvious computational meaning, which we can still define pointwise, like the meet of two relations:

  ```
  _∩_ : {I : Set} {X Y : I → Set} → (X ⇝ Y) → (X ⇝ Y) → (X ⇝ Y)
  (R ∩ S) x y = R x y × S x y
  ```

- Unlike a function, which distinguishes between input and output, inherently a relation treats its domain and codomain symmetrically. This is reflected by the presence of the following **converse** operator:

  ```
  _° : {I : Set} {X Y : I → Set} → (X ⇝ Y) → (Y ⇝ X)
  (R °) y x = R x y
  ```

  A relation can thus be "run backwards" simply by taking its converse. The nondeterministic and bidirectional nature of relations makes them a powerful and concise language for specifications, as will be demonstrated in Sections 5.3.2 and 5.3.3.

- We will also need **relators**, i.e., functorial maps on relations:

  ```
  ℝ : {I : Set} (D : Desc I) {X Y : I → Set} →
  ```

$$(X \rightsquigarrow Y) \to (\mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ Y)$$

If $R : X \rightsquigarrow Y$, the relation $\mathbb{R}\ D\ R : \mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ Y$ applies R to the recursive positions of its input, leaving everything else intact. The definition of $\mathbb{R}$ is shown in Figure 5.1. For example, if $D = \mathit{ListD}\ A$, then $\mathbb{R}\ (\mathit{ListD}\ A)$ is, up to isomorphism,

$$\begin{aligned}
&\mathbb{R}\ (\mathit{ListD}\ A) : \{X\ Y : I \to \mathsf{Set}\} \to \\
&\qquad\qquad (X \rightsquigarrow Y) \to (\mathbb{F}\ (\mathit{ListD}\ A)\ X \rightsquigarrow \mathbb{F}\ (\mathit{ListD}\ A)\ Y) \\
&\mathbb{R}\ (\mathit{ListD}\ A)\ R\ (\text{`nil}\quad ,\qquad \blacksquare) = \mathit{return}\ (\text{`nil}\ ,\ \blacksquare) \\
&\mathbb{R}\ (\mathit{ListD}\ A)\ R\ (\text{`cons}\ ,\ a\ ,\ x\ ,\ \blacksquare) = (\lambda\, y \mapsto \text{`cons}\ ,\ a\ ,\ y\ ,\ \blacksquare)\ \langle\$\rangle\ (R\ x)
\end{aligned}$$

Laws and theorems about relational programs are formulated with relational inclusion:

$$\begin{aligned}
&_\subseteq_ : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\}\ (R\ S : X \rightsquigarrow Y) \to \mathsf{Set} \\
&R \subseteq S = \forall\ \{i\} \to (x : X\ i)\ (y : Y\ i) \to R\ x\ y \to S\ x\ y
\end{aligned}$$

or equivalence of relations, i.e., two-way inclusion:

$$\begin{aligned}
&_\simeq_ : \{I : \mathsf{Set}\}\ \{X\ Y : I \to \mathsf{Set}\}\ (R\ S : X \rightsquigarrow Y) \to \mathsf{Set} \\
&R \simeq S = (R \subseteq S) \times (S \subseteq R)
\end{aligned}$$

With relational inclusion, many concepts can be expressed in a surprisingly concise way. For example, a relation R is a preorder if it is reflexive and transitive. In relational terms, these two conditions are expressed simply as

$$\mathit{id}R \subseteq R \qquad \text{and} \qquad R \bullet R \subseteq R$$

and are easily manipulable in calculations. Another important notion is **monotonic algebras** [Bird and de Moor, 1997, Section 7.2]: an algebra $S : \mathbb{F}\ D\ X \rightsquigarrow X$ is **monotonic** on $R : X \rightsquigarrow X$ (usually an ordering) if

$$S \bullet \mathbb{R}\ D\ R \subseteq R \bullet S$$

which says that if two input values to S have their recursive positions related by R and are otherwise equal, then the output values would still be related by R. In the context of optimisation problems, monotonicity can be used to capture the **principle of optimality**, as will be shown in Section 5.3.3.

probably a simple example of relational calculation here?

```
mutual
  (⟦_⟧) : {I : Set} {D : Desc I} {X : I → Set} → (𝕀 D X ⇝ X) → (μ D ⇝ X)
  (⟦_⟧) {I} {D} R {i} (con ds) = mapFoldR D (D i) R ds >>= R

  mapFoldR : {I : Set} (D : Desc I) (D′ : RDesc I) →
             {X : I → Set} → (𝕀 D X ⇝ X) → ⟦ D′ ⟧ (μ D) → 𝒫 (⟦ D′ ⟧ X)
  mapFoldR D (v [])       R ▪        = return ▪
  mapFoldR D (v (i :: is)) R (d , ds) = _,_ <$>² ((⟦ R ⟧) d)
                                                 (mapFoldR D (v is) f ds)
  mapFoldR D (σ S D′)     R (s , ds) = (_, s) <$> (mapFoldR D (D′ s) f ds)
```

Figure 5.2 Definition of relational folds.

Having defined relations as nondeterministic mappings, it is straightforward to rewrite the datatype-generic *fold* with the subset combinators to obtain a relational version, which is denoted by "banana brackets" [Meijer et al., 1991]:

```
(⟦_⟧) : {I : Set} {D : Desc I} {X : I → Set} → (𝕀 D X ⇝ X) → (μ D ⇝ X)
```

The definition of $(\!|\_|\!)$ is shown in Figure 5.2. For example, the relational fold on lists is, up to isomorphism,

```
(⟦_⟧) {⊤} {ListD A} : {X : ⊤ → Set} →
                      (𝕀 (ListD A) X ⇝ X) → (μ (ListD A) ⇝ X)
(⟦ R ⟧) []       = R (‘nil , ▪)
(⟦ R ⟧) (a :: as) = (⟦ R ⟧) as >>= λ x ↦ R (‘cons , a , x , ▪)
```

The functional and relational fold operators are related by the following lemma:

```
fun-preserves-fold : {I : Set} (D : Desc I) {X : I → Set} →
                     (f : 𝕀 D X ⇉ X) {i : I} (d : μ D i) (x : X i) →
                     fun (fold f) d x ≅ (⟦ fun f ⟧) d x
```

which is a strengthened version of $\mathit{fun}\ (\mathit{fold}\ f) \simeq (\!|\mathit{fun}\ f|\!)$.

5.2 Definition of relational algebraic ornaments

We now turn to relational algebraic ornamentation, the key construct that bridges internalist and relational programming. Let

$$R \;:\; \mathbb{F}\;(\mathit{ListD}\;A)\;(\mathit{const}\;X) \rightsquigarrow \mathit{const}\;X$$

(where X : Set) be a relational algebra for lists. We can define a datatype of "algebraic lists" as

$$\begin{aligned}
&\textbf{indexfirst data}\ \mathsf{AlgList}\ A\ R\ :\ X \to \mathsf{Set}\ \textbf{where}\\
&\quad \mathsf{AlgList}\ A\ R\ x\ \ni\ \mathsf{nil}\ (\mathit{rnil}\ :\ R\ (\text{`}\mathsf{nil}\ ,\ \blacksquare)\ x)\\
&\qquad\qquad\quad \mid\ \mathsf{cons}\ (a\ :\ A)\ (x'\ :\ X)\\
&\qquad\qquad\qquad\qquad (\mathit{rcons}\ :\ R\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ x'\ ,\ \blacksquare)\ x)\\
&\qquad\qquad\qquad\qquad (\mathit{as}\ :\ \mathsf{AlgList}\ A\ R\ x')
\end{aligned}$$

There is an ornament from lists to algebraic lists which marks the fields *rnil*, x', and *rcons* in AlgList as additional and refines the index of the recursive position from $\blacksquare$ to x'. The optimised predicate (Section 3.3.1) for this ornament is

$$\begin{aligned}
&\textbf{indexfirst data}\ \mathsf{AlgListP}\ A\ R\ :\ X \to \mathsf{List}\ A \to \mathsf{Set}\ \textbf{where}\\
&\quad \mathsf{AlgListP}\ A\ R\ x\ [\,]\ \ \ni\ \mathsf{nil}\ (\mathit{rnil}\ :\ R\ (\text{`}\mathsf{nil}\ ,\ \blacksquare)\ x)\\
&\quad \mathsf{AlgListP}\ A\ R\ x\ (a :: \mathit{as})\ \ni\ \mathsf{cons}\ (x'\ :\ X)\\
&\qquad\qquad\qquad\qquad\qquad\qquad (\mathit{rcons}\ :\ R\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ x'\ ,\ \blacksquare)\ x)\\
&\qquad\qquad\qquad\qquad\qquad\qquad (p\ :\ \mathsf{AlgListP}\ A\ R\ x'\ \mathit{as})
\end{aligned}$$

A simple argument by induction shows that $\mathsf{AlgListP}\ A\ R\ x\ \mathit{as}$ is in fact isomorphic to $(\!|\, R \,|\!)\ \mathit{as}\ x$ for any as : List A and x : X. By predicate swapping for the refinement semantics of the ornament from lists to algebraic lists (Section 3.3), we get

$$\mathsf{AlgList}\ A\ R\ x\ \cong\ \Sigma[\,\mathit{as} : \mathsf{List}\ A\,]\ (\!|\, R \,|\!)\ \mathit{as}\ x \tag{5.1}$$

for all x : X. That is, an algebraic list is exactly a plain list and a proof that the list folds to x using the algebra R. The traditional bottom-up vector datatype is a special case of AlgList — define

$$\begin{aligned}
&\mathit{length\text{-}alg}\ :\ \mathbb{F}\ (\mathit{ListD}\ A)\ (\mathit{const}\ \mathsf{Nat}) \rightrightarrows \mathit{const}\ \mathsf{Nat}\\
&\mathit{length\text{-}alg}\ (\text{`}\mathsf{nil}\quad ,\qquad \blacksquare)\ =\ \mathsf{zero}
\end{aligned}$$

$$
\begin{aligned}
&\mathit{algROD} : \{I : \mathsf{Set}\}\ (D : \mathsf{RDesc}\ I)\ \{J : I \to \mathsf{Set}\} \to \\
&\qquad (\llbracket\ D\ \rrbracket\ J \to \mathsf{Set}) \to \mathsf{ROrnDesc}\ (\Sigma\ I\ J)\ \mathsf{outl}\ D \\
&\mathit{algROD}\ (\mathsf{v}\ \mathit{is}) \quad \{J\}\ P\ =\ \Delta[\mathit{js} : \mathbb{P}\ \mathit{is}\ J]\ \Delta[\_ : P\ \mathit{js}] \\
&\qquad\qquad\qquad\qquad \mathsf{v}\ (\mathbb{P}\text{-}\mathit{map}\ (\lambda\ \{i\}\ j \mapsto \mathsf{ok}\ (i\ ,\ j))\ \mathit{is}\ \mathit{js}) \\
&\mathit{algROD}\ (\sigma\ S\ D) \qquad P\ =\ \sigma[s : S]\ \mathit{algROD}\ (D\ s)\ (\mathit{curry}\ P\ s) \\
&\mathit{algOD} : \{I : \mathsf{Set}\}\ (D : \mathsf{Desc}\ I)\ \{J : I \to \mathsf{Set}\} \to \\
&\qquad (\mathbb{F}\ D\ J \rightsquigarrow J) \to \mathsf{OrnDesc}\ (\Sigma\ I\ J)\ \mathsf{outl}\ D \\
&\mathit{algOD}\ D\ R\ (\mathsf{ok}\ (i\ ,\ j))\ =\ \mathit{algROD}\ (D\ i)\ (\lambda\ \mathit{js} \mapsto R\ \mathit{js}\ j)
\end{aligned}
$$

Figure 5.3 Definitions for algebraic ornamentation.

$$\mathit{length\text{-}alg}\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ n\ ,\ \blacksquare)\ =\ \mathsf{suc}\ n$$

and then $\mathsf{AlgList}\ A\ (\mathit{fun}\ \mathit{length\text{-}alg})$ is exactly $\mathsf{Vec}'\ A$. By (5.1) we have the isomorphisms

$$\mathsf{Vec}'\ A\ n\ \cong\ \Sigma[\mathit{as} : \mathsf{List}\ A]\ (\llparenthesis \mathit{fun}\ \mathit{length\text{-}alg} \rrparenthesis)\ \mathit{as}\ n$$

for all $n : \mathsf{Nat}$, from which we can derive

$$\mathsf{Vec}'\ A\ n\ \cong\ \Sigma[\mathit{as} : \mathsf{List}\ A]\ \mathit{length}\ \mathit{as}\ \equiv\ n$$

by *fun-preserves-fold*, where $\mathit{length}\ =\ \mathit{fold}\ \mathit{length\text{-}alg}$.

The above can be generalised to all datatypes encoded by the Desc universe. Let $D : \mathsf{Desc}\ I$ be a description and $R : \mathbb{F}\ D\ X \rightsquigarrow X$ (where $X : I \to \mathsf{Set}$) an algebra. The **algebraic ornamentation** of D with R is an ornamental description

$$\mathit{algOD}\ D\ R : \mathsf{OrnDesc}\ (\Sigma\ I\ X)\ \mathsf{outl}\ D$$

(where $\mathsf{outl} : \Sigma\ I\ X \to I$). The optimised predicate for $\lceil\ \mathit{algOD}\ D\ R\ \rceil$ is pointwise isomorphic to $(\!|\ R\ |\!)$, i.e.,

$$\mathsf{OptP}\ \lceil \mathit{algOD}\ D\ R \rceil\ (\mathsf{ok}\ (i\ ,\ x))\ d\ \cong\ (\!|\ R\ |\!)\ d\ x$$

for all $i : I$, $x : X\ i$, and $d : \mu\ D\ i$. These isomorphisms give rise to a family of predicate swaps for the refinement semantics of $\lceil \mathit{algOD}\ D\ R \rceil$, so we arrive at the following conversion isomorphisms

$$\mu\ \lfloor \mathit{algOD}\ D\ R \rfloor\ (i\ ,\ x)\ \cong\ \Sigma[d : \mu\ D\ i]\ (\!|\ R\ |\!)\ d\ x \tag{5.2}$$

for all $i : I$ and $x : X\ i$. The definition of *algOD*, shown in Figure 5.3, is an adaptation and generalisation of McBride's original definition of functional algebraic ornaments [2011]. Roughly speaking, it retains all the fields of the base description and inserts before every ∨

- a new field of indices for the recursive positions (e.g., the field x' in AlgList) and
- another new field requesting a proof that
  - the indices supplied in the previous field and
  - the values for the fields originally in the base description

  computes to the targeted index through R (e.g., the fields *rnil* and *rcons* in AlgList).

summary and some gluing to the next section

5.3 Examples

5.3.1 The Fold Fusion Theorem

Revise using upgrades.

As a first example of bridging internalist programming with relational calculation through algebraic ornamentation, let us consider the **Fold Fusion Theorem** [Bird and de Moor, 1997, Section 6.2]: Let D : Desc I be a description, $R : X \rightsquigarrow Y$ a relation, and $S : \mathbb{F}\ D\ X \rightsquigarrow X$ and $T : \mathbb{F}\ D\ Y \rightsquigarrow Y$ algebras. If R is a homomorphism from S to T, i.e.,

$$R \bullet S \simeq T \bullet \mathbb{R}\ D\ R$$

which is referred to as the **fusion condition**, then we have

$$R \bullet (\!|\ S\ |\!) \simeq (\!|\ T\ |\!)$$

The above is, in fact, a corollary of two variations of Fold Fusion that replace relational equivalence in the statement of the theorem with relational inclusion. One of the variations is

$$R \bullet S \subseteq T \bullet \mathbb{R}\, D\, R \quad \to \quad R \bullet (\!|\, S \,|\!) \subseteq (\!|\, T \,|\!)$$

This can be used with (5.2) to derive a conversion between algebraically ornamented datatypes:

$$\begin{aligned}
&\mathit{algOD\text{-}fusion\text{-}\subseteq}\ D\ R\ S\ T\ : \\
&\quad R \bullet S \subseteq T \bullet \mathbb{R}\, D\, R \to \\
&\quad \{i : I\}\ (x : X\, i) \to \mu\ \lfloor \mathit{algOD}\ D\ S \rfloor\ (i\,,\, x) \to \\
&\quad (y : Y\, i) \to R\ x\ y \to \mu\ \lfloor \mathit{algOD}\ D\ T \rfloor\ (i\,,\, y)
\end{aligned}$$

The other variation of Fold Fusion simply reverses the direction of inclusion:

$$R \bullet S \supseteq T \bullet \mathbb{R}\, D\, R \quad \to \quad R \bullet (\!|\, S \,|\!) \supseteq (\!|\, T \,|\!)$$

which translates to the conversion

$$\begin{aligned}
&\mathit{algOD\text{-}fusion\text{-}\supseteq}\ D\ R\ S\ T\ : \\
&\quad R \bullet S \supseteq T \bullet \mathbb{R}\, D\, R \to \\
&\quad \{i : I\}\ (y : Y\, i) \to \mu\ \lfloor \mathit{algOD}\ D\ T \rfloor\ (i\,,\, y) \to \\
&\quad \Sigma[\, x : X\, i \,]\ \mu\ \lfloor \mathit{algOD}\ D\ S \rfloor\ (i\,,\, x) \times R\ x\ y
\end{aligned}$$

For a simple example, suppose that we need a "bounded" vector datatype, i.e., lists indexed with an upper bound on their length. A quick thought might lead to this definition

$$\begin{aligned}
&\mathsf{BVec} : \mathsf{Set} \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{BVec}\ A\ m = \mu\ \lfloor \mathit{algOD}\ (\mathit{ListD}\ A)\ (\mathit{geq} \bullet \mathit{fun\ length\text{-}alg}) \rfloor\ (\blacksquare\,,\, m)
\end{aligned}$$

where $\mathit{geq} = \lambda\, x\, y \to x \leqslant y : \mathit{const}\ \mathsf{Nat} \rightsquigarrow \mathit{const}\ \mathsf{Nat}$ maps a natural number x to any natural number that is at least x. The isomorphisms (5.2) specialise for BVec to

$$\mathsf{BVec}\ A\ m \cong \Sigma[\, \mathit{as} : \mathsf{List}\ A \,]\ (\!|\, \mathit{geq} \bullet \mathit{fun\ length\text{-}alg} \,|\!)\ \mathit{as}\ m$$

for all $m : \mathsf{Nat}$. But is BVec really the bounded vectors? Indeed it is, because we can deduce

$$\mathit{geq} \bullet (\!|\, \mathit{fun\ length\text{-}alg} \,|\!) \simeq (\!|\, \mathit{geq} \bullet \mathit{fun\ length\text{-}alg} \,|\!)$$

by Fold Fusion. The fusion condition is

$$\mathit{geq} \bullet \mathit{fun\ length\text{-}alg} \simeq \mathit{geq} \bullet \mathit{fun\ length\text{-}alg} \bullet \mathbb{R}\ (\mathit{ListD}\ A)\ \mathit{geq}$$

The left-to-right inclusion is easily calculated as follows:

relator laws and various monotonicity need to be stated earlier

$$
\begin{array}{ll}
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \\
\subseteq & \quad \{\ \mathit{idR}\ \text{identity}\ \} \\
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \cdot \mathit{idR} \\
\subseteq & \quad \{\ \text{relator preserves identity}\ \} \\
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \cdot \mathbb{R}\ (\mathit{ListD}\ A)\ \mathit{idR} \\
\subseteq & \quad \{\ \mathit{geq}\ \text{reflexive}\ \} \\
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \cdot \mathbb{R}\ (\mathit{ListD}\ A)\ \mathit{geq}
\end{array}
$$

And from right to left:

$$
\begin{array}{ll}
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \cdot \mathbb{R}\ (\mathit{ListD}\ A)\ \mathit{geq} \\
\subseteq & \quad \{\ \mathit{fun}\ \mathit{length\text{-}alg}\ \text{monotonic on}\ \mathit{geq}\ \} \\
 & \mathit{geq} \cdot \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg} \\
\subseteq & \quad \{\ \mathit{geq}\ \text{transitive}\ \} \\
 & \mathit{geq} \cdot \mathit{fun}\ \mathit{length\text{-}alg}
\end{array}
$$

Note that these calculations are good illustrations of the power of relational calculation despite their simplicity — they are straightforward symbolic manipulations, hiding details like quantifier reasoning behind the scenes. As demonstrated by the AoPA library [Mu et al., 2009], they can be faithfully formalised with preorder reasoning combinators in Agda and used to discharge the fusion conditions of *algOD-fusion-*$\subseteq$ and *algOD-fusion-*$\supseteq$. Hence we get two conversions, one of type

$$\mathsf{Vec}\ A\ n \to (n \leqslant m) \to \mathsf{BVec}\ A\ m$$

which relaxes a vector of length n to a bounded vector whose length is bounded above by some m that is at least n, and the other of type

$$\mathsf{BVec}\ A\ m \to \Sigma[\, n : \mathsf{Nat}\,]\ \ \mathsf{Vec}\ A\ n \times (n \leqslant m)$$

which converts a bounded vector whose length is at most m to a vector of length precisely n and guarantees that n is at most m.

Just constraint transformation; base data do not change

5.3.2 The Streaming Theorem for list metamorphisms

A **metamorphism** [Gibbons, 2007b] is an unfold after a fold — it consumes a data structure to compute an intermediate value and then produces a new data structure using the intermediate value as the seed. In this section we will restrict ourselves to metamorphisms consuming and producing lists. As Gibbons noted, (list) metamorphisms in general cannot be automatically optimised in terms of time and space, but under certain conditions it is possible to refine a list metamorphism to a **streaming algorithm** — which can produce an initial segment of the output list without consuming all of the input list — or a parallel algorithm [Nakano, 2013]. In the rest of this section, we prove the **Streaming Theorem** [Bird and Gibbons, 2003, Theorem 30] by implementing the streaming algorithm given by the theorem with algebraic ornamented lists such that the algorithm satisfies its metamorphic specification by construction.

Our first step is to formulate a metamorphism as a relational specification of the streaming algorithm.

- The fold part needs a twist since using the conventional fold — known as the **right fold** for lists since the direction of computation on a list is from right to left (cf. wind direction) — does not easily give rise to a streaming algorithm. This is because we wish to talk about "partial consumption" naturally: for a list, partial consumption means examining and removing some elements of the list to get a sub-list on which we can resume consumption, and the natural way to do this is to consume the list from the left, examining and removing head elements and keeping the tail. We should thus use the **left fold** instead, which is usually defined as

  $$\begin{aligned} &\mathit{foldl} : \{A\ X : \mathsf{Set}\} \to (X \to A \to X) \to X \to \mathsf{List}\ A \to X \\ &\mathit{foldl}\ f\ x\ [\,] \quad\quad\ = x \\ &\mathit{foldl}\ f\ x\ (a :: as) = \mathit{foldl}\ f\ (f\ x\ a)\ as \end{aligned}$$

  The connection to the conventional fold (and thus algebraic ornamentation) is not lost, however — it is well known that a left fold can be alternatively implemented as a right fold by turning a list into a chain of functions of type $X \to X$ transforming the initial value to the final result:

$$
\begin{aligned}
&\mathit{foldl\text{-}alg} \;:\; \{A\ X \;:\; \mathsf{Set}\} \to (X \to A \to X) \to \\
&\qquad\qquad \mathbb{F}\ (\mathit{ListD}\ A)\ (\mathit{const}\ (X \to X)) \rightrightarrows \mathit{const}\ (X \to X) \\
&\mathit{foldl\text{-}alg}\ f\ (\text{`}\mathsf{nil}\quad ,\qquad \blacksquare) \;=\; \mathit{id} \\
&\mathit{foldl\text{-}alg}\ f\ (\text{`}\mathsf{cons}\ ,\ a\ ,\ h\ ,\ \blacksquare) \;=\; h \circ \mathit{flip}\ f\ a \\
&\mathit{foldl} \;:\; \{A\ X \;:\; \mathsf{Set}\} \to (X \to A \to X) \to X \to \mathsf{List}\ A \to X \\
&\mathit{foldl}\ f\ x\ \mathit{as} \;=\; \mathit{fold}\ (\mathit{foldl\text{-}alg}\ f)\ \mathit{as}\ x
\end{aligned}
$$

The left fold can thus be linked to the relational fold by

$$\mathit{fun}\ (\mathit{foldl}\ f\ x) \;\simeq\; \mathit{fun}\ (\lambda\, h \;\mapsto\; h\ x) \bullet (\!|\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f)\ |\!) \tag{5.3}$$

- The unfold part is approximated by the converse of a relational fold: given a list coalgebra $g \;:\; \mathit{const}\ X \rightrightarrows \mathbb{F}\ (\mathit{ListD}\ B)\ (\mathit{const}\ X)$ for some $X \;:\; \mathsf{Set}$, we take its converse, turning it into a relational algebra, and use the converse of the relational fold with this algebra.

  $$(\!|\mathit{fun}\ g^\circ\ |\!)^\circ \;:\; \mathit{const}\ X \rightsquigarrow \mathit{const}\ (\mathsf{List}\ A)$$

  This is only an approximation because, while the relation does produce a list, the resulting list is inductive rather than coinductive, so the relation is actually a **well-founded** unfold, which is incapable of producing an infinite list.

Thus, given a "left algebra" for consuming $\mathsf{List}\ A$

$$f \;:\; X \to A \to X$$

and a coalgebra for producing $\mathsf{List}\ B$

$$g \;:\; \mathit{const}\ X \rightrightarrows \mathbb{F}\ (\mathit{ListD}\ B)\ (\mathit{const}\ X)$$

which together satisfy a **streaming condition** that we will see later, the streaming algorithm we implement, which takes as input the initial value $x \;:\; X$ for the left fold, should be included in the following metamorphic relation:

$$\mathit{meta}\ f\ g\ x \;=\; (\!|\mathit{fun}\ g^\circ\ |\!)^\circ \bullet \mathit{fun}\ (\mathit{foldl}\ f\ x) \;:\; \mathit{const}\ (\mathsf{List}\ A) \rightsquigarrow \mathit{const}\ (\mathsf{List}\ B)$$

Next we devise a type for the streaming algorithm that fully guarantees its correctness. By (5.3), the specification $\mathit{meta}\ f\ g\ x$ is equivalent to

$$(\!|\mathit{fun}\ g^\circ\ |\!)^\circ \bullet \mathit{fun}\ (\lambda\, h \;\mapsto\; h\ x) \bullet (\!|\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f)\ |\!)$$

Inspecting the above relation, we see that a conforming program takes a List A that folds to some $h : X \to X$ with *fun* (*foldl-alg f*) and computes a List B that folds to $h\ x : X$ with *fun* g°. We are thus going to implement the streaming algorithm as

$$\begin{aligned} &\mathit{stream}\ f\ g : (x : X)\ \{h : X \to X\} \to \\ &\qquad \mathsf{AlgList}\ A\ (\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f))\ h \to \mathsf{AlgList}\ B\ (\mathit{fun}\ g^\circ)\ (h\ x) \end{aligned}$$

from which we can extract

$$\mathit{stream}'\ f\ g : X \to \mathsf{List}\ A \to \mathsf{List}\ B$$

which is guaranteed to satisfy

$$\mathit{fun}\ (\mathit{stream}'\ f\ g\ x) \subseteq \mathit{meta}\ f\ g\ x$$

The extraction of *stream′ f g* from *stream f g* is done with the help of the conversion isomorphisms (5.2) for the two algebraic list datatypes involved:

$$\begin{aligned} &\mathit{consumption\text{-}iso} : \\ &\quad (h : X \to X) \to \\ &\quad \mathsf{AlgList}\ A\ (\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f))\ h \cong \Sigma[\, as : \mathsf{List}\ A \,]\ \mathit{fold}\ (\mathit{foldl\text{-}alg}\ f)\ as \equiv h \\ &\mathit{production\text{-}iso} : \\ &\quad (x : X) \to \mathsf{AlgList}\ B\ (\mathit{fun}\ g^\circ)\ x \cong \Sigma[\, bs : \mathsf{List}\ B \,]\ (\!|\mathit{fun}\ g^\circ|\!)\ bs\ x \end{aligned}$$

(where *consumption-iso* has been simplified by *fun-preserves-fold*). Given $x : X$, what *stream′ f g x* does is

- lifting the input $as : \mathsf{List}\ A$ to an algebraic list of type

  $$\mathsf{AlgList}\ A\ (\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f))\ (\mathit{fold}\ (\mathit{foldl\text{-}alg}\ f)\ as)$$

  using the right-to-left direction of *consumption-iso* (*fold* (*foldl-alg f*) *as*) (with the equality proof obligation discharged trivially by refl),
- transforming this algebraic list to a new one of type

  $$\mathsf{AlgList}\ B\ (\mathit{fun}\ g^\circ)\ (\mathit{foldl}\ f\ x\ as)$$

  using *stream f g x*, and
- demoting the new algebraic list to List B using the left-to-right direction of *production-iso* (*foldl f x as*).

The use of *production-iso* in the last step ensures that the result $stream'\ f\ g\ x\ as$: List B satisfies

$$(\!|fun\ g^\circ|\!)\ (stream'\ f\ g\ x\ as)\ (foldl\ f\ x\ as)$$

which easily implies

$$((\!|fun\ g^\circ|\!)^\circ \bullet fun\ (foldl\ f\ x))\ as\ (stream'\ f\ g\ x\ as)$$

i.e., $fun\ (stream'\ f\ g\ x) \subseteq meta\ f\ g\ x$, as required.

What is left is the implementation of $stream\ f\ g$. Operationally, we maintain a state of type X (and hence requires an initial state as an input to the function), and we can try either

- to update the state by consuming elements of A with f, or
- to produce elements of B (and transit to a new state) by applying g to the state.

Since we want $stream\ f\ g$ to be as productive as possible, we should always try to produce elements of B with g first, and only try to consume elements of A with f when g produces nothing. In Agda:

```
stream f g : (x : X) {h : X → X} →
             AlgList A (fun (foldl-alg f)) h → AlgList B (fun g ∘) (h x)
stream f g x      as                  with g x    | inspect g x
stream f g x {h} as                   | next b x' | [ gxeq ] = cons b (h x') { }0
                                                                (stream f g x' as)
stream f g x      (nil       refl   ) | nothing  | [ gxeq ] = nil gxeq
stream f g x      (cons a h' refl as) | nothing  | [ gxeq ] = stream f g (f x a) as
```

We match $g\ x$ with either of the two patterns next $b\ x' = (\text{`cons} , b , x' , \blacksquare)$ and nothing $= (\text{`nil} , \blacksquare)$.

Agda doesn't really allow this, though.

- If the result is next $b\ x'$, we should emit b and and use x' as the new state; the recursively computed algebraic list is indexed with $h\ x'$, and we are left with a proof obligation of type $g\ (h\ x) \equiv \text{next}\ b\ (h\ x')$ at Goal 0; we will come back to this proof obligation later.
- If the result is nothing, we should attempt to consume the input list.



Figure 5.4 State transitions involved in commutativity of production and consumption (cf. Gibbons [2007b, Figures 1 and 2]).

- If the input list is empty, implying that the index h of its type is just id, both production and consumption have ended, so we return an empty list. The nil constructor requires a proof of $(fun\ g\ \circ)$ nothing $(h\ x)$, which reduces to $g\ x \equiv$ nothing and is discharged with the help of the "inspect idiom" in Agda's standard library (which, in a **with**-matching, gives a proof that the term being matched (in this case $g\ x$) is propositionally equal to the matched pattern (in this case nothing)).
- Otherwise the input list is nonempty, implying that h is $h' \circ flip\ f\ a$ where a is the head of the input list, and we should continue with the new state $f\ x\ a$, keeping the tail for further consumption. Typing directly works out because the index of the recursive result $h'\ (f\ x\ a)$ and the required index $(h' \circ flip\ f\ a)\ x$ are definitionally equal.

Now we look at Goal 0. We have

$$gxeq : g\ x \equiv \mathsf{next}\ b\ x'$$

in the context, and need to prove

$$g\ (h\ x) \equiv \mathsf{next}\ b\ (h\ x')$$

This is commutativity of production and consumption (see Figure 5.4): The function $h : X \to X$ is the state transformation resulting from consumption of the input list as. From the initial state x, we can either

- apply g to x to **produce** b and reach a new state x', and then apply h to **consume** the list and update the state to $h\ x'$, or

- apply h to **consume** the list and update the state to $h\ x$, and then apply g to $h\ x$ to **produce** an element and reach a new state,

and we need to prove that the outcomes are the same: doing production using g and consumption using h in whichever order should emit the same element and reach the same final state. This cannot be true in general, so we should impose some commutativity condition on f and g, which is called the **streaming condition**:

$$
\begin{aligned}
&\mathit{StreamingCondition}\ f\ g\ :\ \mathsf{Set}\\
&\mathit{StreamingCondition}\ f\ g\ =\\
&\quad (a\ :\ A)\ (b\ :\ B)\ (x\ x'\ :\ X) \to g\ x\ \equiv\ \mathsf{next}\ b\ x' \to g\ (f\ x\ a)\ \equiv\ \mathsf{next}\ b\ (f\ x'\ a)
\end{aligned}
$$

The streaming condition is commutativity of one step of production and consumption, whereas the proof obligation at Goal 0 is commutativity of one step of production and multiple steps of consumption (of the entire list), so we perform a straightforward induction to extend the streaming condition along the axis of consumption:

$$
\begin{aligned}
&\mathit{streaming\text{-}lemma}\ :\\
&\quad (b\ :\ B)\ (x\ x'\ :\ X) \to g\ x\ \equiv\ \mathsf{next}\ b\ x' \to\\
&\quad \{h\ :\ X \to X\} \to \mathsf{AlgList}\ A\ (\mathit{fun}\ (\mathit{foldl\text{-}alg}\ f))\ h \to g\ (h\ x)\ \equiv\ \mathsf{next}\ b\ (h\ x')\\
&\mathit{streaming\text{-}lemma}\ b\ x\ x'\ \mathit{eq}\ (\mathsf{nil}\qquad\ \mathsf{refl}\quad)\ =\ \mathit{eq}\\
&\mathit{streaming\text{-}lemma}\ b\ x\ x'\ \mathit{eq}\ (\mathsf{cons}\ a\ h\ \mathsf{refl}\ \mathit{as})\ =\\
&\quad \mathit{streaming\text{-}lemma}\ b\ (f\ x\ a)\ (f\ x'\ a)\ (\mathit{streaming\text{-}condition}\ f\ g\ a\ b\ x\ x'\ \mathit{eq})\ \mathit{as}
\end{aligned}
$$

where *streaming-condition* : *StreamingCondition* $f\ g$ is a proof term that should be supplied along with f and g in the beginning. Goal 0 is then discharged by the term *streaming-lemma* $b\ x\ x'$ *gxeq as*.

We have thus completed the implementation of the Streaming Theorem, except that *stream* $f\ g$ is non-terminating, as there is no guarantee that g produces only a finite number of elements. In our setting, where the output list is specified to be finite, we can additionally require that g is well-founded and revise *stream* accordingly (see, e.g., Nordström [1988]); the general way out is to switch to coinductive datatypes to allow the output list to be infinite, which, however, falls outside the scope of this thesis.

It is interesting to compare our implementation with the proofs of Bird and Gibbons [2003]. While their Lemma 29 turns explicitly into our *streaming-lemma*, their Theorem 30 goes implicitly into the typing of *stream* and no longer needs special attention. The structure of *stream* already matches that of Bird and Gibbons's proof of their Theorem 30, and the principled type design using algebraic ornamentation elegantly loads the proof onto the structure of *stream* — this is internalism at its best.

5.3.3 The minimum coin change problem

Suppose that we have an unlimited number of 1-penny, 2-pence, and 5-pence coins, modelled by the following datatype:

$$\textbf{data}\ \mathsf{Coin} : \mathsf{Set}\ \textbf{where}\ \mathsf{1p}\ \mathsf{2p}\ \mathsf{5p} : \mathsf{Coin}$$

Given $n : \mathsf{Nat}$, the **minimum coin change problem** asks for the least number of coins that make up n pence. We can give a relational specification of the problem with the following minimisation operator:

$$\begin{aligned}&min\_\bullet\Lambda\_ : \{I : \mathsf{Set}\}\,\{X\ Y : I \to \mathsf{Set}\}\,(R : Y \rightsquigarrow Y)\,(S : X \rightsquigarrow Y) \to (X \rightsquigarrow Y)\\ &(min\ R \bullet\Lambda\ S)\ x\ y\ =\ S\ x\ y \times (\forall\ y' \to S\ x\ y' \to R\ y'\ y)\end{aligned}$$

An input $x : X\ i$ for some $i : I$ is mapped by $min\ R \bullet\Lambda\ S$ to $y : Y\ i$ if y is a possible result in $S\ x : \mathscr{P}(Y\ i)$ and is the smallest such result under R, in the sense that any y' in $S\ x : \mathscr{P}(Y\ i)$ must satisfy $R\ y'\ y$. (We think of R as mapping larger inputs to smaller outputs.) Intuitively, we can think of $min\ R \bullet\Lambda\ S$ as consisting of two steps: the first step $\Lambda\ S$ computes the set of all possible results yielded by S, and the second step $min\ R$ nondeterministically chooses a minimum result from that set. We use bags of coins as the type of solutions, and represent them as decreasingly ordered lists indexed with an upper bound. (This is a deliberate choice to make the derivation work, but one would naturally turn to this representation having attempted to apply the **Greedy Theorem**, which will be introduced shortly.) If we define the ordering on coins as

```
_⩽C_ : Coin → Coin → Set
c ⩽C d = value c ⩽ value d
```

where the values of the coins are defined by

```
value : Coin → Nat
value 1p = 1
value 2p = 2
value 5p = 5
```

then the datatype of coin bags we use is

```
CoinBagOD : OrnDesc Coin ! (ListD Coin)
CoinBagOD = OrdListOD Coin (flip _⩽C_)

indexfirst data CoinBag : Coin → Set where
  CoinBag c ∋ nil
            | cons (d : Coin) (leq : d ⩽C c) (b : CoinBag d)
```

The base functor for CoinBag is

```
𝔽 ⌊CoinBagOD⌋ : (Coin → Set) → (Coin → Set)
𝔽 ⌊CoinBagOD⌋ X c =
  Σ ListTag λ { 'nil  ↦ ⊤
              ; 'cons ↦ Σ[ d : Coin ] (d ⩽C c) × X d × ⊤ }
```

The total value of a coin bag is the sum of the values of the coins in the bag, which is computed by a (functional) fold:

```
total-value-alg : 𝔽 ⌊CoinBagOD⌋ (const Nat) ⇉ const Nat
total-value-alg ('nil  , ▪) = 0
total-value-alg ('cons , d , _ , n , ▪) = value d + n

total-value : CoinBag ⇉ const Nat
total-value = fold total-value-alg
```

and the number of coins in a coin bag is also computed by a fold:

```
size-alg : 𝔽 ⌊CoinBagOD⌋ (const Nat) ⇉ const Nat
size-alg ('nil  , ▪) = 0
size-alg ('cons , _ , _ , n , ▪) = 1 + n
```

$$
\begin{aligned}
&\mathit{size} \;:\; \mathsf{CoinBag} \rightrightarrows \mathit{const}\ \mathsf{Nat} \\
&\mathit{size} \;=\; \mathit{fold}\ \mathit{size\text{-}alg}
\end{aligned}
$$

The specification of the minimum coin change problem can now be written as

$$
\begin{aligned}
&\mathit{min\text{-}coin\text{-}change} \;:\; \mathit{const}\ \mathsf{Nat} \rightsquigarrow \mathsf{CoinBag} \\
&\mathit{min\text{-}coin\text{-}change} \;=\; \mathit{min}\ (\mathit{fun}\ \mathit{size}^{\circ} \bullet \mathit{leq} \bullet \mathit{fun}\ \mathit{size}) \bullet \Lambda\ (\mathit{fun}\ \mathit{total\text{-}value}^{\circ})
\end{aligned}
$$

where $\mathit{leq} \;=\; \mathit{geq}^{\circ} \;:\; \mathit{const}\ \mathsf{Nat} \rightsquigarrow \mathit{const}\ \mathsf{Nat}$ maps a natural number n to any natural number that is at most n. Intuitively, given an input $n \;:\; \mathsf{Nat}$, the relation $\mathit{fun}\ \mathit{total\text{-}value}^{\circ}$ computes an arbitrary coin bag whose total value is n, so *min-coin-change* first computes the set of all such coin bags and then chooses from the set a coin bag whose size is smallest. Our goal, then, is to write a functional program $f \;:\; \mathit{const}\ \mathsf{Nat} \rightrightarrows \mathsf{CoinBag}$ such that $\mathit{fun}\ f \;\subseteq\; \mathit{min\text{-}coin\text{-}change}$, and then $f\ \{\mathsf{5p}\} \;:\; \mathsf{Nat} \to \mathsf{CoinBag\ 5p}$ would be a solution. (The type CoinBag 5p contains all coin bags, since 5p is the largest denomination and hence a trivial upper bound on the content of bags.) Of course, we may guess what f should look like, but its correctness proof is much harder. Can we construct the program and its correctness proof in a more manageable way?

The plan

In traditional relational program derivation, we would attempt to refine the specification *min-coin-change* to some simpler relational program and then to an executable functional program by applying algebraic laws and theorems. With algebraic ornamentation, however, there is a new possibility: if we can derive that, for some algebra $R \;:\; \mathbb{F}\ \lfloor \mathit{CoinBagOD} \rfloor\ (\mathit{const}\ \mathsf{Nat}) \rightsquigarrow \mathit{const}\ \mathsf{Nat}$,

$$(\!|\, R \,|\!)^{\circ} \;\subseteq\; \mathit{min\text{-}coin\text{-}change} \tag{5.4}$$

then we can manufacture a new datatype

$$
\begin{aligned}
&\mathit{GreedyBagOD} \;:\; \mathsf{OrnDesc}\ (\mathsf{Coin} \times \mathsf{Nat})\ \mathsf{outl}\ \lfloor \mathit{CoinBagOD} \rfloor \\
&\mathit{GreedyBagOD} \;=\; \mathit{algOD}\ \lfloor \mathit{CoinBagOD} \rfloor\ R
\end{aligned}
$$

$$
\begin{aligned}
&\mathsf{GreedyBag} \;:\; \mathsf{Coin} \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{GreedyBag}\ c\ n \;=\; \mu\ \lfloor \mathit{GreedyBagOD} \rfloor\ (c\ ,\ n)
\end{aligned}
$$

and construct a function of type

$$\mathit{greedy} \;:\; (c \;:\; \mathsf{Coin})\;(n \;:\; \mathsf{Nat}) \to \mathsf{GreedyBag}\;c\;n$$

from which we can assemble a solution

$$\begin{aligned}&\mathit{sol} \;:\; \mathsf{Nat} \to \mathsf{CoinBag}\;\mathsf{5p}\\ &\mathit{sol} \;=\; \mathit{forget}\;\lceil \mathit{GreedyBagOD} \rceil \circ \mathit{greedy}\;\mathsf{5p}\end{aligned}$$

The program *sol* satisfies the specification because of the following argument: For any $c \;:\; \mathsf{Coin}$ and $n \;:\; \mathsf{Nat}$, by (5.2) we have

$$\mathsf{GreedyBag}\;c\;n \;\cong\; \Sigma[\, b : \mathsf{CoinBag}\;c \,]\;\; (\!|\, R \,|\!)\;b\;n$$

In particular, since the first half of the left-to-right direction of the isomorphism is $\mathit{forget}\;\lceil \mathit{GreedyBagOD} \rceil$, we have

$$(\!|\, R \,|\!)\;(\mathit{forget}\;\lceil \mathit{GreedyBagOD} \rceil\;g)\;n$$

for any $g \;:\; \mathsf{GreedyBag}\;c\;n$. Substituting g by $\mathit{greedy}\;\mathsf{5p}\;n$, we get

$$(\!|\, R \,|\!)\;(\mathit{sol}\;n)\;n$$

which implies, by (5.4),

$$\mathit{min\text{-}coin\text{-}change}\;n\;(\mathit{sol}\;n)$$

i.e., *sol* satisfies the specification. Thus all we need to do to solve the minimum coin change problem is

- refine the specification *min-coin-change* to the converse of a fold, i.e., find the algebra R in (5.4), and
- construct the internalist program *greedy*.

Refining the specification

The key to refining *min-coin-change* to the converse of a fold lies in the following version of the **Greedy Theorem**, which is essentially a specialisation of Bird and de Moor's Theorem 10.1 [1997]: Let $D \;:\; \mathsf{Desc}\;I$ be a description, $R \;:\; \mu\;D \rightsquigarrow \mu\;D$ a preorder, and $S \;:\; \mathbb{F}\;D\;X \rightsquigarrow X$ an algebra. Consider the specification

$$min\ R \bullet \Lambda\ (([\![\ S\ ]\!])^\circ)$$

That is, given an input value $x : X\ i$ for some $i : I$, we choose a minimum under R among all those elements of $\mu\ D\ i$ that computes to x through $[\![\ S\ ]\!]$. The Greedy Theorem states that, if the initial algebra

$$\alpha\ =\ fun\ \mathsf{con} : \mathbb{F}\ D\ (\mu\ D) \rightsquigarrow \mu\ D$$

is monotonic on R, i.e.,

$$\alpha \bullet \mathbb{R}\ D\ R\ \subseteq\ R \bullet \alpha$$

and there is a relation (ordering) $Q : \mathbb{F}\ D\ X \rightsquigarrow \mathbb{F}\ D\ X$ such that the **greedy condition**

$$\alpha \bullet \mathbb{R}\ D\ (([\![\ S\ ]\!])^\circ) \bullet (Q \cap (S^\circ \bullet S))^\circ\ \subseteq\ R^\circ \bullet \alpha \bullet \mathbb{R}\ D\ (([\![\ S\ ]\!])^\circ)$$

is satisfied, then we have

$$[\![\ (min\ Q \bullet \Lambda\ (S^\circ))^\circ\ ]\!]^\circ\ \subseteq\ min\ R \bullet \Lambda\ (([\![\ S\ ]\!])^\circ)$$

Here we offer an intuitive explanation of the Greedy Theorem, but the theorem admits an elegant calculational proof, which can be faithfully reprised in Agda. The monotonicity condition states that if $ds : \mathbb{F}\ D\ (\mu\ D)\ i$ for some $i : I$ is better than $ds' : \mathbb{F}\ D\ (\mu\ D)\ i$ under $\mathbb{R}\ D\ R$, i.e., ds and ds' are equal except that the recursive positions of ds are all better than the corresponding recursive positions of ds' under R, then $\mathsf{con}\ ds : \mu\ D\ i$ would be better than $\mathsf{con}\ ds' : \mu\ D\ i$ under R. This implies that, when solving the optimisation problem, better solutions to subproblems would lead to a better solution to the original problem, so the **principle of optimality** applies — to reach an optimal solution, it suffices to find optimal solutions to subproblems, and we are entitled to use the converse of a fold to find optimal solutions recursively. The greedy condition further states that there is an ordering Q on the ways of decomposing the problem which has significant influence on the quality of solutions: Suppose that there are two decompositions xs and $xs' : \mathbb{F}\ D\ X\ i$ of some problem $x : X\ i$ for some $i : I$, i.e., both xs and xs' are in $S^\circ\ x : \mathscr{P}(\mathbb{F}\ D\ X\ i)$, and assume that xs is better than xs' under Q. Then for any solution resulting from xs' (computed by $\alpha \bullet \mathbb{R}\ D\ (([\![\ S\ ]\!])^\circ)$) there always exists a better solution resulting from xs, so ignoring xs' would only rule out worse solutions. The greedy condition thus

```
data CoinBag′View : {c : Coin} {n : Nat} {l : Nat} → CoinBag′ c n l → Set where
  empty : {c : Coin} → CoinBag′View {c} {0} {0} bnil
  1p1p  : {m l : Nat} {lep : 1p ⩽C 1p}
          (b : CoinBag′ 1p m l) → CoinBag′View {1p} {1 + m} {1 + l} (bcons 1p lep b)
  1p2p  : {m l : Nat} {lep : 1p ⩽C 2p}
          (b : CoinBag′ 1p m l) → CoinBag′View {2p} {1 + m} {1 + l} (bcons 1p lep b)
  2p2p  : {m l : Nat} {lep : 2p ⩽C 2p}
          (b : CoinBag′ 2p m l) → CoinBag′View {2p} {2 + m} {1 + l} (bcons 2p lep b)
  1p5p  : {m l : Nat} {lep : 1p ⩽C 5p}
          (b : CoinBag′ 1p m l) → CoinBag′View {5p} {1 + m} {1 + l} (bcons 1p lep b)
  2p5p  : {m l : Nat} {lep : 2p ⩽C 5p}
          (b : CoinBag′ 2p m l) → CoinBag′View {5p} {2 + m} {1 + l} (bcons 2p lep b)
  5p5p  : {m l : Nat} {lep : 5p ⩽C 5p}
          (b : CoinBag′ 5p m l) → CoinBag′View {5p} {5 + m} {1 + l} (bcons 5p lep b)
```

Figure 5.5 The view datatype on CoinBag′.

guarantees that we will arrive at an optimal solution by always choosing the best decomposition, which is done by $min\ Q \bullet \Lambda\ (S^{\circ}) : X \rightsquigarrow \mathbb{F}\ D\ X$.

Back to the minimum coin change problem: By *fun-preserves-fold*, the specification *min-coin-change* is equivalent to

$$min\ (fun\ size^{\circ} \bullet leq \bullet fun\ size) \bullet \Lambda\ (([\![fun\ total\text{-}value\text{-}alg]\!])^{\circ})$$

which matches the form of the generic specification given in the Greedy Theorem, so we try to discharge the two conditions of the theorem. The monotonicity condition reduces to monotonicity of *fun size-alg* on *leq*, and can be easily proved either by relational calculation or pointwise reasoning. As for the greedy condition, an obvious choice for Q is an ordering that leads us to choose the largest possible denomination, so we go for

$$\begin{aligned}
&Q : \mathbb{F}\ \lfloor CoinBagOD \rfloor\ (const\ \mathsf{Nat}) \rightsquigarrow \mathbb{F}\ \lfloor CoinBagOD \rfloor\ (const\ \mathsf{Nat}) \\
&Q\ (\text{`nil}\ ,\ \blacksquare) = return\ (\text{`nil}\ ,\ \blacksquare) \\
&Q\ (\text{`cons}\ ,\ d\ ,\ \_) = (\lambda\, e\ rest \mapsto \text{`cons}\ ,\ e\ ,\ rest)\ \langle\$\rangle^2\ (_\leqslant_{\mathsf{C}}_\ d)\ any
\end{aligned}$$

```
greedy-lemma : (c d : Coin) → c ⩽C d → (m n : Nat) → value c + m ≡ value d + n →
               (l : Nat) (b : CoinBag′ c m l) → Σ[ l′ : Nat ] CoinBag′ d n l′ × (l′ ⩽ l)
greedy-lemma c    d  c⩽d m       n       eq   l       b with view-ordered-coin c d c⩽d
greedy-lemma .1p .1p _   .n      n       refl l       b CoinBag′ 1p n l | 1p1p = {Σ[ l′ : Nat ] CoinBag′ 1p n l′ × (l′ ⩽ l)}0
greedy-lemma .1p .2p _   .(1 + n) n      refl l       b | 1p2p with view-CoinBag′ b
greedy-lemma .1p .2p _   .(1 + n) n      refl .(1 + l″) ._ | 1p2p | 1p1p {.n} {l″} b CoinBag′ 1p n l″ =
                                                                {Σ[ l′ : Nat ] CoinBag′ 2p n l′ × (l′ ⩽ 1 + l″)}1
greedy-lemma .1p .5p _   .(4 + n) n      refl l       b | 1p5p with view-CoinBag′ b
greedy-lemma .1p .5p _   .(4 + n) n      refl ._      ._ | 1p5p | 1p1p b with view-CoinBag′ b
greedy-lemma .1p .5p _   .(4 + n) n      refl ._      ._ | 1p5p | 1p1p ._ | 1p1p b with view-CoinBag′ b
greedy-lemma .1p .5p _   .(4 + n) n      refl ._      ._ | 1p5p | 1p1p ._ | 1p1p ._ | 1p1p b with view-CoinBag′ b
greedy-lemma .1p .5p _   .(4 + n) n      refl .(4 + l″) ._ | 1p5p | 1p1p ._ | 1p1p ._ | 1p1p ._ | 1p1p {.n} {l″} b CoinBag′ 1p n l″ =
                                                                {Σ[ l′ : Nat ] CoinBag′ 5p n l′ × (l′ ⩽ 4 + l″)}2
greedy-lemma .2p .2p _   .n      n       refl l       b CoinBag′ 2p n l | 2p2p = {Σ[ l′ : Nat ] CoinBag′ 2p n l′ × (l′ ⩽ l)}3
greedy-lemma .2p .5p _   .(3 + n) n      refl l       b | 2p5p with view-CoinBag′ b
greedy-lemma .2p .5p _   .(3 + n) n      refl ._      ._ | 2p5p | 1p2p b with view-CoinBag′ b
greedy-lemma .2p .5p _   .(3 + n) n      refl ._      ._ | 2p5p | 1p2p ._ | 1p1p b with view-CoinBag′ b
greedy-lemma .2p .5p _   .(3 + n) n      refl .(3 + l″) ._ | 2p5p | 1p2p ._ | 1p1p ._ | 1p1p {.n} {l″} b CoinBag′ 1p n l″ =
                                                                {Σ[ l′ : Nat ] CoinBag′ 5p n l′ × (l′ ⩽ 3 + l″)}4
greedy-lemma .2p .5p _   .(3 + n) n      refl ._      ._ | 2p5p | 2p2p b with view-CoinBag′ b
greedy-lemma .2p .5p _   .(3 + n) n      refl .(2 + l″) ._ | 2p5p | 2p2p ._ | 1p2p {.n} {l″} b CoinBag′ 2p n l″ =
                                                                {Σ[ l′ : Nat ] CoinBag′ 5p n l′ × (l′ ⩽ 2 + l″)}5
greedy-lemma .2p .5p _   .(4 + k) .(1 + k) refl .(2 + l″) ._ | 2p5p | 2p2p ._ | 2p2p {k} {l″} b CoinBag′ 2p k l″ =
                                                                {Σ[ l′ : Nat ] CoinBag′ 5p (1 + k) l′ × (l′ ⩽ 2 + l″)}6
greedy-lemma .5p .5p _   .n      n       refl l       b CoinBag′ 5p n l | 5p5p = {Σ[ l′ : Nat ] CoinBag′ 5p n l′ × (l′ ⩽ l)}7
```

Figure 5.6 Cases of *greedy-lemma*, generated semi-automatically by Agda's interactive case-split mechanism. Goal types are shown in the interaction points, and the types of some pattern variables are shown in subscript beside them.

where, in the cons case, the output is required to be also a cons node, and the coin at its head position must be one that is no smaller than the coin d at the head position of the input. It is non-trivial to prove the greedy condition by relational calculation. Here we offer instead a brute-force yet conveniently expressed case analysis by dependent pattern matching. Define a new datatype $\mathsf{CoinBag}'$ by composing two algebraic ornaments on $\lfloor \mathit{CoinBagOD} \rfloor$ in parallel:

$$\begin{aligned}
&\mathit{CoinBag}'\mathit{OD} \;:\; \mathsf{OrnDesc}\;(\mathsf{outl} \bowtie \mathsf{outl})\;\mathit{pull}\;\lfloor \mathit{CoinBagOD} \rfloor \\
&\mathit{CoinBag}'\mathit{OD} \;=\; \lceil \mathit{algOD}\;\lfloor \mathit{CoinBagOD} \rfloor\;(\mathit{fun}\;\mathit{total\text{-}value\text{-}alg}) \rceil \otimes \\
&\qquad\qquad\qquad \lceil \mathit{algOD}\;\lfloor \mathit{CoinBagOD} \rfloor\;(\mathit{fun}\;\mathit{size\text{-}alg}) \rceil \\
&\mathsf{CoinBag}' \;:\; \mathsf{Coin} \to \mathsf{Nat} \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{CoinBag}' \;=\; \mu\;\lfloor \mathit{CoinBag}'\mathit{OD} \rfloor\;(\mathsf{ok}\;(c\;,\;n)\;,\;\mathsf{ok}\;(c\;,\;l))
\end{aligned}$$

whose two constructors can be specialised to

$$\begin{aligned}
&\mathsf{bnil} \;\;:\; \{c \;:\; \mathsf{Coin}\} \to \mathsf{CoinBag}'\;c\;0\;0 \\
&\mathsf{bcons} \;:\; \{c \;:\; \mathsf{Coin}\}\;\{n\;l \;:\; \mathsf{Nat}\} \to (d \;:\; \mathsf{Coin}) \to d \leqslant_{\mathsf{C}} c \to \\
&\qquad\quad \mathsf{CoinBag}'\;d\;n\;l \to \mathsf{CoinBag}'\;c\;(\mathit{value}\;d + n)\;(1 + l)
\end{aligned}$$

By predicate swapping using the modularity isomorphisms (Section 3.3.2) and *fun-preserves-fold*, $\mathsf{CoinBag}'$ is characterised by the isomorphisms

$$\mathsf{CoinBag}'\;c\;n\;l \;\cong\; \Sigma[\,b : \mathsf{CoinBag}\;c\,]\;\;(\mathit{total\text{-}value}\;b \;\equiv\; n) \times (\mathit{size}\;b \;\equiv\; l) \qquad (5.5)$$

for all $c \;:\; \mathsf{Coin}$, $n \;:\; \mathsf{Nat}$, and $l \;:\; \mathsf{Nat}$. Hence a coin bag of type $\mathsf{CoinBag}'\;c\;n\;l$ contains l coins that are no larger than c and sum up to n pence. The greedy condition then essentially reduces to this lemma:

$$\begin{aligned}
\mathit{greedy\text{-}lemma} \;:\; &(c\;d \;:\; \mathsf{Coin}) \to c \leqslant_{\mathsf{C}} d \to \\
&(m\;n \;:\; \mathsf{Nat}) \to \mathit{value}\;c + m \;\equiv\; \mathit{value}\;d + n \to \\
&(l \;:\; \mathsf{Nat})\;(b \;:\; \mathsf{CoinBag}'\;c\;m\;l) \to \\
&\Sigma[\,l' : \mathsf{Nat}\,]\;\;\mathsf{CoinBag}'\;d\;n\;l' \times (l' \leqslant l)
\end{aligned}$$

That is, given a problem (i.e., a value to be represented by coins), if $c \;:\; \mathsf{Coin}$ is a choice of decomposition (i.e., the first coin used) no better than $d \;:\; \mathsf{Coin}$ (i.e., $c \leqslant_{\mathsf{C}} d$ — recall that we prefer larger denominations), and $b \;:\; \mathsf{CoinBag}'\;c\;m\;l$ is a solution of size l to the remaining subproblem m resulting from choosing c, then there is a solution to the remaining subproblem n resulting from choos-

ing d whose size l' is no greater than l. We define two **views** [McBride and McKinna, 2004] — or "customised pattern matching" — to aid the analysis:

- The first view analyses a proof of $c \leqslant_{\mathsf{C}} d$ and exhausts all possibilities of c and d,

  ```
  data CoinOrderedView : Coin → Coin → Set where
    1p1p : CoinOrderedView 1p 1p
    1p2p : CoinOrderedView 1p 2p
    1p5p : CoinOrderedView 1p 5p
    2p2p : CoinOrderedView 2p 2p
    2p5p : CoinOrderedView 2p 5p
    5p5p : CoinOrderedView 5p 5p

  view-ordered-coin : (c d : Coin) → c ⩽C d → CoinOrderedView c d
  ```

  where the covering function *view-ordered-coin* is written by standard pattern matching on c and d.

- The second view analyses some b : CoinBag′ c n l and exhausts all possibilities of c, n, l, and the first coin in b (if any). The view datatype CoinBag′View is shown in Figure 5.5, and the covering function

  ```
  view-CoinBag′ :
    {c : Coin} {n l : Nat} (b : CoinBag′ c n l) → CoinBag′View b
  ```

  is again written by standard pattern matching.

Given these two views, the function *greedy-lemma* can be split into eight cases by first exhausting all possibilities of c and d with *view-ordered-coin* and then analysing the content of b with *view-CoinBag′*. Figure 5.6 shows the case-split tree generated semi-automatically by Agda; the detail is explained as follows:

- At Goal 0 (and similarly Goals 3 and 7), the input bag is b : CoinBag′ 1p n l, and we should produce a CoinBag′ 1p n l' for some l' : Nat such that $l' \leqslant l$. This is easy because b itself is a suitable bag.
- At Goal 1 (and similarly Goals 2, 4, and 5), the input bag has type CoinBag′ 1p $(1 + n)$ l, i.e., the coins in the bag are no larger than 1p and the total value is $1 + n$. The bag must contain 1p as its first coin; let the rest

of the bag be b : $\mathsf{CoinBag'}\ \mathsf{1p}\ n\ l''$. At this point Agda can deduce that l must be $1 + l''$. Now we can return b as the result after the upper bound on its coins is relaxed from 1p to 2p, which is done by

$$relax : \{c\ d : \mathsf{Coin}\}\ \{n\ l : \mathsf{Nat}\} \to c \leqslant_{\mathsf{C}} d \to \mathsf{CoinBag'}\ c\ n\ l \to \mathsf{CoinBag'}\ d\ n\ l$$

- The remaining Goal 6 is the most interesting one: The input bag has type $\mathsf{CoinBag'}\ \mathsf{2p}\ (3 + n)\ l$, which in this case contains two 2-pence coins, and the rest of the bag is b : $\mathsf{CoinBag'}\ \mathsf{2p}\ k\ l''$. Agda deduces that n must be $1 + k$ and l must be $2 + l''$. We thus need to add a penny to b to increase its total value to $1 + k$, which is done by

  $$add\text{-}penny :$$
  $$\{c : \mathsf{Coin}\}\ \{n\ l : \mathsf{Nat}\} \to \mathsf{CoinBag'}\ c\ n\ l \to \mathsf{CoinBag'}\ c\ (1 + n)\ (1 + l)$$

  and relax the bound of *add-penny b* from 2p to 5p.

The above case analysis may look tedious, but note that Agda is able to

- produce all the cases (modulo some cosmetic revisions) after the programmer decides to use the two views and instructs Agda to do case splitting accordingly, and
- manage all the bookkeeping and deductions about the total value and the size of bags with dependent pattern matching,

so the overhead on the programmer's side is actually less than it seems. The greedy condition can now be discharged by pointwise reasoning, using (5.5) to interface with *greedy-lemma*. We conclude that the Greedy Theorem is applicable, and obtain

$$(\!|\ (min\ Q \bullet \Lambda\ (fun\ total\text{-}value\text{-}alg^\circ))^\circ\ |\!)^\circ \subseteq min\text{-}coin\text{-}change$$

We have thus found the algebra

$$R = (min\ Q \bullet \Lambda\ (fun\ total\text{-}value\text{-}alg^\circ))^\circ$$

which will help us to construct the final internalist program.

Constructing the internalist program

As planned, we synthesise a new datatype by ornamenting CoinBag using the algebra R derived above:

$$\begin{aligned}
&\mathit{GreedyBagOD} \;:\; \mathsf{OrnDesc}\ (\mathsf{Coin} \times \mathsf{Nat})\ \mathsf{outl}\ \lfloor \mathit{CoinBagOD} \rfloor \\
&\mathit{GreedyBagOD} \;=\; \mathit{algOD}\ \lfloor \mathit{CoinBagOD} \rfloor\ R \\
&\mathsf{GreedyBag} \;:\; \mathsf{Coin} \to \mathsf{Nat} \to \mathsf{Set} \\
&\mathsf{GreedyBag}\ c\ n \;=\; \mu\ \lfloor \mathit{GreedyBagOD} \rfloor\ (c\ ,\ n)
\end{aligned}$$

whose two constructors can be given the following types:

$$\begin{aligned}
\mathsf{gnil} \quad &: \{c \;:\; \mathsf{Coin}\}\ \{n \;:\; \mathsf{Nat}\} \to \\
&\quad \mathit{total\text{-}value\text{-}alg}\ (\text{`}\mathsf{nil}\ ,\ \blacksquare) \equiv n \to \\
&\quad ((ns \;:\; \mathbb{F}\ \lfloor \mathit{CoinBagOD} \rfloor\ (\mathit{const}\ \mathsf{Nat})) \to \\
&\qquad \mathit{total\text{-}value\text{-}alg}\ ns \equiv n \to Q\ ns\ (\text{`}\mathsf{nil}\ ,\ \blacksquare)) \to \\
&\quad \mathsf{GreedyBag}\ c\ n \\
\mathsf{gcons} \;:\; &\{c \;:\; \mathsf{Coin}\}\ \{n \;:\; \mathsf{Nat}\}\ (d \;:\; \mathsf{Coin})\ (d{\leqslant}c \;:\; d \leqslant_{\mathsf{C}} c) \to \\
&\quad \{n' \;:\; \mathsf{Nat}\} \to \mathit{total\text{-}value\text{-}alg}\ (\text{`}\mathsf{cons}\ ,\ d\ ,\ d{\leqslant}c\ ,\ n') \equiv n \to \\
&\quad ((ns \;:\; \mathbb{F}\ \lfloor \mathit{CoinBagOD} \rfloor\ (\mathit{const}\ \mathsf{Nat})) \to \\
&\qquad \mathit{total\text{-}value\text{-}alg}\ ns \equiv n \to Q\ ns\ (\text{`}\mathsf{cons}\ ,\ d\ ,\ d{\leqslant}c\ ,\ n')) \to \\
&\quad \mathsf{GreedyBag}\ d\ n' \to \mathsf{GreedyBag}\ c\ n
\end{aligned}$$

and implement the greedy algorithm by

$$\mathit{greedy} \;:\; (c \;:\; \mathsf{Coin})\ (n \;:\; \mathsf{Nat}) \to \mathsf{GreedyBag}\ c\ n$$

Let us first simplify the two constructors of GreedyBag. Each of the two constructors has two additional proof obligations coming from the algebra R:

- For gnil,
  - the first obligation $\mathit{total\text{-}value\text{-}alg}\ (\text{`}\mathsf{nil}\ ,\ \blacksquare) \equiv n$ reduces to $0 \equiv n$, so we may discharge the obligation by specialising n to 0;
  - for the second obligation, ns is necessarily $(\text{`}\mathsf{nil}\ ,\ \blacksquare)$ if $\mathit{total\text{-}value\text{-}alg}\ ns \equiv 0$, and indeed $Q$ maps $(\text{`}\mathsf{nil}\ ,\ \blacksquare)$ to $(\text{`}\mathsf{nil}\ ,\ \blacksquare)$, so the second obligation can be discharged as well.

  We thus obtain a simplified version of gnil:

$\mathsf{gnil}' \;:\; \{c \;:\; \mathsf{Coin}\} \to \mathsf{GreedyBag}\ c\ 0$

- For gcons,
  - the first obligation reduces to $\mathit{value}\ d + n' \equiv n$, so we may just specialise n to $\mathit{value}\ d + n'$ and discharge the obligation;
  - for the second obligation, any ns satisfying $\mathit{total\text{-}value\text{-}alg}\ ns \equiv \mathit{value}\ d + n'$ must be of the form $(\text{`cons}\ ,\ e\ ,\ e{\leqslant}c\ ,\ m'\ ,\ \blacksquare)$ for some $e \;:\; \mathsf{Coin}$, $e{\leqslant}c \;:\; e \leqslant_{\mathsf{C}} c$, and $m' \;:\; \mathsf{Nat}$ since the right-hand side $\mathit{value}\ d + n'$ of the equality is non-zero, and $Q$ maps $ns$ to $(\text{`cons}\ ,\ d\ ,\ d{\leqslant}c\ ,\ n'\ ,\ \blacksquare)$ if $e \leqslant_{\mathsf{C}} d$, so d should be the largest "usable" coin if this obligation is to be discharged. We say that $d \;:\; \mathsf{Coin}$ is **usable** with respect to some $c \;:\; \mathsf{Coin}$ and $n \;:\; \mathsf{Nat}$ if d is bounded above by c and can be part of a solution to the problem for n pence:

    $$\mathit{UsableCoin} \;:\; \mathsf{Nat} \to \mathsf{Coin} \to \mathsf{Coin} \to \mathsf{Set}$$
    $$\mathit{UsableCoin}\ n\ c\ d \;=\; (d \leqslant_{\mathsf{C}} c) \times (\Sigma[\, n' : \mathsf{Nat}\,]\ \mathit{value}\ d + n' \equiv n)$$

    The obligation can then be rewritten as

    $$(e \;:\; \mathsf{Coin}) \to \mathit{UsableCoin}\ (\mathit{value}\ d + n')\ c\ e \to e \leqslant_{\mathsf{C}} d$$

    which requires that d is the largest usable coin with respect to c and $\mathit{value}\ d + n'$. This obligation is the only one that cannot be trivially discharged, since it requires computation of the largest usable coin.

  We thus specialise gcons to

$$\begin{aligned}\mathsf{gcons}' \;:\; &\{c \;:\; \mathsf{Coin}\}\ (d \;:\; \mathsf{Coin}) \to d \leqslant_{\mathsf{C}} c \to \\ &\{n' \;:\; \mathsf{Nat}\} \to \\ &((e \;:\; \mathsf{Coin}) \to \mathit{UsableCoin}\ (\mathit{value}\ d + n')\ c\ e \to e \leqslant_{\mathsf{C}} d) \to \\ &\mathsf{GreedyBag}\ d\ n' \to \mathsf{GreedyBag}\ c\ (\mathit{value}\ d + n')\end{aligned}$$

Because of gcons′, we are directed to implement a function *maximum-coin* that computes the largest usable coin with respect to any $c \;:\; \mathsf{Coin}$ and non-zero $n \;:\; \mathsf{Nat}$:

$$\begin{aligned}&\mathit{maximum\text{-}coin}\ : \\ &\quad(c \;:\; \mathsf{Coin})\ (n \;:\; \mathsf{Nat}) \to n > 0 \to \\ &\quad\Sigma[\, d : \mathsf{Coin}\,]\ \mathit{UsableCoin}\ n\ c\ d \times ((e \;:\; \mathsf{Coin}) \to \mathit{UsableCoin}\ n\ c\ e \to e \leqslant_{\mathsf{C}} d)\end{aligned}$$

This takes some theorem proving but is overall a typical Agda exercise in dealing with natural numbers and ordering. Finally, the greedy algorithm is implemented as the following internalist program, which repeatedly uses *maximum-coin* to find the largest usable coin and unfolds a GreedyBag:

```
greedy : (c : Coin) (n : Nat) → GreedyBag c n
greedy c n = <-rec P f n c
  where
    P : Nat → Set
    P n = (c : Coin) → GreedyBag c n
    f : (n : Nat) → ((n′ : Nat) → n′ < n → P n′) → P n
    f n          rec c with compare-with-zero n
    f .0         rec c | is-zero = gnil′
    f n          rec c | above-zero n>z with maximum-coin c n n>z
    f .(value d + n′) rec c | above-zero n>z | d , (d⩽c , n′ , refl) , guc =
                                  gcons′ d d⩽c guc (rec n′ { }8 d)
```

In *greedy*, the combinator

```
<-rec : (P : Nat → Set) →
        ((n : Nat) → ((n′ : Nat) → n′ < n → P n′) → P n) →
        (n : Nat) → P n
```

is for well-founded recursion on $\_<\_$, and the function

```
compare-with-zero : (n : Nat) → ZeroView n
```

is a covering function for the view

```
data ZeroView : Nat → Set where
  is-zero    : ZeroView 0
  above-zero : {n : Nat} → n > 0 → ZeroView n
```

At Goal 8, Agda deduces that n is *value* $d + n'$ and demands that we prove $n' < \mathit{value}\ d + n'$ in order to make the recursive call, which is easily discharged since $\mathit{value}\ d > 0$.

5.4 Discussion

compare the McBride [2011] version (compatible with the two-constructor universe) and the Dagand and McBride [2012b] version of algebraic ornamentation in terms of "quality" (amount of σ's used); proof-relevant Algebra of Programming (e.g., *fun-preserves-fold*; linking to the next chapter); related work: Atkey et al. [2012]

Chapter 6

Categorical equivalence of ornaments and relational algebras

Consider the AlgList datatype in Section 5.2 again. The way it is refined relative to the plain list datatype looks canonical, in the sense that any variation of the list datatype can be programmed as a special case of AlgList: we can choose whatever index set we want by setting the carrier of the algebra R; and by carefully programming R, we can insert fields into the list datatype that add more information or put restriction on fields and indices. For example, if we want some new information in the nil case, we can program R such that $R\ (nil - tag\ ,\ \blacksquare)\ x$ contains a field requesting that information; if, in the cons case, we need the targeted index x, the head element a, and the index x' of the recursive position to be related in some way, we can program R such that $R\ (cons - tag\ ,\ a\ ,\ x')\ x$ expresses that relationship.

The above observation leads to the following general theorem: Let O : Orn $e\ D\ E$ be an ornament from D : Desc I to E : Desc J. There is a classifying algebra for O

$$clsAlg\ O\ :\ \mathbb{F}\ D\ (\mathsf{InvImage}\ e) \rightsquigarrow \mathsf{InvImage}\ e$$

such that there are isomorphisms

$$\mu\ \lfloor algOrn\ D\ (clsAlg\ O) \rfloor\ (e\ j\ ,\ \mathsf{ok}\ j)\ \cong\ \mu\ E\ j$$

for all $j : J$. That is, the algebraic ornamentation of D using the classifying algebra derived from O produces a datatype isomorphic to $\mu\ E$, so intuitively the algebraic ornament has the same content as O. We may interpret this theorem as saying that algebraic ornaments are "complete" for the ornament language: any relationship between datatypes that can be described by an ornament can be described up to isomorphism by an algebraic ornament.

The completeness theorem brings up a nice algebraic intuition about inductive families. Consider the ornament from lists to vectors, for example. This ornament specifies that the type $\mathsf{List}\ A$ is refined by the collection of types $\mathsf{Vec}\ A\ n$ for all $n : \mathsf{Nat}$. A list, say $a :: b :: [] : \mathsf{List}\ A$, can be reconstructed as a vector by starting in the type $\mathsf{Vec}\ A\ \mathsf{zero}$ as $[]$, jumping to the next type $\mathsf{Vec}\ A\ (\mathsf{suc}\ \mathsf{zero})$ as $b :: []$, and finally landing in $\mathsf{Vec}\ A\ (\mathsf{suc}\ (\mathsf{suc}\ \mathsf{zero}))$ as $a :: b :: []$. The list is thus classified as having length 2, as computed by the fold function *length*, and the resulting vector is a fused representation of the list and the classification proof. In the case of vectors, this classification is total and deterministic: every list is classified under one and only one index. But in general, classifications can be partial and nondeterministic. For example, promoting a list to an ordered list is classifying the list under an index that is a lower bound of the list. The classification process checks at each jump whether the list is still ordered; this check can fail, so an unordered list would "disappear" midway through the classification. Also there can be more than one lower bound for an ordered list, so the list can end up being classified under any one of them. Algebraic ornamentation in its original functional form can only capture part of this intuition about classification, namely those classifications that are total and deterministic. By generalising algebraic ornamentation to accept relational algebras, bringing in partiality and nondeterminacy, this idea about classification is captured in its entirety — a classification is just a relational fold computing the index that classifies an element. All ornaments specify classifications, and thus can be transformed into algebraic ornaments.

For more examples, let us first look at the classifying algebra for the ornament from natural numbers to lists. The base functor for natural numbers is

$$\mathbb{F}\ \mathit{NatD} : (\top \to \mathsf{Set}) \to (\top \to \mathsf{Set})$$
$$\mathbb{F}\ \mathit{NatD}\ X\ \_ = \Sigma\ \mathit{LTag}\ (\lambda\ \{\mathit{nil} - \mathit{tag} \to \top; \mathit{cons} - \mathit{tag} \to X\ \blacksquare\})$$

And the classifying algebra for the ornament *NatD-ListD A* is essentially

$$\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A) : \mathbb{F}\ \mathit{NatD}\ (\mathsf{InvImage}\ !) \rightsquigarrow \mathsf{InvImage}\ !$$
$$\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathit{nil} - \mathit{tag}\ \ ,\ \_\ \ )\ (\mathsf{ok}\ \blacksquare) = \top$$
$$\mathit{clsAlg}\ (\mathit{NatD\text{-}ListD}\ A)\ (\mathit{cons} - \mathit{tag}\ ,\ \mathsf{ok}\ t)\ (\mathsf{ok}\ \blacksquare) = A \times (t \equiv \blacksquare)$$

The result of folding a natural number n with this algebra is uninteresting, as it can only be $\mathsf{ok}\ \blacksquare$. The fold, however, requires an element of A for each successor node it encounters, so a proof that n goes through the fold consists of n elements of A. Another example is the ornament $\mathit{OL} = \lceil \mathit{OrdListOD}\ A\ \_\leqslant_A\_ \rceil$ from lists to ordered lists, whose classifying algebra is essentially

$$\mathit{clsAlg}\ \mathit{OL} : \mathbb{F}\ (\mathit{ListD}\ A)\ (\mathsf{InvImage}\ !) \rightsquigarrow \mathsf{InvImage}\ !$$
$$\mathit{clsAlg}\ \mathit{OL}\ (\mathit{nil} - \mathit{tag}\ \ ,\ \_\ \ \ \ \ \ )\ (\mathsf{ok}\ b) = \top$$
$$\mathit{clsAlg}\ \mathit{OL}\ (\mathit{cons} - \mathit{tag}\ ,\ a\ ,\ \mathsf{ok}\ b')\ (\mathsf{ok}\ b) = (b \leqslant_A a) \times (b' \equiv a)$$

In the nil case, the empty list can be mapped to any $\mathsf{ok}\ b$ because any $b : A$ is a lower bound of the empty list; in the cons case, where $a : A$ is the head and $\mathsf{ok}\ b'$ is a result of classifying the tail, i.e., $b' : A$ is a lower bound of the tail, the list can be mapped to $\mathsf{ok}\ b$ if $b : A$ is a lower bound of a and a is exactly b'.

Perhaps the most important consequence of the completeness theorem (in its present form) is that it provides a new perspective on the expressive power of ornaments and inductive families. We showed in a previous paper Ko and Gibbons [2013] that every ornament induces a promotion predicate and a corresponding family of isomorphisms. But one question was untouched: can we determine (independently from ornaments) the range of predicates induced by ornaments? An answer to this question would tell us something about the expressive power of ornaments, and also about the expressive power of inductive families in general, since the inductive families we use are usually ornamentations of simpler algebraic datatypes from traditional functional programming. The completeness theorem offers such an answer: ornament-induced promotion predicates are exactly those expressible as relational folds (up to pointwise isomorphism). In other words, a predicate can be baked into a datatype by

ornamentation if and only if it can be thought of as a nondeterministic classification of the elements of the datatype with a relational fold. This is more a guideline than a precise criterion, though, as the closest work about characterisation of the expressive power of folds discusses only functional folds Gibbons et al. [2001] (however, we believe that those results generalise to relations too). But this does encourage us to think about ornamentation computationally and to design new datatypes with relational algebraic methods. We illustrate this point with a solution to the minimum coin change problem in the next section.

6.1 Ornaments and horizontal transformations

6.2 Ornaments and relational algebras

6.3 Consequences

6.3.1 Parallel composition and the banana-split law

algebras corresponding to singleton ornaments and ornaments for optimised predicates

6.3.2 Ornamental algebraic ornaments

6.4 Discussion

bad computational behaviour; ornaments for optimised representation

Chapter 7

Conclusion

type computation — easy one like upgrades, swaps and more intricate one relying on universe construction; computational formalism — examples: ornaments, universal property of pullbacks; non-examples: relational calculus

7.1 Future work

fibred category theory for unifying similar categorical constructions; measure of representational efficiency; impact of homotopy type theory (e.g., functor equality); future of internalism (highly structural; scalability might lie in, e.g., hierarchical typing)

Bibliography

Thorsten Altenkirch, James Chapman, and Tarmo Uustalu [2010]. Monads need not be endofunctors. In *Foundations of Software Science and Computational Structures*, volume 6014 of *Lecture Notes in Computer Science*, pages 297–311. Springer-Verlag. doi: 10.1007/978-3-642-12032-9_21. ↰ page 118

Thorsten Altenkirch and Conor McBride [2003]. Generic programming within dependently typed programming. In *IFIP TC2/WG2.1 Working Conference on Generic Programming*, pages 1–20. Kluwer, B.V. doi: 10.1007/978-0-387-35672-3_1. ↰ page 20

Thorsten Altenkirch, Conor McBride, and James McKinna [2005]. Why dependent types matter. Available at http://www.cs.nott.ac.uk/~txa/publ/ydtm.pdf. ↰ page 1

Thorsten Altenkirch, Conor McBride, and Wouter Swierstra [2007]. Observational equality, now! In *Programming Languages meets Program Verification*, PLPV'07, pages 57–68. ACM. doi: 10.1145/1292597.1292608. ↰ page 18

Thorsten Altenkirch and Peter Morris [2009]. Indexed containers. In *Logic in Computer Science*, LICS'09, pages 277–285. IEEE. doi: 10.1109/LICS.2009.33. ↰ page 29

Robert Atkey, Patricia Johann, and Neil Ghani [2012]. Refining inductive types. *Logical Methods in Computer Science*, 8(2:09). doi: 10.2168/LMCS-8(2:9)2012. ↰ pages 147 and 163

Gilles Barthe, Venanzio Capretta, and Olivier Pons [2003]. Setoids in type

theory. *Journal of Functional Programming*, 13(2):261–293. doi: 10.1017/S0956796802004501. ↰ pages 19 and 94

Jean-Philippe Bernardy and Moulin Guilhem [2013]. Type theory in color. In *International Conference on Functional Programming*, ICFP'13, pages 61–72. ACM. doi: 10.1145/2500365.2500577. ↰ pages 88, 90, and 162

Yves Bertot and Pierre Castéran [2004]. *Interactive Theorem Proving and Program Development — Coq'Art: The Calculus of Inductive Constructions*. Texts in Theoretical Computer Science. Springer-Verlag. ↰ page 30

Richard Bird and Oege de Moor [1997]. *Algebra of Programming*. Prentice-Hall. ↰ pages 37, 121, 125, and 137

Richard Bird and Jeremy Gibbons [2003]. Arithmetic coding with folds and unfolds. In *Advanced Functional Programming*, volume 2638 of *Lecture Notes in Computer Science*, pages 1–26. Springer-Verlag. doi: 10.1007/978-3-540-44833-4_1. ↰ pages 128 and 134

Errett Bishop and Douglas Bridges [1985]. *Constructive Analysis*. Springer-Verlag. ↰ page 1

Ana Bove and Peter Dybjer [2009]. Dependent types at work. In *Language Engineering and Rigorous Software Development*, volume 5520 of *Lecture Notes in Computer Science*, pages 57–99. Springer-Verlag. doi: 10.1007/978-3-642-03153-3_2. ↰ page 1

Edwin Brady, Conor McBride, and James McKinna [2004]. Inductive families need not store their indices. In *Types for Proofs and Programs*, volume 3085 of *Lecture Notes in Computer Science*, pages 115–129. Springer-Verlag. doi: 10.1007/978-3-540-24849-1_8. ↰ page 22

James Chapman, Pierre-Évariste Dagand, Conor McBride, and Peter Morris [2010]. The gentle art of levitation. In *International Conference on Functional Programming*, ICFP'10, pages 3–14. ACM. doi: 10.1145/1863543.1863547. ↰ pages 20 and 59

R. L. Constable, S. F. Allen, H. M. Bromley, W. R. Cleaveland, J. F. Cremer, R. W. Harper, D. J. Howe, T. B. Knoblock, N. P. Mendler, P. Panangaden, J. T. Sasaki, and S. F. Smith [1985]. *Implementing Mathematics with the Nuprl Proof Development System*. Prentice-Hall. ↰ page 17

Thierry Coquand and Gérard Huet [1988]. The calculus of constructions. *Information and Computation*, 76(2–3):95–120. doi: 10.1016/0890-5401(88)90005-3. ↰ page 30

Thierry Coquand and Christine Paulin-Mohring [1990]. Inductively defined types. In *International Conference on Computer Logic*, volume 417 of *Lecture Notes in Computer Science*, pages 50–66. Springer-Verlag. doi: 10.1007/3-540-52335-9_47. ↰ page 30

Pierre-Évariste Dagand and Conor McBride [2012a]. Elaborating inductive definitions. arXiv:1210.6390. ↰ page 27

Pierre-Évariste Dagand and Conor McBride [2012b]. Transporting functions across ornaments. In *International Conference on Functional Programming*, ICFP'12, pages 103–114. ACM. doi: 10.1145/2364527.2364544. ↰ pages 20, 44, 47, 58, 147, 161, and 163

Michael Dummett [2000]. *Elements of Intuitionism*. Oxford University Press, second edition. ↰ pages 3 and 14

Peter Dybjer [1994]. Inductive families. *Formal Aspects of Computing*, 6(4):440–465. ↰ page 8

Peter Dybjer [1998]. A general formulation of simultaneous inductive-recursive definitions in type theory. *Journal of Symbolic Logic*, 65(2):525–549. doi: 10.2307/2586554. ↰ page 23

Jeremy Gibbons [2007a]. Datatype-generic programming. In *Spring School on Datatype-Generic Programming*, volume 4719 of *Lecture Notes in Computer Science*, pages 1–71. Springer-Verlag. doi: 10.1007/978-3-540-76786-2_1. ↰ page 19

Jeremy Gibbons [2007b]. Metamorphisms: Streaming representation-changers. *Science of Computer Programming*, 65(2):108–139. doi: 10.1016/j.scico.2006.01.006. ↰ pages 128 and 132

Jeremy Gibbons, Graham Hutton, and Thorsten Altenkirch [2001]. When is a function a fold or an unfold? *Electronic Notes in Theoretical Computer Science*, 44(1):146–160. doi: 10.1016/S1571-0661(04)80906-X. ↰ page 151

Healfdene Goguen, Conor McBride, and James McKinna [2006]. Eliminating dependent pattern matching. In *Algebra, Meaning, and Computation*, volume 4060 of *Lecture Notes in Computer Science*, pages 521–540. Springer-Verlag. doi: 10.1007/11780274_27. ↰ page 8

Robert Harper and Robert Pollack [1991]. Type checking with universes. *Theoretical Computer Science*, 89(1):107–136. doi: 10.1016/0304-3975(90)90108-T. ↰ page 94

Arend Heyting [1971]. *Intuitionism: An Introduction*. Amsterdam: North-Holland Publishing, third revised edition. ↰ page 3

Paul Hudak, John Hughes, Simon Peyton Jones, and Philip Wadler [2007]. A history of Haskell: Being lazy with class. In *History of Programming Languages*, pages 1–55. ACM. doi: 10.1145/1238844.1238856. ↰ page 7

Hsiang-Shang Ko and Jeremy Gibbons [2011]. Modularising inductive families. In *Workshop on Generic Programming*, WGP'11, pages 13–24. ACM. doi: 10.1145/2036918.2036921. ↰ page 31

Hsiang-Shang Ko and Jeremy Gibbons [2013]. Modularising inductive families. *Progress in Informatics*, 10:65–88. doi: 10.2201/NiiPi.2013.10.5. ↰ pages 20, 21, and 150

Xavier Leroy [2009]. A formally verified compiler back-end. *Journal of Automated Reasoning*, 43(4):363–446. doi: 10.1007/s10817-009-9155-4. ↰ page 30

Pierre Letouzey [2003]. A new extraction for Coq. In *Types for Proofs and Programs*, volume 2646 of *Lecture Notes in Computer Science*, pages 200–219. Springer-Verlag. doi: 10.1007/3-540-39185-1_12. ↰ page 30

Zhaohui Luo [1994]. *Computation and Reasoning: A Type Theory for Computer Science*. Clarendon Press. ↰ pages 16 and 30

Saunders Mac Lane [1998]. *Categories for the Working Mathematician*. Springer-Verlag, second edition. ↰ pages 92 and 106

Per Martin-Löf [1975]. An intuitionistic theory of types: Predicative part. In *Logic Colloquium '73*, volume 80 of *Studies in Logic and the Foundations of Mathematics*, pages 73–118. Elsevier B.V. doi: 10.1016/S0049-237X(08)71945-1. ↰ page 1

Per Martin-Löf [1984a]. Constructive mathematics and computer programming. *Philosophical Transactions of the Royal Society of London*, 312(1522):501–518. doi: 10.1098/rsta.1984.0073. ↰ page 6

Per Martin-Löf [1984b]. *Intuitionistic Type Theory*. Bibliopolis, Napoli. ↰ pages 1, 19, and 23

Per Martin-Löf [1987]. Truth of a proposition, evidence of a judgement, validity of a proof. *Synthese*, 73(3):407–420. ↰ page 4

Conor McBride [1999]. *Dependently Typed Functional Programs and their Proofs*. Ph.D. thesis, University of Edinburgh. ↰ pages 8 and 97

Conor McBride [2004]. Epigram: Practical programming with dependent types. In *Advanced Functional Programming*, volume 3622 of *Lecture Notes in Computer Science*, pages 130–170. Springer-Verlag. doi: 10.1007/11546382_3. ↰ pages 1, 9, and 31

Conor McBride [2011]. Ornamental algebras, algebraic ornaments. To appear in *Journal of Functional Programming*. ↰ pages 28, 39, 58, 125, 147, and 163

Conor McBride [2012]. A polynomial testing principle. Available at https://personal.cis.strath.ac.uk/conor.mcbride/PolyTest.pdf. ↰ pages 32 and 33

Conor McBride and James McKinna [2004]. The view from the left. *Journal of Functional Programming*, 14(1):69–111. doi: 10.1017/S0956796803004829. ↰ pages 9, 11, 12, 13, and 142

Conor McBride and Ross Paterson [2008]. Applicative programming with effects. *Journal of Functional Programming*, 18(1):1–13. doi: 10.1017/S0956796807006326. ↰ page 119

Erik Meijer, Maarten Fokkinga, and Ross Paterson [1991]. Functional programming with bananas, lenses, envelopes and barbed wire. In *Functional Programming Languages and Computer Architecture*, number 523 in Lecture Notes in Computer Science, pages 124–144. Springer-Verlag. doi: 10.1007/3540543961_7. ↰ pages 7 and 122

Eugenio Moggi [1991]. Notions of computation and monads. *Information and Computation*, 93(1):55–92. doi: 10.1016/0890-5401(91)90052-4. ↰ page 118

Stefan Monnier and David Haguenauer [2010]. Singleton types here, singleton types there, singleton types everywhere. In *Programming Languages meets Program Verification*, PLPV'10, pages 1–8. ACM. doi: 10.1145/1707790.1707792. ↰ page 57

Shin-Cheng Mu, Hsiang-Shang Ko, and Patrik Jansson [2009]. Algebra of Programming in Agda: Dependent types for relational program derivation. *Journal of Functional Programming*, 19(5):545–579. doi: 10.1017/S0956796809007345. ↰ pages 96 and 127

Keisuke Nakano [2013]. Metamorphism in jigsaw. *Journal of Functional Programming*, 23(2):161–173. doi: 10.1017/S0956796812000391. ↰ page 128

Bengt Nordström [1988]. Terminating general recursion. *BIT Numerical Mathematics*, 28(3):605–619. doi: 10.1007/BF01941137. ↰ page 133

Bengt Nordström, Kent Peterson, and Jan M. Smith [1990]. *Programming in Martin-Löf's Type Theory: An Introduction*. Oxford University Press. ↰ pages 1, 17, 19, and 30

Ulf Norell [2007]. *Towards a practical programming language based on dependent type theory*. Ph.D. thesis, Chalmers University of Technology. ↰ pages 1 and 12

Ulf Norell [2009]. Dependently typed programming in Agda. In *Advanced Functional Programming*, volume 5832 of *Lecture Notes in Computer Science*, pages 230–266. Springer-Verlag. doi: 10.1007/978-3-642-04652-0_5. ↰ page 1

Chris Okasaki [1999]. *Purely functional data structures*. Cambridge University Press. ↰ pages 73, 74, 82, and 87

Christine Paulin-Mohring [1989]. Extracting F_ω's programs from proofs in the Calculus of Constructions. In *Principles of Programming Languages*, pages 89–104. ACM. doi: 10.1145/75277.75285. ↰ page 30

Simon Peyton Jones [1997]. A new view of guards. Available at http://research.microsoft.com/en-us/um/people/simonpj/Haskell/guards.html. ↰ page 11

John C. Reynolds [2000]. The meaning of types — from intrinsic to extrinsic semantics. Basic Research in Computer Science (BRICS) Report Series RS-00-32, Department of Computer Science, University of Aarhus. ↰ page 32

Tim Sheard and Nathan Linger [2007]. Programming in Ωmega. In *Central-European Functional Programming School*, volume 5161 of *Lecture Notes in Computer Science*, pages 158–227. Springer-Verlag. doi: 10.1007/978-3-540-88059-2_5. ↰ page 18

Jan M. Smith [1988]. The independence of Peano's fourth axiom from Martin-Löf's type theory without universes. *Journal of Symbolic Logic*, 53(3):840–845. ↰ page 19

Thomas Streicher [1993]. Investigations into intensional type theory. Habilitation thesis, Ludwig Maximilian Universität. ↰ pages 8 and 16

Wouter Swierstra [2008]. Data types à la carte. *Journal of Functional Programming*, 18(4):423–436. doi: 10.1017/S0956796808006758. ↰ page 59

The Univalent Foundations Program [2013]. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Institute for Advanced Study, Princeton. ↰ pages 18 and 19

Philip Wadler [1987]. Views: A way for pattern matching to cohabit with data abstraction. In *Principles of Programming Languages*, pages 307–313. ACM. doi: 10.1145/41625.41653. ↰ page 12

Philip Wadler [1992]. The essence of functional programming. In *Principles of Programming Languages*, POPL'92, pages 1–14. ACM. doi: 10.1145/143165.143169. ↰ page 118

Todo list

program correctness by construction (from specifications to programs); type theory (unification of logic and computation and/vs types as classification/specification); new direction of program derivation, while inheriting problems . . . 1

"datatypes" for inductive families . . . 1

XD . . . 40

Explain the meaning of this (scoping). . . . 43

definition of $*$ . . . 43

definition of pointwise equality . . . 44

Dagand and McBride [2012b], origin of coherence property, no need to construct a universe (open for easy extension) . . . 47

relationship with ornaments . . . 49

connection to refinement families . . . 51

coproduct-related definitions . . . 59

intro — analysis for composability . . . 60

Chapter 4 . . . 60

Chapter 4 . . . 64

intro . . . 66

Optimised in what sense? 66

Postulate operations on *Val* like $\_\leqslant?\_$, $\leqslant$-*refl*, $\leqslant$-*trans*, and $\not\leqslant$-*invert* in Chapter 2. 73

use an ornament-parametrised upgrade 87

Do we need a summary here? 90

summary of the three-level architecture of ornaments, refinements, and upgrades; bundle; why ornaments?; functor-level computation and recursion schemes; compare with Bernardy and Guilhem [2013] . . . 90

functors preserve isomorphisms (a quick demonstration of preorder reasoning); TBC 103

name scoping 104

diagram commutativity not yet defined 106

morphism equivalence and proof irrelevance 106

product diagram; "morphism relevance"? 106

diagram of pullback 109

pullback diagram; pullback preservation 109

mention commutativity, associativity; should probably refactor the pullback square in $\mathbb{F}\textsc{am}$ into the pullback square in $\mathbb{O}\textsc{rn}$ and pullback preservation of $\textsc{Ind}$ 110

elimination of arbitrariness of type-theoretic constructions; functor-level abstraction; compare with purely categorical approach 116

the synthetic direction of the conversion isomorphism; emphasis no longer only on program derivation (relational calculus) but also on relational specifications 117

intro needs revision to de-emphasise program derivation a bit 117

relations vs families of relations 119

probably a simple example of relational calculation here? 121

summary and some gluing to the next section 125

Revise using upgrades. 125

relator laws and various monotonicity need to be stated earlier 127

Just constraint transformation; base data do not change 127

Agda doesn't really allow this, though. 131

compare the McBride [2011] version (compatible with the two-constructor universe) and the Dagand and McBride [2012b] version of algebraic ornamentation in terms of "quality" (amount of σ's used); proof-relevant Algebra of Programming (e.g., *fun-preserves-fold*; linking to the next chapter); related work: Atkey et al. [2012] 147

algebras corresponding to singleton ornaments and ornaments for optimised predicates 151

bad computational behaviour; ornaments for optimised representation . . 151

type computation — easy one like upgrades, swaps and more intricate one relying on universe construction; computational formalism — examples: ornaments, universal property of pullbacks; non-examples: relational calculus 152

fibred category theory for unifying similar categorical constructions; measure of representational efficiency; impact of homotopy type theory (e.g., functor equality); future of internalism (highly structural; scalability might lie in, e.g., hierarchical typing) 152